



**HAL**  
open science

## Exercices de probabilités avec Mathematica

Christophe Chesneau

► **To cite this version:**

Christophe Chesneau. Exercices de probabilités avec Mathematica. Licence. France. 2016. cel-01389944

**HAL Id: cel-01389944**

**<https://cel.hal.science/cel-01389944v1>**

Submitted on 30 Oct 2016


**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

# Exercices de probabilités avec Mathematica



Christophe Chesneau





## Table des matières

<b>1</b>	<b>Présentation</b>	<b>5</b>
1.1	Sur Mathematica . . . . .	5
1.2	Sur le livre . . . . .	6
<b>2</b>	<b>Première approche</b>	<b>7</b>
<b>3</b>	<b>Prise en main</b>	<b>11</b>
<b>4</b>	<b>Probabilités : principales commandes Mathematica</b>	<b>51</b>
4.1	Commandes de base . . . . .	51
4.2	Commandes associées aux probabilités . . . . .	52
4.3	Lois discrètes usuelles . . . . .	54
4.4	Lois à densité usuelles . . . . .	56
<b>5</b>	<b>Variables aléatoires réelles (<i>var</i>) discrètes</b>	<b>59</b>
<b>6</b>	<b>Couples de <i>var</i> discrètes</b>	<b>93</b>
<b>7</b>	<b>Vecteurs de <i>var</i> discrètes et convergence</b>	<b>121</b>
<b>8</b>	<b>Simulations de <i>var</i> discrètes</b>	<b>135</b>
<b>9</b>	<b><i>Var</i> à densité</b>	<b>159</b>
<b>10</b>	<b>Couples de <i>var</i> à densité</b>	<b>207</b>
<b>11</b>	<b>Vecteurs de <i>var</i> à densité et convergence</b>	<b>259</b>
<b>12</b>	<b>Simulation de <i>var</i> à densité</b>	<b>281</b>
<b>13</b>	<b>Ressources</b>	<b>311</b>



# 1 Présentation


## 1.1 Sur Mathematica

Mathematica est un logiciel de

- **calcul formel** : manipulation des expressions mathématiques sous leur forme symbolique (dérivées, primitives, équations. . . Par exemple, si on demande la dérivée de  $x^2$ , on obtiendra  $2x$ ),
- **calcul numérique** : opération ou ensemble d'opérations portant sur des nombres ou des symboles numériques,
- **visualisation graphique**.

Parmi ses qualités, Mathematica propose

- une prise en main simple et intuitive,
- des temps de calculs rapides,
- des graphiques précis et interactifs ; on peut en faire ce que l'on veut (éditer, pivoter. . . ) :



- une multitude de fonctions prédéfinies,
- une aide logicielle et une entraide utilisateurs en ligne très développées.

Il est développé par Wolfram Research :

<https://www.wolfram.com/france/>

Une licence payante est requise. On peut toutefois l'utiliser gratuitement avec WolframAlpha :

<https://www.wolframalpha.com/>



## 1.2 Sur le livre

Ce livre propose 271 exercices corrigés sur le thème des **Probabilités**. Une partie de ces exercices est issue de sujets de concours ou d'examens (niveau Licence 1, 2, 3 et Master 1), une autre a été créée pour l'occasion, profitant de nouvelles perspectives offertes par **Mathematica**. Les corrections donnent les commandes **Mathematica** requises et affichent les résultats obtenus. À eux seuls, ces résultats présentent un intérêt certain pour les étudiants de toutes filières. Le livre dispose d'une centaine de graphiques, allant du classique à l'original, de l'uni au multidimensionnel.

Ce livre est un support idéal pour compléter des cours magistraux et travaux dirigés. Il propose une approche originale, pédagogique et moderne pour illustrer des notions incontournables de Probabilités.

Pour qu'il profite au plus grand nombre, ce livre est en téléchargement libre.

**Sur l'auteur :** Je suis enseignant-chercheur à l'Université de Caen, également auteur des livres :




Pour tout complément d'information, n'hésitez pas à me contacter : [christophe.chesneau@gmail.com](mailto:christophe.chesneau@gmail.com)


**Merci et bonne lecture !**


Version du 01 Novembre 2016

## 2 Première approche


Dans ce livre, on utilise Mathematica 11.0 installé sous Windows.

Une fois le logiciel installé, double-cliquer sur le logo Mathematica : , puis cliquer sur

New Notebook :  Une fenêtre de ce type s'affiche :





Dans cette fenêtre, on y trouve une feuille vierge dans laquelle on peut écrire un calcul. Par exemple, on fait :  $1 + 2 + 3$  :



**Important :** Pour exécuter ce calcul :


- soit on utilise la combinaison de touches "shift entrée" du clavier alphanumérique :

Maintenir  (Shift) puis 

- soit on utilise la touche  du clavier numérique.




En utilisant la première solution, on obtient alors :




La feuille est divisée en cellules symbolisées par des crochets à droite de la feuille. Il y a une cellule d'entrée `In[1]` (pour "input" la 1-ère ligne de commande) et une cellule de sortie `Out[1]` (pour "output" associé à la 1-ère ligne de commande).

S'il n'est pas nécessaire d'avoir la cellule de sortie, on termine le calcul par un point virgule, puis on exécute pour passer au suivant :




On peut également écrire une ligne de calcul, passer à la ligne en appuyant sur la touche `Entrée` du clavier numérique, et écrire une autre ligne de calcul. Après exécution, le logiciel va considérer 2 cellules d'entrées correspondants aux 2 lignes de calculs et 2 cellules de sorties correspondantes aux résultats de ses 2 calculs :



Deux règles fondamentales :

- Toutes les fonctions prédéfinies par Mathematica commencent par une lettre majuscule (**Exp**, **Log**, **Cos**, **Sinh**, **Integrate**, **Plot**...).
- Le logiciel est sensible aux espaces. Par exemple :  $23 = 23$  et  $2\ 3 = 6$ .

Le tableau ci-dessous présente quelques manipulations et syntaxes de base :

Exécuter un calcul	Maintenir $\boxed{\uparrow}$ (Shift) puis $\boxed{\text{Entrée}}$
Arrêter un calcul en cours	Maintenir $\boxed{\text{Alt}}$ et faire un point ( $\boxed{\uparrow}$ puis $\boxed{; .}$ )
Aide sur une commande <b>Com</b>	Faire $\text{?Com}$ , puis cliquer sur $\gg$ en bas à droite pour avoir les détails 
symbole $\leq$	Faire $\leq$
symbole $\geq$	Faire $\geq$
symbole $\rightarrow$	Faire $\rightarrow$
symbole $\alpha$	Faire $\boxed{\text{Échap}}$ a $\boxed{\text{Échap}}$
symbole $\beta$	Faire $\boxed{\text{Échap}}$ b $\boxed{\text{Échap}}$
symbole $\lambda$	Faire $\boxed{\text{Échap}}$ l $\boxed{\text{Échap}}$
symbole $\mu$	Faire $\boxed{\text{Échap}}$ m $\boxed{\text{Échap}}$
symbole $\infty$	Faire $\boxed{\text{Échap}}$ inf $\boxed{\text{Échap}}$
symbole $x_i$	Faire $\boxed{x}$ puis maintenir $\boxed{\text{Ctrl}}$ puis $\boxed{8 \_ \backslash}$ et enfin $\boxed{i}$

**Dorénavant, on se familiarise à Mathematica avec le format "exercices corrigés".**

- La première partie concerne la prise en main des commandes,
- La deuxième partie rentre dans le vif du sujet avec des applications aux Probabilités.



### 3 Prise en main

#### Exercice 1

Utiliser Mathematica pour calculer :

$$3 + 2 + 1, \quad 3 \times 2 + 3 - 15, \quad 5 \times (2 + 15), \quad \left(\frac{238}{7}\right)^2.$$

Commandes (élémentaires) : +, (\* \*), \*, -, /, ^.

**Solution 1.** On propose :

```
In[1]: 3 + 2 + 1 (* commentaire : le premier calcul *)
Out[1]: 6
In[2]: 3 * 2 + 3 - 15
Out[2]: -6
In[3]: 5 (2 + 15)
Out[3]: 85
In[4]: (238 / 7)^2
Out[4]: 1156
```

#### Exercice 2

Utiliser Mathematica pour calculer les valeurs numériques de :

$$\frac{27}{4}, \quad \frac{22}{7}, \quad \frac{22}{7} \text{ avec 12 décimales}, \quad \pi, \quad \pi \text{ avec 9 décimales.}$$

Commandes : N, Pi, //N.

**Solution 2.** On propose :

```
In[1]: N[27 / 4]
Out[1]: 6.75
In[2]: N[22 / 7]
Out[2]: 3.14286
```

```
In[3]: N[22 / 7, 12]
```

```
Out[3]: 3.14285714286
```

```
In[4]: N[Pi]
```

```
Out[4]: 3.14159
```

```
In[5]: N[Pi, 9]
```

```
Out[5]: 3.14159265
```

!! On aurait aussi pu utiliser la commande //N. Par exemple :

```
In[1]: 27 / 4 //N
```

```
Out[1]: 6.75
```

### Exercice 3

Utiliser Mathematica pour

- mettre en mémoire les affectations :  $x = 2$ ,  $a = 2$ ,  $b = 3$  et  $c = 4$ ,
- calculer les valeurs numériques de :

$$x + c, \quad ax^2 + bx + c, \quad \frac{b}{(x + c)^a} \quad x^{a+b+c},$$

- effacer  $x$  de la mémoire,
- effacer simultanément  $a$ ,  $b$  et  $c$  de la mémoire.

Commandes : `Clear`, `//N`.

**Solution 3.** On propose :

```
In[1]: x = 2;
```

```
In[2]: a = 2;
```

```
In[3]: b = 3;
```

```
In[4]: c = 4;
```

```
In[5]: x + c
```

```
Out[5]: 6
```

```
In[6]: a x^2 + b x + c
```

```
Out[6]: 18
```

```
In[7]: b / (x + c)^a //N
```

```
Out[7]: 0.0833333
```

```
In[8]: x^(a + b + c)
Out[8]: 512
In[9]: Clear[x]
In[10]: Clear[a, b, c]
```

**Exercice 4**

Utiliser Mathematica pour calculer les valeurs numériques de :

$$\sqrt{34}, \quad |-6|, \quad e^{2.7}, \quad \ln(8.2), \quad \sin\left(\frac{3\pi}{2}\right), \quad \cos\left(\frac{\pi}{5}\right), \quad \tan(3.4).$$

Commandes : Sqrt, //N, Abs, Exp, Log, Sin, Cos, Tan.

**Solution 4.** On propose :

```
In[1]: Sqrt[34.] (* ou Sqrt[34.0] ou Sqrt[34] //N ou N[Sqrt[34]] *)
Out[1]: 5.83095
In[2]: Abs[-6]
Out[2]: 6
In[3]: Exp[2.7]
Out[3]: 14.8797
In[4]: Log[8.2]
Out[4]: 2.10413
In[5]: Sin[3 Pi / 2] (* ou Sin[3 * Pi / 2] *)
Out[5]: -1
In[6]: Cos[Pi / 5] //N (* sans le //N on obtient  $\frac{1}{4}(1 + \sqrt{5})$  *)
Out[6]: 0.809017
In[7]: Tan[3.4]
Out[7]: 0.264317
```

**Exercice 5**

Utiliser Mathematica pour calculer les valeurs numériques de :

$$\arctan(1), \quad \arcsin(0.5), \quad \arccos(0.7), \quad \cosh(3.8), \quad \sinh(2.1), \quad \tanh(1.2).$$

Commandes : ArcTan, ArcSin, ArcCos, Cosh, Sinh, Tanh.

**Solution 5.** On propose :

```
In[1]: ArcTan[1.]
Out[1]: 0.785398
In[2]: ArcSin[0.5]
Out[2]: 0.523599
In[3]: ArcCos[0.7]
Out[3]: 0.795399
In[4]: Cosh[3]
Out[4]: 2.3618
In[5]: Sinh[2.1]
Out[5]: 4.02186
In[6]: Tanh[1.2]
Out[6]: 0.833655
```

### Exercice 6

Exécuter les commandes suivantes une à une et comprendre la différence entre le symbole (classique) de l'affectation = et le symbole de l'affectation différée := :

```
x = Pi / 5
y1 = Cos[x]
y2 := Cos[x]
x = Pi / 2
y1
y2
```

Commandes : Pi, Cos, :=.

**Solution 6.** On propose :

```
In[1]: x = Pi / 5
Out[1]:  $\frac{\pi}{5}$ 
In[2]: y1 = Cos[x]
Out[2]:  $\frac{1}{4}(1 + \sqrt{5})$ 
In[3]: y2 := Cos[x] (* rien ne se passe après exécution *)
```

```
In[4]: x = Pi / 2
Out[4]:  $\frac{\pi}{2}$ 
In[5]: y1
Out[5]:  $\frac{1}{4}(1 + \sqrt{5})$ 
In[6]: y2 (* on va obtenir la valeur de Cos[x] avec x = Pi / 2 *)
Out[6]: 0
```

### Exercice 7

On définit deux éléments a, b par les commandes :

```
a = Exp[p]
b = Cos[u + v]
```

Les variables p, u, v sont muettes. Créer a et b, et

- calculer  $e^{2.5}$  à partir de a,
- calculer  $\cos(3.14525 + 0.25\pi)$  à partir de b.

Commandes : Exp, Cos, /. →.

**Solution 7.** On propose :

```
In[1]: a = Exp[p];
In[2]: a /. p → 2.5
Out[2]: 12.1825
In[3]: b = Cos[u + v];
In[4]: b /. {u → 3.14525, v → 0.25 Pi}
Out[4]: -0.704516
```

### Exercice 8

On considère les input (entrées) et les output (sorties) suivantes :

```
In[1]: 10
Out[1]: 10
In[2]: 1
Out[2]: 1
In[3]: 4
Out[3]: 4
In[4]: 5
Out[4]: 5
```

Déterminer la sortie **Out[5]** de l'input : In[5]: % + 2 %% + %%% + Out[1]^2

Commandes : %, %, %%, Out.



**Solution 8.** !! La commande :

- % appelle le résultat du dernier output,
- %% appelle le résultat de l'avant dernier output,
- %%% appelle le résultat de l'avant avant dernier output,
- Out[n] appelle le résultat obtenu au  $n$ -ème output.

Par conséquent, on a :

```
In[5]: % + 2 %% + %%% + Out[1]^2      (* 5 + 2 * 4 + 1 + 10^2 = 114 *)
```

```
Out[5]: 114
```

### Exercice 9

Utiliser l'aide en ligne de **Mathematica** pour la commande **Sin**, chercher dans Basic Examples et copier-coller (sans comprendre les détails à ce stade) les commandes permettant

- d'afficher le développement limité de  $\sin(x)$  au voisinage de  $x = 0$  à l'ordre 9 :

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + \frac{x^9}{362880} + O[x]^{11}$$

- de tracer le graphe de  $\sin(x)$  pour  $x \in [0, 2\pi]$ .

Commandes : Sin, ?, ??, Series, Plot, Pi

**Solution 9.** On propose :


```
In[1]: ?Sin      (* une fois exécuté, on clique sur >> et on trouve : *)
```

```
In[2]: Series[Sin[x], {x, 0, 10}]
```

```
Out[2]: x - x^3/6 + x^5/120 - x^7/5040 + x^9/362880 + O[x]^11
```

```
In[3]: Plot[Sin[x], {x, 0, 2 Pi}]
```

```
Out[3]:
```



---

**Exercice 10**

Utiliser *Mathematica* pour

- renvoyer `True` (Vrai) ou `False` (Faux) à l'affirmation " $\sin(\pi) = 0$ ",
- renvoyer `True` ou `False` à l'affirmation " $\sin(\pi) \neq 0$ ",
- renvoyer `True` ou `False` à l'affirmation " $\sin(2.6789) > 0.5$ ",
- renvoyer `True` ou `False` à l'affirmation " $1 > 2$  et  $\pi > 3$ ",
- renvoyer `True` ou `False` à l'affirmation " $1 > 2$  ou  $\pi > 3$ ".

Commandes : `Sin`, `Pi`, `==`, `!=`, `&&`, `||`.

**Solution 10.** On propose :

```
In[1]: Sin[Pi] == 0
```

```
Out[1]: True
```

```
In[2]: Sin[Pi] != 0
```

```
Out[2]: False
```

```
In[3]: Sin[2.6789] > 0.5
```

```
Out[3]: False
```

```
In[4]: 1 > 2 && Pi > 3
```

```
Out[4]: False
```

```
In[5]: 1 > 2 || Pi > 3
```

```
Out[5]: True
```

---

**Exercice 11**

Utiliser *Mathematica* pour

- créer le vecteur  $x = (a, b, c, d, e)$ ,
- créer le vecteur  $y = (\text{Tom}, \text{Bob}, \text{Mat}, \text{Jim})$ ,
- extraire le deuxième élément de  $x$ ,
- extraire le premier et dernier éléments de  $y$ ,
- créer un vecteur  $z$  construit par la concaténation de  $x$  et  $y$ .

Commandes : `{}`, `[[ ]]`, `Join`

**Solution 11.** On propose :

```
In[1]: x = {a, b, c, d, e}
```

```
Out[1]: {a, b, c, d, e}
```

```
In[2]: y = {"Tom", "Bob", "Mat", "Jim"}
```

```
Out[2]: {Tom, Bob, Mat, Jim}
```

```
In[3]: x[[2]]
Out[3]: b
In[4]: y[{{1, 4}}]
Out[4]: {Tom, Jim}
In[5]: z = Join[x, y]
Out[5]: {a, b, c, d, e, Tom, Bob, Mat, Jim}
```

### Exercice 12

Utiliser Mathematica pour

- créer le vecteur  $x = (1, 2, 3, 4, 5, 6, 7, 8, 9)$ ,
- créer le vecteur  $y$  contenant tous les entiers pairs de 2 à 36,
- créer la matrice :

$$A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}.$$

Commandes : `Range`, `//MatrixForm`, `Table`, `TableForm`.

**Solution 12.** On propose :

```
In[1]: x = Range[9]
Out[1]: {1, 2, 3, 4, 5, 6, 7, 8, 9}
In[2]: y = Range[2, 36, 2]
Out[2]: {2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34, 36}
In[3]: A = {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}} //MatrixForm
```

```
Out[3]: 
$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$$

```

!! Une autre possibilité est :

```
In[3]: A = Table[{1 + 3 x, 2 + 3 x, 3 + 3 x}, {x, 0, 2}] //TableForm
Out[3]: 
$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$$

```

**Exercice 13**

On considère la matrice :

$$A = \begin{pmatrix} 6 & 3 & 1 \\ 2 & 1 & 6 \\ 3 & 1 & 2 \end{pmatrix}.$$

Utiliser **Mathematica** pour

- créer  $A$ ,
- afficher  $A$  avec l'écriture matricielle (sans accolade dans la sortie),
- extraire la troisième ligne de  $A$ ,
- extraire la deuxième colonne de  $A$ ,
- extraire l'élément de valeur 1 à l'intersection de la deuxième ligne et de la deuxième colonne de  $A$ .

Commandes : `//MatrixForm, [[]]`.

**Solution 13.** On propose :

```
In[1]: A = {{6, 3, 1}, {2, 1, 6}, {3, 1, 2}};
```

```
In[2]: % //MatrixForm
```

```
Out[2]:  $\begin{pmatrix} 6 & 3 & 1 \\ 2 & 1 & 6 \\ 3 & 1 & 2 \end{pmatrix}$ 
```

```
In[3]: A[[3]]
```

```
Out[3]: {3, 1, 2}
```

```
In[4]: A[[All, 2]]
```

```
Out[4]: {3, 1, 1}
```

```
In[5]: A[[2, 2]]
```

```
Out[5]: 1
```

**Exercice 14**

On considère les matrices :

$$A = \begin{pmatrix} 1 & -1 & 2 \\ 2 & 1 & 4 \\ 3 & 0 & 2 \end{pmatrix}, \quad I_3 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \quad 1_3 = \begin{pmatrix} 1 & 1 & 1 \\ 1 & 1 & 1 \\ 1 & 1 & 1 \end{pmatrix}.$$

Utiliser Mathematica pour

- créer  $A$ ,
- déterminer la matrice transposée  $A^t$  de  $A$ ,
- vérifier que  $A I_3 \neq I_3 A$ ,
- vérifier que  $A I_3 = I_3 A$ ,
- calculer  $(I_3 + A)^2 + A A^t - I_3$ ,
- calculer le déterminant  $\det(A)$  de  $A$ ,
- calculer la matrice inverse  $A^{-1}$  de  $A$ ,
- vérifier que  $A^{-1} A = A A^{-1} = I_3$ .

Commandes : //MatrixForm, Transpose, . (Dot), 1, IdentityMatrix, Det, Inverse.

**Solution 14.** On propose :

```
In[1]: A = {{1, -1, 2}, {2, 1, 4}, {3, 0, 2}};
```

```
In[2]: Transpose[A] //MatrixForm
```

```
Out[2]: 
$$\begin{pmatrix} 1 & 2 & 3 \\ -1 & 1 & 0 \\ 2 & 4 & 2 \end{pmatrix}$$

```

```
In[3]: A.1 != 1.A
```

```
Out[3]: True
```

```
In[4]: A.IdentityMatrix[3] == IdentityMatrix[3].A
```

```
Out[4]: True
```

```
In[5]: (1 + A)^2 + A.Transpose[A] - IdentityMatrix[3] //MatrixForm
```

```
Out[5]: 
$$\begin{pmatrix} 9 & 9 & 16 \\ 18 & 24 & 39 \\ 23 & 15 & 21 \end{pmatrix}$$

```

```
In[6]: Det[A]
```

```
Out[6]: -12
```

```
In[7]: Inverse[A] //MatrixForm
```

```
Out[7]: 
$$\begin{pmatrix} -\frac{1}{6} & -\frac{1}{6} & \frac{1}{2} \\ \frac{2}{3} & \frac{1}{3} & 0 \\ \frac{1}{4} & \frac{1}{4} & -\frac{1}{4} \end{pmatrix}$$

```

```
In[8]: Inverse[A].A == A.Inverse[A] == IdentityMatrix[3]
```

```
Out[8]: True
```

### Exercice 15

Utiliser Mathematica pour

- créer le vecteur  $x = \left(3, \frac{3^2}{2}, \frac{3^3}{3}, \dots, \frac{3^{16}}{16}\right)$ ,
- calculer la somme des éléments du vecteur  $x$  avec la commande `Total`,
- retrouver le résultat précédent avec la commande `sum`.

Commandes : `Table`, `Total`, `Sum`.

**Solution 15.** On propose :

```
In[1]: x = Table[3^i / i, {i, 1, 16}];
```

```
In[2]: Total[x] //N
```

```
Out[2]: 4.18112 * 10^6
```

```
In[3]: Sum[3^i / i, {i, 1, 16}] //N
```

```
Out[3]: 4.18112 * 10^6
```

### Exercice 16

Utiliser Mathematica pour créer un vecteur dont les valeurs des éléments sont données par :

$e^i \sin(i) \cos(i) \log(i + \pi) \sqrt{i}$ , avec  $i = 2, 2.1, 2.2, \dots, 7.9, 8$ .

Commandes : `Table`, `Exp`, `Sin`, `Cos`, `Log`, `Pi`, `Sqrt`, `Range`.

**Solution 16.** On propose :

```
In[1]: Table[Exp[i] Sin[i] Cos[i] Log[i + Pi] Sqrt[i], {i, Range[2, 8, 0.1]}];
```

### Exercice 17

Utiliser Mathematica pour calculer les sommes et produits suivants :

$$\sum_{i=1}^{10} i^2, \quad \sum_{i \in I} i^2 \text{ avec } I = \{1, 2, 5, 9\}, \quad 9! = 9 \times 8 \times \dots \times 2 \times 1, \quad \prod_{i=1}^5 \frac{i}{i+2^i}, \quad \prod_{i=1}^{2018} \left(1 + \frac{1}{i}\right).$$

Commandes : `Sum`, `!`, `Product`.

**Solution 17.** On propose :

```
In[1]: Sum[i^2, {i, 1, 10}]
```

```
Out[1]: 385
```

```
In[2]: Sum[i^2, {i, {1, 2, 5, 9}}]
```

```
Out[2]: 111
```

```
In[3]: 9!
Out[3]: 362880
In[4]: Product[i / (i + 2^i), {i, 1, 5}]
Out[4]:  $\frac{1}{1221}$ 
In[5]: N[%] (* pour avoir la valeur numérique de cette fraction *)
Out[5]: 0.000819001
In[6]: Product[(1 + 1 / i), {i, 1, 2018}]
Out[6]: 2019
```

### Exercice 18

Utiliser Mathematica pour

- choisir au hasard (ou tirer aléatoirement) un réel appartenant à l'intervalle  $[0, 1]$ ,
- choisir au hasard un réel appartenant à l'intervalle  $[0, 50]$ ,
- choisir au hasard 10 réels appartenant à l'intervalle  $[0, 50]$ ,
- créer une matrice à 3 lignes et 3 composantes ayant pour éléments des réels choisis au hasard dans l'intervalle  $[0, 50]$ ,
- choisir au hasard et avec remise 10 entiers appartenant à  $\{0, \dots, 9\}$ ,
- choisir au hasard et avec remise 6 éléments de l'ensemble :  
 $\{\text{"fraise"}, \text{"pomme"}, \text{"cerise"}, 1, 10, 100\}$ ,
- choisir au hasard et sans remise 6 éléments de l'ensemble :  
 $\{\text{"fraise"}, \text{"pomme"}, \text{"cerise"}, 1, 10, 100\}$ .

Commandes : `RandomReal`, `//MatrixForm`, `RandomInteger`, `RandomChoice`, `RandomSample`, `SeedRandom`.

**Solution 18.** !! Comme les valeurs sont choisies au hasard, les résultats que vous allez obtenir seront très probablement différents des sorties ci-dessous.

On propose :

```
In[1]: RandomReal[]
Out[1]: 0.224138
In[2]: RandomReal[{0, 50}]
Out[2]: 34.7481
In[3]: RandomReal[{0, 50}, 10]
Out[3]: {29.6435, 8.53591, 26.1151, 34.2052, 14.2002, 6.72449, 49.7777,
37.1308, 46.5789, 21.1333}
```

```
In[4]: RandomReal[{0, 50}, {3, 3}] //MatrixForm
Out[4]: 
$$\begin{pmatrix} 49.4828 & 42.0522 & 29.2701 \\ 46.4203 & 3.16164 & 4.12868 \\ 25.0922 & 21.0083 & 29.7965 \end{pmatrix}$$

```

```
In[5]: RandomInteger[{0,9},10]
Out[5]: {3, 1, 5, 0, 4, 7, 7, 5, 7, 6}
```

```
In[6]: RandomChoice[{"fraise", "pomme", "cerise"}, 1, 10, 100], 6]
Out[6]: {"cerise", 100, "pomme", 1, 10, "pomme"}
```

```
In[7]: RandomSample[{"fraise", "pomme", "cerise"}, 1, 10, 100], 6]
Out[7]: {"pomme", 10, "fraise", 100, "cerise", 1}
```

!! On peut utiliser la commande `SeedRandom` pour travailler à différentes occasions avec les valeurs obtenues lors d'un choix au hasard.

### Exercice 19

Utiliser `Mathematica` pour

- simplifier l'expression :  $x(2 - x) - 7x^2 + (x - 2)(2x + 1)$ ,
- simplifier l'expression :  $\frac{x^3 + 6x^2 + 11x + 6}{x^2 + 4x + 3}$ ,
- simplifier l'expression :  $\cos^2(x) + \frac{1 - \cos(2x)}{2}$ ,
- utiliser la commande `FullSimplify` pour vérifier que :  $\log(\sqrt{2} + \sqrt{3}) = \frac{1}{2} \log(5 + 2\sqrt{6})$ .

Commandes : `Simplify`, `Cos`, `Log`, `Sqrt`, `//Simplify`.

**Solution 19.** On propose :

```
In[1]: Simplify[x(2 - x) - 7x^2 + (x-2) (2x + 1)]
Out[1]: -2 - x - 6 x^2
```

```
In[2]: Simplify[(x^3 + 6 x^2 + 11 x + 6) / (x^2 + 4 x + 3)]
Out[2]: 2 + x
```

```
In[3]: Simplify[Cos[x]^2 + (1 - Cos[2 x]) / 2]
Out[3]: 1
```

```
In[4]: FullSimplify[Log[Sqrt[2] + Sqrt[3]] == 1 / 2 Log[5 + 2 Sqrt[6]]]
Out[4]: True
```



!! On aurait aussi pu faire :

```
In[1]: x(2 - x) - 7x^2 + (x-2) (2x + 1) //Simplify
```

```
Out[1]: -2 - x - 6 x^2
```

### Exercice 20

Utiliser Mathematica pour

- développer l'expression :  $(x + 3)^3(2x - 1)^2$ ,
- développer l'expression :  $(1 + x - y)(2 - x + y)^3$ .

Commande : Expand.

**Solution 20.** On propose :

```
In[1]: Expand[(x + 3)^3 (2x - 1)^2]
```

```
Out[1]: 27 - 81x + 9x^2 + 73x^3 + 32x^4 + 4x^5
```

```
In[2]: Expand[(1 + x - y) (2 - x + y)^2]
```

```
Out[2]: 4 - 3x^2 + x^3 + 6xy - 3x^2y - 3y^2 + 3xy^2 - y^3
```

### Exercice 21

Utiliser Mathematica pour

- factoriser le polynôme :  $3 - 3x - 7x^2 + 7x^3 + 5x^4 - 5x^5 - x^6 + x^7$ ,
- mettre sur le même dénominateur l'expression :

$$1 + \frac{3}{(2+x)^2} - \frac{2(5+3x)}{(2+x)^3}.$$

Commande : Factor.

**Solution 21.** On propose :

```
In[1]: Factor[3 - 3 x - 7 x^2 + 7 x^3 + 5 x^4 - 5 x^5 - x^6 + x^7]
```

```
Out[1]: (-1 + x)^3(1 + x)^2(-3 + x^2)
```

```
In[2]: Factor[1 + 3 / (2 + x)^2 - 2 (5 + 3 x) / (2 + x)^3]
```

```
Out[2]: (1 + x)^2(4 + x) / (2 + x)^3
```

**Exercice 22**

Utiliser **Mathematica** pour calculer les sommes infinies suivantes :

$$\sum_{i=1}^{\infty} \frac{1}{i^2}, \quad \sum_{i=0}^{\infty} \left(\frac{1}{2}\right)^i, \quad \sum_{i=1}^{\infty} i \left(\frac{1}{2}\right)^{i-1}, \quad \sum_{i=0}^{\infty} \frac{2^i}{i!}, \quad \sum_{i=1}^{\infty} (-1)^{i-1} \frac{(0.3)^i}{i}.$$

Commandes : `Sum`, `∞`.

**Solution 22.** !! Pour utiliser le symbole `∞` :

- soit on écrit les commandes `\[Infinity]`,
- soit on tape sur la touche `Échap`, on écrit `inf`, puis on tape de nouveau sur `Échap`.

Ou alors, sans symbole, on écrit directement `Infinity`.

On propose :

```
In[1]: Sum[1 / i^2, {i, 1, ∞}]
Out[1]:  $\frac{\pi^2}{6}$ 
In[2]: N[%]
Out[2]: 1.64493
In[3]: Sum[(1 / 2)^i, {i, 0, ∞}]
Out[3]: 2
In[4]: Sum[2^i / i!, {i, 0, ∞}]
Out[4]:  $e^2$ 
In[5]: Sum[(-1)^(i - 1) (0.3)^i / i, {i, 1, ∞}]
Out[5]: 0.262364
```

**Exercice 23**

Utiliser **Mathematica** pour résoudre les équations suivantes :

$$x^2 + 4x + 1 = 0, \quad x^3 - 6x^2 - 13x + 42 = 0, \quad (x^2 + 2)(x^2 - 2) = 0.$$

Commande : `Solve`.

**Solution 23.** On propose :

```
In[1]: Solve[x^2 + 4 x + 1 == 0, x]
Out[1]: {{x -> -2 - √3}, {x -> -2 + √3}}
```

In[2]: `Solve[x^3 - 6 x^2 - 13 x + 42 == 0, x]`

Out[2]: `{x -> -3}, {x -> 2}, {x -> 7}`

In[3]: `Solve[(x^2 + 2) (x^2 - 2) == 0, x]`

Out[3]: `{x -> -sqrt(2)}, {x -> -i sqrt(2)}, {x -> i sqrt(2)}, {x -> sqrt(2)}`

### Exercice 24

Utiliser Mathematica pour résoudre l'équation suivante en fonction de  $\alpha$  et  $\mu$  :  $x^2 - \alpha x + \mu = 0$ .

Commandes clés : `Solve`,  $\alpha$ ,  $\mu$ .

**Solution 24.** !! Pour utiliser le symbole  $\alpha$  :

- soit on tape la commande `\[Alpha]`,
- soit on tape sur la touche `Échap`, on écrit `a`, puis on tape de nouveau sur `Échap`,
- soit on tape sur la touche `Échap`, on écrit `\alpha`, puis on tape de nouveau sur `Échap`.

!! De même pour  $\mu$  (`\[Mu]`...) et la plupart des autres lettres grecques.

On propose :

In[1]: `Solve[(x^2 - \alpha x + \mu) == 0, x]`

Out[1]: `{ {x -> 1/2(\alpha - sqrt(\alpha^2 - 4\mu))}, {x -> 1/2(\alpha + sqrt(\alpha^2 - 4\mu))} }`

### Exercice 25

Utiliser Mathematica pour

- résoudre numériquement les équations suivantes :

$$2x^2 + x - 2 = 0, \quad x^3 + x^2 - 2x - 2 = 0, \quad x^3 - x + 2 = 0,$$

- résoudre numériquement les systèmes d'équations suivants :

$$\begin{cases} x + 2y = 2 \\ 2x - 3y = 1 \end{cases}, \quad \begin{cases} x + y + z = 5 \\ x - 3y + 2z = 0 \\ 2x + y - 2z = 3 \end{cases}, \quad \begin{cases} 2x - y + 3z + 2t = -3 \\ x - 2y - 2z - 3t = 2 \\ y + 4t + 4z = 4 \\ 3x - 2y + 3z - t = 5 \end{cases}.$$

Commande : `NSolve`.

**Solution 25.** On propose :

```
In[1]: NSolve[2 x^2 + x - 2 == 0, x]
Out[1]: {{x -> -1.28078}, {x -> 0.780776}}

In[2]: NSolve[x^3 + x^2 - 2 x - 2 == 0, x]
Out[2]: {{x -> -1.41421}, {x -> -1.}, {x -> 1.41421}}

In[3]: NSolve[x^3 - x + 2 == 0, x]
Out[3]: {{x -> -1.52138}, {x -> 0.76069 - 0.857874 i},
{x -> 0.76069 + 0.857874 i}}

In[4]: NSolve[{x + 2 y == 2, 2 x - 3 y == 1}, {x, y}]
Out[4]: {{x -> 1.14286, y -> 0.428571}}

In[5]: NSolve[{x + y + z == 5, x - 3 y + 2 z == 0, 2 x + y - 2 z == 3},
{x, y, z}]
Out[5]: {{x -> 2.05882, y -> 1.58824, z -> 1.35294}}

In[6]: NSolve[{2 x - y + 3 z + 2 t == -3, x - 2 y - 2 z - 3 t == 2,
y + 4 t + 4 z == 4, 3 x - 2 y + 3 z - t == 5}, {x, y, z, t}]
Out[6]: {{x -> -12., y -> -8., z -> 7., t -> -4.}}
```

### Exercice 26

Utiliser Mathematica pour réduire les inégalités suivantes :

$$x^2 + 3x > 2, \quad (x - 1)(x - 2)(x - 3)(x - 4) > 0, \quad \sin(x) > 0 \text{ avec } 0 < x < 20.$$

Commandes : Reduce, Sin.

**Solution 26.** On propose :

```
In[1]: Reduce[x^2 + 3 x > 2, x]
Out[1]: x < 1/2(-3 - sqrt(17)) || x > 1/2(-3 + sqrt(17))

In[2]: Reduce[(x - 1)(x - 2)(x - 3)(x - 4) > 0, x]
Out[2]: x < 1 || 2 < x < 3 || x > 4

In[3]: Reduce[{Sin[x] > 0, 0 < x < 20}, x]
Out[3]: 0 < x < pi || 2pi < x < 3pi || 4pi < x < 5pi || 6pi < x < 20
```

---

**Exercice 27**


---

On considère la fonction :

$$f(x, y) = 1 - (xy - 3)^2, \quad (x, y) \in \mathbb{R}^2.$$

Utiliser **Mathematica** pour

- déterminer la valeur maximale de  $g(x) = f(x, 1)$ ,
- déterminer le réel  $x$  qui maximise  $g(x)$ ,
- déterminer la valeur numérique minimale de  $h(y) = f(e^{-|y|}, y)$ ,
- déterminer le réel  $y$  qui minimise  $h(y)$ ,
- déterminer la valeur maximale de  $f(x, y)$ ,
- déterminer le couple de réels  $(x, y)$  qui maximise  $f(x, y)$ .

Commandes : `MaxValue`, `ArgMax`, `MinValue`, `Abs`, `//N`, `ArgMin`.

**Solution 27.** On propose :

```
In[1]:  MaxValue[1 - (x - 3)^2, x]
```

```
Out[1]:  1
```

```
In[2]:  ArgMax[1 - (x - 3)^2, x]
```

```
Out[2]:  3
```

```
In[3]:  MinValue[1 - (y Exp[-Abs[y]] - 3)^2, y] //N
```

```
Out[3]:  -10.3426
```

```
In[4]:  ArgMin[1 - (y Exp[-Abs[y]] - 3)^2, y]
```

```
Out[4]:  -1
```

```
In[5]:  MaxValue[1 - (x y - 3)^2, {x, y}]
```

```
Out[5]:  1
```

```
In[6]:  ArgMax[1 - (x y - 3)^2, {x, y}]
```

```
Out[6]:  {-1, -3}
```

---

**Exercice 28**


---

Utiliser **Mathematica** pour

- créer la fonction :

$$f(x) = 3 - 4x + 5x^2, \quad x \in \mathbb{R},$$

- calculer  $f(2.56)$ ,
- tracer le graphe de  $f(x)$  pour  $x \in [0, 6]$ .

Commandes : `x_`, `:=`, `Plot`.

**Solution 28.** On propose :


```
In[1]: f[x_] := 3 - 4 x + 5 x^2; (* x_ désigne la variable x d'une fonction *)
```

```
In[2]: f[2.56]
```

```
Out[2]: 25.528
```

```
In[3]: Plot[f[x], {x, 0, 6}]
```

```
Out[3]:
```



### Exercice 29

Utiliser Mathematica pour

- créer la fonction :

$$f(x) = |x^2 - |6x| + 2|, \quad x \in \mathbb{R},$$

- tracer le graphe de  $f(x)$  pour  $x \in [-6, 6]$  avec les couleurs de l'arc-en-ciel et un quadrillage.


Commandes : `x_`, `:=`, `Abs`, `Plot`, `ColorFunction`, `GridLines`.

**Solution 29.** On propose :

```
In[1]: f[x_] := Abs[x^2 - Abs[6 x] + 2];
```

```
In[2]: Plot[f[x], {x, -6, 6}, ColorFunction -> "Rainbow", GridLines -> Automatic]
```

```
Out[2]:
```



## Exercice 30

Utiliser Mathematica pour

- créer la fonction :

$$f(x) = \frac{|x|}{5 + e^x}, \quad x \in \mathbb{R},$$

- calculer  $f(2k)$  avec  $k \in \{1, 2, \dots, 10\}$ ,

- tracer, sur un même graphique, les graphes de  $f(x)$  et  $g(x) = \frac{x^2}{10}$  pour  $x \in [-4, 4]$  avec des couleurs différentes, une légende et un titre.

Commandes : `x_`, `:=`, `Abs`, `Exp`, `Range`, `//N`, `Plot`, `PlotLabels`.

**Solution 30.** On propose :

```
In[1]: f[x_] := Abs[x] / (5 + Exp[x]);
```

```
In[2]: a = Range[2, 20, 2];
```


```
In[3]: f[a] //N
```

```
Out[3]: {0.161433, 0.0671162, 0.0146904, 0.00267921, 0.000453896,
0.0000737283, 0.0000116414, 1.80056 * 10-6, 2.7414 * 10-7, 4.12231 * 10-8}
```

```
In[4]: g[x_] := x^2 / 10;
```

```
In[5]: Plot[{f[x], g[x]}, {x, -4, 4}, PlotLabel -> "Graphes de f(x) et g(x)",
PlotLabels -> "Expressions"]
```

```
Out[5]:
```



## Exercice 31

Utiliser Mathematica pour

- créer la fonction :

$$f(x) = \log(2 + \cos(x)), \quad x \in \mathbb{R},$$

- tracer, sur un même graphique, les graphes de  $f(x)$ ,  $f(2x)$  et  $f(3x)$  pour  $x \in [0, 2\pi]$  avec des couleurs différentes et une légende,

○ reprendre les commandes précédentes en ajoutant l'option : `PlotTheme` → "Business".


Commandes : `x_`, `:=`, `Log`, `Cos`, `Range`, `//N`, `Plot`, `Pi`, `PlotLegends`, `PlotTheme`.

**Solution 31.** On propose :

```
In[1]: f[x_] := Log[2 + Cos[x]];
```


```
In[2]: Plot[{f[x], f[2 x], f[3 x]}, {x, 0, 2 Pi}, PlotLegends → "Expressions"]
```

Out[2]:



```
In[3]: Plot[{f[x], f[2 x], f[3 x]}, {x, 0, 2 Pi}, PlotLegends → "Expressions",
PlotTheme → "Business"]
```

Out[3]:



### Exercice 32

Utiliser `Mathematica` pour tracer, sur un même graphique, le graphe de  $\log(x)$  en bleu, le graphe de  $e^x$  en rouge et le graphe de  $x$  en pointillés verts pour  $x \in [-6, 6]$  (on fera le plus joli possible avec les commandes `PlotRange`, `AspectRatio` et `PlotLegends`).


Commandes : `Plot`, `Exp`, `Log`, `PlotRange`, `AspectRatio`, `PlotLegends`.



**Solution 32.** On propose :

```
In[1]: Plot[{Exp[x], Log[x], x}, {x, -6, 6}, PlotRange -> 6, PlotStyle -> {Red,
Blue, Dashed}, AspectRatio -> Automatic, PlotLegends -> "Expressions"]
```

Out[1]:



### Exercice 33


Utiliser *Mathematica* pour tracer, sur un même graphique, le graphe de  $1 + \cos(x)$ , le graphe de  $1 + \cos(x) + \log(1 + x)$  et l'aire entre ces deux graphes pour  $x \in [0, 20]$ .

Commandes : `Plot`, `Cos`, `Log`, `Filling`, `PlotLegends`.

**Solution 33.** On propose :

```
In[1]: Plot[{1 + Cos[x], 1 + Cos[x] + Log[1 + x]}, {x, 0, 20},
Filling -> {1 -> {2}}, PlotLegends -> "Expressions"]
```

Out[1]:



**Exercice 34**

Exécuter les commandes suivantes une à une et comprendre leurs enjeux :

```
p = {{2, 6}, {4, 11}, {8, 15}, {10, 20}, {24, 39}, {40, 62}, {52, 85}};
ListPlot[p]
Show[Plot[Evaluate[Fit[p {1, x}, x]], {x, 0, 55}, ColorFunction →
"BlueGreenYellow"], ListPlot[p]]
```


Commandes : ListPlot, Show, Plot, Evaluate, Fit, ColorFunction.

**Solution 34.** On propose :

```
In[1]: p = {{2, 6}, {4, 11}, {8, 15}, {10, 20}, {24, 39}, {40, 62}, {52, 85}};
```


```
In[2]: ListPlot[p]
```

Out [2]:



```
In[3]: Show[Plot[Evaluate[Fit[p, {1, x}, x]], {x, 0, 55}, ColorFunction →
"BlueGreenYellow"], ListPlot[p]]
```

Out [3]:



Ainsi, on a tracé un nuage de points :  $M_1$  de coordonnées (2,6),  $M_2$  de coordonnées (4,11), ...,  $M_7$  de coordonnées (52,85). Puis on a ajusté ce nuage de points par une droite (avec la méthode des moindres carrés ordinaires ; c'est la droite de régression).

### Exercice 35

Exécuter les commandes suivantes une à une et comprendre leurs enjeux :


```
a = ParametricPlot[{u, v}, {u, 1, 3}, {v, 3, 4}];
b = ParametricPlot[{v Cos[u], v Sin[u]}, {u, 0, 2 Pi}, {v, 0, 1}];
c = PolarPlot[{u Cos[u], u Sin[u]}, {u, 0, 4 Pi}];
{a, b, c}
```

Commandes : ParametricPlot, Cos, Sin, PolarPlot.

**Solution 35.** On propose :

```
In[1]: a = ParametricPlot[{u, v}, {u, 1, 3}, {v, 3, 4}];
In[2]: b = ParametricPlot[{v Cos[u], v Sin[u]}, {u, 0, 2 Pi}, {v, 0, 1}];
In[3]: c = PolarPlot[{u Cos[u], u Sin[u]}, {u, 0, 4 Pi}];
In[4]: {a, b, c}
```

Out [4] :



On a alors tracé trois graphiques : un rectangle avec la commande ParametricPlot, un cercle avec la commande ParametricPlot et deux spirales sur un même graphique avec la commande PolarPlot.

### Exercice 36

Utiliser Mathematica pour

- créer la fonction :

$$f(x, y) = \sin(x^2 + y^2), \quad (x, y) \in \mathbb{R}^2,$$

- tracer le graphe de  $f(x, y)$  pour  $x \in [-2.5, 2.5]$  et  $y \in [-2, 2]$ .


Commandes : x\_, :=, Sin, Plot3D.

**Solution 36.** On propose :


```
In[1]: f[x_, y_] := Sin[x^2 + y^2];
```

```
In[2]: Plot3D[f[x,y], {x, -2.5, 2.5}, {y, -2, 2}]
```

Out[2]:



!! Le graphique est interactif; on peut le faire pivoter avec la souris pour mieux voir les détails :



### Exercice 37

Utiliser Mathematica pour

- créer la fonction :

$$f(x, y) = e^{-(x^2+y^2)}, \quad (x, y) \in \mathbb{R}^2,$$

- comprendre les enjeux des commandes suivantes :

```
Plot3D[f[x,y], {x, -2, 2}, {y, -2, 2}, ColorFunction -> "BlueGreenYellow",  
Mesh -> None]
```


Commandes : x\_, :=, Exp, Plot3D, ColorFunction, Mesh.

**Solution 37.** On propose :

```
In[1]: f[x_, y_] := Exp[-(x^2 + y^2)];
```

```
In[2]: Plot3D[f[x, y], {x, -2, 2}, {y, -2, 2}, ColorFunction -> "BlueGreenYellow",
      Mesh -> None]
```

Out [2]:



On a tracé le graphe de la fonction  $f(x, y)$  pour  $(x, y) \in [-2, 2]^2$ , avec le dégradé de couleurs BlueGreenYellow et sans quadrillage.

### Exercice 38

Utiliser Mathematica pour

- créer la fonction  $f$  définie par

$$f(x, y) = \cos(2x) \sin(2y), \quad (x, y) \in \mathbb{R}^2,$$

- comprendre les enjeux des commandes suivantes :

```
Plot3D[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction -> Function[{x, y, z},
      Hue[z]]]
```


Commandes : x\_, :=, Cos, Sin, Plot3D, ColorFunction, Hue.

**Solution 38.** On propose :

```
In[1]: f[x_, y_] := Cos[2 x] Sin[2 y];
```

```
In[2]: Plot3D[f[x, y], {x, -5, 5}, {y, -5, 5},
ColorFunction -> Function[{x, y, z}, Hue[z]]]
```

Out[2]:



On a tracé le graphe de la fonction  $f(x, y)$  pour  $(x, y) \in [-5, 5]^2$ , avec un dégradé de couleurs en fonction des valeurs de  $z = f(x, y)$ .

### Exercice 39

Utiliser Mathematica pour

- créer la fonction  $f$  définie par

$$f(x, y) = \cos(x^2) \sin(y^2), \quad (x, y) \in \mathbb{R}^2,$$

- comprendre les enjeux des commandes suivantes :

```
Plot3D[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction -> "AvocadoColors"]
```

- comprendre les enjeux des commandes suivantes :

```
DensityPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction -> "AvocadoColors"]
```

- comprendre les enjeux des commandes suivantes :

```
ContourPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction -> "AvocadoColors"]
```


Commandes : `x_`, `:=`, `Cos`, `Sin`, `Plot3D`, `ColorFunction`, `DensityPlot`, `Hue`.

**Solution 39.** On propose :

```
In[1]: f[x_, y_] := Cos[x^2] Sin[y^2];
```

```
In[2]: Plot3D[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction -> "AvocadoColors"]
```


Out [2]:



On a tracé le graphe de la fonction  $f(x, y)$  pour  $(x, y) \in [-5, 5]^2$ , avec les couleurs de AvocadoColors.


```
In[3]: DensityPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction ->
"AvocadoColors"]
```

Out [3]:



```
In[4]: ContourPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, ColorFunction →
"AvocadoColors"]
```

Out[4]:



Ces graphiques permettent de mieux localiser les zones "denses" *i.e.* en relief, du graphique initial.



#### Exercice 40

Comprendre les enjeux des commandes Mathematica suivantes :


```
Manipulate[Plot3D[Cos[n x y], {x, 1, 2}, {y, 1, 2}, ColorFunction → "Rainbow"],
{n, 1, 5}]
```

Commandes : Manipulate, Plot3D, Cos, ColorFunction.

**Solution 40.** Cela renvoie le graphe de la fonction  $f(x, y) = \cos(nxy)$  avec les couleurs de l'arc-en-ciel. Dans ce graphique, on a un curseur qui permet de choisir plusieurs valeurs de  $n$  :







!! En cliquant sur le minuscule + situé au bout de la ligne du curseur, on active le "Animation Control".  
Dès lors, notre graphique s'anime de manière fluide suivant les valeurs de  $n$ .

#### Exercice 41

Utiliser Mathematica pour

- créer la fonction :

$$f(x) = \begin{cases} 1 - x & \text{si } x \geq 0, \\ x^2 & \text{si } x < 0, \end{cases}$$

- tracer le graphe de  $f(x)$  pour  $x \in [-1, 1]$ .


Commandes : `x_`, `:=`, `Piecewise`, `Plot`, `Boole`.

**Solution 41.** On propose :

```
In[1]: f[x_] := Piecewise[{{x^2, x < 0}, {1 - x, x ≥ 0}}];
```

```
In[2]: Plot[f[x], {x, -1, 1}]
```

Out [2]:



!! On aurait aussi pu utiliser la commande `Boole` (on privilégiera toutefois `Piecewise` par la suite) :

```
In[1]: f[x_] := (1 - x) Boole[x ≥ 0] + x^2 Boole[x < 0];
```

### Exercice 42

Utiliser `Mathematica` pour

- créer la fonction :

$$f(x) = \begin{cases} \frac{x^2}{10} & \text{si } x \in [0, 2[, \\ \frac{2}{5} & \text{si } x \in [2, 5[, \\ \frac{1}{10}(9 - x) & \text{si } x \in [5, 9[, \\ 0 & \text{sinon,} \end{cases}$$

- tracer le graphe de  $f(x)$  pour  $x \in [0, 9]$ .


Commandes : `x_`, `:=`, `Piecewise`, `Plot`.

**Solution 42.** On propose :

```
In[1]: f[x_] := Piecewise[{{x^2 / 10, 0 ≤ x < 2}, {2 / 5, 2 ≤ x < 5},
{(9 - x) / 10, 5 ≤ x < 9}}];
```

```
In[2]: Plot[f[x], {x, 0, 9}]
```

Out[2]:



### Exercice 43

Utiliser `Mathematica` pour

- créer la fonction :

$$f(x, y) = \begin{cases} \frac{1}{18}e^{-\frac{x}{3}} & \text{si } x + y \geq 0 \text{ et } x - y \geq 0, \\ 0 & \text{sinon,} \end{cases}$$

avec la commande `Boole`,

○ tracer le graphe de  $f(x, y)$  pour  $(x, y) \in [-1, 6] \times [-1, 2]$ .


Commandes : `x_`, `:=`, `Exp`, `Boole`, `Plot3D`.

**Solution 43.** On propose :

```
In[1]: f[x_, y_] := 1 / 18 Exp[-x / 3] Boole[x + y ≥ 0 && x - y ≥ 0];
```

```
In[2]: Plot3D[f[x,y], {x, -1, 6}, {y, -1, 2}]
```

Out[2]:



#### Exercice 44

Utiliser Mathematica pour déterminer les limites suivantes :

$$\lim_{x \rightarrow 0} \log(1 + x), \quad \lim_{x \rightarrow \infty} \frac{2x^2 + 3}{x^2 + 6x + 18}, \quad \lim_{x \rightarrow 1} (x - 1) \log(x - 1), \quad \lim_{x \rightarrow \infty} x^{2017} e^{-x}.$$

Commandes : `Limit`, `log`, `∞`, `Exp`.

**Solution 44.** On propose :

```
In[1]: Limit[Log[1 + x], x → 0]
```

Out[1]: 0

```
In[2]: Limit[(2x^2 + 3) / (x^2 + 6x + 18), x → ∞]
```

Out[2]: 2

```
In[3]: Limit[(x - 1) Log[x - 1], x → 1]
```

Out[3]: 0

```
In[4]: Limit[x^2017 Exp[-x], x → ∞]
```

Out[4]: 0

**Exercice 45**

Utiliser Mathematica pour dériver les fonctions suivantes :

$$f(x) = x^2, \quad f(x) = (x + \log(x))^2, \quad f(x) = x \log(2 + x), \quad f(x) = x^3 e^x, \quad f(x) = \frac{1}{1 + \sqrt{x}}.$$

Commandes : D, log, Exp, Sqrt.

**Solution 45.** On propose :

In[1]: D[x^2, x]

Out[1]: 2x

In[2]: D[(x + Log[x])^2, x]

Out[2]:  $2 \left(1 + \frac{1}{x}\right) (x + \text{Log}[x])$

In[3]: D[x Log[2 + x], x]

Out[3]:  $\frac{x}{2 + x} + \text{Log}[2 + x]$

In[4]: D[x^3 Exp[x], x]

Out[4]:  $3e^x x^2 + x^3 e^x$

In[5]: D[1 / (1 + Sqrt[x]), x]

Out[5]:  $-\frac{1}{2(1 + \sqrt{x})^2 \sqrt{x}}$

**Exercice 46**

Utiliser Mathematica pour dériver les fonctions suivantes :

$$f(x) = \cos(x), \quad f(x) = (\sin(x) + 2)^2, \quad f(x) = \arctan(1 + \sqrt{x}), \quad f(x) = \frac{1}{\cos(x) + \cosh(x)}.$$

Commandes : D, Cos, Sin, ArcTan, Cosh.

**Solution 46.** On propose :

In[1]: D[Cos[x], x]

Out[1]: -Sin[x]

In[2]: D[(Sin[x] + 2)^2, x]

Out[2]: 2Cos[x](2 + Sin[x])

```
In[3]: D[ArcTan[1 + Sqrt[x]], x]
Out[3]:  $\frac{1}{2(1 + (1 + \sqrt{x})^2)\sqrt{x}}$ 
In[4]: D[1 / (Cos[x] + Cosh[x]), x]
Out[4]:  $-\frac{-\text{Sin}[x] + \text{Sinh}[x]}{(\text{Cos}[x] + \text{Cosh}[x])^2}$ 
```

### Exercice 47

Utiliser Mathematica pour

- créer la fonction :


$$f(x) = (1 + x)^{-\frac{3}{2}}, \quad x \in \mathbb{R},$$

- déterminer  $\lim_{x \rightarrow \infty} f(x)$ ,
- calculer la dérivée première  $f'(x)$  et la dérivée 8-ème  $f^{(8)}(x)$ ,
- tracer le graphe de  $f(x)$  pour  $x \in [0, 4]$  en pointillé vert épais.

Commandes : `x_`, `:=`, `Limit`, `∞`, `D`, `Plot`, `PlotStyle`, `Dashing`, `Thickness`.

**Solution 47.** On propose :

```
In[1]: f[x_] := (1 + x)^(-3 / 2);
In[2]: Limit[f[x], x -> ∞]
Out[2]: 0
In[3]: D[f[x], x]
Out[3]:  $-\frac{3}{2(1 + x)^{5/2}}$ 
In[4]: D[f[x], {x, 8}]
Out[4]:  $\frac{34459425}{256(1 + x)^{19/2}}$ 
In[5]: Plot[f[x], {x, 0, 4}, PlotStyle -> {Green, Dashing[0.02], Thickness[0.01]}]
Out[5]:
```



## Exercice 48

Utiliser Mathematica pour

- créer la fonction :

$$f(x, y) = \sqrt{x^2 + y^2}, \quad (x, y) \in \mathbb{R}^2,$$

- calculer les dérivées partielles :  $\frac{\partial}{\partial x}f(x, y)$ ,  $\frac{\partial^2}{\partial x^2}f(x, y)$ ,  $\frac{\partial}{\partial y}f(x, y)$  et  $\frac{\partial^2}{\partial x \partial y}f(x, y)$ ,

- tracer le graphe de  $f(\cos(x), \cos(y))$  pour  $(x, y) \in [-2, 2]^2$ .

Commandes : `x_`, `:=`, `Sqrt`, `D`, `Plot3D`.

**Solution 48.** On propose :

In[1]: `f[x_,y_] := Sqrt[x^2 + y^2];`

In[2]: `D[f[x,y], x]`

Out [2]:  $\frac{x}{\sqrt{x^2 + y^2}}$

In[3]: `D[f[x,y], {x, 2}]`

Out [3]:  $-\frac{x^2}{(x^2 + y^2)^{3/2}} + \frac{1}{\sqrt{x^2 + y^2}}$

In[4]: `D[f[x, y], y]`


Out [4]:  $\frac{y}{\sqrt{x^2 + y^2}}$

In[5]: `D[f[x, y], x, y]`

Out [5]:  $-\frac{xy}{(x^2 + y^2)^{3/2}}$

In[6]: `Plot3D[f[Cos[x], Cos[y]], {x, -2, 2}, {y, -2, 2}]`

Out [6]:



---

**Exercice 49**

Utiliser Mathematica pour

- vérifier que, pour tout  $x \in \mathbb{R}$ ,

$$x(1 - x) \leq 0.25,$$

- vérifier que, pour tout  $(x, y) \in \mathbb{R}^2$ ,

$$x^2 + xy + \frac{y^2}{4} \geq 0.$$

Commandes : Refine, Element, Reals.

**Solution 49.** On propose :

```
In[1]: Refine[x (1 - x) ≤ 1 / 4, Element[x, Reals]]
```

```
Out[1]: True
```

```
In[2]: Refine[x^2 + x y + y^2 / 4 ≥ 0, Element[{x, y}, Reals]]
```

```
Out[2]: True
```

---

**Exercice 50**

Utiliser Mathematica pour déterminer "les primitives" (sans addition de constante) suivantes :

$$\int x^{2017} dx, \quad \int \frac{1}{\sqrt{x}} dx, \quad \int \cos(x) \sin(x) dx, \quad \int \frac{1}{1+x} dx, \quad \int \frac{1}{\sqrt{x}(1+x)} dx.$$

Commandes : Integrate, Sqrt, Cos, Sin.

**Solution 50.** On propose :

```
In[1]: Integrate[x^2017, x]
```

```
Out[1]:  $\frac{x^{2018}}{2018}$ 
```

```
In[2]: Integrate[1 / Sqrt[x], x]
```

```
Out[2]:  $2\sqrt{x}$ 
```

```
In[3]: Integrate[Cos[x] Sin[x], x]
```

```
Out[3]:  $-\frac{1}{2}\text{Cos}[x]^2$ 
```

```
In[4]: Integrate[1 / (1 + x), x]
```

```
Out[4]: Log[1 + x]
```

In[5]: Integrate[1 / (Sqrt[x] (1 + x)), x]

Out[5]: 2ArcTan[ $\sqrt{x}$ ]

### Exercice 51

Utiliser Mathematica pour calculer les intégrales suivantes :

$$\int_0^1 \frac{x}{1+x} dx, \quad \int_1^e x \log(x) dx, \quad \int_0^\pi \cos^2(x) dx, \quad \int_0^1 \arctan(x) dx.$$

Commandes : Integrate, Log, Cos, ArcTan, N[%].

**Solution 51.** On propose :

In[1]: Integrate[x / (1 + x), {x, 0, 1}]

Out[1]: 1 - Log[2]

In[2]: Integrate[x Log[x], {x, 1, E}]

Out[2]:  $\frac{1}{4}(1 + e^2)$

In[3]: Integrate[Cos[x]^2, {x, 0, Pi}]

Out[3]:  $\frac{\pi}{2}$

In[4]: Integrate[ArcTan[x], {x, 0, 1}]

Out[4]:  $\frac{1}{4}(\pi - \text{Log}[4])$

In[5]: N[%] (\* pour avoir la valeur numérique de la quantité précédente \*)

Out[5]: 0.438825

### Exercice 52

Utiliser Mathematica pour calculer les intégrales suivantes :

$$\int_0^\infty \frac{1}{1+e^x} dx, \quad \int_0^\infty x e^{-x} dx, \quad \int_0^1 \frac{1}{\sqrt{x(1-x)}} dx, \quad \int_0^\infty \frac{1}{1+x^4} dx, \quad \int_0^\infty \frac{1}{\cosh(x)} dx.$$

Commandes : Integrate, Exp,  $\infty$ , Sqrt, ArcTan, N[%], Cosh.

**Solution 52.** On propose :

In[1]: Integrate[1 / (1 + Exp[x]), {x, 0,  $\infty$ }]

Out[1]: Log[2]



```

In[2]: Integrate[x Exp[-x], {x, 0, ∞}]
Out[2]: 1
In[3]: Integrate[1 / Sqrt[x (1 - x)], {x, 0, 1}]
Out[3]: π
In[4]: Integrate[ArcTan[x], {x, 0, 1}]
Out[4]:  $\frac{\pi}{2\sqrt{2}}$ 
In[5]: N[%] (* pour avoir la valeur numérique de la quantité précédente *)
Out[5]: 1.11072
In[6]: Integrate[1 / Cosh[x], {x, 0, ∞}]
Out[6]:  $\frac{\pi}{2}$ 

```

**Exercice 53**

Utiliser Mathematica pour calculer les intégrales suivantes :

$$\int_0^1 \int_0^1 (1+x+y)^2 dx dy, \quad \int_0^\infty \int_0^\infty xy e^{-(x+y)} dx dy, \quad \int_0^{\frac{\pi}{2}} \left( \int_0^\pi \sin(x-y) dx \right) dy.$$

Commandes : Integrate, Exp, ∞, Sin, Pi.

**Solution 53.** On propose :

```

In[1]: Integrate[(1 + x + y)^2, {x, 0, 1}, {y, 0, 1}]
Out[1]:  $\frac{25}{6}$ 
In[2]: Integrate[x y Exp[-(x + y)], {x, 0, ∞}, {y, 0, ∞}]
Out[2]: 1
In[3]: Integrate[Sin[x - y], {x, 0, Pi}, {y, 0, Pi / 2}]
Out[3]: 2

```

**Exercice 54**

Soient  $a > 0$  et  $b > 0$ . Utiliser Mathematica pour déterminer les intégrales suivantes en fonction de  $a$  et  $b$  :

$$\int_0^a \left( \int_0^b (x^2 + y^2) dy \right) dx, \quad \int_0^a \left( \int_0^x (x^2 + y^2) dy \right) dx, \quad \int_0^a \left( \int_{\frac{\pi}{2}}^x (x^2 + y^2) dy \right) dx.$$

Commande : Integrate.

**Solution 54.** On propose :

In[1]: `Integrate[x^2 + y^2, {x, 0, a}, {y, 0, b}]`

Out[1]:  $\frac{1}{3}ab(a^2 + b^2)$

In[2]: `Integrate[x^2 + y^2, {x, 0, a}, {y, 0, x}]`

Out[2]:  $\frac{a^4}{3}$

In[3]: `Integrate[x^2 + y^2, {x, 0, a}, {y, x / 2, x}]`

Out[3]:  $\frac{19a^4}{96}$

### Exercice 55

Utiliser Mathematica pour déterminer les intégrales suivantes, fonctions par rapport à  $y$  :

$$\int_0^1 x^{2017} y^3 dx, \quad \int_0^1 \frac{1}{\sqrt{1+x+y}} dy, \quad \int_{-\infty}^{\infty} e^{-(x-y)^2} dx, \quad \int_0^{\infty} e^{-x(1+y)} dx \text{ avec } y > -1.$$

Commandes : `Integrate`, `Sqrt`, `Exp`, `∞`, `Assumptions`.

**Solution 55.** On propose :

In[1]: `Integrate[x^2017 y^3, {x, 0, 1}]`

Out[1]:  $\frac{y^3}{2018}$

In[2]: `Integrate[1/ Sqrt[1 + x + y], {y, 0, 1}]`

Out[2]:  $-2\sqrt{1+x} + 2\sqrt{2+x}$

In[3]: `Integrate[Exp[-(x - y)^2], {x, -∞, ∞}]`

Out[3]:  $\sqrt{\pi}$

In[4]: `Integrate[Exp[-x (1 + y)], {x, 0, ∞}, Assumptions → y > -1]`

Out[4]:  $\frac{1}{1+y}$

### Exercice 56

Utiliser Mathematica pour calculer les intégrales triples suivantes :

$$\int_0^1 \int_0^1 \int_0^1 z e^{-\sqrt{x+y-z^2}} dx dy dz, \quad \int_0^{\infty} \int_0^1 \int_0^{\infty} \frac{\arctan(x) \log(y)}{(1+x^2)(1+z^2)} dx dy dz.$$

Commandes : `Integrate`, `Exp`, `Sqrt`, `∞`, `ArcTan`, `Log`.

**Solution 56.** On propose :

In[1]: `Integrate[z Exp[-Sqrt[x] + y - z^2], {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]`

Out[1]:  $\frac{(-2 + e)(-1 + e)^2}{e^2}$

In[2]: `Integrate[ArcTan[x] Log[y] / ((1 + x^2) (1 + z^2)), {x, 0, ∞}, {y, 0, 1}, {z, 0, ∞}]`

Out[2]:  $-\frac{\pi^3}{16}$

### Exercice 57

Utiliser Mathematica pour

- créer la fonction :

$$f(x, y, z) = \begin{cases} \frac{1}{(x + y + z + 1)^3} & \text{si } (x, y, z) \in \mathcal{D}, \\ 0 & \text{sinon,} \end{cases}$$

avec  $\mathcal{D} = \{(x, y, z) \in [0, 1]^3; x + y + z \leq 1\}$ , de deux façons :

- ◊ en utilisant la commande `Boole`,
- ◊ en utilisant la commande `Piecewise`,

- calculer

$$\int_0^1 \int_0^1 \int_0^1 f(x, y, z) dx dy dz.$$

Commandes : `x_`, `:=`, `Boole`, `Piecewise`, `Integrate`.

**Solution 57.** On propose :

In[1]: `f[x_, y_, z_] := 1 / (x + y + z + 1)^3 Boole[x + y + z ≤ 1];`

In[2]: `f[x_, y_, z_] := Piecewise[{{1 / (x + y + z + 1)^3, x + y + z ≤ 1}}];`

In[3]: `Integrate[f[x, y, z], {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]`

Out[3]:  $\frac{1}{16}(-5 + 8\text{Log}[2])$

## 4 Probabilités : principales commandes Mathematica

### 4.1 Commandes de base


#### Commandes fondamentales

Exécuter un calcul	Maintenir $\boxed{\uparrow}$ (Shift) puis $\boxed{\text{Entrée}}$
Arrêter un calcul en cours	Maintenir $\boxed{\text{Alt}}$ et faire un point ( $\boxed{\uparrow}$ puis $\boxed{; \cdot}$ )

#### Commande incontournable

Pour obtenir de l'aide sur une commande `Com`, on fait `?Com` et on exécute.

Voici quelques exemples de sorties avec `?Integrate`, `?Max` et `?Boole` :



Pour obtenir plus de détails sur la commande, on clique sur le symbole >> en bas à droite.

## Symboles récurrents

symbole $\leq$	Faire <code>&lt;=</code>
symbole $\geq$	Faire <code>&gt;=</code>
symbole $\rightarrow$	Faire <code>-&gt;</code>
symbole $\alpha$	Faire <code>Échap a Échap</code>
symbole $\beta$	Faire <code>Échap b Échap</code>
symbole $\lambda$	Faire <code>Échap l Échap</code>
symbole $\mu$	Faire <code>Échap m Échap</code>
symbole $\infty$	Faire <code>Échap inf Échap</code>
symbole $x_i$	Faire <code>x</code> puis maintenir <code>Ctrl</code> puis <code>8 _ \</code> et enfin <code>i</code>

## 4.2 Commandes associées aux probabilités

## Symboles spécifiques aux probabilités

symbole $\approx$	signifie "suit la loi"	Faire <code>Échap dist Échap</code>
symbole <code>&amp;&amp;</code>	signifie "inter (et)"	Faire <code>&amp;&amp;</code>
symbole <code>  </code>	signifie "union (ou)"	Faire <code>  </code>
symbole $\triangleright$	signifie "conditionnellement à"	Faire <code>Échap cond Échap</code>

## Commandes utiles à la construction de lois de probabilités

<code>ProbabilityDistribution</code>	Construction d'une loi de probabilité (général)
<code>EmpiricalDistribution</code>	Construction d'une loi de probabilité discrète
<code>TransformedDistribution</code>	Construction d'une loi de probabilité à partir d'une autre
<code>MarginalDistribution</code>	Loi marginale d'un vecteur de <i>var</i>

**Principales fonctions associées aux lois de probabilités**

PDF	Densité
CDF	Fonction de répartition
MomentGeneratingFunction	Fonction génératrice des moments
CharacteristicFunction	Fonction caractéristique

**Principaux outils/paramètres associés aux lois de probabilités**

Probability	Probabilité
NProbability	Probabilité ; valeur numérique
Mean	Espérance
Moment	Moment
Expectation	Espérance de $g(X)$
NExpectation	Espérance de $g(X)$ ; valeur numérique
Variance	Variance
Quantile	Quantile
StandardDeviation	Écart-type
Median	Médiane
Kurtosis	Kurtosis
Skewness	Coefficient d'asymétrie
Covariance	Matrice de covariance
Correlation	Matrice de corrélation

### Principaux outils graphiques associés aux lois de probabilités

Plot	Graphique (général)
Plot3D	Graphique en trois dimensions (général)
DiscretePlot	Diagramme en bâtons
DiscretePlot3D	Diagramme en bâtons en trois dimensions
Histogram	Histogramme
Histogram3D	Histogramme en trois dimensions
Show	Tracer plusieurs courbes sur un même graphique

### Simulation d'une *var*

RandomVariate	Réalisations de variables aléatoires réelles
---------------	--

## 4.3 Lois discrètes usuelles

### Principales lois discrètes (univariées)


DiscreteUniformDistribution	Loi uniforme sur un ensemble fini
BernoulliDistribution	Loi de Bernoulli
BinomialDistribution	Loi binomiale
GeometricDistribution	Loi géométrique
PoissonDistribution	Loi de Poisson
HypergeometricDistribution	Loi hypergéométrique
NegativeBinomialDistribution	Loi binomiale négative

Deux exemples sont décrits ci-dessous :

Loi	binomiale	Poisson
paramètres	$n \in \mathbb{N}, p \in ]0, 1[$	$\lambda > 0$
loi de $X$	$\mathcal{B}(n, p)$	$\mathcal{P}(\lambda)$
$X(\Omega)$	$\{0, \dots, n\}$	$\mathbb{N}$
$\mathbb{P}(X = x)$	$\binom{n}{x} p^x (1-p)^{n-x}$	$e^{-\lambda} \frac{\lambda^x}{x!}$
commandes	<code>PDF[BinomialDistribution[n, p], x]</code>	<code>PDF[PoissonDistribution[λ], x]</code>

Un exemple de représentation graphique de la loi Binomiale  $\mathcal{B}(n, p)$  avec  $n = 10$  et  $p = 0.3$  :

```
DiscretePlot[PDF[BinomialDistribution[10, 0.3], x], {x, 0, 10}]
```



### Principales lois discrètes multivariées

<code>MultinomialDistribution</code>	Loi multinomiale
<code>MultivariatePoissonDistribution</code>	Loi de Poisson multivariée
<code>MultivariateHypergeometricDistribution</code>	Loi hypergéométrique multivariée



## 4.4 Lois à densité usuelles

## Principales lois à densité (univariées)


UniformDistribution	Loi uniforme sur un intervalle
ExponentialDistribution	Loi exponentielle
NormalDistribution	Loi normale
StudentTDistribution	Loi de Student
ChiSquareDistribution	Loi du Chi-deux
GammaDistribution	Loi gamma
ErlangDistribution	Loi de Erlang
BetaDistribution	Loi bêta
LaplaceDistribution	Loi de Laplace
CauchyDistribution	Loi de Cauchy
LogNormalDistribution	Loi log-normale

Deux exemples sont décrits ci-dessous :

Loi	exponentielle	normale
paramètres	$\lambda > 0$	$\mu \in \mathbb{R}, \sigma > 0$
loi de $X$	$\mathcal{E}(\lambda)$	$\mathcal{N}(\mu, \sigma)$
$X(\Omega)$	$[0, \infty[$	$\mathbb{R}$
$f_X(x)$	$\lambda e^{-\lambda x}$	$\frac{1}{\sqrt{2\pi}\sigma^2} e^{-\frac{(x-\mu)^2}{2\sigma^2}}$
commandes	PDF [ExponentialDistribution[ $\lambda$ ], x]	PDF [NormalDistribution[ $\mu$ , $\sigma$ ], x]

Ci-dessous, un exemple de représentation graphique de la loi normale  $\mathcal{N}(\mu, \sigma)$  avec  $\mu = 0$  et  $\sigma = 0.6$  :

```
Plot[PDF[NormalDistribution[0, 0.6], x], {x, -2.5, 2.5}]
```




### Principales lois à densité multivariées

MultinormalDistribution	Loi normale multivariée
MultivariateTDistribution	Loi de Student multivariée
MultivariateHypergeometricDistribution	Loi hypergéométrique multivariée

Ci-dessous, un exemple de représentation graphique de la loi normale multivariée  $\mathcal{N}_2(\mu, \Sigma)$  avec  $\mu = (0, 0)^t$  et  $\Sigma = \begin{pmatrix} 0.7 & 0.3 \\ 0.3 & 1.1 \end{pmatrix}$  :

```
Plot3D[PDF[MultinormalDistribution[{0, 0}, {{0.7, 0.3},{0.3, 1.1}}], {x, y}],  
{x, -2.5, 2.5}, {y, -2.5, 2.5}, PlotRange -> All]
```





## 5 Variables aléatoires réelles (*var*) discrètes

### Exercice 58

Soient  $p \in ]0, 1[$  et  $X$  une *var* suivant la loi de Bernoulli  $\mathcal{B}(p)$  :

$$\mathbb{P}(X = 0) = 1 - p, \quad \mathbb{P}(X = 1) = p.$$

Utiliser *Mathematica* pour

- déterminer la fonction de répartition de  $X$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : `BernoulliDistribution`, `CDF`, `Mean`, `Variance`.

**Solution 58.** !! Quand la loi utilisée porte un nom, il y a des commandes prédéfinies qui la caractérise.

On propose :

```
In[1]: loi = BernoulliDistribution[p];
```

```
In[2]: CDF[loi, x]
```

```
Out[2]: { 0      x < 0
         { 1 - p  0 ≤ x < 1
         { 1      True
```

```
In[3]: Mean[loi]
```

```
Out[3]: p
```

```
In[4]: Variance[loi]
```

```
Out[4]: (1 - p)p
```

### Exercice 59

Soient  $(a, b) \in (\mathbb{N}^*)^2$  avec  $b > a$  et  $X$  une *var* suivant la loi uniforme discrète  $\mathcal{U}(\{a, \dots, b\})$  :

$$\mathbb{P}(X = k) = \frac{1}{b - a + 1}, \quad k \in \{a, \dots, b\}.$$

Utiliser *Mathematica* pour

- déterminer la fonction de répartition de  $X$ ,

- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : `DiscreteUniformDistribution`, `CDF`, `Mean`, `Variance`.

**Solution 59.** On propose :

In[1]: `loi = DiscreteUniformDistribution[{a,b}];`

In[2]: `CDF[loi, x]`

Out[2]: 
$$\begin{cases} \frac{1 - a + \text{Floor}[x]}{1 - a + b} & a \leq x < b \\ 1 & x \geq b \\ 0 & \text{True} \end{cases}$$

In[3]: `Mean[loi]`

Out[3]: 
$$\frac{a + b}{2}$$

In[4]: `Variance[loi]`

Out[4]: 
$$\frac{1}{12} (-1 + (1 - a + b)^2)$$

### Exercice 60

Soient  $n \in \mathbb{N}^*$ ,  $p \in ]0, 1[$  et  $X$  une var suivant la loi binomiale  $\mathcal{B}(n, p)$  :

$$\mathbb{P}(X = k) = \binom{n}{k} p^k (1 - p)^{n-k}, \quad k \in \{0, \dots, n\}.$$

Utiliser `Mathematica` pour

- calculer  $\mathbb{P}(X = 4)$  avec  $n = 18$  et  $p = 0.31$ ,
- représenter graphiquement la loi de  $X$  avec  $n = 20$  et  $p = 0.4$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- calculer  $\sigma(X)$ .

Commandes : `PDF`, `BinomialDistribution`, `DiscretePlot`, `PDF`, `Mean`, `Variance`, `StandardDeviation`, `/.`, `→`.


**Solution 60.** On propose :

In[1]: `PDF[BinomialDistribution[18, 0.31], 4]`

Out[1]: `0.156695`

```
In[2]: DiscretePlot[PDF[BinomialDistribution[20, 0.4], k], {k, 0, 20}]
```

Out[2]:



```
In[3]: loi = BinomialDistribution[n, p];
```

```
In[4]: Mean[loi]
```

Out[4]: np

```
In[5]: Variance[loi]
```

Out[5]: n(1 - p)p

```
In[6]: StandardDeviation[loi]
```

Out[6]:  $\sqrt{n(1 - p)p}$

!! Dès le début, on aurait pu travailler avec la loi binomiale  $\mathcal{B}(n, p)$  dans sa généralité puis lui substituer des valeurs précises de  $n$  et  $p$  :

```
In[1]: loi = BinomialDistribution[n, p];
```

```
In[2]: PDF[loi /. {n -> 18, p -> 0.31}, 4]
```

Out[2]: 0.156695

!! Cela marche aussi pour le graphique.

### Exercice 61

Reproduire et comprendre les enjeux des commandes Mathematica suivantes :


```
ListPlot[Table[{k, PDF[BinomialDistribution[50, p], k]}, {p, {0.3, 0.5, 0.8}},
{k, 0, 50}], Filling -> Axis]
```

Commandes : ListPlot, Table, PDF, BinomialDistribution, Filling.

**Solution 61.** On propose :

```
In[1]: ListPlot[Table[{k, PDF[BinomialDistribution[50, p], k]},
{p, {0.3, 0.5, 0.8}}, {k, 0, 50}], Filling -> Axis]
```

Out [1] :



Ce graphique représente sous forme de bâtons trois "densités" différentes associées à la loi binomiale  $\mathbb{B}(50, p)$  suivant trois valeurs de  $p$ . Plus précisément, on a les valeurs de  $\mathbb{P}(X = k)$ ,  $k \in \{0, \dots, 50\}$ , avec  $X$  suivant la loi binomiale  $\mathcal{B}(50, p)$ , *i.e.*

$$\mathbb{P}(X = k) = \binom{50}{k} p^k (1-p)^{50-k}, \quad k \in \{0, \dots, 50\}.$$

Les bâtons bleus correspondent à  $p = 0.3$ , les oranges à  $p = 0.5$  et les verts à  $p = 0.8$ .

### Exercice 62

Soient  $p \in ]0, 1[$  et  $X$  une *var* suivant la loi géométrique  $\mathcal{G}(p)$  (notation française usuelle :  $\mathcal{G}_0(p)$ , elle est appelée "loi géométrique modifiée en 0") :

$$\mathbb{P}(X = k) = (1-p)^k p, \quad k \in \mathbb{N}.$$

Utiliser **Mathematica** pour


- représenter graphiquement  $\mathbb{P}(X = k)$  pour  $p = 0.2$  et  $k \in \{0, \dots, 19\}$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- calculer  $\sigma(X)$ .

Commandes : `DiscretePlot`, `PDF`, `GeometricDistribution`, `Mean`, `Variance`, `StandardDeviation`.

**Solution 62.** On propose :

```
In[1]: DiscretePlot[PDF[GeometricDistribution[0.2], k], {k, 0, 19}]
```

Out [1] :

In[2]: `loi = GeometricDistribution[p];`In[3]: `Mean[loi]`Out [3]:  $-1 + \frac{1}{p}$ In[4]: `Variance[loi]`Out [4]:  $\frac{1-p}{p^2}$ In[5]: `StandardDeviation[loi]`Out [5]:  $\frac{\sqrt{1-p}}{p}$ **Exercice 63**Soient  $\lambda > 0$  et  $X$  une *var* suivant la loi de Poisson  $\mathcal{P}(\lambda)$  :

$$\mathbb{P}(X = k) = e^{-\lambda} \frac{\lambda^k}{k!}, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `PoissonDistribution`,
- utiliser la loi précédente pour représenter graphiquement  $\mathbb{P}(X = k)$  pour  $\lambda = 5.3$  et  $k \in \{0, \dots, 17\}$ , avec les couleurs de `BrightBands`,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- calculer  $\sigma(X)$ ,
- déterminer la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- déterminer la fonction caractéristique de  $X$  :  $\varphi(t) = \mathbb{E}(e^{itX})$ .

Commandes : `PoissonDistribution`,  $\lambda$ , `DiscretePlot`, `PDF`, `/.`, `→`, `ColorFunction`, `Mean`, `Variance`, `StandardDeviation`, `MomentGeneratingFunction`, `CharacteristicFunction`.



**Solution 63.** !! Pour utiliser le symbole  $\lambda$  :


- soit on tape la commande `\[Lambda]`,
- soit on tape sur la touche `Échap`, on écrit `1`, puis on tape de nouveau sur `Échap`,
- soit on tape sur la touche `Échap`, on écrit `\lambda`, puis on tape de nouveau sur `Échap`.

On propose :

```
In[1]: loi = PoissonDistribution[λ];
```

```
In[2]: DiscretePlot[PDF[loi, k] /. λ → 5.3, {k, 0, 17}, ColorFunction →
      "BrightBands"]
```

Out[2]:



```
In[3]: Mean[loi]
```

Out[3]:  $\lambda$

```
In[4]: Variance[loi]
```

Out[4]:  $\lambda$

```
In[5]: StandardDeviation[loi]
```

Out[5]:  $\sqrt{\lambda}$

```
In[6]: MomentGeneratingFunction[loi, t]
```

Out[6]:  $e^{(-1+e^t)\lambda}$

```
In[7]: CharacteristicFunction[loi, t]
```

Out[T]:  $e^{(-1+e^{it})\lambda}$

**Exercice 64**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = 1) = 0.24, \quad \mathbb{P}(X = 2) = 0.38, \quad \mathbb{P}(X = 3) = 0.11, \quad \mathbb{P}(X = 4) = 0.27.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `EmpiricalDistribution`,
- déterminer la fonction de répartition de  $X$ ,
- tracer le graphe de la fonction de répartition de  $X$  avec la commande `DiscretePlot` et l'option : `ExtentSize → Right`,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .


Commandes : `EmpiricalDistribution`, `DiscretePlot`, `CDF`, `ExtentSize`, `Mean`, `Variance`, `Boole`.

**Solution 64.** On propose :

```
In[1]: loi = EmpiricalDistribution[{0.24, 0.38, 0.11, 0.27} → {1, 2, 3, 4}];
```

```
In[2]: DiscretePlot[CDF[loi, x], {x, 0, 4}, ExtentSize → Right]
```

Out [2]:



```
In[3]: Mean[loi]
```

Out [3]: 2.41

```
In[4]: Variance[loi]
```

Out [4]: 1.2619

!! On aurait aussi pu définir la loi de  $X$  avec les commandes `Boole`, `ProbabilityDistribution` :

```
In[1]: masses = 0.24 Boole[k == 1] + 0.38 Boole[k == 2] +
0.11 Boole[k == 3] + 0.27 Boole[k == 4];
```

```
In[2]: loi = ProbabilityDistribution[masses, {k, 1, 4, 1}]
```

### Exercice 65

On dispose d'un dé cubique truqué dont les faces sont numérotées de 1 à 6. La particularité de ce dé est la suivante : les numéros pairs sont équiprobables, les numéros impairs sont équiprobables, et la probabilité d'obtenir un numéro pair est égale au double de celle d'obtenir un numéro impair. Soit  $X$  le numéro affiché après un lancer. Alors la loi de  $X$  est donnée par

$$\mathbb{P}(X = 1) = \mathbb{P}(X = 3) = \mathbb{P}(X = 5) = \frac{1}{9}, \quad \mathbb{P}(X = 2) = \mathbb{P}(X = 4) = \mathbb{P}(X = 6) = \frac{2}{9}.$$

Utiliser Mathematica pour

- tracer le graphe de la fonction de répartition de  $X$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .


Commandes : EmpiricalDistribution, DiscretePlot, CDF, ExtentSize, Mean, Variance.

**Solution 65.** On propose :

```
In[1]: loi = EmpiricalDistribution[{1 / 9, 2 / 9, 1 / 9, 2 / 9, 1 / 9, 2 / 9}
→ {1, 2, 3, 4, 5, 6}];
```

```
In[2]: DiscretePlot[CDF[loi, x], {x, 1, 6}, ExtentSize → Right]
```

Out [2] :



```
In[3]: Mean[loi]
```

Out [3] :  $\frac{11}{3}$

```
In[4]: Variance[loi]
```

Out [4] :  $\frac{26}{9}$

**Exercice 66**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = 10^{i-1}) = \frac{i}{28}, \quad i \in \{1, \dots, 7\}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `EmpiricalDistribution`,
- calculer la valeur numérique de  $\mathbb{P}(X \leq 60000)$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\sigma(X)$ .

Commandes : `EmpiricalDistribution`, `Table`, `CDF`, `//N`, `Mean`, `StandardDeviation`.

**Solution 66.** On propose :

```
In[1]: loi = EmpiricalDistribution[Table[i / 28, {i, 1, 7}] → Table[10^(i - 1),
      {i, 1, 7}]]
```

```
In[2]: CDF[loi, 60000] //N
```

```
Out[2]: 0.535714
```

```
In[3]: Mean[loi] //N
```

```
Out[3]: 273369.
```

```
In[4]: StandardDeviation[loi] //N
```

```
Out[4]: 421225.
```

**Exercice 67**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{k}{6}, \quad k \in \{1, 2, 3\}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `ProbabilityDistribution`,
- déterminer la fonction de répartition de  $X$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- déterminer la fonction caractéristique de  $X$  :  $\varphi(t) = \mathbb{E}(e^{itX})$ .

Commandes : `ProbabilityDistribution`, `CDF`, `Mean`, `Variance`, `MomentGeneratingFunction`, `CharacteristicFunction`.

**Solution 67.** !! Lorsque l'on veut créer une loi de probabilité (de nature quelconque) définie par une formule mathématique, on peut utiliser la commande `ProbabilityDistribution`. Si la loi considérée est discrète, il faut ajouter l'argument `1` à la fin de la définition de son support, sinon, on n'ajoute rien.

On propose :

```
In[1]: loi = ProbabilityDistribution[k / 6, {k, 1, 3, 1}]; (* 1 à la fin *)
In[2]: CDF[loi, x]
Out[2]: 
$$\begin{cases} \frac{1}{6} & 1 \leq x < 2 \\ \frac{1}{2} & 2 \leq x < 3 \\ 1 & x \geq 3 \\ 0 & \text{True} \end{cases}$$

In[3]: Mean[loi]
Out[3]:  $\frac{7}{3}$ 
In[4]: Variance[loi]
Out[4]:  $\frac{5}{9}$ 
In[5]: MomentGeneratingFunction[loi, t]
Out[5]:  $\frac{e^t}{6} + \frac{e^{2t}}{3} + \frac{e^{3t}}{2}$ 
In[6]: CharacteristicFunction[loi, t]
Out[6]:  $\frac{e^{it}}{6} + \frac{1}{3}e^{2it} + \frac{1}{2}e^{3it}$ 
```

### Exercice 68

Soient  $\theta \in ]0, 1[$  et  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = c(1 - \theta) \max(k, k^2 - 5k + 6), \quad k \in \{1, \dots, 9\}, \quad \mathbb{P}(X = 10) = \theta,$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie :  $\sum_{k=1}^{10} \mathbb{P}(X = k) = 1$ .

Utiliser `Mathematica` pour

- donner l'expression de  $c$ ,
- créer la loi de  $X$  avec la constante  $c$  déterminée précédemment,
- représenter  $\mathbb{P}(X \leq x)$  avec  $\theta = 0.1$ , pour  $x \in [-1, 12]$ , avec les couleurs de l'arc-en-ciel,

- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : Join, Table,  $\theta$ , Max, Solve, Total, EmpiricalDistribution, Range, DiscretePlot, CDF, /.  
 →, ExtentSize, ColorFunction, Mean, Variance.

**Solution 68.** On propose :

```
In[1]: a = Join[Table[c (1 -  $\theta$ ) Max[k, k^2 - 5 k + 6], {k, 1, 9}], { $\theta$ }];
```


```
In[2]: Solve[Total[a] == 1, c]
```

```
Out[2]:  $\left\{ \left\{ c \rightarrow \frac{1}{121} \right\} \right\}$ 
```

```
In[3]: loi = EmpiricalDistribution[a → Range[1, 10, 1]] /. c → 1 / 121;
```

```
In[4]: DiscretePlot[CDF[loi, x] /.  $\theta \rightarrow 0.1$ , {x, -1, 12}, ExtentSize → Right,  
ColorFunction → "Rainbow"]
```

```
Out[4]:
```



```
In[5]: Mean[loi]
```

```
Out[5]:  $\frac{81(1 - \theta)}{11} + 10\theta$ 
```

```
In[6]: Variance[loi]
```

```
Out[6]:  $\frac{1}{121}(424 + 417\theta - 841\theta^2)$ 
```

**Exercice 69**

Soient  $\lambda > 0$  et  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{e^{-\lambda}}{\lambda(\lambda + 1)} k^2 \frac{\lambda^k}{k!}, \quad k \in \mathbb{N}^*.$$

Utiliser **Mathematica** pour

- vérifier que  $\sum_{k=1}^{\infty} \mathbb{P}(X = k) = 1$ ,
- calculer  $\mathbb{P}(X \geq 3)$ ,
- calculer  $\mathbb{P}(\{X \text{ est impaire}\})$  avec  $\lambda = 0.2$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : Sum, Exp,  $\lambda$ ,  $\infty$ , //Simplify, ProbabilityDistribution, Probability,  $\approx$ , Mod, Mean, Variance.

**Solution 69.** !! Pour utiliser le symbole  $\approx$  signifiant "suit la loi caractérisée par" :

- soit on tape la commande `\[Distributed]`,
- soit on tape sur la touche `Échap`, on écrit `dist`, puis on tape de nouveau sur `Échap`.

On propose :

```
In[1]: Sum[Exp[-λ] / (λ (λ + 1)) k^2 λ^k / k!, {k, 1, ∞}] == 1 //Simplify
Out[1]: True
In[2]: loi = ProbabilityDistribution[Exp[-λ] / (λ (λ + 1)) k^2 λ^k / k!,
    {k, 1, ∞, 1}];
In[3]: Probability[x ≥ 3, x ≈ loi]
Out[3]: 1 -  $\frac{e^{-\lambda}(1 + 2\lambda)}{1 + \lambda}$ 
In[4]: Probability[Mod[x, 2] == 1, x ≈ loi] /. λ → 0.2
Out[4]: 0.72344
In[5]: Mean[loi]
Out[5]:  $\frac{1 + 3\lambda + \lambda^2}{1 + \lambda}$ 
In[6]: Variance[loi]
Out[6]:  $\frac{2\lambda + 2\lambda^2 + \lambda^3}{(1 + \lambda)^2}$ 
```

**Exercice 70**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{16^{2k+1}}{\sinh(16)(2k+1)!}, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- vérifier que  $\sum_{k=0}^{\infty} \mathbb{P}(X = k) = 1$ ,
- représenter graphiquement  $\mathbb{P}(X = k)$  et  $\mathbb{P}(X \leq k)$  pour  $k \in \{0, \dots, 15\}$  sur un même graphique.

Commandes : `Sum`, `Sinh`, `∞`, `ProbabilityDistribution`, `DiscretePlot`, `PDF`, `CDF`, `ExtentSize`.

**Solution 70.** On propose :


```
In[1]: Sum[16^(2 k + 1) / (Sinh[16] (2 k + 1)!), {k, 0, ∞}] == 1
```

```
Out[1]: True
```

```
In[2]: loi = ProbabilityDistribution[16^(2 k + 1) / (Sinh[16] (2 k + 1)!),
      {k, 0, ∞, 1}];
```

```
In[3]: DiscretePlot[{PDF[loi, x], CDF[loi, x]}, {x, 0, 15}, ExtentSize → 0.5]
```

```
Out[3]:
```

**Exercice 71**

Soit  $X$  une *var* suivant la loi binomiale  $\mathcal{B}(19, 0.3)$  :

$$\mathbb{P}(X = k) = \binom{19}{k} 0.3^k (1 - 0.3)^{19-k}, \quad k \in \{0, \dots, 19\}.$$

Utiliser *Mathematica* pour

- calculer  $\mathbb{P}(X = 5)$ ,
- calculer  $\mathbb{P}(X \leq 12)$ ,



- calculer  $\mathbb{P}(X^2 + X + 1 \leq 14)$ ,
- calculer  $\mathbb{P}(\{X > 11\} \cup \{X < 5\})$ ,
- calculer  $\mathbb{P}_{\{X > 8\}}(X \leq 10)$ ,
- calculer le plus petit entier  $k$  tel que  $\mathbb{P}(X \leq k) \geq 0.76$ .

Commandes : BinomialDistribution, PDF, CDF, Probability,  $\approx$ ,  $||$ ,  $\triangleright$  et Quantile.

**Solution 71.** !! Pour utiliser le symbole  $\triangleright$  signifiant "conditionnellement à l'événement" :

- soit on tape la commande `\[Conditioned]`,
- soit on tape sur la touche `Échap`, on écrit `cond`, puis on tape de nouveau sur `Échap`.

On propose :

```
In[1]: loi = BinomialDistribution[19, 0.3];
In[2]: PDF[loi, 5]
Out[2]: 0.191639
In[3]: CDF[loi, 12]
Out[3]: 0.999383
In[4]: Probability[x^2 + x + 1 ≤ 14, x ≈ loi]
Out[4]: 0.133171
In[5]: Probability[x > 11 || x < 5, x ≈ loi]
Out[5]: 0.285046
In[6]: Probability[x ≤ 10 ▷ x > 8, x ≈ loi]
Out[6]: 0.874383
In[7]: Quantile[loi, 0.76]
Out[7]: 7
```

### Exercice 72

Soit  $X$  une *var* suivant la loi géométrique  $\mathcal{G}(0.2)$  (notation française usuelle :  $\mathcal{G}_0(0.2)$ , elle est appelée "loi géométrique modifiée en 0") :

$$\mathbb{P}(X = k) = 0.2(1 - 0.2)^k, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- calculer  $\mathbb{P}(X \geq 5)$ ,
- calculer  $\mathbb{P}(3X^2 + X + 3 \leq 20)$ ,
- calculer  $\mathbb{P}_{\{X > 10\}}(X \leq 15)$ ,

- vérifier que  $\mathbb{P}(\{X \text{ est paire}\}) > 0.5$ ,
- calculer le plus petit entier  $k$  tel que  $\mathbb{P}(X \leq k) \geq 0.84$ ,
- représenter graphiquement  $\mathbb{P}(X \leq x)$  pour  $x \in [0, 12]$  avec les couleurs de `AvocadoColors`.

Commandes : `GeometricDistribution`, `Probability`, `≈`, `▷`, `Mod`, `Quantile`, `DiscretePlot`, `CDF`, `ColorFunction`, `ExtentSize`.

**Solution 72.** On propose :

```
In[1]: loi = GeometricDistribution[0.2];
```

```
In[2]: Probability[x ≥ 5, x ≈ loi]
```

```
Out[2]: 0.32768
```

```
In[3]: Probability[3 x^2 + x + 3 ≤ 20, x ≈ loi]
```

```
Out[3]: 0.488
```

```
In[4]: Probability[x^2 + x + 1 ≤ 14, x ≈ loi]
```

```
Out[4]: 0.133171
```

```
In[5]: Probability[x ≤ 15 ▷ x > 10, x ≈ loi]
```

```
Out[5]: 0.67232
```

```
In[6]: Probability[Mod[x, 2] == 0, x ≈ loi] > 0.5
```


```
Out[6]: True
```

```
In[7]: Quantile[loi, 0.84]
```

```
Out[7]: 8
```

```
In[8]: DiscretePlot[CDF[loi, x], {x, 0, 12}, ColorFunction → "AvocadoColors",
ExtentSize → Right]
```

```
Out[8]:
```



**Exercice 73**

Soient  $n \in \mathbb{N}^*$ ,  $p \in ]0, 1[$  et  $X$  une var suivant la loi binomiale  $\mathcal{B}(n, p)$  :

$$\mathbb{P}(X = k) = \binom{n}{k} p^k (1-p)^{n-k}, \quad k \in \{0, \dots, n\}.$$

On pose  $Y = n - X$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $Y$ ,
- déterminer la fonction génératrice des moments de  $Y$  :  $M(t) = \mathbb{E}(e^{tY})$ ,
- déterminer la fonction caractéristique de  $Y$  :  $\varphi(t) = \mathbb{E}(e^{itY})$ .

Commandes : `TransformedDistribution`, `≈`, `PDF`, `MomentGeneratingFunction`, `CharacteristicFunction`.

**Solution 73.** On propose :

```
In[1]: loi = TransformedDistribution[n - x, x ≈ BinomialDistribution[n, p]]
```

```
Out[1]: BinomialDistribution[n, 1 - p]
```

```
In[2]: PDF[loi, k]
```

```
Out[2]: { (1 - p)^k p^{-k+n} Binomial[n, k]  0 ≤ k ≤ n
          0                                 True
```

!! On reconnaît la loi binomiale  $\mathcal{B}(n, 1 - p)$ .

```
In[3]: MomentGeneratingFunction[loi, t]
```

```
Out[3]: (1 + (-1 + e^t)(1 - p))^n
```

```
In[4]: CharacteristicFunction[loi, t]
```

```
Out[4]: (e^{it}(1 - p) + p)^n
```

**Exercice 74**

Soit  $X$  une var suivant la loi binomiale  $\mathcal{B}(18, 0.4)$  :

$$\mathbb{P}(X = k) = \binom{18}{k} 0.4^k (1 - 0.4)^{18-k}, \quad k \in \{0, \dots, 18\}.$$

On pose  $Y = (-1)^X$ .

Utiliser **Mathematica** pour déterminer la loi de  $Y$ .

Commandes : `TransformedDistribution`, `≈`, `PDF`.

**Solution 74.** On propose :

```
In[1]: loi = TransformedDistribution[(-1)^x, x ≈ BinomialDistribution[18, 0.4]];
In[2]: PDF[loi, k]
Out[2]: { 0.5 k == -1
          0.5 k == 1
          0 True
```

!! On reconnaît la loi de Rademacher  $\mathcal{R}(1)$  :  $\mathbb{P}(Y = -1) = 0.5$ ,  $\mathbb{P}(Y = 1) = 0.5$ .

### Exercice 75

Soient  $p \in ]0, 1[$  et  $X$  une var suivant la loi géométrique  $\mathcal{G}(p)$  :

$$\mathbb{P}(X = k) = (1 - p)^k p, \quad k \in \mathbb{N}.$$

On pose  $Y = \inf(X, 2)$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $Y$ ,
- calculer  $\mathbb{E}(Y)$ ,
- calculer  $\mathbb{V}(Y)$ .

Commandes : `TransformedDistribution`, `Min`, `≈`, `GeometricDistribution`, `PDF`, `Mean`, `Variance`.

**Solution 75.** On propose :

```
In[1]: loi = TransformedDistribution[Min[x, 2], x ≈ GeometricDistribution[p]];
In[2]: PDF[loi, k]
Out[2]: { p k == 0
          - (-1 + p)p k == 1
          (-1 + p)^2 k == 2
          0 True

In[3]: Mean[loi]
Out[3]: 2 - 3p + p^2

In[4]: Variance[loi]
Out[4]: 5p - 10p^2 + 6p^3 - p^4
```

**Exercice 76**

Soient  $n \in \mathbb{N}^*$ ,  $p \in ]0, 1[$  et  $X$  une *var* suivant la loi binomiale  $\mathcal{B}(n, p)$  :

$$\mathbb{P}(X = k) = \binom{n}{k} p^k (1-p)^{n-k}, \quad k \in \{0, \dots, n\}.$$

Utiliser **Mathematica** pour

- créer la fonction `m` définie par la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- calculer  $M'(0)$ ,
- calculer  $M''(0) - (M'(0))^2$ ,
- vérifier que  $M''(1) = \mathbb{E}(X^2 e^X)$ .

Commandes : `BinomialDistribution`, `x_`, `:=`, `MomentGeneratingFunction`, `D`, `/. t ->`, `//Simplify`, `Expectation`, `≈`, `Assumptions`.

**Solution 76.** On propose :

```
In[1]: loi = BinomialDistribution[n, p];
```

```
In[2]: m[t_] := MomentGeneratingFunction[loi, t];
```

```
In[3]: m[t]
```

```
Out[3]: (1 + (-1 + e^t)p)^n
```

```
In[4]: D[m[t], t] /. t -> 0
```

```
Out[4]: np
```

```
In[5]: D[m[t], {t, 2}] - (D[m[t], t])^2 /. t -> 0 //Simplify
```

```
Out[5]: -n(-1 + p)p
```

!! On obtient  $M'(0) = \mathbb{E}(X) = np$  et  $M''(0) - (M'(0))^2 = \mathbb{E}(X^2) - (\mathbb{E}(X))^2 = \mathbb{V}(X) = np(1-p)$ .

```
In[6]: D[m[t], {t, 2}] == Expectation[x^2 Exp[x], x ≈ loi, Assumptions -> n ≥ 1]
      /. t -> 1 //Simplify
```

```
Out[6]: True
```

**Exercice 77**

Soient  $\lambda > 0$  et  $X$  une *var* suivant la loi de Poisson  $\mathcal{P}(\lambda)$  :

$$\mathbb{P}(X = k) = e^{-\lambda} \frac{\lambda^k}{k!}, \quad k \in \mathbb{N}.$$

Utiliser **Mathematica** pour

- créer la fonction `m` définie par la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- vérifier que  $M'(0) = \mathbb{E}(X) = \lambda$ ,

○ vérifier que  $M''(0) - (M'(0))^2 = \mathbb{V}(X) = \lambda$ .

Commandes : `PoissonDistribution`,  $\lambda$ ,  $x_$ , `:=`, `MomentGeneratingFunction`, `D`, `Mean`, `/.`  $\rightarrow$ , `Variance`, `//Simplify`.

**Solution 77.** On propose :

```
In[1]: loi = PoissonDistribution[λ];
```

```
In[2]: m[t_] := MomentGeneratingFunction[loi, t];
```

```
In[3]: m[t]
```

```
Out[3]: e(-1+et)λ
```

```
In[4]: D[m[t], t] == Mean[loi] == λ /. t → 0
```

```
Out[4]: True
```

```
In[5]: D[m[t], {t, 2}] - (D[m[t], t])^2 == Variance[loi] == λ /. t → 0 //Simplify
```

```
Out[5]: True
```

### Exercice 78

Soit  $X$  une *var* suivant la loi hypergéométrique  $\mathcal{H}(2, 6, 10)$  :

$$\mathbb{P}(X = k) = \frac{1}{45} \binom{4}{2-k} \binom{6}{k}, \quad k \in \{0, 1, 2\}.$$

Utiliser `Mathematica` pour

- créer la loi de  $X$  avec la commande `HypergeometricDistribution`,
- expliciter  $\mathbb{P}(X = 0)$ ,  $\mathbb{P}(X = 1)$  et  $\mathbb{P}(X = 2)$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\mathbb{V}(X)$ ,

- calculer la valeur numérique du coefficient d'asymétrie :  $\gamma_1 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^3 \right)$ ,

- calculer la valeur numérique du kurtosis :  $\beta_2 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^4 \right)$ ,

- déterminer la fonction génératrice des moments de  $X$ .

Commandes : `HypergeometricDistribution`, `PDF`, `Mean`, `//N`, `Variance`, `Skewness`, `Kurtosis`, `MomentGeneratingFunction`.

**Solution 78.** On propose :

```
In[1]: loi = HypergeometricDistribution[2, 6, 10];
```

```

In[2]: PDF[loi, {0, 1, 2}]
Out[2]:  $\left\{ \frac{2}{15}, \frac{8}{15}, \frac{1}{3} \right\}$ 
In[3]: Mean[loi] //N
Out[3]: 1.2
In[4]: Variance[loi] //N
Out[4]: 0.426667
In[5]: Skewness[loi] //N
Out[5]: -0.22964
In[6]: Kurtosis[loi] //N
Out[6]: 2.27344
In[7]: MomentGeneratingFunction[loi, t]
Out[7]:  $\frac{1}{15}(2 + 8e^t + 5e^{2t})$ 

```

**Exercice 79**

Soit  $X$  une *var* suivant la loi binomiale  $\mathcal{B}(10, 0.5)$  :

$$\mathbb{P}(X = k) = \binom{10}{k} 0.5^{10}, \quad k \in \{0, \dots, 10\}.$$

Utiliser Mathematica pour

- calculer  $\mathbb{E}(X^4)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{1+X}\right)$ ,
- calculer  $\mathbb{E}(\sin(X))$ ,
- vérifier que  $\mathbb{E}(\sin(X)) = (\cos(0.5))^{10} \sin(5)$ ,
- calculer  $\mathbb{E}(3X^3 + \log(1+X))$ ,
- vérifier que  $|\mathbb{E}(X)| \leq \sqrt{\mathbb{E}(X^2)}$  (c'est un cas particulier de l'inégalité de Cauchy-Schwarz),
- vérifier que  $|\mathbb{E}(X)| \leq (\mathbb{E}(X^{2018}))^{\frac{1}{2018}}$  (c'est un cas particulier de l'inégalité de Hölder).

Commandes : BinomialDistribution, Moment, Expectation,  $\approx$ , Sin, %, Cos, Log, Abs.

**Solution 79.** On propose :

```
In[1]: loi = BinomialDistribution[10, 0.5];
```

```

In[2]: Moment[loi, 4]
Out[2]: 1017.5
In[3]: Expectation[1 / (1 + x), x ≈ loi]
Out[3]: 0.181729
In[4]: Expectation[Sin[x], x ≈ loi]
Out[4]: -0.259815
In[5]: % - Cos[0.5]^10 Sin[5] //N
Out[5]: 0.
In[6]: Expectation[3 x^3 + Log[1 + x], x ≈ loi]
Out[6]: 489.253
In[7]: Abs[Mean[loi]] ≤ Sqrt[Moment[loi, 2]]
Out[7]: True
In[8]: Abs[Mean[loi]] ≤ Moment[loi, 2018]^(1 / 2018)
Out[8]: True

```

**Exercice 80**

Soient  $n \in \mathbb{N}^*$  et  $X$  une *var* suivant la loi binomiale  $\mathcal{B}(n, \frac{1}{2})$  :

$$\mathbb{P}(X = k) = \binom{n}{k} \frac{1}{2^n}, \quad k \in \{0, \dots, n\}.$$

Utiliser **Mathematica** pour calculer, pour tout  $m \in \mathbb{N}$ ,  $\mathbb{E} \left( \frac{(-1)^X}{X + m + 1} \right)$ .

Commandes : **BinomialDistribution**, **Expectation**,  $\approx$ , **Assumptions**.

**Solution 80.** On propose :

```

In[1]: loi = BinomialDistribution[n, 1 / 2];
In[2]: Expectation[(-1^x / (x + m + 1), x ≈ loi, Assumptions → n ≥ 1]
Out[2]:  $\frac{2^{-n} m! n!}{(1 + m + n)!}$ 

```



**Exercice 81**

Soient  $p \in ]0, 1[$  et  $X$  une *var* suivant la loi géométrique  $\mathcal{G}(p)$  :

$$\mathbb{P}(X = k) = (1 - p)^k p, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- calculer  $\mathbb{E}(X^3)$ ,
- calculer  $\mathbb{E}_{\{X > 10\}}(X)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{1 + X}\right)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{(1 + X)(2 + X)}\right)$ .

Commandes : `GeometricDistribution`, `Moment`, `Expectation`, `▷`, `≈`.

**Solution 81.** On propose :

`In[1]: loi = GeometricDistribution[p];`

`In[2]: Moment[loi, 3]`

`Out[2]:`  $\frac{1 + 4(1 - p)^2 + (1 - p)^3 - p}{p^3}$

`In[3]: Expectation[x ▷ x > 10, x ≈ loi]`

`Out[3]:`  $\frac{1 + 10p}{p}$

`In[4]: Expectation[1 / (1 + x), x ≈ loi]`

`Out[4]:`  $\frac{p \text{Log}[p]}{-1 + p}$

`In[5]: Expectation[1 / ((1 + x) (2 + x)), x ≈ loi]`

`Out[5]:`  $\frac{p(1 - p + p \text{Log}[p])}{(-1 + p)^2}$

**Exercice 82**

Soient  $\lambda > 0$  et  $X$  une *var* suivant la loi de Poisson  $\mathcal{P}(\lambda)$  :

$$\mathbb{P}(X = k) = e^{-\lambda} \frac{\lambda^k}{k!}, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- calculer  $\mathbb{E}(X^8)$ ,

- calculer  $\mathbb{E}\left(\frac{1}{1+X}\right)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{(1+X)(2+X)}\right)$ ,
- calculer  $\mathbb{E}(9^X)$ .

Commandes : `PoissonDistribution`,  $\lambda$ , `Moment`, `Expectation`,  $\approx$ .

**Solution 82.** On propose :

```
In[1]: loi = PoissonDistribution[λ];
In[2]: Moment[loi, 8]
Out[2]: λ + 127λ2 + 966λ3 + 1701λ4 + 1050λ5 + 266λ6 + 28λ7 + λ8
In[3]: Expectation[1 / (1 + x), x ≈ loi]
Out[3]:  $\frac{e^{-\lambda}(-1 + e^\lambda)}{\lambda}$ 
In[4]: Expectation[1 / ((1 + x) (2 + x)), x ≈ loi]
Out[4]:  $\frac{e^{-\lambda}(-1 + e^\lambda - \lambda)}{\lambda^2}$ 
In[5]: Expectation[9~x, x ≈ loi]
Out[5]: e8λ
```

### Exercice 83

Soit  $n \in \mathbb{N} \setminus \{0, 1, 2\}$ . Une urne contient  $n$  boules numérotés de 1 à  $n$ . Un joueur extrait au hasard une poignée de 2 boules. Soient  $X$  la *var* égale au plus petit des numéros obtenus. Alors la loi de  $X$  est donnée par

$$\mathbb{P}(X = k) = \frac{2(n - k)}{n(n - 1)}, \quad k \in \{1, \dots, n - 1\}.$$

Utiliser `Mathematica` pour

- créer la loi de  $X$  avec la commande `ProbabilityDistribution`,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- calculer la valeur numérique de  $\mathbb{E}(4X^2 + 3X - 2)$  pour  $n = 45$ .

Commandes : `ProbabilityDistribution`, `Assumptions`, `Mean`, `Variance`, `NExpectation`, `/.`  $\rightarrow$ .

**Solution 83.** On propose :

```
In[1]: loi = ProbabilityDistribution[2 (n - k) / (n (n - 1)),
{k, 1, n - 1, 1}, Assumptions -> n ≥ 3];
```

```

In[2]: Mean[loi]
Out[2]:  $\frac{1+n}{3}$ 
In[3]: Variance[loi]
Out[3]:  $\frac{1}{18}(-2-n+n^2)$ 
In[4]: NExpectation[4 x^2 + 3 x - 2, x ≈ loi] /. n → 45
Out[4]: 1424.

```

**Exercice 84**

Soient  $n \in \mathbb{N} \setminus \{0, 1\}$ ,  $p \in ]0, 1[$  et  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{1}{np} k \binom{n}{k} p^k (1-p)^{n-k}, \quad k \in \{1, \dots, n\}.$$

Utiliser **Mathematica** pour

- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{X}\right)$ ,
- déterminer la fonction génératrice des moments de  $X$ .

Commandes : ProbabilityDistribution, Binomial, Assumptions, Mean, Expectation, ≈, MomentGeneratingFunction.

**Solution 84.** On propose :

```

In[1]: loi = ProbabilityDistribution[1 / np k Binomial[n, k] p^k (1 - p)^(n - k),
      {k, 1, n, 1}, Assumptions → n ≥ 2];
In[2]: Mean[loi]
Out[2]:  $\frac{p(n - np + n^2 p)}{np}$ 
In[3]: Expectation[1 / x, x ≈ loi]
Out[3]:  $\frac{\left(-1 + \left(-\frac{1}{-1+p}\right)^n\right) (1-p)^n}{np}$ 
In[4]: MomentGeneratingFunction[loi, t]
Out[4]:  $\frac{e^t np(1 + (-1 + e^t)p)^n}{np(1 - p + e^t p)}$ 

```

**Exercice 85**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{1}{3} \times \frac{1}{7} + \frac{2}{3} \binom{7}{k} 0.3^k (1 - 0.3)^{7-k}, \quad k \in \{0, \dots, 7\}.$$

Utiliser *Mathematica* pour


- créer la loi de  $X$  avec la commande `MixtureDistribution`,
- représenter graphiquement la loi de  $X$ ,
- calculer  $\mathbb{P}(X^2 - 2X \leq 3)$ ,
- calculer  $\mathbb{P}(\{X \geq 5\} \cup \{X \leq 2\})$ ,
- calculer  $\mathbb{P}(\{X \text{ est paire}\})$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$ .

Commandes : `DiscreteUniformDistribution`, `BinomialDistribution`, `MixtureDistribution`, `DiscretePlot`, `PDF`, `ExtentSize`, `Probability`, `≈`, `||`, `Mod`, `Mean`, `Variance`, `MomentGeneratingFunction`.

**Solution 85.** On propose :

```
In[1]: loi1 = DiscreteUniformDistribution[{0, 7}];
In[2]: loi2 = BinomialDistribution[7, 0.3];
In[3]: loi3 = MixtureDistribution[{1, 2}, {loi1, loi2}];
In[4]: DiscretePlot[PDF[loi3, k], {k, 0, 7}, ExtentSize -> 0.5]
```

Out[4]:



```
In[5]: Probability[x^2 - 2 x ≤ 3, x ≈ loi3]
```

Out[5]: 0.749309

```

In[6]: Probability[x ≥ 5 || x ≤ 2, x ≈ loi3]
Out[6]: 0.700577
In[7]: Probability[Mod[x, 2] == 0, x ≈ loi3]
Out[7]: 0.500546
In[8]: Mean[loi3]
Out[8]: 2.56667
In[9]: Variance[loi3]
Out[9]: 3.16556
In[10]: MomentGeneratingFunction[loi3, t]
Out[10]:  $\frac{1}{3} \left( -\frac{1 - e^{8t}}{8(-1 + e^t)} + 2(1 + 0.3(-1 + e^t))^7 \right)$ 

```

**Exercice 86**

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{4}{7} \times e^{-7} \frac{7^k}{k!} + \frac{3}{7} \times 0.4(1 - 0.4)^k, \quad k \in \mathbb{N}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$ ,
- représenter graphiquement  $\mathbb{P}(X = k)$  pour  $k \in \{0, \dots, 15\}$ ,
- calculer  $\mathbb{P}_{\{X \leq 14\}}(X \geq 6)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\sigma(X)$ .

Commandes : *PoissonDistribution*, *GeometricDistribution*, *MixtureDistribution*, *DiscretePlot*, *PDF*, *ExtentSize*, *Probability*,  $\triangleright$ ,  $\approx$ , *Mean*, *StandardDeviation*.


**Solution 86.** On propose :

```

In[1]: loi1 = PoissonDistribution[7];
In[2]: loi2 = GeometricDistribution[0.4];
In[3]: loi3 = MixtureDistribution[{4, 3}, {loi1, loi2}];
In[4]: DiscretePlot[PDF[loi3, k], {k, 0, 15}, ExtentSize → 0.5]

```

Out [4] :



In[5]: Probability[x ≥ 6 > x ≤ 14, x ≈ loi3]

Out [5]: 0.417571

In[6]: Mean[loi3]

Out [6]: 4.64286

In[7]: Variance[loi3]

Out [7]: 13.0153

### Exercice 87

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{e - 1}{e + 1} e^{-|k|}, \quad k \in \mathbb{Z}.$$

Utiliser Mathematica pour

- représenter graphiquement  $\mathbb{P}(X = k)$  pour  $k \in \{-7, \dots, 7\}$ ,
- calculer  $\mathbb{P}(2X \leq X^2)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{E}(|X|)$ ,
- calculer  $\mathbb{V}(X)$ .


Commandes : ProbabilityDistribution, E, Exp, Abs, ∞, DiscretePlot, PDF, ExtentSize, Probability, ≈, Mean, Expectation, Variance.

**Solution 87.** On propose :

```
In[1]: loi = ProbabilityDistribution[(E - 1) / (E + 1) Exp[-Abs[k]],
    {k, -∞, ∞, 1}];
```

In[2]: `DiscretePlot[PDF[loi, k], {k, -7, 7}, ExtentSize -> 0.5]`

Out[2]:



In[3]: `Probability[2 x ≤ x^2, x ≈ loi]`

Out[3]:  $\frac{1 + e^2}{e(1 + e)}$

In[4]: `Mean[loi]`

Out[4]: 0

In[5]: `Expectation[Abs[x], x ≈ loi]`

Out[5]:  $\frac{2e}{(-1 + e)(1 + e)}$

In[6]: `Variance[loi]`

Out[6]:  $\frac{2e}{(-1 + e)^2}$

### Exercice 88

Soit  $X$  une *var* suivant la loi binomiale tronquée  $\mathcal{B}_{\{3, \dots, 10\}}(10, 0.4)$  :

$$\mathbb{P}(X = k) = 1.2009 \times \binom{10}{k} 0.4^k (1 - 0.4)^{10-k}, \quad k \in \{3, \dots, 10\}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `TruncatedDistribution`,
- calculer le plus petit entier  $k$  tel que  $\mathbb{P}(X \leq k) \geq 0.81$ ,
- calculer  $\mathbb{P}(X^2 - 3X + 1 \geq 7)$ ,
- calculer  $\mathbb{E}(X)$ ,

- calculer  $\mathbb{E}\left(\frac{1}{X^2}\right)$ ,
- calculer  $\mathbb{E}(\cos(X))$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : BinomialDistribution, TruncatedDistribution, Quantile, probability,  $\approx$ , Mean, Expectation, Cos, Variance.

**Solution 88.** On propose :

```
In[1]: loi = BinomialDistribution[10, 0.4];
In[2]: loi2 = TruncatedDistribution[{2, 10}, loi];
In[3]: Quantile[loi2, 0.81]
Out[3]: 6
In[4]: Probability[x^2 - 3x + 1 ≥ 7, x ≈ loi2]
Out[4]: 0.440606
In[5]: Mean[loi2]
Out[5]: 4.46473
In[6]: Expectation[1 / x^2, x ≈ loi2]
Out[6]: 0.0621347
In[7]: Expectation[Cos[x], x ≈ loi2]
Out[7]: -0.220823
In[8]: Variance[loi2]
Out[8]: 1.53341
```

### Exercice 89

Soit  $X$  une *var* suivant la loi de poisson tronquée en zéro  $\mathcal{P}_{\geq 1}(1.2)$  :

$$\mathbb{P}(X = k) = 0.431013 \times \frac{1.2^k}{k!}, \quad k \in \mathbb{N}^*.$$

Utiliser *Mathematica* pour

- calculer le plus petit entier  $k$  tel que  $\mathbb{P}(X \leq k) \geq 0.99$ ,
- calculer  $\mathbb{P}(\{X < 4\} \cup \{X > 100\})$ ,
- calculer  $\mathbb{P}(\max(X^2 - X + 1, X^3 - 2) \geq 7)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\sigma(X)$ ,



- calculer  $\mathbb{E}(e^X + 3)$ ,
- calculer  $\mathbb{E}_{\{(X-2)^2 \leq 5\}}(X)$ .

Commandes : PoissonDistribution, TruncatedDistribution, Quantile, probability, ||, ≈, Max, Mean, StandardDeviation, Expectation, Exp, ▷.

**Solution 89.** On propose :

```
In[1]: loi = PoissonDistribution[1.2];
In[2]: loi2 = TruncatedDistribution[{0, ∞}, loi];
In[3]: Quantile[loi2, 0.99]
Out[3]: 5
In[4]: Probability[x < 4 || x > 100, x ≈ loi2]
Out[4]: 0.951676
In[5]: Probability[Max[x^2 - x + 1, x^3 - 2] ≥ 7, x ≈ loi2]
Out[5]: 0.172455
In[6]: Mean[loi2]
Out[6]: 1.71722
In[7]: StandardDeviation[loi2]
Out[7]: 0.910519
In[8]: Expectation[Exp[x] + 3, x ≈ loi2]
Out[8]: 13.8185
In[9]: Expectation[x ▷ (x - 2)^2 ≤ 5, x ≈ loi2]
Out[9]: 1.67782
```

### Exercice 90





Comprendre les enjeux des commandes Mathematica suivantes :

```
Manipulate[DiscretePlot[PDF[PoissonDistribution[λ], x], {x, 0, 12, 1},
ExtentSize → 0.5, PlotRange → {0, 0.4}], {{λ, 5}, 1, 10, 1, Appearance →
"Labeled"}]
```

Commandes : Manipulate, DiscretePlot, PDF, PoissonDistribution, λ, ExtentSize, PlotRange, Appearance.

**Solution 90.** Cela renvoie un graphique qui représente la "densité" associée à la loi de Poisson.

Dans ce graphique, on a un curseur qui permet de choisir plusieurs valeurs de  $\lambda$  :



### Exercice 91

*Inégalité de Bienaymé-Tchebychev* : Soit  $X$  une var admettant un moment d'ordre 2. Alors, pour tout  $\epsilon > 0$ , on a

$$\mathbb{P}(|X - \mathbb{E}(X)| > \epsilon) \leq \frac{\mathbb{V}(X)}{\epsilon^2}.$$

L'objectif de cet exercice est d'illustrer cette inégalité.

Utiliser **Mathematica** pour

- créer une fonction  $f$  à deux arguments : `eps` et `loi`, telle que, pour  $X$  suivant la loi `loi`, on a

$$f(\text{eps}, \text{loi}) = \mathbb{P}(|X - \mathbb{E}(X)| > \text{eps}),$$

- créer une fonction  $g$  a deux arguments :  $\text{eps}$  et  $\text{loi}$ , telle que, pour  $X$  suivant la loi  $\text{loi}$ , on a

$$g(\text{eps}, \text{loi}) = \frac{\mathbb{V}(X)}{\text{eps}^2},$$

- représenter sur un même graphique les graphes de  $f(\text{eps}, \text{loi})$  et  $g(\text{eps}, \text{loi})$  avec  $\text{eps} \in [0.1, 6]$  et successivement :

- ◊  $\text{loi} = \text{PoissonDistribution}[2]$ ,
- ◊  $\text{loi} = \text{BinomialDistribution}[13, 0.5]$ ,
- ◊  $\text{loi} = \text{NegativeBinomialDistribution}[4, 0.8]$ .

Commandes :  $x_+$ ,  $:=$ ,  $\text{NProbability}$ ,  $\text{Abs}$ ,  $\text{Expectation}$ ,  $\approx$ ,  $\text{Plot}$ ,  $\text{PoissonDistribution}$ ,  $\text{PlotLabels}$ ,  $\text{BinomialDistribution}$ ,  $\text{NegativeBinomialDistribution}$ .


**Solution 91.** On propose :

```
In[1]: f[eps_, loi_] := NProbability[Abs[x - Expectation[x, x ≈ loi]] ≥ eps,
      x ≈ loi];
```

```
In[2]: g[eps_, loi_] := Expectation[(x - Expectation[x, x ≈ loi])^2,
      x ≈ loi] / eps^2;
```


```
In[3]: Plot[{f[eps, PoissonDistribution[2]], g[eps, PoissonDistribution[2]]},
      {eps, 0.1, 6}, PlotLabels → {"f(eps, loi)", "g(eps, loi)"}]
```

Out[3]:




```
In[4]: Plot[{f[eps, BinomialDistribution[13, 0.5]],
      g[eps, BinomialDistribution[13, 0.5]]}, {eps, 0.1, 6},
      PlotLabels → {"f(eps, loi)", "g(eps, loi)"}]
```

Out [4] :



```
In[5]: Plot[{f[eps, NegativeBinomialDistribution[4, 0.8]],  
g[eps, NegativeBinomialDistribution[4, 0.8]]}, {eps, 0.1, 6},  
PlotLabels -> {"f(eps, loi)", "g(eps, loi)"}]
```

Out [5] :





## 6 Couples de *var* discrètes

### Exercice 92

On sait qu'il existe un unique réel  $b \in ]0, 1[$  tel que la loi d'un couple de *var*  $(X, Y)$  soit donnée par le tableau :

		Y	
		0	1
X	0	$b^2$	$\frac{b}{2}$
	1	$\frac{1}{2}$	$b$

avec

		Y	
		0	1
X	0	$\mathbb{P}(\{X = 0\} \cap \{Y = 0\})$	$\mathbb{P}(\{X = 0\} \cap \{Y = 1\})$
	1	$\mathbb{P}(\{X = 1\} \cap \{Y = 0\})$	$\mathbb{P}(\{X = 1\} \cap \{Y = 1\})$

Utiliser *Mathematica* pour déterminer  $b$ .

Commandes : `Solve`, `Total`, `&&`.

**Solution 92.** On propose :

```
In[1]: Solve[Total[{b^2, b / 2, 1 / 2, b}] == 1 && 0 < b < 1, b]
```

```
Out[1]: {{b -> 1/4(-3 + Sqrt[17])}}
```

### Exercice 93

On sait qu'il existe un unique réel  $b \in ]0, 1[$  tel que la loi d'un couple de *var*  $(X, Y)$  soit donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = b \frac{2^{-(i+j)}}{j!}, \quad (i, j) \in (\mathbb{N}^*)^2.$$

Utiliser *Mathematica* pour déterminer  $b$ .

Commande : `Sum`, `!`, `N[%]`.

**Solution 93.** On propose :

```
In[1]: 1 / Sum[2^(-(i + j)) / j!, {i, 1, Infinity}, {j, 1, Infinity}]
```

```
Out[1]: 1 / (-1 + Sqrt[e])
```

In[2]: N[%]

Out[2]: 1.54149

### Exercice 94

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par le tableau :

		Y	
		0	1
X	0	0.2	0.3
	1	0.1	0.4

Utiliser Mathematica pour

- vérifier que  $\sum_{i=0}^1 \sum_{j=0}^1 \mathbb{P}(\{X = i\} \cap \{Y = j\}) = 1$ ,
- créer la loi de  $(X, Y)$  avec la commande `EmpiricalDistribution`,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer simultanément  $\sigma(X)$  et  $\sigma(Y)$ ,
- calculer  $\mathbb{C}(X, Y)$ .

Commandes : `Total`, `EmpiricalDistribution`, `PDF`, `MarginalDistribution`, `Boole`, `Mean`, `Variance`, `StandardDeviation`, `Covariance`.

**Solution 94.** On propose :

In[1]: `Total[{0.2, 0.3, 0.1, 0.4}] == 1`

Out[1]: `True`

In[2]: `loi = EmpiricalDistribution[{0.2, 0.3, 0.1, 0.4} → {{0, 0}, {0, 1}, {1, 0}, {1, 1}}];`

In[3]: `PDF[MarginalDistribution[loi, 1], k]`

Out[3]: `0.5 Boole[0 == k] + 0.5 Boole[1 == k]`

In[4]: `PDF[MarginalDistribution[loi, 2], k]`

Out[4]: `0.3 Boole[0 == k] + 0.7 Boole[1 == k]`

```

In[5]: Mean[loi]
Out[5]: {0.5, 0.7}
In[6]: Variance[loi]
Out[6]: {0.25, 0.21}
In[7]: StandardDeviation[loi]
Out[7]: {0.5, 0.458258}
In[8]: Covariance[loi, 1, 2]
Out[8]: 0.05

```

**Exercice 95**

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par le tableau :

$X \backslash Y$	0.5	0.7	0.9	1
1	0.1	0.1	0.15	0.05
2	0.1	0.2	0.25	0.05

Utiliser *Mathematica* pour

- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer simultanément  $\sigma(X)$  et  $\sigma(Y)$ ,
- calculer la matrice de covariance  $\Sigma$  de  $(X, Y)$ ,
- extraire de  $\Sigma$  la covariance  $\mathbb{C}(X, Y)$ .

Commandes : `EmpiricalDistribution`, `PDF`, `MarginalDistribution`, `Boole`, `Mean`, `Variance`, `StandardDeviation`, `Covariance`, `//MatrixForm`.

**Solution 95.** On propose :

```

In[1]: loi = EmpiricalDistribution[{0.1, 0.1, 0.15, 0.05, 0.1, 0.2, 0.25, 0.05}
→ {{1, 0.5}, {1, 0.7}, {1, 0.9}, {1, 1}, {2, 0.5}, {2, 0.7}, {2, 0.9},
{2, 1}}];
In[2]: PDF[MarginalDistribution[loi, 1], k]
Out[2]: 0.4 Boole[1 == k] + 0.6 Boole[2 == k]

```



```

In[3]: PDF[MarginalDistribution[loi, 2], k]
Out[3]: 0.2 Boole[0.5 == k] + 0.3 Boole[0.7 == k] + 0.4 Boole[0.9 == k] +
0.1 Boole[1 == k]

In[4]: Mean[loi]
Out[4]: {1.6, 0.77}

In[5]: Variance[loi]
Out[5]: {0.24, 0.0281}

In[6]: StandardDeviation[loi]
Out[6]: {0.489898, 0.167631}

In[7]: Covariance[loi] //MatrixForm
Out[7]: 
$$\begin{pmatrix} 0.24 & 0.003 \\ 0.003 & 0.281 \end{pmatrix}$$


In[8]: Covariance[loi, 1, 2]
Out[8]: 0.003

```

**Exercice 96**

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par le tableau :

		Y		
		-1	0	1
X	-1	$\frac{1}{4}$	$\frac{1}{6}$	$\frac{1}{12}$
	0	$\frac{1}{18}$	$\frac{1}{9}$	$\frac{1}{6}$
	1	$\frac{1}{12}$	$\frac{1}{18}$	$\frac{1}{36}$

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$ ,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes,
- créer la loi de  $(X^2, Y^2)$ ,
- déterminer la loi de  $X^2$ ,

- déterminer la loi de  $Y^2$ ,
- vérifier que  $X^2$  et  $Y^2$  sont indépendantes.

Commandes : EmpiricalDistribution, MarginalDistribution, PDF, Boole, Sum, TransformedDistribution.

**Solution 96.** On propose :

```
In[1]: loi = EmpiricalDistribution[{1 / 4, 1 / 6, 1 / 12, 1 / 18, 1 / 9,
1 / 6, 1 / 12, 1 / 18, 1 / 36} → {{-1, -1}, {0, -1}, {1, -1}, {-1, 0},
{0, 0}, {1, 0}, {-1, 1}, {0, 1}, {1, 1}}];
```

```
In[2]: loi1 = MarginalDistribution[loi, 1];
```

```
In[3]: PDF[loi1, k]
```

```
Out[3]:  $\frac{1}{18} \text{Boole}[-1 == k] + \frac{1}{3} \text{Boole}[0 == k] + \frac{5}{18} \text{Boole}[1 == k]$ 
```

```
In[4]: loi2 = MarginalDistribution[loi, 2];
```

```
In[5]: PDF[loi2, k]
```

```
Out[5]:  $\frac{1}{2} \text{Boole}[-1 == k] + \frac{1}{3} \text{Boole}[0 == k] + \frac{1}{6} \text{Boole}[1 == k]$ 
```

```
In[6]: Sum[Boole[PDF[loi, {i, j}] != PDF[loi1, i] PDF[loi2, j]], {i, -1, 1},
{j, -1, 1}] == 0
```

```
Out[6]: False
```

!! Ainsi, il existe un couple de valeurs  $(i, j)$  tel que  $\mathbb{P}(\{X = i\} \cap \{Y = j\}) \neq \mathbb{P}(X = i)\mathbb{P}(Y = j)$ ;

$X$  et  $Y$  ne sont pas indépendantes.

```
In[7]: loi3 = TransformedDistribution[{x^2, y^2}, {x, y} ≈ loi];
```

```
In[8]: loi4 = MarginalDistribution[loi3, 1]
```

```
In[9]: PDF[loi4, k]
```

```
Out[9]:  $\frac{1}{3} \text{Boole}[0 == k] + \frac{2}{3} \text{Boole}[1 == k]$ 
```

!! On reconnaît la loi de Bernoulli  $\mathcal{B}(\frac{2}{3})$ .

```
In[10]: loi5 = MarginalDistribution[loi3, 2];
```

```
In[11]: PDF[loi5, k]
```

```
Out[11]:  $\frac{1}{3} \text{Boole}[0 == k] + \frac{2}{3} \text{Boole}[1 == k]$ 
```

!! On reconnaît la loi de Bernoulli  $\mathcal{B}(\frac{2}{3})$ .

```
In[12]: Sum[Boole[PDF[loi3, {i, j}] != PDF[loi4, i] PDF[loi5, j]], {i, 0, 1},
        {j, 0, 1}] == 0
```

Out[12]: True

!! Ainsi, on a  $\mathbb{P}(\{X^2 = i\} \cap \{Y^2 = j\}) = \mathbb{P}(X^2 = i)\mathbb{P}(Y^2 = j)$  pour tout  $(i, j) \in (X, Y)(\Omega) = \{0, 1\}^2$ ;  $X^2$  et  $Y^2$  sont indépendantes.

### Exercice 97

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \frac{1}{18}(i + 2j), \quad (i, j) \in \{1, 2\}^2.$$

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$  avec la commande `ProbabilityDistribution`,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer simultanément  $\sigma(X)$  et  $\sigma(Y)$ ,
- calculer  $\mathbb{C}(X, Y)$ ,
- créer la fonction `m` définie par la fonction génératrice des moments de  $(X, Y)$  :  $M(s, t) = \mathbb{E}(e^{sX+tY})$ ,
- vérifier que  $\frac{\partial^2}{\partial s \partial t} M(0, 0) - \frac{\partial}{\partial s} M(0, 0) \frac{\partial}{\partial t} M(0, 0) = \mathbb{C}(X, Y)$ .

Commandes : `ProbabilityDistribution`, `PDF`, `MarginalDistribution`, `Mean`, `Variance`, `StandardDeviation`, `Covariance`, `x_`, `:=`, `MomentGeneratingFunction`, `D`, `%` / . .

**Solution 97.** On propose :

```
In[1]: loi = ProbabilityDistribution[1 / 18 (i + 2j), {i, 1, 2, 1}, {j, 1, 2, 1}];
```

```
In[2]: PDF[MarginalDistribution[loi, 1], k]
```

```
Out[2]: {
  3+k / 9  1 <= k <= 2
  0       True
```

```
In[3]: PDF[MarginalDistribution[loi, 2], k]
```

```
Out[3]: {
  1/6 + 2k/9  1 <= k <= 2
  0          True
```

```

In[4]: Mean[loi]
Out[4]:  $\left\{ \frac{14}{9}, \frac{29}{18} \right\}$ 
In[5]: Variance[loi]
Out[5]:  $\left\{ \frac{20}{81}, \frac{77}{324} \right\}$ 
In[6]: StandardDeviation[loi]
Out[6]:  $\left\{ \frac{2\sqrt{5}}{9}, \frac{\sqrt{77}}{18} \right\}$ 
In[7]: Covariance[loi, 1, 2]
Out[7]:  $-\frac{1}{162}$ 
In[8]: M[s_, t_] := MomentGeneratingFunction[loi, {s, t}]
Out[8]:  $\frac{1}{18}e^{s+t}(3 + 4e^s + 5e^t + 6e^{s+t})$ 
In[9]: D[M[s, t], {s, 1}, {t, 1}] - D[M[s, t], {s, 1}] * D[M[s, t], {t, 1}];
In[10]: % /. {s -> 0, t -> 0}
Out[10]:  $-\frac{1}{162}$ 

```

### Exercice 98

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \frac{1}{6i}, \quad j \in \{1, \dots, i\}, i \in \{1, \dots, 6\}.$$

Utiliser **Mathematica** pour

- créer la loi de  $(X, Y)$  avec la commande `ProbabilityDistribution`,
- représenter graphiquement la loi de  $(X, Y)$  avec la commande `DiscretePlot3D`,
- calculer  $\mathbb{P}(X^2 + Y^2 \leq 10)$ ,
- calculer  $\mathbb{P}(XY \leq 20)$ ,
- calculer  $\mathbb{E}(3XY + 5)$ ,
- calculer la matrice de covariance de  $(X, Y)$ .


Commandes : `ProbabilityDistribution`, `DiscretePlot3D`, `PDF`, `ExtentSize`, `Probability`, `≈`, `Expectation`, `Covariance`, `//MatrixForm`.

**Solution 98.** On propose :

```
In[1]: loi = ProbabilityDistribution[1 / (6 i), {j, 1, i, 1}, {i, 1, 6, 1}];
```

```
In[2]: DiscretePlot3D[PDF[loi, {i, j}], {i, 1, 6}, {j, 1, 6}, ExtentSize -> Full]
```

Out[2]:



```
In[3]: Probability[x^2 + y^2 ≤ 10, {x, y} ≈ loi]
```

Out[3]:  $\frac{7}{18}$

```
In[4]: Probability[x y ≤ 10, {x, y} ≈ loi]
```

Out[4]:  $\frac{53}{60}$

```
In[5]: Expectation[3 x y + 5, {x, y} ≈ loi]
```

Out[5]: 33

```
In[6]: Covariance[loi] //MatrixForm
```

Out[6]: 
$$\begin{pmatrix} \frac{275}{24} & \frac{35}{12} \\ \frac{144}{35} & \frac{24}{35} \end{pmatrix}$$

### Exercice 99

Soit  $X$  une var dont la loi est donnée par

$$\mathbb{P}(X = -1) = 0.3, \quad \mathbb{P}(X = 0) = 0.6, \quad \mathbb{P}(X = 1) = 0.3.$$

On pose  $Y = 1 - |X|$ .

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$ ,
- calculer la matrice de covariance de  $(X, Y)$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes.

Commandes : EmpiricalDistribution, TransformedDistribution, Abs,  $\approx$ , PDF, Boole, Covariance, //MatrixForm, MarginalDistribution.

**Solution 99.** On propose :

```
In[1]: loi = EmpiricalDistribution[{0.3, 0.4, 0.3} → {-1, 0, 1}];
```

```
In[2]: loi2 = TransformedDistribution[{x, 1 - Abs[x]}, x ≈ loi];
```

```
In[3]: PDF[loi2, {i, j}]
```

```
Out[3]: 0.3 Boole[{i, j} == {-1, 0}] + 0.4 Boole[{i, j} == {0, 1}]
        + 0.3 Boole[{i, j} == {1, 0}]
```

```
In[4]: Covariance[loi2] //MatrixForm
```

```
Out[4]:  $\begin{pmatrix} 0.6 & 0. \\ 0. & 0.24 \end{pmatrix}$ 
```

```
In[5]: PDF[loi2, {1, 1}] == PDF[MarginalDistribution[loi2, 1], 1] *
        PDF[MarginalDistribution[loi2, 2], 1]
```

```
Out[5]: False
```

!! Ainsi,  $\mathbb{P}(\{X = 1\} \cap \{Y = 1\}) \neq \mathbb{P}(X = 1)\mathbb{P}(Y = 1)$ ;  $X$  et  $Y$  ne sont pas indépendantes.

### Exercice 100

Soient  $n \in \mathbb{N}^*$  et  $(X, Y)$  un couple de *var* suivant la loi multinomiale  $\mathcal{M}(n, 0.6, 0.4)$  :

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \binom{n}{j} 0.6^i 0.4^j, \quad (i, j) \in \mathcal{D},$$

avec  $\mathcal{D} = \{(i, j) \in \{0, \dots, n\}^2; i + j = n\}$ .

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$  avec la commande MultinomialDistribution,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- représenter graphiquement la fonction de répartition de  $(X, Y)$  pour  $n = 10$  avec les couleurs de SolarColors,
- calculer  $\mathbb{P}(\sqrt{|X - Y|} \leq 1)$  pour  $n = 6$ ,

◦ calculer  $\mathbb{C}(X, Y)$ .

Commandes : `MultinomialDistribution`, `MarginalDistribution`, `DiscretePlot3D`, `CDF`, `/. →`, `ColorFunction`, `ExtentSize`, `Probability`, `Sqrt`, `Abs`, `Covariance`.

**Solution 100.** On propose :

```
In[1]: loi = MultinomialDistribution[n, {0.6, 0.4}];
```

```
In[2]: MarginalDistribution[loi, 1]
```

```
Out[2]: BinomialDistribution[n, 0.6]
```

```
In[3]: MarginalDistribution[loi, 2]
```


```
Out[3]: BinomialDistribution[n, 0.4]
```

```
In[4]: MarginalDistribution[loi, 2]
```

```
Out[4]: BinomialDistribution[n, 0.4]
```

```
In[5]: DiscretePlot3D[CDF[loi /. n → 10, {i, j}], {i, 0, 10}, {j, 0, 10},
ColorFunction → "SolarColors", ExtentSize → 0.5]
```

```
Out[5]:
```



```
In[6]: Probability[Sqrt[Abs[x - y]] ≤ 1, {x, y} ≈ loi /. n → 6]
```

```
Out[6]: 0.27648
```

```
In[7]: Covariance[loi, 1, 2]
```

```
Out[7]: -0.24n
```

**Exercice 101**

Soit  $X$  une *var* dont la loi est donnée par le tableau :

$i$	-2	-1	0	1	2
$\mathbb{P}(X = i)$	$\frac{1}{6}$	$\frac{1}{4}$	$\frac{1}{6}$	$\frac{1}{4}$	$\frac{1}{6}$

On pose  $Y = X^2$ .

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,
- calculer  $\mathbb{P}(X + Y \geq 1)$ ,
- calculer  $\mathbb{C}(X, Y)$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes.

Commandes : `EmpiricalDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Boole`, `Probability`, `Covariance`, `MarginalDistribution`.

**Solution 101.** On propose :

```
In[1]: loi = EmpiricalDistribution[{1/6, 1/4, 1/6, 1/4, 1/6} → {-2, -1, 0, 1, 2}];
```

```
In[2]: loi2 = TransformedDistribution[{x, x^2}, x ≈ loi];
```

```
In[3]: PDF[loi2, {i, j}]
```

```
Out[3]:  $\frac{1}{6} \text{Boole}[\{i, j\} == \{-2, 4\}] + \frac{1}{4} \text{Boole}[\{i, j\} == \{-1, 1\}] +$   

 $\frac{1}{6} \text{Boole}[\{i, j\} == \{0, 0\}] + \frac{1}{4} \text{Boole}[\{i, j\} == \{1, 1\}] +$   

 $\frac{1}{6} \text{Boole}[\{i, j\} == \{2, 4\}]$ 
```

```
In[4]: Probability[x + y ≥ 1, {x, y} ≈ loi2]
```

```
Out[4]:  $\frac{7}{12}$ 
```

```
In[5]: Covariance[loi2, 1, 2]
```

```
Out[5]: 0
```

```
In[6]: PDF[loi2, {0, 1}] == PDF[MarginalDistribution[loi2, 1], 0]  

PDF[MarginalDistribution[loi2, 2], 1]
```

```
Out[6]: False
```

!! Ainsi,  $\mathbb{P}(\{X = 0\} \cap \{Y = 1\}) \neq \mathbb{P}(X = 0)\mathbb{P}(Y = 1)$ ;  $X$  et  $Y$  ne sont pas indépendantes.



**Exercice 102**

Soient  $(p, q) \in ]0, 1[^2$ , et  $X$  et  $Y$  deux *var* indépendantes telles que

◊  $X$  suit la loi de Bernoulli  $\mathcal{B}(p)$  :

$$\mathbb{P}(X = 0) = 1 - p, \quad \mathbb{P}(X = 1) = p.$$

◊  $Y$  suit la loi de Bernoulli  $\mathcal{B}(1 - p)$  :

$$\mathbb{P}(Y = 0) = p, \quad \mathbb{P}(Y = 1) = 1 - p.$$

On pose  $W = \cos(\pi XY)$  et  $Z = X - Y$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $W$ ,
- déterminer la loi de  $Z$ ,
- calculer la matrice de covariance de  $(W, Z)$ .

Commandes : `TransformedDistribution`, `Cos`, `Pi`, `BernoulliDistribution`, `PDF`, `Covariance`,  
`//MatrixForm`.

**Solution 102.** On propose :

```
In[1]: loi1 = TransformedDistribution[Cos[Pi x y],
    {x ≈ BernoulliDistribution[p], y ≈ BernoulliDistribution[1 - p]}];
```

```
In[2]: PDF[loi1, i]
```

```
Out[2]: {
  -(-1 + p)p   i == -1
  1 + (-1 + p)p i == 1
  0           True
```

```
In[3]: loi2 = TransformedDistribution[x - y,
```

```
{x ≈ BernoulliDistribution[p], y ≈ BernoulliDistribution[1 - p]}];
```

```
In[4]: PDF[loi2, j]
```

```
Out[4]: {
  (-1 + p)2   j == -1
  -2(-1 + p)p j == 0
  p2         j == 1
  0           True
```

```
In[5]: loi3 = TransformedDistribution[{Cos[Pi x y], x - y},
    {x ≈ BernoulliDistribution[p], y ≈ BernoulliDistribution[1 - p]};

In[6]: Covariance[loi3] //MatrixForm

Out[6]: 
$$\begin{pmatrix} -4p(-1 + 2p - 2p^2 + p^3) & -2p(1 - 3p + 2p^2) \\ -2p(1 - 3p + 2p^2) & -2(-p + p^2) \end{pmatrix}$$

```

**Exercice 103**

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \frac{i + j}{e^{2^{i+j}} i! j!}, \quad (i, j) \in \mathbb{N}^2.$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,
- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  pour  $(i, j) \in \{0, \dots, 4\}^2$ ,
- calculer  $\mathbb{E}(2^{X+Y})$ ,
- calculer  $\mathbb{C}(X, Y)$ .


Commandes : *ProbabilityDistribution*, *E*!, *DiscretePlot3D*, *PDF*, *ExtentSize*, *Expectation*, *≈*, *Covariance*.

**Solution 103.** On propose :

```
In[1]: loi = ProbabilityDistribution[(i + j) / (E (2^(i + j) i! j!)),
    {i, 0, ∞, 1}, {j, 0, ∞, 1}];

In[2]: DiscretePlot3D[PDF[loi, {i, j}], {i, 0, 4}, {j, 0, 4}, ExtentSize → Full]

Out[2]:
```



In[3]: Expectation[2^(x + y), {x, y} ≈ loi]

Out[3]: 2e

In[4]: Covariance[loi, 1, 2]

Out[4]:  $-\frac{1}{4}$

### Exercice 104

Soient  $n \in \mathbb{N}$  et  $X$  et  $Y$  deux *var iid* suivant chacune la loi binomiale  $\mathcal{B}(n, \frac{1}{2})$  :

$$\mathbb{P}(X = k) = \binom{n}{k} \frac{1}{2^n}, \quad k \in \{0, \dots, n\}.$$

Utiliser *Mathematica* pour calculer

$$\mathbb{E} \left( \frac{(-1)^X}{1 + X + Y + XY} \right).$$

Commandes : *BinomialDistribution*, *Expectation*, *≈* et *Timing*.

**Solution 104.** On propose :

In[1]: loi = BinomialDistribution[n, 1/2];

In[2]: Expectation[(-1)^x / (1 + x + y + x y), {x ≈ loi, y ≈ loi}]

Out[2]:  $\frac{4^{-n}(-1 + 2^{1+n})}{(1 + n)^2}$

!! Intéressons nous au temps de calcul :

In[3]: Timing[Expectation[(-1)^x / (1 + x + y + x y), {x ≈ loi, y ≈ loi}]]

Out[3]:  $\left\{ 0.0780005, \frac{4^{-n}(-1 + 2^{1+n})}{(1 + n)^2} \right\}$

*Mathematica* a donc mis 0.078 secondes. Exceptionnellement, pour juger du niveau de complexité du calcul, regardons la preuve "à la main" :

Par l'indépendance de  $X$  et  $Y$ , on a

$$\mathbb{E} \left( \frac{(-1)^X}{1 + X + Y + XY} \right) = \mathbb{E} \left( \frac{(-1)^X}{(1 + X)(1 + Y)} \right) = \mathbb{E} \left( \frac{(-1)^X}{1 + X} \right) \mathbb{E} \left( \frac{1}{1 + Y} \right).$$

Par l'égalité :  $\frac{1}{k+1} \binom{n}{k} = \frac{1}{n+1} \binom{n+1}{k+1}$ , un changement d'indice et la formule du binôme de Newton, il vient

$$\begin{aligned}
\mathbb{E} \left( \frac{(-1)^X}{1+X} \right) &= \sum_{k \in X(\Omega)} \frac{(-1)^k}{1+k} \mathbb{P}(X = k) = \sum_{k=0}^n \frac{(-1)^k}{1+k} \binom{n}{k} \frac{1}{2^n} \\
&= \frac{1}{2^n(n+1)} \sum_{k=0}^n (-1)^k \binom{n+1}{k+1} = \frac{1}{2^n(n+1)} \sum_{k=1}^{n+1} (-1)^{k-1} \binom{n+1}{k} \\
&= \frac{1}{2^n(n+1)} \left( \sum_{k=0}^{n+1} \binom{n+1}{k} (-1)^{k-1} + 1 \right) \\
&= \frac{1}{2^n(n+1)} \left( - \sum_{k=0}^{n+1} \binom{n+1}{k} (-1)^k 1^{(n+1)-k} + 1 \right) \\
&= \frac{1}{2^n(n+1)} ( -(-1+1)^{n+1} + 1 ) = \frac{1}{2^n(n+1)}.
\end{aligned}$$

En utilisant des techniques similaires au calcul ci-dessus, on obtient

$$\begin{aligned}
\mathbb{E} \left( \frac{1}{1+Y} \right) &= \sum_{k \in Y(\Omega)} \frac{1}{1+k} \mathbb{P}(Y = k) = \sum_{k=0}^n \frac{1}{1+k} \binom{n}{k} \frac{1}{2^n} \\
&= \frac{1}{2^n(n+1)} \sum_{k=0}^n \binom{n+1}{k+1} = \frac{1}{2^n(n+1)} \sum_{k=1}^{n+1} \binom{n+1}{k} \\
&= \frac{1}{2^n(n+1)} \left( \sum_{k=0}^{n+1} \binom{n+1}{k} - 1 \right) \\
&= \frac{1}{2^n(n+1)} \left( \sum_{k=0}^{n+1} \binom{n+1}{k} 1^k 1^{(n+1)-k} - 1 \right) \\
&= \frac{1}{2^n(n+1)} ((1+1)^{n+1} - 1) = \frac{2^{n+1} - 1}{2^n(n+1)}.
\end{aligned}$$

Au final, on a

$$\mathbb{E} \left( \frac{(-1)^X}{1+X+Y+XY} \right) = \frac{1}{2^n(n+1)} \times \frac{2^{n+1} - 1}{2^n(n+1)} = \frac{2^{n+1} - 1}{2^{2n}(n+1)^2} = \frac{4^{-n}(-1+2^{1+n})}{(1+n)^2}.$$

□

**Exercice 105**

Soient  $p \in ]0, 1[$ , et  $X$  et  $Y$  deux *var* indépendantes telles que

◊  $X$  suit la loi de Bernoulli  $\mathcal{B}(p)$  :

$$\mathbb{P}(X = 0) = 1 - p, \quad \mathbb{P}(X = 1) = p,$$

◊ la loi de  $Y$  est donnée par

$$\mathbb{P}(Y = k) = \frac{k}{55}, \quad k \in \{1, \dots, 10\}.$$

On pose  $Z = (-X)^Y$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $Z$ ,
- calculer  $\mathbb{E}(Z)$ ,
- calculer  $\mathbb{V}(Z)$ .

Commandes : BernoulliDistribution, ProbabilityDistribution, TransformedDistribution,  $\approx$ , PDF, Mean, Variance.

**Solution 105.** On propose :

```
In[1]: loi1 = BernoulliDistribution[p];
```

```
In[2]: loi2 = ProbabilityDistribution[k / 55, {k, 1, 10, 1}];
```

```
In[3]: loi3 = TransformedDistribution[(-x)^y, {x ≈ loi1, y ≈ loi2}];
```

```
In[4]: PDF[loi3, k]
```

```
Out[4]: {
  1 - p   k == 0
  5p/11   k == -1
  6p/11   k == 1
  0       True
}
```

```
In[5]: Mean[loi3]
```

```
Out[5]: p/11
```

```
In[6]: Variance[loi3]
```

```
Out[6]: p - p^2/121
```

**Exercice 106**

Soient  $X$  et  $Y$  deux *var* indépendantes telles que

◊  $X$  suit la loi uniforme  $\mathcal{U}(\{1, 2, 3\})$  :

$$\mathbb{P}(X = k) = \frac{1}{3}, \quad k \in \{1, 2, 3\},$$

◊ la loi de  $Y$  est donnée par

$$\mathbb{P}(Y = k) = \frac{2+k}{10}, \quad k \in \{-1, 0, 1, 2\}.$$

Utiliser *Mathematica* pour déterminer la loi de  $Z = X + Y$ .

Commandes : `DiscreteUniformDistribution`, `ProbabilityDistribution`, `TransformedDistribution`, `≈`, `PDF`.

**Solution 106.** On propose :

```
In[1]: loi1 = DiscreteUniformDistribution[{1, 3}];
In[2]: loi2 = ProbabilityDistribution[(2 + k) / 10, {k, -1, 2, 1}];
In[3]: loi3 = TransformedDistribution[x + y, {x ≈ loi1, y ≈ loi2}]
In[4]: PDF[loi3, k]
```

```
Out[4]: {
  1/30 k == 0
  1/10 k == 1
  2/15 k == 5
  1/5 k == 2
  7/30 k == 4
  3/10 k == 3
  0 True
```

**Exercice 107**

Soient  $p \in ]0, 1[$ , et  $X$  et  $Y$  deux *var iid* suivant chacune la loi géométrique  $\mathcal{G}(p)$  :

$$\mathbb{P}(X = k) = p(1-p)^k, \quad k \in \mathbb{N}.$$

Utiliser Mathematica pour

- calculer  $\mathbb{P}(Y < X)$ ,
- calculer  $\mathbb{E}((X + Y)^3)$ ,
- calculer  $\mathbb{E}(e^{X+Y})$ .

Commandes : GeometricDistribution, Probability,  $\approx$ , Expectation, Exp.

**Solution 107.** On propose :

In[1]: `loi = GeometricDistribution[p];`

In[2]: `Probability[x < y, {x  $\approx$  loi, y  $\approx$  loi}]`

Out[2]: 
$$\frac{-1 + p}{-2 + p}$$

In[3]: `Expectation[(x + y)^3, {x  $\approx$  loi, y  $\approx$  loi}]`

Out[3]: 
$$-\frac{2(-12 + 27p - 19p^2 + 4p^3)}{p^3}$$

In[4]: `Expectation[Exp[x + y], {x  $\approx$  loi, y  $\approx$  loi}]`

Out[4]: 
$$\frac{p^2}{(1 + e(-1 + p))^2}$$

### Exercice 108

Soient  $X$  et  $Y$  deux *var* indépendantes. Utiliser Mathematica pour vérifier les résultats en loi du tableau :

loi de $X$	loi de $Y$	loi de $X + Y$
binomiale $\mathcal{B}(n_1, p)$	binomiale $\mathcal{B}(n_2, p)$	binomiale $\mathcal{B}(n_1 + n_2, p)$
Poisson $\mathcal{P}(\lambda_1)$	Poisson $\mathcal{P}(\lambda_2)$	Poisson $\mathcal{P}(\lambda_1 + \lambda_2)$

Commandes : TransformedDistribution,  $\approx$ ,  $n_1$ ,  $\lambda_1$ , BinomialDistribution, PoissonDistribution.

**Solution 108.** !! Pour écrire  $x_i$  ( $x$  indice  $i$ ) :

- soit on tape `[x]`, on maintient `[Ctrl]`, puis `[8 _ \]` et enfin on tape `[i]`,
- soit on utilise la combinaison de touches "Ctrl moins" du clavier alphanumérique : `[Ctrl] + [-]`,
- soit on écrit `Subscript[x, i]`.

On propose :

In[1]: `TransformedDistribution[x + y, {x  $\approx$  BinomialDistribution[n1, p],  
y  $\approx$  BinomialDistribution[n2, p]}]`

Out[1]: `BinomialDistribution[n1 + n2, p]`

```
In[2]: TransformedDistribution[x + y, {x ≈ PoissonDistribution[λ1],
      y ≈ PoissonDistribution[λ2]}]
```

```
Out[2]: PoissonDistribution[λ1 + λ2]
```

**Exercice 109**

Soient  $p \in ]0, 1[$ , et  $X$  et  $Y$  deux *var iid* suivant chacune la loi géométrique  $\mathcal{G}(p)$  :

$$\mathbb{P}(X = k) = p(1 - p)^k, \quad k \in \mathbb{N}.$$

On pose  $Z = \inf(X, Y)$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $Z$ ,
- calculer  $\mathbb{E}(Z)$ ,
- calculer  $\mathbb{V}(Z)$ .

Commandes : `GeometricDistribution`, `TransformedDistribution`, `Min`, `≈`, `Expectation`, `Variance`, `OrderDistribution`.

**Solution 109.** On propose :

```
In[1]: loi = GeometricDistribution[p];
```

```
In[2]: loi2 = TransformedDistribution[Min[x, y], {x ≈ loi, y ≈ loi}]
```

```
Out[2]: GeometricDistribution[1 - (1 - p)2]
```

```
In[3]: Expectation[loi2]
```

```
Out[3]: -1 +  $\frac{1}{p}$ 
```

```
In[4]: Variance[loi2]
```

```
Out[4]:  $\frac{1 - p}{p^2}$ 
```

!! On aurait aussi pu faire :

```
In[1]: loi = GeometricDistribution[p];
```

```
In[2]: loi2 = OrderDistribution[{loi, 2}, 1]
```

**Exercice 110**

Soient  $\lambda > 0$ , et  $X$  et  $Y$  deux *var iid* suivant chacune la loi de Poisson  $\mathcal{P}(\lambda)$  :

$$\mathbb{P}(X = k) = e^{-\lambda} \frac{\lambda^k}{k!}, \quad k \in \mathbb{N}.$$



Utiliser *Mathematica* pour déterminer la loi de  $Z = \begin{cases} 1 & \text{si } XY = 1, \\ 0 & \text{si } XY \neq 1 \end{cases}$

Commandes : `PoissonDistribution`,  $\lambda$ , `TransformedDistribution`, `Boole`,  $\approx$  et `PDF`.

**Solution 110.** On propose :

```
In[1]: loi = PoissonDistribution[λ];
```

```
In[2]: loi2 = TransformedDistribution[Boole[x y == 1], {x ≈ loi, y ≈ loi}];
```

```
In[3]: PDF[loi2, k]
```

```
Out[3]: { e-2λλ2      k == 1
          1 - e-2λλ2  k == 0
          0              True
```

!! On reconnaît la loi de Bernoulli  $\mathcal{B}(e^{-2\lambda}\lambda^2)$ .

### Exercice 111

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \frac{i + j}{2e^2 i!j!}, \quad (i, j) \in \mathbb{N}^2.$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,
- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  pour  $(i, j) \in \{0, \dots, 4\}^2$ ,
- calculer  $\mathbb{P}(\{X \leq 2\} \cap \{Y \leq 2\})$  avec
  - ◇ la commande `CDF`,
  - ◇ la commande `Probability`,
- calculer la valeur numérique de  $\mathbb{P}(\{1 \leq X \leq 3\} \cap \{2 \leq Y \leq 4\})$ ,
- calculer la valeur numérique de  $\mathbb{P}(\{X > 6\} \cup \{Y < 4\})$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\sigma(X)$  et  $\sigma(Y)$ ,
- calculer le coefficient de corrélation linéaire :  $\rho(X, Y)$ .


Commandes : `ProbabilityDistribution`, `!`,  $\infty$ , `DiscretePlot3D`, `PDF`, `ExtentSize`, `CDF`, `Probability`, `&&`,  $\approx$ , `NProbability`, `||` `Mean`, `StandardDeviation`, `Correlation`.

**Solution 111.** On propose :

```
In[1]: loi = ProbabilityDistribution[(i + j) / (2 Exp[2] i! j!),
  {i, 0, ∞, 1}, {j, 0, ∞, 1}];
```

```
In[2]: DiscretePlot3D[PDF[loi, {i, j}], {i, 0, 4}, {j, 0, 4}, ExtentSize → 0.5]
```

Out [2]:



```
In[3]: CDF[loi, {2, 2}]
```

Out [3]:  $\frac{5}{e^2}$

```
In[4]: Probability[x ≤ 2 && y ≤ 2, {x, y} ≈ loi]
```

Out [4]:  $\frac{5}{e^2}$

```
In[5]: NProbability[1 ≤ x ≤ 3 && 2 ≤ y ≤ 4, {x, y} ≈ loi]
```

Out [5]: 0.307794

```
In[6]: NProbability[x > 6 || y < 4, {x, y} ≈ loi]
```

Out [6]: 0.950364

```
In[7]: Mean[loi]
```

Out [7]:  $\left\{ \frac{3}{2}, \frac{3}{2} \right\}$

```
In[8]: StandardDeviation[loi]
```

Out [8]:  $\left\{ \frac{\sqrt{5}}{2}, \frac{\sqrt{5}}{2} \right\}$

```
In[9]: Correlation[loi, 1, 2]
```

```
Out[9]:  $-\frac{1}{5}$ 
```

### Exercice 112

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par la commande Mathematica :

```
MultivariatePoissonDistribution[2, {1, 3}]
```

Utiliser Mathematica pour

- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  pour  $(i, j) \in \{0, \dots, 8\}^2$ ,
- déterminer la loi (usuelle) de  $X$ ,
- déterminer la loi (usuelle) de  $Y$ ,
- calculer le coefficient de corrélation linéaire :  $\rho(X, Y)$ ,
- calculer la valeur numérique de  $\mathbb{P}(\sup(2X, Y) \geq 3)$ .


Commandes : MultivariatePoissonDistribution, DiscretePlot3D, PDF, ExtentSize, MarginalDistribution, Correlation, NProbability, Max,  $\approx$ .

**Solution 112.** On propose :

```
In[1]: loi = MultivariatePoissonDistribution[2, {1, 3}];
```

```
In[2]: DiscretePlot3D[PDF[loi, {i, j}], {i, 0, 8}, {j, 0, 8}, ExtentSize  $\rightarrow$  0.5]
```

```
Out[2]:
```



```
In[3]: loi1 = MarginalDistribution[loi, 1]
```

```
Out[3]: PoissonDistribution[3]
```

```

In[4]: loi2 = MarginalDistribution[loi, 2]
Out[4]: PoissonDistribution[5]
In[5]: Correlation[loi, 1, 2]
Out[5]:  $\frac{2}{\sqrt{15}}$ 
In[6]: NProbability[Max[2 x, y] ≥ 3, {x, y} ≈ loi]
Out[6]: 0.608334

```

**Exercice 113**

Soient  $n \in \mathbb{N}^*$ ,  $p \in ]0, 1[$ ,  $\lambda > 0$  et  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \binom{n}{i} p^i (1-p)^{n-i} e^{-\lambda} \frac{\lambda^j}{j!}, \quad (i, j) \in \{0, \dots, n\} \times \mathbb{N}.$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$  avec la commande `ProductDistribution`,
- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  avec  $n = 5$ ,  $p = 0.2$  et  $\lambda = 2$  pour  $(i, j) \in \{0, \dots, 8\}^2$ , et les couleurs de l'arc-en-ciel,
- déterminer la fonction génératrice des moments de  $(X, Y)$ ,
- calculer la valeur numérique de  $\mathbb{P}(X < Y)$  avec  $n = 10$ ,  $p = 0.48$  et  $\lambda = 2$ ,
- calculer la valeur numérique de  $\mathbb{P}(X - Y^2 \geq 5)$  avec  $n = 6$ ,  $p = 0.36$  et  $\lambda = 3$ .

Commandes : `BinomialDistribution`, `PoissonDistribution`, `λ`, `ProductDistribution`, `DiscretePlot3D`, `PDF`, `/.`, `→`, `ExtentSize`, `ColorFunction`, `MomentGeneratingFunction`, `NProbability`, `≈`.


**Solution 113.** On propose :

```

In[1]: loi1 = BinomialDistribution[n, p];
In[2]: loi2 = PoissonDistribution[λ];
In[3]: loi3 = ProductDistribution[loi1, loi2];
In[4]: DiscretePlot3D[PDF[loi3, {i, j}] /. {n → 5, p → 0.2, λ → 2},
  {i, 0, 8}, {j, 0, 8}, ExtentSize → 0.5, ColorFunction → "Rainbow"]

```

Out [4] :



In[5]: `MomentGeneratingFunction[loi3, {s, t}]`

Out [5]:  $e^{(-1+e^t)\lambda}(1 + (-1 + e^s)p)^n$

In[6]: `NProbability[x < y, {x, y} ≈ loi3 /. {n → 10, p → 0.48, λ → 2}]`

Out [6]: 0.063207

In[7]: `NProbability[x - y^2 ≥ 5, {x, y} ≈ loi3 /. {n → 6, p → 0.36, λ → 3}]`

Out [7]: 0.00158951

#### Exercice 114

Soit  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = e^{-2} \frac{2^i}{i!} e^{-1} (1 - e^{-1})^j, \quad (i, j) \in \mathbb{N}^2.$$

Utiliser *Mathematica* pour


- créer la loi de  $(X, Y)$  avec la commande `ProductDistribution`,
- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  pour  $(i, j) \in \{0, \dots, 5\}^2$ , et les couleurs de `MintColors`,
- calculer la valeur numérique de  $\mathbb{P}(X^2 + |XY| \geq 5)$ ,
- calculer la valeur numérique de  $\mathbb{E}((X + Y)^2)$ ,
- déterminer la fonction génératrice des moments de  $(X, Y)$ .

Commandes : `PoissonDistribution`, `GeometricDistribution`, `Exp`, `ProductDistribution`, `DiscretePlot3D`, `PDF`, `ExtentSize`, `ColorFunction`, `NProbability`, `Abs`, `≈`, `Expectation`, `MomentGeneratingFunction`.

**Solution 114.** On propose :

```
In[1]: loi1 = PoissonDistribution[2];
In[2]: loi2 = GeometricDistribution[Exp[-1]];
In[3]: loi3 = ProductDistribution[loi1, loi2];
In[4]: DiscretePlot3D[PDF[loi3, {i, j}], {i, 0, 5}, {j, 0, 5},
      ExtentSize → 0.5, ColorFunction → "MintColors"]
```

Out[4]:



```
In[5]: NProbability[x^2 + Abs[x y] ≥ 5, {x, y} ≈ loi3]
```

Out[5]: 0.537636

```
In[6]: Expectation[(x + y)^2, {x, y} ≈ loi3] //N
```

Out[6]: 20.4964

```
In[7]: MomentGeneratingFunction[loi3, {s, t}]
```

Out[7]: 
$$\frac{e^{-1+2(-1+e^s)}}{1 - \left(1 - \frac{1}{e}\right) e^t}$$

### Exercice 115

Soient  $p \in ]0, 1[$  et  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = p^2(1 - p)^{j-2}, \quad (i, j) \in \mathcal{D},$$

où  $\mathcal{D} = \{(i, j) \in \mathbb{N}^2; 1 \leq i < j\}$ .

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,

- créer la loi de  $X + Y$ ,
- vérifier que  $\mathbb{E}(X + Y) = \mathbb{E}(X) + \mathbb{E}(Y)$ ,
- vérifier que  $\mathbb{V}(X + Y) = \mathbb{V}(X) + \mathbb{V}(Y) + 2\mathbb{C}(X, Y)$ .

Commandes : ProbabilityDistribution, Boole, Assumptions, TransformedDistribution, Mean, Variance, Covariance, //Simplify.

**Solution 115.** On propose :

```
In[1]: loi = ProbabilityDistribution[p^2 (1 - p)^(j-2) Boole[i < j],
      {i, 1, ∞, 1}, {j, 1, ∞, 1}, Assumptions → 0 < p < 1];
```

```
In[2]: loi2 = TransformedDistribution[x + y, {x, y} ≈ loi];
```

```
In[3]: Mean[loi2] == Mean[loi][[1]] + Mean[loi][[2]]
```

Out [3]: True

```
In[4]: Variance[loi2] == Covariance[loi, 1, 1] + Covariance[loi, 2, 2] +
      2 Covariance[loi, 1, 2] //Simplify
```

Out [4]: True

### Exercice 116

Soient  $p \in ]0, 1[$  et  $(X, Y)$  un couple de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\}) = \frac{(1-p)^2}{p^2} p^j, \quad (i, j) \in \mathcal{D},$$

où  $\mathcal{D} = \{(i, j) \in \mathbb{N}^2; 1 \leq i < j\}$ .

Utiliser Mathematica pour

- vérifier que  $\sum_{i=1}^{\infty} \sum_{j=1}^{\infty} \mathbb{P}(\{X = i\} \cap \{Y = j\}) = 1$ ,
- créer la loi de  $(X, Y)$ ,
- représenter graphiquement  $\mathbb{P}(\{X = i\} \cap \{Y = j\})$  pour  $(i, j) \in \{1, \dots, 7\}^2$ , avec  $p = 0.55$  et les couleurs de NeonColors,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- calculer la matrice de covariance de  $(X, Y)$ .

Commandes : Sum, Boole, ∞, ProbabilityDistribution, Assumptions, DiscretePlot3D, PDF, /. →, ExtentSize, ColorFunction, MarginalDistribution, Covariance, //MatrixForm.

**Solution 116.** On propose :


```
In[1]: Sum[(1 - p)^2 / p^2 p^j Boole[i < j], {i, 1, ∞, 1}, {j, 1, ∞, 1}] == 1
```

```
Out[1]: True
```

```
In[2]: loi = ProbabilityDistribution[(1 - p)^2 / p^2 p^j Boole[i < j],
{i, 1, ∞, 1}, {j, 1, ∞, 1}, Assumptions → 0 < p < 1];
```

```
In[3]: DiscretePlot3D[PDF[loi, {i, j}] /. p → 0.55, {i, 1, 7}, {j, 1, 7},
ExtentSize → 0.5, ColorFunction → "NeonColors"]
```

```
Out[3]:
```



```
In[4]: PDF[MarginalDistribution[loi, 1], i]
```

```
Out[4]: 
$$\begin{cases} -(-1 + p)p^{-1+i} & i \geq 1 \\ 0 & \text{True} \end{cases}$$

```

```
In[5]: PDF[MarginalDistribution[loi, 2], j]
```

```
Out[5]: 
$$\begin{cases} (-1 + j)(-1 + p)^2 p^{-2+j} & j > 1 \\ 0 & \text{True} \end{cases}$$

```

```
In[6]: Covariance[loi] //MatrixForm
```

```
Out[6]: 
$$\begin{pmatrix} \frac{p}{(-1 + p)^2} & \frac{p}{(-1 + p)^2} \\ \frac{p}{(-1 + p)^2} & \frac{2p}{(-1 + p)^2} \end{pmatrix}$$

```





## 7 Vecteurs de var discrètes et convergence

### Exercice 117

On sait qu'il existe un unique réel  $b \in ]0, 1[$  tel que la loi d'un vecteur de var  $(X, Y, Z)$  soit donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = b \left(\frac{1}{2}\right)^{i+j+k} \frac{1}{k!}, \quad (i, j, k) \in \mathbb{N}^3.$$

Utiliser *Mathematica* pour déterminer  $b$ .

Commandes : `Sum, !, ∞, N[%]`.

**Solution 117.** On propose :

`In[1]: b = 1 / Sum[(1 / 2)^(i + j + k) 1 / k!, {i, 0, ∞}, {j, 0, ∞}, {k, 0, ∞}]`

`Out[1]:  $\frac{1}{4\sqrt{e}}$`

`In[2]: N[%]`

`Out[2]: 0.151633`

### Exercice 118

Soit  $n \in \mathbb{N}^*$ . On sait qu'il existe un unique réel  $b \in ]0, 1[$  tel que la loi d'un vecteur de var  $(X, Y, Z)$  soit donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = b \left( j + \frac{k}{i+1} \binom{n}{i} \binom{n}{k} \right), \quad (i, j, k) \in \{0, \dots, n\}^3.$$

Utiliser *Mathematica* pour déterminer  $b$ .

Commandes : `Sum, Binomial`.

**Solution 118.** On propose :

`In[1]: b = 1 / Sum[j + k / (i + 1) Binomial[n, i] Binomial[n, k], {i, 0, n}, {j, 0, n}, {k, 0, n}]`

`Out[1]:  $\frac{2}{n(1 - 2^n + 2^{1+2n} + 3n + 3n^2 + n^3)}$`

**Exercice 119**

Soient  $n \in \mathbb{N} \setminus \{0, 1\}$  et  $(X, Y, Z)$  un vecteur de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = ci(n - j)(2n - j - k), \quad (i, j, k) \in \{0, \dots, n\}^3,$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\sum_{i=0}^n \sum_{j=0}^n \sum_{k=0}^n \mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = 1$ .

Utiliser **Mathematica** pour

- déterminer  $c$ ,
- créer la loi de  $(X, Y, Z)$  avec la constante  $c$  déterminée précédemment,
- calculer  $\mathbb{P}(X = Y = Z)$ ,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- déterminer la loi de  $Z$ ,
- déterminer la loi de  $(X, Y)$ ,
- calculer la matrice de covariance de  $(X, Y, Z)$  pour  $n = 4$ .

Commandes : `Sum`, `ProbabilityDistribution`, `Assumptions`, `Probability`, `≈`, `PDF`, `MarginalDistribution`, `Covariance`, `/.`, `→`, `//MatrixForm`.

**Solution 119.** On propose :

```
In[1]: c = 1 / Sum[i (n - j) (2 n - j - k), {i, 0, n}, {j, 0, n}, {k, 0, n}]
```

```
Out[1]: 
$$\frac{24}{n^2(1+n)^3(2+7n)}$$

```

```
In[2]: loi = ProbabilityDistribution[ c i (n - j) (2 n - j - k), {i, 0, n, 1},
    {j, 0, n, 1}, {k, 0, n, 1}, Assumptions → n ≥ 2]
```

```
In[3]: Probability[x == y == z, {x, y, z} ≈ loi]
```

```
Out[3]: 
$$\frac{4(-1+n)}{(1+n)^2(2+7n)}$$

```

```
In[4]: PDF[MarginalDistribution[loi, 1], i]
```

```
Out[4]: 
$$\begin{cases} \frac{2i}{n+n^2} & i \geq 0 \ \&\& \ n \geq i \\ 0 & \text{True} \end{cases}$$

```

```
In[5]: PDF[MarginalDistribution[loi, 2], j]
```

```
Out[5]: 
$$\begin{cases} \frac{6(-j+n)(-2j+3n)}{n(1+n)(2+7n)} & n \geq 2 \ \&\& \ j \geq 0 \ \&\& \ j - n \leq 0 \\ 0 & \text{True} \end{cases}$$

```

```

In[6]: PDF[MarginalDistribution[loi, 3], k]
Out[6]: 
$$\begin{cases} \frac{2 - 6k + 10n}{2 + 9n + 7n^2} & k \geq 0 \ \&\& \ k \leq n \\ 0 & \text{True} \end{cases}$$

In[7]: PDF[MarginalDistribution[loi, {1, 2}], {i, j}]
Out[7]: 
$$\begin{cases} \frac{12i(-j+n)(-2j+3n)}{n^2(1+n)^2(2+7n)} & n \geq 2 \ \&\& \ i \geq 0 \ \&\& \ i - n \leq 0 \ \&\& \ j \geq 0 \ \&\& \ j - n \leq 0 \\ 0 & \text{True} \end{cases}$$

In[8]: Covariance[loi] /. n -> 4 //MatrixForm
Out[8]: 
$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{21}{25} & -\frac{2}{25} \\ 0 & -\frac{2}{25} & \frac{46}{25} \end{pmatrix}$$


```

**Exercice 120**

Soit  $(X, Y, Z)$  un vecteur de *var* dont la loi est donnée par

$$\mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = 0.31(1 - 0.31)^i \times 0.23(1 - 0.23)^j e^{0.63} \frac{0.63^k}{k!}, \quad (i, j, k) \in \mathbb{N}^3.$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y, Z)$  avec la commande `ProductDistribution`,
- calculer  $\mathbb{P}(\{X = 1\} \cap \{Y = 1\} \cap \{Z = 1\})$ ,
- calculer  $\mathbb{P}(\{X \geq 1\} \cup \{Y \geq 1\} \cup \{Z \geq 1\})$ ,
- calculer  $\mathbb{P}(X + Y + Z = 8)$ ,
- calculer  $\mathbb{P}(\inf(X, Y, Z) \geq 2)$ ,
- calculer  $\mathbb{P}(XYZ = 8)$ ,
- calculer  $\mathbb{P}_{\{X+Y \geq 3\}}(X + Y + Z \leq 5)$ ,
- déterminer la fonction caractéristique de  $(X, Y, Z)$ .

Commandes : `GeometricDistribution`, `PoissonDistribution`, `ProductDistribution`, `PDF`, `Probability`, `||`, `≈`, `Min`, `▷`, `CharacteristicFunction`.

**Solution 120.** On propose :

```

In[1]: loi1 = GeometricDistribution[0.31];
In[2]: loi2 = GeometricDistribution[0.23];
In[3]: loi3 = PoissonDistribution[0.63];

```

```

In[4]: loi4 = ProductDistribution[loi1, loi2, loi3];
In[5]: PDF[loi4, {1, 1, 1}]
Out[5]: 0.0127106
In[6]: Probability[x ≥ 1 || y ≥ 1 || z ≥ 1, {x, y, z} ≈ loi4]
Out[6]: 0.962026
In[7]: Probability[x + y + z == 8, {x, y, z} ≈ loi4]
Out[7]: 0.0604289
In[8]: Probability[Min[x, y, z] ≥ 2, {x, y, z} ≈ loi4]
Out[8]: 0.0372257
In[9]: Probability[x y z == 8, {x, y, z} ≈ loi4]
Out[9]: 0.0156692
In[10]: Probability[x + y + z ≤ 5 & x + y ≥ 3, {x, y, z} ≈ loi4]
Out[10]: 0.34411
In[11]: CharacteristicFunction[loi4, {t1, t2, t3}]
Out[11]: 
$$\frac{0.0713e^{0.63(-1+e^{it_3})}}{(1 - 0.69e^{it_1})(1 - 0.77e^{it_2})}$$


```

**Exercice 121**

Soient  $n \in \mathbb{N}^*$ ,  $(p_1, p_2, p_3) \in ]0, 1[^3$  tels que  $p_1 + p_2 + p_3 = 1$  et  $(X, Y, Z)$  un vecteur de *var* suivant la loi multinomiale  $\mathcal{M}(n, p_1, p_2, p_3)$  :

$$\mathbb{P}(\{X = i\} \cap \{Y = j\} \cap \{Z = k\}) = \binom{n}{k} \binom{i+j}{j} p_1^i p_2^j p_3^k, \quad (i, j, k) \in \mathcal{D},$$

avec  $\mathcal{D} = \{(i, j, k) \in \{0, \dots, n\}^3; i + j + k = n\}$ .

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y, Z)$  avec la commande `MultinomialDistribution`,
- déterminer la loi de  $X$ ,
- déterminer la loi de  $Y$ ,
- déterminer la loi de  $Z$ ,
- calculer  $\mathbb{E}((X + Y + Z)^2)$  pour  $n = 10$ ,
- déterminer la fonction caractéristique de  $(X, Y, Z)$ .

Commandes : `MultinomialDistributio`, `MarginalDistribution`, `Expectation`, `≈`, `/.`, `→`, `CharacteristicFunction`.

**Solution 121.** On propose :

```

In[1]: loi = MultinomialDistribution[n, {p1, p2, p3}];
In[2]: MarginalDistribution[loi, 1]
Out[2]: BinomialDistribution[n, p1]
In[3]: MarginalDistribution[loi, 2]
Out[3]: BinomialDistribution[n, p2]
In[4]: MarginalDistribution[loi, 3]
Out[4]: BinomialDistribution[n, p3]
In[5]: Expectation[(x + y + z)^2, {x,y,z} ≈ loi] /. n -> 10
Out[5]: 10p1 + 90p1^2 + 10p2 + 180p1p2 + 90p2^2 + 10p3 + 180p1p3 + 180p2p3 + 90p3^2
In[6]: CharacteristicFunction[loi, {t1, t2, t3}]
Out[6]: (e^{it1}p1 + e^{it2}p2 + e^{it3}p3)^n

```

### Exercice 122

Soient  $p \in ]0, 1[$  et  $X, Y, Z$  3 var *iid* suivant chacune la loi de Bernoulli  $\mathcal{B}(p)$  :

$$\mathbb{P}(X = 0) = 1 - p, \quad \mathbb{P}(X = 1) = p.$$

On pose  $M = \sup(X, Y) + \sup(X, Z) + \sup(Y, Z)$ .

Utiliser *Mathematica* pour

- créer la loi de  $M$ ,
- déterminer la loi de  $M$ ,
- calculer  $\mathbb{E}(M)$ ,
- calculer  $\mathbb{V}(M)$ .

Commandes : *ProductDistribution*, *BernoulliDistribution*, *TransformedDistribution*, *Max*, *≈*, *PDF*, *Mean*, *Variance*.

**Solution 122.** On propose :

```

In[1]: loi = ProductDistribution[{BernoulliDistribution[p], 3}];
In[2]: loi2 = TransformedDistribution[Max[x, y] + Max[x, z] + Max[y, z],
{x, y, z} ≈ loi]

```

```

In[3]: PDF[loi2, k]
Out[3]: 
$$\begin{cases} -(-1+p)^3 & k == 0 \\ 3(-1+p)^2 p & k == 2 \\ (3-2p)p^2 & k == 3 \\ 0 & \text{True} \end{cases}$$

In[4]: Mean[loi2]
Out[4]:  $-3(-2+p)p$ 
In[5]: Variance[loi2]
Out[5]:  $-3(-1+p)^2 p(-4+3p)$ 

```

**Exercice 123**

Soient  $p \in ]0, 1[$  et  $(X_1, \dots, X_5)$  un vecteur de 5 var dont la loi est donnée par

$$\mathbb{P}\left(\bigcap_{i=1}^5 \{X_i = k_i\}\right) = \prod_{i=1}^5 p(1-p)^{k_i}, \quad (k_1, \dots, k_5) \in \mathbb{N}^5.$$

Utiliser Mathematica pour

- déterminer la loi de  $(X_3, X_5)$ ,
- déterminer la fonction génératrice des moments de  $(X_1, \dots, X_5)$ ,
- calculer  $\mathbb{P}(X_1 \leq X_2 \leq X_3 \leq X_4 \leq X_5)$ ,
- calculer  $\mathbb{P}(X_1 < X_2 < X_3 < X_4 < X_5)$ ,
- calculer  $\mathbb{E}(X_1 X_2^2 X_3^2 X_4^4 X_5^5)$ ,
- calculer  $\mathbb{E}\left(\frac{X_1 X_2 X_3 X_4 X_5}{X_1 + X_2 + X_3 + X_4 + X_5}\right)$ ,
- calculer la matrice de covariance de  $(X_1, X_2, X_3, X_4, X_5)$ ,
- calculer la matrice de covariance de  $(X_1 - X_2, X_2 + X_3, X_3 - X_4, X_4 + X_5, X_5 - X_1)$ .

Commandes : GeometricDistribution, ProductDistribution, PDF, MarginalDistribution, MomentGeneratingFunction, Probability,  $\approx$ , Moment, Expectation, Covariance, //MatrixForm, TransformedDistribution.

**Solution 123.** On propose :

```

In[1]: loi = GeometricDistribution[p];
In[2]: loi2 = ProductDistribution[{loi, 5}];

```

```

In[3]: PDF[MarginalDistribution[loi2, {3, 5}], {k3, k5}]
Out[3]: 
$$\begin{cases} (1-p)^{k_3+k_5} p^2 & k_3 \geq 0 \ \&\& \ k_5 \geq 0 \\ 0 & \text{True} \end{cases}$$

In[4]: MomentGeneratingFunction[loi2, {t1, t2, t3, t4, t5}]
Out[4]: 
$$\frac{p^5}{(1-e^{t_1}(1-p))(1-e^{t_2}(1-p))(1-e^{t_3}(1-p))(1-e^{t_4}(1-p))(1-e^{t_5}(1-p))}$$

In[5]: Probability[x1 ≤ x2 ≤ x3 ≤ x4 ≤ x5, {x1, x2, x3, x4, x5} ≈ loi2]
Out[5]: 
$$\frac{1}{(-2+p)^2(3-3p+p^2)(2-2p+p^2)(5-10p+10p^2-5p^3+p^4)}$$

In[6]: Probability[x1 < x2 < x3 < x4 < x5, {x1, x2, x3, x4, x5} ≈ loi2]
Out[6]: 
$$\frac{(-1+p)^{10}}{(-2+p)^2(3-3p+p^2)(2-2p+p^2)(5-10p+10p^2-5p^3+p^4)}$$

In[7]: Moment[loi2, {1, 2, 3, 4, 5}] //Simplify
Out[7]: 
$$-\frac{(-2+p)^2(-1+p)^5(12-12p+p^2)(6-6p+p^2)(120-240p+150p^2-30p^3+p^4)}{p^{15}}$$

In[8]: Expectation[(x1 x2 x3 x4 x5) / (x1 + x2 + x3 + x4 + x5),
{x1, x2, x3, x4, x5} ≈ loi2]
Out[8]: 
$$-\frac{(-1+p)^5(70+35p+15p^2+5p^3+p^4)}{630p^4}$$

In[9]: Covariance[loi2] //MatrixForm
Out[9]: 
$$\begin{pmatrix} \frac{1-p}{p^2} & 0 & 0 & 0 & 0 \\ 0 & \frac{1-p}{p^2} & 0 & 0 & 0 \\ 0 & 0 & \frac{1-p}{p^2} & 0 & 0 \\ 0 & 0 & 0 & \frac{1-p}{p^2} & 0 \\ 0 & 0 & 0 & 0 & \frac{1-p}{p^2} \end{pmatrix}$$

In[10]: loi3 = TransformedDistribution[{x1 - x2, x2 + x3, x3 - x4, x4 + x5,
x5 - x1}, {x1, x2, x3, x4, x5} ≈ loi2];

```



In[11]: Covariance[loi3] //MatrixForm

$$\text{Out[11]: } \begin{pmatrix} -\frac{2(-1+p)}{p^2} & \frac{-1+p}{p^2} & 0 & 0 & \frac{-1+p}{p^2} \\ \frac{-1+p}{p^2} & -\frac{2(-1+p)}{p^2} & \frac{1-p}{p^2} & 0 & 0 \\ 0 & \frac{1-p}{p^2} & -\frac{2(-1+p)}{p^2} & \frac{-1+p}{p^2} & 0 \\ 0 & 0 & \frac{-1+p}{p^2} & -\frac{2(-1+p)}{p^2} & \frac{1-p}{p^2} \\ \frac{-1+p}{p^2} & 0 & 0 & \frac{1-p}{p^2} & -\frac{2(-1+p)}{p^2} \end{pmatrix}$$

### Exercice 124

Soient  $n \in \mathbb{N}^*$  et  $X_1, \dots, X_n$   $n$  var indépendantes. On pose :  $S_n = \sum_{i=1}^n X_i$ .

Utiliser Mathematica pour

- créer une fonction `sumloi` à deux arguments : `loi` et `n`, qui caractérise la loi de  $S_n$  avec  $X_1, \dots, X_n$  suivant chacune la loi `loi`,
- utiliser la fonction `sumloi` pour vérifier les résultats en loi du tableau avec  $n = 10$  :

loi de $X_i$	loi de $S_n$
binomiale $\mathcal{B}(m, p)$	binomiale $\mathcal{B}(nm, p)$
Poisson $\mathcal{P}(\lambda)$	Poisson $\mathcal{P}(n\lambda)$

- proposer des commandes pour vérifier les résultats en loi du tableau avec  $n = 10$  :

loi de $X_i$	loi de $S_n$
binomiale $\mathcal{B}(n_i, p)$	binomiale $\mathcal{B}\left(\sum_{i=1}^n n_i, p\right)$
Poisson $\mathcal{P}(\lambda_i)$	Poisson $\mathcal{P}\left(\sum_{i=1}^n \lambda_i\right)$

Commandes : `x_`, `:=`, `TransformedDistribution`, `Sum`, `Table`, `≈`, `BinomialDistribution`, `PoissonDistribution`, `ni`, `λi`.

**Solution 124.** On propose :

```
In[1]: sumloi[loi_, n_] := TransformedDistribution[Sum[x[i], {i, 1, n}],
Table[x[i] ≈ loi, {i, 1, n}]];
```

```
In[2]: sumloi[BinomialDistribution[m, p], 10]
```

```
Out[2]: BinomialDistribution[10m, p]
```

```

In[3]: sumloi[PoissonDistribution[λ], 10]
Out[3]: PoissonDistribution[10λ]

In[4]: TransformedDistribution[Sum[x[i], {i, 10}], Table[x[i] ≈
BinomialDistribution[ni, p], {i, 10}]]
Out[4]: BinomialDistribution[n1 + n2 + n3 + n4 + n5 + n6 + n7 + n8 + n9 + n10, p]

In[5]: TransformedDistribution[Sum[x[i], {i, 10}], Table[x[i] ≈
PoissonDistribution[λi], {i, 10}]]
Out[5]: PoissonDistribution[λ1 + λ2 + λ3 + λ4 + λ5 + λ6 + λ7 + λ8 + λ9 + λ10]

```

**Exercice 125**

Soient  $p \in ]0, 1[$  et  $X_1, \dots, X_{20}$  20 var indépendantes suivant chacune la loi de Bernoulli  $\mathcal{B}(p)$  :

$$\mathbb{P}(X_1 = 0) = 1 - p, \quad \mathbb{P}(X_1 = 1) = p.$$

On pose :  $P = \prod_{i=1}^{20} X_i$ .

Utiliser Mathematica pour déterminer la loi de  $P$ .

Commandes : BernoulliDistribution, TransformedDistribution, Boole, Product, Table, ≈, PDF.

**Solution 125.** On propose :

```

In[1]: loi = BernoulliDistribution[p];
In[2]: loi2 = TransformedDistribution[Boole[Product[x[i], {i, 1, 20}] == 1],
Table[x[i] ≈ loi, {i, 1, 20}]];
In[3]: PDF[loi2, k]
Out[3]: { p20      k == 1
         { 1 - p20 k == 0
         { 0        True

```

**Exercice 126**

Soient  $p \in ]0, 1[$ , et  $X_1, \dots, X_n$   $n$  var iid suivant chacune la loi géométrique  $\mathcal{G}(p)$  :

$$\mathbb{P}(X_1 = k) = p(1 - p)^k, \quad k \in \mathbb{N}.$$

On pose  $Z = \inf(X_1, \dots, X_n)$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $Z$  en utilisant la commande `OrderDistribution`,
- calculer  $\mathbb{P}(Z \geq m)$  avec  $m \in \mathbb{N}^*$ ,
- calculer  $\mathbb{E}(Z)$ ,
- calculer  $\mathbb{V}(Z)$ .

Commandes : `GeometricDistribution`, `OrderDistribution`, `PDF`, `Probability`, `≈`, `Assumptions`, `∈`, `Integers`, `&&`, `Mean`, `Variance`.

**Solution 126.** On propose :

`In[1]: loi = GeometricDistribution[p];`

`In[2]: loi2 = OrderDistribution[{loi, n}, 1];`

`In[3]: PDF[loi2, k]`

`Out[3]: GeometricDistribution[1 - (1 - p)^n]`

`In[4]: Probability[z ≥ m, z ≈ loi2, Assumptions → m ∈ Integers && m ≥ 1]`

`Out[4]: (1 - p)^mn`

`In[5]: Mean[loi2]`

`Out[5]:  $-1 + \frac{1}{1 - (1 - p)^n}$`

`In[6]: Variance[loi2]`

`Out[6]:  $\frac{(1 - p)^n}{(1 - (1 - p)^n)^2}$`

### Exercice 127

Soient  $n \in \mathbb{N}^*$  et  $X_1, \dots, X_n$   $n$  var iid suivant chacune la loi uniforme  $\mathcal{U}(\{1, \dots, 6\})$  :

$$\mathbb{P}(X_1 = k) = \frac{1}{6}, \quad k \in \{1, \dots, 6\}.$$

On pose  $W = \sup(X_1, \dots, X_n)$ .

Utiliser **Mathematica** pour

- déterminer la loi de  $W$ ,
- représenter graphiquement la loi de  $W$  pour  $n = 4$ ,
- calculer  $\mathbb{P}(W \leq m)$  avec  $m \in \{1, \dots, 5\}$ .

Commandes : `DiscreteUniformDistribution`, `OrderDistribution`, `PDF`, `DiscretePlot`, `/.`, `→`, `ExtentSize`, `Probability`, `≈`, `Assumptions`, `∈`, `Integers`, `&&`.

**Solution 127.** On propose :

```
In[1]: loi = DiscreteUniformDistribution[{1, 6}];
```


```
In[2]: loi2 = OrderDistribution[{loi, n}, n];
```

```
In[3]: PDF[loi2, k]
```

$$\text{Out [3]: } \begin{cases} 1 - \left(\frac{5}{6}\right)^n & k == 6 \\ -\left(-\frac{1}{6} + \frac{k}{6}\right)^n + 6^{-n}k^n & 1 \leq k < 6 \\ 0 & \text{True} \end{cases}$$

```
In[4]: DiscretePlot[PDF[loi2, k] /. n -> 4, {k, 1, 6}, ExtentSize -> Right]
```

Out [4]:



```
In[5]: Probability[z ≤ m, z ≈ loi2, Assumptions -> m ∈ Integers && 1 ≤ m ≤ 5]
```

Out [5]:  $6^{-n}m^n$

### Exercice 128

Certaines lois de *var* peuvent être approchées par d'autres lois plus simples à calculer. Notamment, la loi binomiale  $\mathcal{B}(n, p)$  peut être approchée par la loi de Poisson  $\mathcal{P}(np)$  si  $n \geq 31$  et  $np \leq 10$ , *i.e.*  $\mathcal{B}(n, p) \approx \mathcal{P}(np)$ . L'objectif de cet exercice est de visualiser cette approximation.

Utiliser *Mathematica* pour

- créer une fonction **error** à deux arguments : **n** et **p**, telle que

$$\text{error}(n, p) = \max_{k \in \{0, \dots, n\}} |\mathbb{P}(X = k) - \mathbb{P}(Y = k)|,$$

où  $X \sim \mathcal{B}(n, p)$  et  $Y \sim \mathcal{P}(np)$ .


- représenter graphiquement  $\{\text{error}(n,p), n \in \{1, \dots, 20\}, p \in \{0.1, 0.2, \dots, 0.9, 1\}\}$ . Que remarque-t-on ?

Commandes : `x_`, `:=`, `Max`, `Table`, `≈`, `PDF`, `BinomialDistribution`, `PoissonDistribution`, `DiscretePlot3D`, `Range`, `AxesLabel`, `ColorFunction`, `ExtentSize`.

**Solution 128.** On propose :

```
In[1]: error[n_, p_] := Max[Table[Abs[PDF[BinomialDistribution[n, p], k] -
    PDF[PoissonDistribution[n p], k]], {k, 0, n}]];
In[2]: DiscretePlot3D[error[n, p], {n, 1, 20}, {p, Range[0.1, 1, 0.1]},
    AxesLabel → {"n", "p"}, ColorFunction → "BrightBands", ExtentSize → Full]
```

Out[2]:



On remarque que plus  $n$  est grand et  $p$  est petit, plus l'erreur est petite. Cela illustre l'approximation  $\mathcal{B}(n,p) \approx \mathcal{P}(np)$  pour  $n \geq 31$  et  $np \leq 10$ .

### Exercice 129

Soient  $\lambda > 0$ ,  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var* indépendantes telles que, pour tout  $n \in \mathbb{N}^*$ ,  $X_n$  suit la loi binomiale  $\mathcal{B}(n, \frac{\lambda}{n})$  et  $X$  une *var* suivant la loi de Poisson  $\mathcal{P}(\lambda)$ .

Utiliser *Mathematica* pour

- déterminer la fonction génératrice des moments de  $X$  :  $M_X(t) = \mathbb{E}(e^{tX})$ ,
- déterminer la fonction génératrice des moments de  $X_n$  :  $M_{X_n}(t) = \mathbb{E}(e^{tX_n})$ ,

◦ montrer que

$$\lim_{n \rightarrow \infty} M_{X_n}(t) = M_X(t), \quad t \in \mathbb{R}.$$

Commandes : `x_`, `:=`, `MomentGeneratingFunction`, `PoissonDistribution`, `λ`, `BinomialDistribution`, `Limit`, `∞`.

**Solution 129.** On propose :

In[1]: `m1[t_] := MomentGeneratingFunction[PoissonDistribution[λ], t]`

Out[1]:  $e^{(-1+e^t)\lambda}$

In[2]: `m2[t_, n_] := MomentGeneratingFunction[BinomialDistribution[n, λ / n], t]`

Out[2]:  $\left(1 + \frac{(-1 + e^t)\lambda}{n}\right)^n$

In[3]: `Limit[m2[t, n], n → ∞] == m1[t]`

Out[3]: **True**

!! Cela caractérise la convergence en loi de  $(X_n)_{n \in \mathbb{N}^*}$  vers  $X$ . En outre, pour  $n$  suffisamment grand, on a l'approximation  $\mathcal{B}(n, \frac{\lambda}{n}) \approx \mathcal{P}(\lambda)$  (rejoignant ainsi le résultat de l'exercice précédent).

### Exercice 130

*Théorème central limite : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right).$$

*Alors la suite de var  $(Z_n)_{n \in \mathbb{N}^*}$  converge en loi vers une var  $Z$  suivant la loi normale  $\mathcal{N}(0, 1)$ .*

En termes mathématiques, le théorème central limite est équivalent à la convergence simple :

$$\lim_{n \rightarrow \infty} \varphi_{Z_n}(t) = e^{-\frac{t^2}{2}}, \quad t \in \mathbb{R},$$

où  $\varphi_{Z_n}(t) = \mathbb{E}(e^{itZ_n})$  est la fonction caractéristique de  $Z_n$  (notons que  $e^{-\frac{t^2}{2}} = \varphi_Z(t) = \mathbb{E}(e^{itZ})$  est la fonction caractéristique de  $Z$ ). Après développement et utilisant les hypothèses, on peut montrer que

$$\varphi_{Z_n}(t) = \left( \varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}} \left( \frac{t}{\sqrt{n}} \right) \right)^n,$$

où  $\varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}}(t)$  est la fonction caractéristique de  $\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}$ .

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser Mathematica pour

- créer une fonction `tcl` ayant deux arguments : `loi` et `t`, telle que, pour  $X_1$  suivant la loi `loi` :

$$\text{tcl}(\text{loi}, t) = \lim_{n \rightarrow \infty} \left( \varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}} \left( \frac{t}{\sqrt{n}} \right) \right)^n .$$

- vérifier que  $\text{tcl}(\text{loi}, t) = e^{-\frac{t^2}{2}}$  avec l'argument `loi` correspondant à :
  - ◇ la loi uniforme  $\mathcal{U}(\{-2, \dots, 9\})$ ,
  - ◇ la loi binomiale  $\mathcal{B}(n, p)$ ,
  - ◇ la loi géométrique  $\mathcal{G}(p)$ ,
  - ◇ la loi de poisson  $\mathcal{P}(\lambda)$ ,
  - ◇ la loi binomiale négative  $\mathcal{B}_{neg}(n, p)$ .

Commandes : `x_`, `:=`, `CharacteristicFunction`, `TransformedDistribution`, `Mean`, `StandardDeviation`, `≈`, `Sqrt`, `∞`, `DiscreteUniformDistribution`, `BinomialDistribution`, `GeometricDistribution`, `PoissonDistribution`, `NegativeBinomialDistribution`.

**Solution 130.** On propose :

```
In[1]:  tcl[loi_, t_] := Limit[(CharacteristicFunction[TransformedDistribution[
      (x - Mean[loi]) / StandardDeviation[loi], x ≈ loi], t / Sqrt[n]]]^n,
      n → ∞];

In[2]:  tcl[DiscreteUniformDistribution[{-2, 9}], t]
Out[2]:  e-t2/2

In[3]:  tcl[BinomialDistribution[n, p], t]
Out[3]:  e-t2/2

In[4]:  tcl[GeometricDistribution[p], t]
Out[4]:  e-t2/2

In[5]:  tcl[PoissonDistribution[λ], t]
Out[5]:  e-t2/2

In[6]:  tcl[NegativeBinomialDistribution[n, p], t]
Out[6]:  e-t2/2
```

## 8 Simulations de var discrètes

### Exercice 131

Utiliser Mathematica pour

- simuler 6 réalisations d'une var  $X$  suivant la loi de Bernoulli  $\mathcal{B}(0.6)$ ,
- simuler 4 réalisations d'une var  $X$  suivant la loi binomiale  $\mathcal{B}(10, 0.2)$ ,
- simuler 5 réalisations d'une var  $X$  suivant la loi de Poisson  $\mathcal{P}(2)$ ,
- simuler 6 réalisations d'une var  $X$  suivant la loi géométrique  $\mathcal{G}(0.6)$ ,
- simuler 4 réalisations d'une var  $X$  suivant la loi hypergéométrique  $\mathcal{H}(10, 50, 100)$ ,
- simuler 5 réalisations d'une var  $X$  suivant la loi binomiale négative  $\mathcal{B}_{neg}(2, 0.3)$ .

Commandes : `RandomVariate`, `BernoulliDistribution`, `BinomialDistribution`, `PoissonDistribution`, `GeometricDistribution`, `HypergeometricDistribution`, `NegativeBinomialDistribution`.

**Solution 131.** On propose :

```
In[1]: RandomVariate[BernoulliDistribution[0.6], 6]
```

```
Out[1]: {0, 1, 1, 1, 1, 1}
```

```
In[2]: RandomVariate[BinomialDistribution[10, 0.2], 4]
```

```
Out[2]: {1, 3, 3, 1}
```

```
In[3]: RandomVariate[PoissonDistribution[2], 5]
```

```
Out[3]: {0, 2, 2, 4, 3}
```

```
In[4]: RandomVariate[GeometricDistribution[0.6], 6]
```

```
Out[4]: {0, 3, 0, 1, 0, 0}
```

```
In[5]: RandomVariate[HypergeometricDistribution[10, 50, 100], 4]
```

```
Out[5]: {5, 3, 7, 5}
```

```
In[6]: RandomVariate[NegativeBinomialDistribution[2, 0.3], 5]
```

```
Out[6]: {3, 4, 7, 11, 16}
```

### Exercice 132

Utiliser Mathematica pour

- simuler 3000 réalisations d'une var  $X$  suivant la loi binomiale  $\mathcal{B}(13, 0.4)$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de `CherryTones`.

Commandes : `RandomVariate`, `BinomialDistribution`, `Histogram`, `ColorFunction`.




**Solution 132.** On propose :

```
In[1]: simu = RandomVariate[BinomialDistribution[13, 0.4], 3000];
```

```
In[2]: Histogram[simu, ColorFunction -> "CherryTones"]
```

Out[2]:



### Exercice 133

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = 0) = 0.2, \quad \mathbb{P}(X = 2) = 0.5, \quad \mathbb{P}(X = 5) = 0.3.$$

Utiliser Mathematica pour

- simuler 1000 réalisations de  $X$ ,
- afficher les effectifs associés à ces valeurs.

Commandes : EmpiricalDistribution, RandomVariate, Counts.

**Solution 133.** On propose :

```
In[1]: loi = EmpiricalDistribution[{0.2, 0.5, 0.3} -> {0, 2, 5}];
```

```
In[2]: simu = RandomVariate[loi, 1000];
```

```
In[3]: Counts[simu]
```

Out[3]: <|2 -> 480, 5 -> 321, 0 -> 199|>

### Exercice 134

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{1}{32} \min(k, 8 - k), \quad k \in \{1, \dots, 7\}, \quad \mathbb{P}(X = 8) = \frac{1}{2}.$$

Utiliser Mathematica pour

- simuler 10000 réalisations de  $X$ ,
- afficher les effectifs associés à ces valeurs.

Commandes : Join, Table, Min, EmpiricalDistribution, Range, RandomVariate, Counts.

**Solution 134.** On propose :

```
In[1]: a = Join[Table[1 / 32 Min[k, 8 - k], {k, 1, 7}], {1 / 2}];
```

```
In[2]: loi = EmpiricalDistribution[a → Range[8]];
```

```
In[3]: simu = RandomVariate[loi, 10000];
```

```
In[4]: Counts[simu]
```

```
Out[4]: <|2 → 630, 1 → 316, 8 → 4964, 4 → 1254, 3 → 915, 5 → 973,
        7 → 330, 6 → 618|>
```

### Exercice 135

Soit  $X$  une *var* suivant la loi de poisson tronquée  $\mathcal{P}_{\geq 5}(7)$  :

$$\mathbb{P}(X = k) = ce^{-7} \frac{7^k}{k!}, \quad k \in \{5, 6, \dots, \infty\},$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\sum_{k=5}^{\infty} \mathbb{P}(X = k) = 1$ .

Utiliser Mathematica pour

- simuler 7000 réalisations de  $X$ ,
- représenter les valeurs obtenues par un histogramme des fréquences à 14 classes avec les couleurs de FruitPunchColors.

Commandes : TruncatedDistribution,  $\infty$ , PoissonDistribution, RandomVariate, Histogram, ColorFunction.


**Solution 135.** On propose :

```
In[1]: loi = TruncatedDistribution[{4,  $\infty$ }, PoissonDistribution[7]];
```

```
In[2]: simu = RandomVariate[loi, 7000];
```

```
In[3]: h = Histogram[simu, 14, "ProbabilityDensity", ColorFunction →
        "FruitPunchColors"]
```

Out [3] :

**Exercice 136**

Soit  $X$  une var suivant la loi uniforme  $\mathcal{U}(\{1, \dots, 10\})$  :

$$\mathbb{P}(X = k) = 0.1, \quad k \in \{1, \dots, 10\}.$$

On pose  $Y = \log(1 + X)$ .

Utiliser Mathematica pour simuler 10 réalisations de  $Y$ .

Commandes : `DiscreteUniformDistribution`, `TransformedDistribution`, `Log`, `≈`, `RandomVariate`, `//N`.

**Solution 136.** On propose :

```
In[1]: loi = DiscreteUniformDistribution[{1, 10}];
```

```
In[2]: loi2 = TransformedDistribution[Log[1 + x], x ≈ loi];
```

```
In[3]: RandomVariate[loi2, 10] //N
```

```
Out[3]: {1.60944, 0.693147, 2.30259, 1.79176, 1.60944, 1.09861, 1.60944,
2.30259, 2.3979, 1.60944}
```

**Exercice 137**

Donner des commandes Mathematica utilisant `RandomVariate`, `BinomialDistribution` équivalentes à

```
Total[RandomChoice[{0, 0, 0, 1}, 15]].
```

Commandes : `Total`, `RandomChoice`,

**Solution 137.** Les commandes : `Total[RandomChoice[{0, 0, 0, 1}, 15]]` effectuent 15 tirages avec remise d'un chiffre parmi ceux du vecteur  $\{0, 0, 0, 1\}$ .

La commande `sum` calcule la somme des chiffres obtenus, donc le nombre de 1 obtenus. C'est donc une réalisation de la somme de  $n = 15$  *var iid* suivant chacune la loi de Bernoulli  $\mathcal{B}(p)$  avec  $p = \frac{1}{4} = 0.25$  (il y a 1 chance sur 4 d'avoir le 1 à chaque tirage), laquelle suit la loi binomiale  $\mathcal{B}(n, p)$ . Les commandes équivalentes sont : `RandomVariate[BinomialDistribution[15, 0.25], 1][[1]]` (on précise `[[1]]` pour avoir un réel, et non un vecteur à un élément).

### Exercice 138

Utiliser *Mathematica* pour

- simuler 10000 réalisations d'une *var*  $X$  suivant la loi de Poisson  $\mathcal{P}(8)$ ,
- représenter ces 10000 réalisations par un histogramme des fréquences à 22 classes avec les couleurs de `DeepSeaColors`,
- représenter graphiquement la loi de  $X : \mathbb{P}(X = x)$ , pour  $x \in \{0, \dots, 20\}$ , en rouge,
- superposer à l'historgramme le graphe de la loi de  $X$  obtenus.


Commandes : `RandomVariate`, `PoissonDistribution`, `Histogram`, `ColorFunction`, `DiscretePlot`, `PDF`, `PoissonDistribution`, `PlotStyle`, `PointSize`, `Show`.

**Solution 138.** On propose :

```
In[1]: simu = RandomVariate[PoissonDistribution[8], 10000];
```


```
In[2]: h = Histogram[simu, 22, "ProbabilityDensity", ColorFunction →
      "DeepSeaColors"]
```

Out [2] :




```
In[3]: g = DiscretePlot[PDF[PoissonDistribution[8], x], {x, 0, 20},
      PlotStyle → {Yellow, PointSize[0.02]}]
```

Out [3] :



In [4] : Show[h, g]

Out [4] :

**Exercice 139**

Soient  $X$  et  $Y$  deux *var* indépendantes telles que

- ◊  $X$  suit la loi géométrique  $\mathcal{G}(0.61)$ ,
- ◊  $Y$  suit la loi de Poisson  $\mathcal{P}(1.34)$ .

On pose  $Z = X - Y$ .

Utiliser *Mathematica* pour


- simuler 100000 réalisations de  $Z$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de *MintColors*.

Commandes : `GeometricDistribution`, `PoissonDistribution`, `TransformedDistribution`, `≈`, `RandomVariate`, `Histogram`, `ColorFunction`.

**Solution 139.** On propose :

```
In[1]: loi1 = GeometricDistribution[0.61];
In[2]: loi2 = PoissonDistribution[1.34];
In[3]: loi3 = TransformedDistribution[x - y, {x ≈ loi1, y ≈ loi2}];
In[4]: simu = RandomVariate[loi3, 100000];
In[5]: Histogram[simu, ColorFunction → "MintColors"]
```

Out [5]:



### Exercice 140

Soit  $X$  une *var* dont la loi est donnée par

$$\mathbb{P}(X = k) = \frac{7}{9} \times e^{-6} \frac{6^k}{k!} + \frac{2}{9} \times 0.2(1 - 0.2)^k, \quad k \in \mathbb{N}.$$

Utiliser Mathematica pour


- simuler 10000 réalisations de  $X$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de BrightBands.

Commandes : PoissonDistribution, GeometricDistribution, MixtureDistribution, RandomVariate, Histogram, ColorFunction.

**Solution 140.** On propose :

```
In[1]: loi1 = PoissonDistribution[6];
In[2]: loi2 = GeometricDistribution[0.2];
In[3]: loi3 = MixtureDistribution[{7, 2}, {loi1, loi2}];
In[4]: simu = RandomVariate[loi3, 10000];
In[5]: Histogram[simu, ColorFunction → "BrightBands"]
```

Out [5] :



### Exercice 141

Les 35 réalisations d'une var  $X$  sont :

0, 1, 0, 0, 2, 3, 1, 0, 1, 0, 1, 0, 0, 2, 1, 1, 1, 6, 2, 0, 0, 1, 2, 2, 0, 1, 3, 0, 1, 2, 0, 1, 0, 1, 1

Utiliser la commande `ProbabilityPlot` de `Mathematica` pour déterminer si la loi de  $X$  est plutôt :

- la loi géométrique  $\mathcal{G}(0.64)$ ,
- la loi binomiale  $\mathcal{B}(6, 0.22)$ ,
- la loi de Poisson  $\mathcal{P}(1)$ .


Commandes : `ProbabilityPlot`, `GeometricDistribution`, `BinomialDistribution`, `PoissonDistribution`.

**Solution 141.** On propose :

```
In[1]: x = {0, 1, 0, 0, 2, 3, 1, 0, 1, 0, 1, 0, 0, 2, 1, 1, 1, 6, 2, 0, 0, 1,
2, 2, 0, 1, 3, 0, 1, 2, 0, 1, 0, 1, 1};
```


```
In[2]: ProbabilityPlot[x, GeometricDistribution[0.64]]
```

Out [2] :




```
In[3]: ProbabilityPlot[x, BinomialDistribution[6, 0.22]]
```

Out [3] :



In[4]: `ProbabilityPlot[x, PoissonDistribution[1]]`

Out [4] :



Seul le dernier graphique affiche un nuage de points bien ajusté par la droite diagonale  $y = x$ . On en déduit que  $X$  suit plutôt la loi de Poisson  $\mathcal{P}(1)$ .

### Exercice 142

Soient  $X$  et  $Y$  deux *var* indépendantes telles que

- ◊  $X$  suit la loi binomiale  $\mathcal{B}(18, 0.34)$ ,
- ◊  $Y$  suit la loi de Poisson  $\mathcal{P}(0.62)$ .

Utiliser *Mathematica* pour

- simuler 60000 réalisations de  $(X, Y)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de `LightTerrain`,
- simuler 60000 réalisations de  $(X - Y, X + Y)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de `LightTerrain`.


Commandes : `BinomialDistribution`, `PoissonDistribution`, `ProductDistribution`, `RandomVariate`, `Histogram3D`, `ColorFunction`, `TransformedDistribution`,  $\approx$ .



**Solution 142.** On propose :


```
In[1]: loi1 = BinomialDistribution[18, 0.34];
In[2]: loi2 = PoissonDistribution[0.62];
In[3]: loi3 = ProductDistribution[loi1, loi2];
In[4]: simu = RandomVariate[loi3, 60000];
In[5]: Histogram3D[simu, ColorFunction -> "LightTerrain"]
```

Out [5]:



```
In[6]: loi4 = TransformedDistribution[{x - y, x + y}, {x ≈ loi1, y ≈ loi2}]
In[7]: simu2 = RandomVariate[loi4, 60000];
In[8]: Histogram3D[simu2, ColorFunction -> "LightTerrain"]
```

Out [8]:



**Exercice 143**

Soient  $X$  et  $Y$  deux *var* indépendantes telles que  $X$  suit la loi binomiale  $\mathcal{B}(5, 0.6)$  et  $Y$  suit la loi binomiale  $\mathcal{B}(8, 0.6)$ . Alors on sait que  $Z = X + Y$  suit la loi binomiale  $\mathcal{B}(5 + 8, 0.6) = \mathcal{B}(13, 0.6)$ .

Utiliser *Mathematica* pour illustrer ce résultat avec

- les commandes `RandomVariate` et `QuantilePlot`,
- les commandes `RandomVariate` et `ProbabilityPlot`.

Commandes : `RandomVariate`, `BinomialDistribution`, `QuantilePlot`, `ProbabilityPlot`.

**Solution 143.** On propose :


```
In[1]: x = RandomVariate[BinomialDistribution[5, 0.6], 10000];
```

```
In[2]: y = RandomVariate[BinomialDistribution[8, 0.6], 10000];
```

```
In[3]: z = x + y;
```


```
In[4]: QuantilePlot[z, BinomialDistribution[13, 0.6]]
```

Out [4]:



```
In[5]: ProbabilityPlot[z, BinomialDistribution[13, 0.6]]
```

Out [5]:



!! Dans les deux graphiques, le nuage de points est bien ajusté par la droite diagonale  $y = x$ . Cela illustre le fait que  $X + Y$  suit la loi binomiale  $\mathcal{B}(13, 0.6)$ .

**Exercice 144**

Soient  $X$  et  $Y$  deux *var* indépendantes telles que  $X$  suit la loi de Poisson  $\mathcal{P}(5)$  et  $Y$  suit la loi de Poisson  $\mathcal{P}(3)$ . Alors on sait que  $Z = X + Y$  suit la loi de Poisson  $\mathcal{P}(5 + 3) = \mathcal{P}(8)$ .

Utiliser *Mathematica* pour illustrer ce résultat avec

- les commandes `RandomVariate` et `QuantilePlot`,
- les commandes `RandomVariate` et `ProbabilityPlot`.

Commandes : `RandomVariate`, `PoissonDistribution`, `QuantilePlot`, `ProbabilityPlot`.

**Solution 144.** On propose :


```
In[1]: x = RandomVariate[PoissonDistribution[5], 10000];
```

```
In[2]: y = RandomVariate[PoissonDistribution[3], 10000];
```

```
In[3]: z = x + y;
```


```
In[4]: QuantilePlot[z, PoissonDistribution[8]]
```

Out [4]:



```
In[5]: ProbabilityPlot[z, PoissonDistribution[8]]
```

Out [5]:



!! Dans les deux graphiques, le nuage de points est bien ajusté par la droite diagonale  $y = x$ . Cela illustre le fait que  $X + Y$  suit la loi de Poisson  $\mathcal{P}(8)$ .

**Exercice 145**

*Loi forte des grands nombres : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 1. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i.$$

*Alors  $(\bar{X}_n)_{n \in \mathbb{N}^*}$  converge presque sûrement vers  $\mathbb{E}(X_1)$ .*

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- créer une fonction `fsimu1` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $\bar{X}_n$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- tester graphiquement cette fonction en faisant :

```
Histogram[fsimu1[PoissonDistribution[2], 100]]
```

```
Histogram[fsimu1[PoissonDistribution[2], 1000]]
```

```
Histogram[fsimu1[PoissonDistribution[2], 10000]]
```

- créer une fonction `fsimu2` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $|\bar{X}_n - \mathbb{E}(X_1)|$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- tester graphiquement cette fonction en faisant :

```
Histogram[fsimu2[PoissonDistribution[2], 100]]
```

```
Histogram[fsimu2[PoissonDistribution[2], 1000]]
```

```
Histogram[fsimu2[PoissonDistribution[2], 10000]]
```


Commandes : `x_`, `:=`, `Table`, `Mean`, `RandomVariate`, `Histogram`, `PoissonDistribution`, `Abs`.

**Solution 145.** On propose :

```
In[1]: fsimu1[loi_, n_] := Table[Mean[RandomVariate[loi, n]], 1000];
```


```
In[2]: Histogram[fsimu1[PoissonDistribution[2], 100]]
```

Out [2]:




```
In[3]: Histogram[fsimu1[PoissonDistribution[2], 1000]]
```

Out[3]:



```
In[4]: Histogram[fsimu1[PoissonDistribution[2], 10000]]
```

Out[4]:




!! En regardant l'axe des abscisses pour chaque graphique, on remarque que plus  $n$  est grand, plus les réalisations de  $\bar{X}_n$  se concentrent en  $\mathbb{E}(X_1) = 2$ .

```
In[5]: fsimu2[loi_, n_] := Table[Abs[Mean[RandomVariate[loi, n]] - Mean[loi]],  
1000];
```


```
In[6]: Histogram[fsimu2[PoissonDistribution[2], 100]]
```

Out [6] :




In[7]: `Histogram[fsimu2[PoissonDistribution[2], 1000]]`

Out [7] :



In[8]: `Histogram[fsimu2[PoissonDistribution[2], 10000]]`

Out [8] :



!! En regardant l'axe des abscisses pour chaque graphique, on remarque que plus  $n$  est grand, plus les différences entre les réalisations de  $\bar{X}_n$  et  $\mathbb{E}(X_1) = 2$  s'amenuisent (en moyenne).

### Exercice 146

Résultat : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad S_n^2 = \frac{1}{n-1} \sum_{i=1}^n (X_i - \bar{X}_n)^2.$$

Alors  $(S_n^2)_{n \in \mathbb{N}^*}$  converge presque sûrement vers  $\mathbb{V}(X_1)$  (et  $\mathbb{E}(S_n^2) = \mathbb{V}(X_1)$ ).

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- créer une fonction `fsimu` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $|S_n^2 - \mathbb{V}(X_1)|$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- tester graphiquement cette fonction en faisant :

```
Histogram[fsimu[GeometricDistribution[0.7], 100], ColorFunction -> "Pastel"]
```

```
Histogram[fsimu[GeometricDistribution[0.7], 1000], ColorFunction -> "Pastel"]
```

```
Histogram[fsimu[GeometricDistribution[0.7], 10000], ColorFunction -> "Pastel"]
```


Commandes : `x_`, `:=`, `Table`, `Abs`, `Variance`, `RandomVariate`, `Histogram`, `GeometricDistribution`, `ColorFunction`.

**Solution 146.** On propose :

```
In[1]: fsimu[loi_, n_] := Table[Abs[Variance[RandomVariate[loi, n]] -
Variance[loi]], 1000];
```


```
In[2]: Histogram[fsimu[GeometricDistribution[0.7], 100], ColorFunction -> "Pastel"]
```

Out [2]:




```
In[3]: Histogram[fsimu[GeometricDistribution[0.7], 1000], ColorFunction -> "Pastel"]
```

Out [3] :



In[4]: `Histogram[fsimu[GeometricDistribution[0.7], 10000], ColorFunction → "Pastel"]`

Out [4] :



!! En regardant l'axe des abscisses pour chaque graphique, on remarque que plus  $n$  est grand, plus les différences entre les réalisations de  $S_n^2$  et  $\mathbb{V}(X_1) = 0.612245$  s'amenuisent (en moyenne).

#### Exercice 147

*Théorème central limite : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right) = \frac{1}{\sqrt{n}} \sum_{i=1}^n \left( \frac{X_i - \mathbb{E}(X_1)}{\sigma(X_1)} \right).$$

*Alors la suite de var  $(Z_n)_{n \in \mathbb{N}^*}$  converge en loi vers une var  $Z$  suivant la loi normale  $\mathcal{N}(0, 1)$ . L'objectif de cet exercice est d'illustrer ce résultat.*



Utiliser Mathematica pour

- créer une fonction `fsimutcl` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $Z_n$ , avec  $X_1, \dots, X_n$  *iid* suivant chacune la loi `loi`,
- appliquer la fonction `fsimutcl` avec  $n = 100000$  et, pour `loi`, la loi géométrique  $\mathcal{G}(0.2)$ ,
- représenter ces 100000 réalisations par un histogramme des fréquences à 30 classes avec les couleurs de `LakeColors`,
- superposer à l'histogramme le graphe de la densité  $f(x)$  associée à la loi normale  $\mathcal{N}(0,1)$  pour  $x \in [-4, 4]$  en rouge,
- reprendre les points précédents avec la loi binomiale  $\mathcal{B}(15, 0.76)$ .

Commandes : `x_`, `:=`, `Table`, `Sqrt`, `Total`, `RandomVariate`, `TransformedDistribution`, `Mean`, `StandardDeviation`, `≈`, `//N`, `GeometricDistribution`, `Histogram`, `ColorFunction`, `Show`, `Plot`, `PDF`, `NormalDistribution`, `PlotStyle`, `BinomialDistribution`.

**Solution 147.** On propose :


```
In[1]: fsimutcl[loi_, n_] := Table[1 / Sqrt[n] Total[RandomVariate[
      TransformedDistribution[(x - Mean[loi]) / StandardDeviation[loi],
      x ≈ loi], n] //N], 1000];

In[2]: a = fsimutcl[GeometricDistribution[0.2], 100000];

In[3]: h = Histogram[a, 30, "ProbabilityDensity", ColorFunction → "LakeColors"];

In[4]: Show[h, Plot[PDF[NormalDistribution[0, 1], x], {x, -4, 4}, PlotStyle → Red]]
```

Out [4]:




```
In[5]: b = fsimutcl[BinomialDistribution[15, 0.76], 100000];

In[6]: h2 = Histogram[b, 30, "ProbabilityDensity", ColorFunction → "LakeColors"];
```

```
In[7]: Show[h2, Plot[PDF[NormalDistribution[0, 1], x], {x, -4, 4},
PlotStyle → Red]]
```

Out[7]:



!! On constate que le graphe de  $f$  ajuste bien les sommets des bâtons de l'histogramme. Cela illustre le théorème central limite et plus particulièrement l'approximation :  $Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right) \approx \mathcal{N}(0, 1)$  pour  $n$  suffisamment grand.

#### Exercice 148

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* admettant un moment d'ordre 2. Le théorème central limite nous assure que, pour  $n$  suffisamment grand, on a l'approximation :

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i \approx \mathcal{N} \left( \mathbb{E}(X_1), \frac{\sigma(X_1)}{\sqrt{n}} \right).$$

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- reproduire et comprendre l'enjeu de la fonction `tclhist` (sophistiquée) à un argument `loi` :

```
tclhist[loi_] := Table[Show[Histogram[(1 / n) Plus @@@
RandomVariate[ProductDistribution[{loi, n}], 10^5], Automatic, "PDF",
ColorFunction → "IslandColors", PlotLabel → n],
Plot[Boole[n ≥ 2] PDF[NormalDistribution[Mean[loi],
StandardDeviation[loi] / Sqrt[n]], x], {x, -20, 20}, PlotStyle → Red,
PlotRange → Full]], {n, Range[1, 56, 5]}]
```
- appliquer la fonction `fsimutcl` avec, pour `loi` :
  - ◇ la loi uniforme  $\mathcal{U}(\{1, \dots, 6\})$ ,
  - ◇ la loi géométrique  $\mathcal{G}(0.27)$ ,

◇ la loi de Poisson  $\mathcal{P}(0.86)$ .


Commandes : Table, Show, Histogram, Plus RandomVariate, ProductDistribution, ColorFunction, PlotLabel, Plot, Boole, PDF, NormalDistribution, Mean, StandardDeviation, Sqrt, PlotStyle, PlotRange, Range, DiscreteUniformDistribution, GeometricDistribution, PoissonDistribution.

**Solution 148.** On propose :

```
In[1]: tclhist[loi_] := Table[Show[Histogram[(1 / n) Plus @@
RandomVariate[ProductDistribution[{loi, n}], 10^5], Automatic, "PDF",
ColorFunction -> "IslandColors", PlotLabel -> n],
Plot[Boole[n ≥ 2] PDF[NormalDistribution[Mean[loi],
StandardDeviation[loi] / Sqrt[n]], x], {x, -20, 20}, PlotStyle -> Red,
PlotRange -> Full]], {n, Range[1, 56, 5]}];


In[2]: tclhist[DiscreteUniformDistribution[{1, 6}]]
```

Out [2] :




In[3]: `tclhist[GeometricDistribution[0.27]]`

Out[3]:



```
In[4]: tclhist[PoissonDistribution[0.86]]
```

```
Out[4]:
```



L'enjeu de la fonction `tclhist` est de simuler  $10^5$  réalisations de  $\bar{X}_n$  pour plusieurs valeurs de  $n$  ( $n \in \{1, 6, 11, 16, \dots, 50, 56\}$ ), tracer l'histogramme associé pour chacune de ces valeurs et superposer la loi normale  $\mathcal{N}\left(\mathbb{E}(X_1), \frac{\sigma(X_1)}{\sqrt{n}}\right)$ . Le théorème central limite nous assure que, plus  $n$  est grand, plus les sommets de l'histogramme seront bien ajustés par le graphe d'une densité de la loi normale considérée. À cela s'ajoute plusieurs options graphiques secondaires (tailles, couleurs...).

**Exercice 149**

Soient  $X_1, \dots, X_5$  5 var iid suivant chacune la loi uniforme  $\mathcal{U}(\{0, \dots, 16\})$ . On pose  $U = \inf(X_1, \dots, X_5)$  et  $V = \sup(X_1, \dots, X_5)$ .

Utiliser Mathematica pour


- créer la loi de  $(U, V)$ ,
- simuler 10000 réalisations de  $(U, V)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de CandyColors,
- vérifier que les moyennes des valeurs obtenues pour  $U$  et  $V$  sont proches de  $(\mathbb{E}(U), \mathbb{E}(V))$ .

Commandes : `DiscreteUniformDistribution`, `OrderDistribution`, `RandomVariate`, `Histogram3D`, `ColorFunction`, `Mean`, `//N`.

**Solution 149.** On propose :

```
In[1]: loi = DiscreteUniformDistribution[{0, 16}];
In[2]: loi2 = OrderDistribution[{loi, 5}, {1, 5}]
In[3]: simu = RandomVariate[loi2, 10000];
In[4]: Histogram3D[simu, ColorFunction -> "CandyColors"]
```

Out [4]:



```
In[5]: Mean[simu] - Mean[loi2] //N
```

Out [5]: {0.0279738, 0.00822618}



## 9 Var à densité

### Exercice 150

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x) = \begin{cases} x + 3 & \text{si } x \in [-b, b], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser la commande `Solve` de Mathematica.

Commandes : `Solve`, `Integrate`, `&&`.

**Solution 150.** On propose :

```
In[1]: Solve[Integrate[Integrate[x + 3, {x, -b, b}] == 1 && b > 0, b]
```

```
Out[1]: {{b -> 1/6}}
```

### Exercice 151

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x) = \begin{cases} b(1-x)^2 + \frac{b}{2} & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser la commande `Solve` de Mathematica.

Commandes : `Solve`, `Integrate`, `&&`.

**Solution 151.** On propose :

```
In[1]: Solve[Integrate[b (1 - x)^2 + b / 2, {x, 0, 1}] == 1 && b > 0 , b]
```

```
Out[1]: {{b -> 6/5}}
```



**Exercice 152**

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x) = \begin{cases} b^2 x^2 + \frac{4b}{3} x & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser la commande `Solve` de Mathematica.

Commandes : `Solve`, `Integrate`, `&&`.

**Solution 152.** On propose :

```
In[1]: Solve[Integrate[b^2 x^2 + 4 b x / 3, {x, 0, 1}] == 1 && b > 0, b]
```

```
Out[1]: {{b -> 1}}
```

**Exercice 153**

Soient  $(a, b) \in \mathbb{R}^2$  avec  $b > a$  et  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([a, b])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{b-a} & \text{si } x \in [a, b], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- déterminer la fonction de répartition de  $X$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : `UniformDistribution`, `CDF`, `Mean`, `Variance`.

**Solution 153.** On propose :

```
In[1]: loi = UniformDistribution[{a, b}];
```

```
In[2]: CDF[loi, x]
```

```
Out[2]: { -a + x / -a + b  a <= x <= b
          1                x > b
          0                True }
```

```
In[3]: Mean[loi]
```

```
Out[3]: (a + b) / 2
```

In[4]: Variance[loi]

Out[4]:  $\frac{1}{12}(-a + b)^2$

### Exercice 154

Soit  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([0, 10])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{10} & \text{si } x \in [0, 10], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la loi de  $X$  avec la commande **UniformDistribution**,
- tracer le graphe de la fonction de répartition de  $X$  :  $F(x) = \mathbb{P}(X \leq x)$ , pour  $x \in [-3, 13]$ ,
- déterminer la médiane de  $X$ , *i.e.* le réel  $x$  tel que  $F(x) = 0.5$ ,
- calculer la valeur numérique de  $\mathbb{P}(3 \leq X \leq 6)$ ,
- calculer la valeur numérique de  $\mathbb{P}(\{X \leq 2\} \cup \{X > 3\})$ ,
- calculer la valeur numérique de  $\mathbb{P}_{\{X \geq 4\}}(X \leq 8)$ ,
- calculer la valeur numérique de  $\mathbb{P}(X^2 - 4X + 6 \leq 8)$ ,
- calculer la valeur numérique de  $\mathbb{E}(X^5)$ ,
- déterminer la fonction caractéristique de  $X$  :  $\varphi(t) = \mathbb{E}(e^{itX})$ .


Commandes : **UniformDistribution**, **Plot**, **CDF**, **Median**, **NProbability**,  $\approx$ , **||**, **▷**, **Moment**, **//N**, **CharacteristicFunction**.

**Solution 154.** On propose :

In[1]: loi = UniformDistribution[{0, 10}];

In[2]: Plot[CDF[loi, x], {x, -3, 13}]

Out[2]:



```

In[3]: Median[loi]
Out[3]: 5
In[4]: NProbability[3 ≤ x ≤ 6, x ≈ loi]
Out[4]: 0.3
In[5]: NProbability[x ≤ 2 || x > 3, x ≈ loi]
Out[5]: 0.9
In[6]: NProbability[x ≤ 8 > x ≥ 4, x ≈ loi]
Out[6]: 0.666667
In[7]: NProbability[x^2 - 4 x + 6 ≤ 8, x ≈ loi]
Out[7]: 0.444949
In[8]: Moment[loi, 5] //N
Out[8]: 0.666667
In[9]: CharacteristicFunction[loi, t]
Out[9]: 
$$-\frac{i(-1 + e^{10it})}{10t}$$


```

### Exercice 155

Soit  $X$  une *var* suivant la loi exponentielle  $\mathcal{E}(8)$ , *i.e.* de densité :

$$f(x) = \begin{cases} 8e^{-8x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `ExponentialDistribution`,
- tracer le graphe de  $f(x)$  pour  $x \in [0, 1]$ ,
- déterminer la fonction de répartition  $F$  de  $X$ ,
- tracer le graphe de  $F(x)$  pour  $x \in [0, 1]$ ,
- déterminer la médiane de  $X$ , *i.e.* le réel  $x$  tel que  $F(x) = 0.5$ ,
- vérifier que  $\mathbb{P}_{\{X \geq b\}}(X \geq a + b) = \mathbb{P}(X \geq a)$  pour tout  $a > 0$  et  $b > 0$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{E}(e^{2X})$ .


Commandes : `ExponentialDistribution`, `Plot`, `PDF`, `CDF`, `Median`, `Probability`, `≈`, `Assumptions`, `Mean`, `Expectation`, `Exp`.

**Solution 155.** On propose :

```
In[1]: loi = ExponentialDistribution[8];
```

```
In[2]: Plot[PDF[loi, x], {x, 0, 1}]
```

Out[2]:




```
In[3]: CDF[loi, x]
```

Out[3]:

$$\begin{cases} 1 - e^{-8x} & x \geq 0 \\ 0 & \text{True} \end{cases}$$

```
In[4]: Plot[CDF[loi, x], {x, 0, 1}]
```

Out[4]:



```
In[5]: Median[loi]
```

Out[5]:  $\frac{\text{Log}[2]}{8}$

```
In[6]: Probability[x ≥ a + b > x ≥ b, x ≈ loi, Assumptions → a > 0
&& b > 0] == Probability[x ≥ a, x ≈ loi, Assumptions → a > 0]
```

Out[6]: True

```
In[7]: Mean[loi]
Out[7]:  $\frac{1}{8}$ 
In[8]: Expectation[Exp[2x], x ≈ loi]
Out[8]:  $\frac{4}{3}$ 
```

**Exercice 156**

Soient  $\lambda > 0$  et  $X$  une *var* suivant la loi exponentielle  $\mathcal{E}(\lambda)$ , *i.e.* de densité :

$$f(x) = \begin{cases} \lambda e^{-\lambda x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction *m* définie par la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- calculer  $M'(0)$ ,
- calculer  $M''(0) - (M'(0))^2$ .

Commandes : `ExponentialDistribution`, `λ`, `x_`, `:=`, `MomentGeneratingFunction`, `D`, `//Simplify`.

**Solution 156.** On propose :

```
In[1]: loi = ExponentialDistribution[λ];
In[2]: m[t_] := MomentGeneratingFunction[loi, t];
In[3]: m[t]
Out[3]:  $\frac{\lambda}{-\text{t} + \lambda}$ 
In[4]: D[m[t], t] /. t → 0
Out[4]:  $\frac{1}{\lambda}$ 
In[5]: D[m[t], {t, 2}] - (D[m[t], t])^2 /. t → 0 //Simplify
Out[5]:  $\frac{1}{\lambda^2}$ 
```

!! On obtient  $M'(0) = \mathbb{E}(X) = \frac{1}{\lambda}$  et  $M''(0) - (M'(0))^2 = \mathbb{E}(X^2) - (\mathbb{E}(X))^2 = \mathbb{V}(X) = \frac{1}{\lambda^2}$ .

**Exercice 157**

Soient  $n \in \mathbb{N}^*$ ,  $\beta > 0$  et  $X$  une *var* suivant la loi gamma  $\Gamma(n, \beta)$  (notation française plus usuelle :  $\Gamma(n, \lambda)$  avec  $\lambda = \frac{1}{\beta}$ ), *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{(n-1)!} \beta^{-n} x^{n-1} e^{-\frac{x}{\beta}} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la loi de  $X$  avec la commande **GammaDistribution**,
- tracer le graphe de  $f$  avec  $\beta = 2$  pour plusieurs valeurs de  $n$  avec la commande **Manipulate**,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ .



Commandes : **GammaDistribution**,  $\beta$ , **Manipulate**, **Plot**, **PDF**, **ColorFunction**, **Range**, **Mean**, **Variance**, **MomentGeneratingFunction**.


**Solution 157.** On propose :

```
In[1]: loi = GammaDistribution[n, β];
```

```
In[2]: Manipulate[Plot[PDF[GammaDistribution[n, 2], x], {x, 0, 45},
  ColorFunction -> "Rainbow", {n, 1, 10}, {n, Range[10]}]
```

**Out [2] :**





In[3]: Mean[loi]


Out[3]:  $n\beta$

In[4]: Variance[loi]

Out[4]:  $n\beta^2$

In[5]: MomentGeneratingFunction[loi, t]

Out[5]:  $(1 - t\beta)^{-n}$



### Exercice 158

Soient  $\mu \in \mathbb{R}$ ,  $\sigma > 0$  et  $X$  une *var* suivant la loi normale  $\mathcal{N}(\mu, \sigma)$  (notation française usuelle :  $\mathcal{N}(\mu, \sigma^2)$ , le  $\sigma$  désignant l'écart-type dans les deux notations), *i.e.* de densité :

$$f(x) = \frac{1}{\sqrt{2\pi\sigma^2}} e^{-\frac{(x-\mu)^2}{2\sigma^2}}, \quad x \in \mathbb{R}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `NormalDistribution`,
- tracer le graphe de  $f(x)$  pour  $\mu = 0$ ,  $\sigma \in \{0.75, 1, 2\}$  (soit trois courbes) et  $x \in [-6, 6]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée,
- tracer le graphe de  $f(x)$  pour  $\sigma = 1$ ,  $\mu \in \{-1, 1, 2\}$  (soit trois courbes) et  $x \in [-5, 5]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ ,
- déterminer la fonction caractéristique de  $X$  :  $\varphi(t) = \mathbb{E}(e^{itX})$ .


Commandes : `NormalDistribution`,  $\mu$ ,  $\sigma$ , `Plot`, `Table`, `PDF`, `/.`, `→`, `//Evaluate`, `LineLegend`, `LegendLabel`, `Filling`, `Mean`, `Variance`, `MomentGeneratingFunction`, `CharacteristicFunction`.

**Solution 158.** On propose :

```
In[1]: loi = NormalDistribution[ $\mu$ ,  $\sigma$ ];
```


```
In[2]: Plot[Table[PDF[loi, x] /.  $\mu \rightarrow 0$ , { $\sigma$ , {0.75, 1, 2}}] //Evaluate,
{x, -6, 6}, PlotLegends  $\rightarrow$  LineLegend[{0.75, 1, 2}, LegendLabel  $\rightarrow \sigma$ ],
Filling  $\rightarrow$  Axis]
```

Out[2]:



```
In[3]: Plot[Table[PDF[loi, x] /.  $\sigma \rightarrow 1$ , { $\mu$ , {-1, 1, 2}}] //Evaluate,
{x, -6, 6}, PlotLegends  $\rightarrow$  LineLegend[{-1, 1, 2}, LegendLabel  $\rightarrow \mu$ ],
Filling  $\rightarrow$  Axis]
```

Out[3]:



```
In[4]: Mean[loi]
```

Out[4]:  $\mu$



In[5]: Variance[loi]

Out[5]:  $\sigma^2$

In[6]: MomentGeneratingFunction[loi, t]

Out[6]:  $e^{t\mu + \frac{t^2\sigma^2}{2}}$

In[7]: CharacteristicFunction[loi, t]

Out[7]:  $e^{it\mu - \frac{t^2\sigma^2}{2}}$

!! On peut remarquer que " $\varphi(t) = M(it)$ ".

### Exercice 159

Soit  $X$  une *var* suivant la loi normale  $\mathcal{N}(2, 3)$ , *i.e.* de densité :

$$f(x) = \frac{1}{\sqrt{2\pi 3^2}} e^{-\frac{(x-2)^2}{2 \times 3^2}}, \quad x \in \mathbb{R}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `NormalDistribution`,
- tracer le graphe de la fonction de répartition de  $X$  :  $F(x) = \mathbb{P}(X \leq x)$ , pour  $x \in [-16, 16]$ ,
- calculer  $\mathbb{P}(X \leq 1.2)$ ,
- tracer, sur un même graphique, le graphe de  $f(x)$  pour  $x \in [-10, 14]$ , ainsi que l'aire sous la courbe correspondante à  $\mathbb{P}(X \leq 1.2)$ ,
- calculer  $\mathbb{P}(-0.7 \leq X \leq 1.7)$ ,
- calculer  $\mathbb{P}(|X - 2.5| \leq 0.9)$ ,
- calculer  $\mathbb{P}_{\{X \leq 2.7\}}(|X| \geq 1.7)$ ,
- calculer  $\mathbb{P}(|e^{2X} - 5.1| \geq 2.7)$ ,
- calculer  $\mathbb{P}(X^2 + 3.6X + 1.6 \leq 1)$ ,
- déterminer le réel  $x$  tel que  $\mathbb{P}(X \leq x) = 0.76$ .


Commandes : `NormalDistribution`, `Plot`, `CDF`, `Show`, `PDF`, `Filling`, `PlotRange`, `Probability`, `≈`, `Abs`, `▷`, `Exp`, `Quantile`, `With`, `PlotStyle`.

**Solution 159.** On propose :

In[1]: loi = NormalDistribution[2, 3];

In[2]: Plot[CDF[loi, x], {x, -16, 16}]

Out[2]:




In[3]: `CDF[loi, 1.2]`

Out[3]: `0.394863`

In[4]: `Show[Plot[PDF[loi, x], {x, -10, 14}], Plot[PDF[loi, x], {x, -10, 1.2},  
Filling -> Axis, PlotRange -> All]`

Out[4]:



In[5]: `Probability[-0.7 ≤ x ≤ 1.7, x ≈ loi]`

Out[5]: `0.276112`

In[6]: `Probability[Abs[x - 2.5] ≤ 0.9, x ≈ loi]`

Out[6]: `0.232666`

In[7]: `Probability[Abs[x] ≥ 1.7 > x ≤ 2.7, x ≈ loi]`

Out[7]: `0.40659`

```
In[8]: Probability[Abs[Exp[2 x] - 5.1] ≥ 2.7, x ≈ loi]
```

```
Out[8]: 0.928418
```

```
In[9]: Probability[x^2 + 3.6 x + 1.6 ≤ 1, x ≈ loi]
```

```
Out[9]: 0.198924
```


```
In[10]: Quantile[loi, 0.76]
```

```
Out[10]: 4.11891
```

!! Une autre solution (offrant plus d'options) pour tracer, sur un même graphique, le graphe de  $f(x)$  pour  $x \in [-10, 14]$ , ainsi que l'aire sous la courbe correspondante à  $\mathbb{P}(X \leq 1.2)$  :

```
In[4]: With[{dens = PDF[loi]}, Plot[{If[-10 < x < 1.2, dens[x]], dens[x]},
{x, -10, 14}, Filling → {1 → Axis}, PlotStyle → ColorData[1, 6]]]
```

```
Out[4]:
```



### Exercice 160

Soient  $\nu > 0$  et  $X$  une *var* suivant la loi du Chi-deux  $\chi^2(\nu)$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{2^{\frac{\nu}{2}} \Gamma(\frac{\nu}{2})} x^{\frac{\nu}{2}-1} e^{-\frac{x}{2}} & \text{si } x \geq 0, \\ 0 & \text{sinon,} \end{cases}$$

où  $\Gamma$  désigne la fonction gamma d'Euler définie par :  $\Gamma(x) = \int_0^\infty t^{x-1} e^{-t} dt$ ,  $x > 0$ .

Utiliser **Mathematica** pour

- créer la loi de  $X$  avec la commande **ChiSquareDistribution**,
- tracer le graphe de  $f(x)$  pour  $\nu \in \{0.5, 3, 5\}$  (soit trois courbes) et  $x \in [0, 6]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,

- déterminer la fonction génératrice des moments de  $X$  :  $M(t) = \mathbb{E}(e^{tX})$ .


Commandes : `ChiSquareDistribution`,  $\nu$ , `Plot`, `Table`, `PDF`, `//Evaluate`, `PlotLegends`, `LineLegend`, `LegendLabel`, `Filling`, `Mean`, `Variance`, `MomentGeneratingFunction`.

**Solution 160.** On propose :

```
In[1]: loi = ChiSquareDistribution[ν];
```

```
In[2]: Plot[Table[PDF[loi, x], {ν, {0.5, 3, 5}}] //Evaluate, {x, 0, 6},
PlotLegends → LineLegend[{0.5, 3, 5}, LegendLabel → ν], Filling → Axis]
```

Out [2] :



```
In[3]: Mean[loi]
```

Out [3] :  $\nu$

```
In[4]: Variance[loi]
```

Out [4] :  $2\nu$

```
In[5]: MomentGeneratingFunction[loi, t]
```

Out [5] :  $(1 - 2t)^{-\nu/2}$

### Exercice 161

Soient  $\nu > 0$  et  $X$  une *var* suivant la loi de Student  $\mathcal{T}(\nu)$ , *i.e.* de densité :

$$f(x) = \frac{1}{\sqrt{\nu\pi}} \frac{\Gamma\left(\frac{\nu+1}{2}\right)}{\Gamma\left(\frac{\nu}{2}\right)} \left(1 + \frac{x^2}{\nu}\right)^{-\frac{\nu+1}{2}}, \quad x \in \mathbb{R},$$

où  $\Gamma$  désigne la fonction gamma d'Euler définie par :  $\Gamma(x) = \int_0^\infty t^{x-1} e^{-t} dt$ ,  $x > 0$ .

Utiliser `Mathematica` pour

- créer la loi de  $X$  avec la commande `StudentTDistribution`,

- tracer le graphe de  $f(x)$  pour  $\nu \in \{0.1, 0.5, 4\}$  (soit trois courbes) et  $x \in [-5, 5]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ .


Commandes : StudentTDistribution,  $\nu$ , Plot, Table, PDF, //Evaluate, PlotLegends, LineLegend, LegendLabel, Filling, Mean, Variance.

**Solution 161.** On propose :

```
In[1]: loi = StudentTDistribution[ $\nu$ ];
```

```
In[2]: Plot[Table[PDF[loi, x], { $\nu$ , {0.1, 0.5, 4}}] //Evaluate, {x, -5, 5},
PlotLegends  $\rightarrow$  LineLegend[{0.1, 0.5, 4}, LegendLabel  $\rightarrow$   $\nu$ ], Filling  $\rightarrow$  Axis]
```

Out[2]:



```
In[3]: Mean[loi]
```

```
Out[3]: { 0           $\nu > 1$ 
         { Indeterminate True
```

```
In[4]: Variance[loi]
```

```
Out[4]: {  $\frac{\nu}{-2 + \nu}$      $\nu > 2$ 
         { Indeterminate True
```

### Exercice 162

Soient  $\beta > 0$  et  $X$  une *var* suivant la loi de Laplace  $\mathcal{L}(0, \beta)$ , *i.e.* de densité :

$$f(x) = \frac{1}{2\beta} e^{-\frac{|x|}{\beta}}, \quad x \in \mathbb{R}.$$

Utiliser `Mathematica` pour

- calculer  $\mathbb{E}(X)$ ,
- calculer  $\sigma(X)$ ,
- déterminer la loi de  $Y = |X|$ .

Commandes : `LaplaceDistribution`,  $\beta$ , `Mean`, `StandardDeviation`, `TransformedDistribution`, `Abs`,  $\approx$ .

**Solution 162.** On propose :

```
In[1]: loi = LaplaceDistribution[0,  $\beta$ ];
```

```
In[2]: Mean[loi]
```

```
Out[2]: 0
```

```
In[3]: StandardDeviation[loi]
```

```
Out[3]:  $\sqrt{2}\beta$ 
```

```
In[4]: loi2 = TransformedDistribution[Abs[x], x  $\approx$  loi]
```

```
Out[4]: ExponentialDistribution  $\left[\frac{1}{\beta}\right]$ 
```

### Exercice 163

Soit  $X$  une *var* suivant la loi de Weibull  $\mathcal{W}(3,4)$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{3}{4} \left(\frac{x}{4}\right)^2 e^{-\left(\frac{x}{4}\right)^3} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser `Mathematica` pour

- créer la loi de  $X$  avec la commande `WeibullDistribution`,
- tracer le graphe de  $f(x)$  pour  $x \in [0, 5]$ ,
- déterminer la fonction de répartition de  $X$ ,
- déterminer la médiane de  $X$ , *i.e.* le réel  $x$  tel que  $F(x) = 0.5$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer une densité de  $Y = |X|^3$ .


Commandes : `WeibullDistribution`, `Plot`, `PDF`, `CDF`, `Median`, `Mean`, `Variance`, `TransformedDistribution`, `Abs`,  $\approx$ .

**Solution 163.** On propose :

```
In[1]: loi = WeibullDistribution[3, 4];
```

```
In[2]: Plot[PDF[loi, x], {x, 0, 9}]
```

```
Out[2]:
```



```
In[3]: CDF[loi, x]
```

```
Out[3]: 
$$\begin{cases} 1 - e^{-\frac{x^3}{64}} & x > 0 \\ 0 & \text{True} \end{cases}$$

```

```
In[4]: Median[loi, x]
```

```
Out[4]:  $4\text{Log}[2]^{1/3}$ 
```

```
In[5]: Mean[loi]
```

```
Out[5]:  $4\text{Gamma}\left[\frac{4}{3}\right]$ 
```

```
In[6]: Variance[loi]
```

```
Out[6]:  $16 \left( -\text{Gamma}\left[\frac{4}{3}\right]^2 + \text{Gamma}\left[\frac{5}{3}\right] \right)$ 
```

```
In[7]: loi2 = TransformedDistribution[Abs[x]^3, x ≈ loi];
```

```
In[8]: PDF[loi2, x]
```

```
Out[8]: 
$$\begin{cases} \frac{e^{-x/64}}{64} & x > 0 \\ 0 & \text{True} \end{cases}$$

```

!! On reconnaît une densité associée à la loi exponentielle  $\mathcal{E}\left(\frac{1}{64}\right)$ .

**Exercice 164**

Soient  $\mu \in \mathbb{R}$ ,  $\beta > 0$  et  $X$  une *var* suivant la loi logistique  $\mathcal{L}_{ogi}(\mu, \beta)$ , i.e. de densité :

$$f(x) = \frac{e^{-\frac{x-\mu}{\beta}}}{\beta(1 + e^{-\frac{x-\mu}{\beta}})^2}, \quad x \in \mathbb{R}.$$

Utiliser **Mathematica** pour

- créer la loi de  $X$  avec la commande **LogisticDistribution**,
- tracer le graphe de  $f(x)$  pour  $x \in [-6, 6]$ ,  $\mu \in [-3, 3]$  et  $\beta \in [0.5, 3]$  avec la commande **Manipulate** (sur  $\mu$  et  $\beta$ ) et les couleurs de **NeonColors**,
- déterminer la fonction de répartition  $F$  de  $X$ ,
- tracer le graphe de  $F(x)$  pour  $x \in [-6, 6]$ ,  $\mu \in [-3, 3]$  et  $\beta \in [0.5, 3]$  avec la commande **Manipulate** (sur  $\mu$  et  $\beta$ ) et les couleurs de **NeonColors**,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$  (on verra apparaître la fonction  $\text{Sinc}(x) = \frac{\sin(x)}{x}$ ,  $x \neq 0$ ).



Commandes : **LogisticDistribution**,  $\mu$ ,  $\beta$ , **Manipulate**, **Plot**, **PDF**, **ColorFunction**, **CDF**, **Mean**, **Variance**, **MomentGeneratingFunction**.

**Solution 164.** On propose :



```
In[1]: loi = LogisticDistribution[μ, β];
```

```
In[2]: Manipulate[Plot[PDF[LogisticDistribution[μ, β], x], {x, -6, 6},
ColorFunction -> "NeonColors"], {μ, -3, 3}, {σ, 0.5, 3}]
```

Out [2] :











In[3]: `CDF[loi, x]`

Out[3]: 
$$\frac{1}{1 + e^{-\frac{x-\mu}{\beta}}}$$

In[4]: `Manipulate[Plot[CDF[LogisticDistribution[ $\mu$ ,  $\beta$ ], x], {x, -6, 6},  
ColorFunction  $\rightarrow$  "NeonColors"], { $\mu$ , -3, 3}, { $\beta$ , 0.5, 3}]`

Out[4]:



```

In[5]: Mean[loi]
Out[5]:  $\mu$ 
In[6]: Variance[loi]
Out[6]:  $\frac{\pi^2\beta^2}{3}$ 
In[7]: MomentGeneratingFunction[loi, t]
Out[6]:  $\frac{e^{t\mu}}{\text{Sinc}[\pi t\beta]}$ 

```

**Exercice 165**

Soit  $X$  une *var* de densité :

$$f(x) = \begin{cases} \frac{1}{x} & \text{si } x \in [1, e], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- vérifier que  $f$  est bien une densité,
- déterminer la fonction de répartition  $F$  de  $X$ ,
- déterminer le réel  $x$  tel que  $F(x) = 0.5$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- calculer  $\mathbb{P}(X^2 - 3X + 2 \leq 0)$ .

Commandes : Integrate, E, ProbabilityDistribution, CDF, Quantile, Mean, Variance, Probability,  $\approx$ .

**Solution 165.** On propose :

```

In[1]: Integrate[1 / x, {x, 1, E}] == 1
Out[1]: True
In[2]: loi = ProbabilityDistribution[1 / x, {x, 1, E}];
In[3]: CDF[loi, x]
Out[3]:  $\begin{cases} 1 & x > e \\ \text{Log}[x] & 1 < x \leq e \\ 0 & \text{True} \end{cases}$ 
In[4]: Quantile[loi, 0.5]
Out[4]: 1.64872

```

In[5]: Mean[loi]

Out[5]:  $-1 + e$

In[6]: Variance[loi]

Out[6]:  $-\frac{1}{2}(-3 + e)(-1 + e)$

In[7]: Probability[x^2 + -3 x + 2 ≤ 0, x ≈ loi]

Out[7]: Log[2]

### Exercice 166

Soit  $X$  une *var* suivant la loi bêta  $B(2, 2)$  *i.e.* de densité :

$$f(x) = \begin{cases} 6x(1-x) & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- créer la loi de  $X$  avec la commande `BetaDistribution`,
- tracer le graphe de la fonction de répartition de  $X$  :  $F(x) = \mathbb{P}(X \leq x)$  pour  $x \in [-1, 2]$ ,
- déterminer le réel  $x$  tel que  $F(x) = 0.64$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\sigma(X)$ ,
- calculer la valeur numérique de  $\mathbb{E}(X^{20})$ .


Commandes : `BetaDistribution`, `Plot`, `CDF`, `Quantile`, `Mean`, `//N`, `StandardDeviation`, `Moment`.

**Solution 166.** On propose :

In[1]: loi = BetaDistribution[2, 2];

In[2]: Plot[CDF[loi, x], {x, -1, 2}]

Out[2]:



```

In[3]: Quantile[loi, 0.64]
Out[3]: 0.594457
In[4]: Mean[loi] //N
Out[4]: 0.5
In[5]: StandardDeviation[loi] //N
Out[5]: 0.223607
In[6]: Moment[loi, 20] //N
Out[6]: 0.0118577

```

**Exercice 167**

Soit  $X$  une *var* de densité :

$$f(x) = ce^{-x^2(1+x^2)}, \quad x \in \mathbb{R},$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\int_{-\infty}^{\infty} f(x)dx = 1$ .

Utiliser *Mathematica* pour

- calculer la valeur numérique de  $c$ ,
- créer la loi de  $X$  en utilisant la valeur numérique de  $c$  dans la définition de  $f$ ,
- tracer le graphe de  $f(x)$  pour  $x \in [-2.5, 2.5]$ ,
- calculer la valeur numérique de  $\mathbb{P}_{\{X > -0.6\}}(|X| \leq 0.4)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\sigma(X)$ ,
- calculer la valeur numérique de  $\mathbb{E}_{\{X \geq 0.5\}}(X^2)$ .

Commandes : `NIntegrate`, `Exp`, `∞`, `ProbabilityDistribution`, `Plot`, `PDF`, `NProbability`, `▷`, `≈`, `Abs`, `Mean`, `StandardDeviation`, `NExpectation`.


**Solution 167.** On propose :

```

In[1]: c = 1 / NIntegrate[Exp[-x^2 (1 + x^2)], {x, -∞, ∞}]
Out[1]: 0.730766
In[2]: loi = ProbabilityDistribution[c Exp[-x^2 (1 + x^2)] , {x, -∞, ∞}];
In[3]: Plot[PDF[loi, x], {x, -2.5, 2.5}]

```

Out [3] :



In[4]: `NProbability[Abs[x] ≤ 0.4 & x > -0.6, x ≈ loi]`

Out [4]: `0.625709`

In[5]: `Mean[loi]`

Out [5]: `0.`

In[6]: `StandardDeviation[loi]`

Out [6]: `0.483694`

In[7]: `NExpectation[x^2 & x ≥ 0.5, x ≈ loi]`

Out [7]: `0.548229`

### Exercice 168

Soit  $X$  une var de densité :

$$f(x) = c \frac{1}{\cosh(x)}, \quad x \in \mathbb{R},$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\int_{-\infty}^{\infty} f(x) dx = 1$ .

Utiliser **Mathematica** pour

- donner l'expression de  $c$ ,
- créer la loi de  $X$  en utilisant l'expression de  $c$  dans la définition de  $f$ ,
- donner l'expression de la fonction de répartition de  $X$ ,
- calculer la valeur numérique de  $\mathbb{P}(-1.4 \leq X \leq 2.1)$ ,
- calculer la valeur numérique de  $\mathbb{P}_{\{|X| \leq 2.1\}}(X^3 + 2X \leq 0.9)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\sigma(X)$ .

Commandes : `Integrate`, `Cosh`, `∞`, `ProbabilityDistribution`, `CDF`, `NProbability`, `≈`, `▷`, `Abs`, `Mean`, `StandardDeviation`.

**Solution 168.** On propose :

```
In[1]: c = 1 / Integrate[1 / Cosh[x], {x, -∞, ∞}]
Out[1]:  $\frac{1}{\pi}$ 
In[2]: loi = ProbabilityDistribution[c 1 / Cosh[x], {x, -∞, ∞}];
In[3]: CDF[loi, x]
Out[3]:  $\frac{2\text{ArcTan}[e^x]}{\pi}$ 
In[4]: NProbability[-1.4 ≤ x ≤ 2.1, x ≈ loi]
Out[4]: 0.76851
In[5]: NProbability[x^3 + 2x ≤ 0.9 & Abs[x] ≤ 2.1, x ≈ loi]
Out[5]: 0.651849
In[6]: Mean[loi]
Out[6]: 0
In[7]: StandardDeviation[loi]
Out[7]:  $\frac{\pi}{2}$ 
```

### Exercice 169

Soient  $\theta > 0$  et  $X$  une *var* de densité :

$$f(x) = \begin{cases} (\theta + 1)(\theta + 2)(1 - x)x^\theta & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $f$  avec  $\theta$  en argument,
- tracer le graphe de  $f$  pour plusieurs valeurs de  $\theta$  avec la commande *Manipulate*, les couleurs de *NeonColors* et l'option : *GridLines* → *Automatic*,
- déterminer la fonction de répartition de  $X$ ,
- calculer la valeur numérique de  $\mathbb{P}(X^3 + 2X > 0.4)$  avec  $\theta = 0.51$ ,
- calculer  $\mathbb{E}(X)$ .





Commandes : `x_`, `θ`, `:=`, `Manipulate`, `Plot`, `ColorFunction`, `GridLines`, `Range`, `ProbabilityDistribution`, `CDF`, `Probability`, `≈`, `/.` →, `Mean`.

**Solution 169.** On propose :

```
In[1]: f[x_, θ_] := (θ + 1) (θ + 2) (1 - x) x^θ;
```

```
In[2]: Manipulate[Plot[f[x, θ], {x, 0, 1}, ColorFunction -> "NeonColors",
  GridLines -> Automatic], {θ, 1, 16}, {θ, Range[16]}]
```

Out [2]:



```
In[3]: loi = ProbabilityDistribution[f[x, θ], {x, 0, 1}];
```

```
In[4]: CDF[loi, x]
```

Out [4]:

$$\begin{cases} 1 & x \geq 1 \\ x^{1+\theta}(2 + \theta - x(1 + \theta)) & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

```
In[5]: Probability[x^3 + 2 x > 0.4, x ≈ loi] /. θ -> 0.51
```

Out [5]: 0.810692

In[6]: Mean[loi]

Out[6]:  $\frac{1+\theta}{3+\theta}$

### Exercice 170

Soit  $X$  une *var* de densité :

$$f(x) = \begin{cases} \frac{1}{2} & \text{si } x \in ]1, 2], \\ \frac{1}{4\sqrt{x}} & \text{si } x \in ]0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la fonction  $f$ ,
- tracer le graphe de  $f(x)$  pour  $x \in [-0.5, 2.5]$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\mathbb{V}(X)$ ,
- calculer la valeur numérique du coefficient d'asymétrie :  $\gamma_1 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^3 \right)$ ,
- calculer la valeur numérique du kurtosis :  $\beta_2 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^4 \right)$ .


Commandes : `x_`, `:=`, `Piecewise`, `Sqrt`, `Plot`, `ProbabilityDistribution`, `Mean`, `//N`, `Variance`, `Skewness`, `Kurtosis`.

**Solution 170.** On propose :

In[1]: `f[x_] := Piecewise[{{1 / 2, 1 < x ≤ 2}, {1 / (4 Sqrt[x]), 0 < x ≤ 1}}];`

In[2]: `Plot[f[x], {x, -0.5, 2.5}]`

Out[2]:





```

In[3]: loi = ProbabilityDistribution[f[x], {x, 0, 2}];
In[4]: Mean[loi] //N
Out[4]: 0.916667
In[5]: Variance[loi] //N
Out[5]: 0.426389
In[6]: Skewness[loi] //N
Out[6]: 0.012946
In[7]: Kurtosis[loi] //N
Out[7]: 1.57618

```

### Exercice 171

Soit  $X$  une *var* de densité :

$$f(x) = \begin{cases} (x-2)^5 & \text{si } x \in [2, 3[, \\ x-1 & \text{si } x \in [1, 2[, \\ x^2 & \text{si } x \in [0, 1[, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- tracer le graphe de  $f(x)$  pour  $x \in [-0.5, 3.5]$ ,
- déterminer la fonction de répartition  $F$  de  $X$ ,
- tracer le graphe de  $F(x)$  pour  $x \in [-0.5, 3.5]$ ,
- déterminer la médiane de  $X$ , *i.e.* le réel  $x$  tel que  $F(x) = 0.5$ ,
- déterminer le réel  $x$  tel que  $F(x) = 0.34$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\mathbb{V}(X)$ ,
- calculer la valeur numérique du coefficient d'asymétrie :  $\gamma_1 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^3 \right)$ ,
- calculer la valeur numérique du kurtosis :  $\beta_2 = \mathbb{E} \left( \left( \frac{X - \mathbb{E}(X)}{\sigma(X)} \right)^4 \right)$ .


Commandes : `x_`, `:=`, `Piecewise`, `Plot`, `ProbabilityDistribution`, `CDF`, `//Simplify`, `Median`, `//N`, `Quantile`, `Mean`, `Variance`, `Skewness`, `Kurtosis`.

**Solution 171.** On propose :

```
In[1]: f[x_] := Piecewise[{{(x - 2)^5, 2 ≤ x < 3}, {x - 1, 1 ≤ x < 2},
{x^2, 0 ≤ x < 1}}];
```

```
In[2]: Plot[f[x], {x, -0.5, 3.5}]
```

Out[2]:



```
In[3]: loi = ProbabilityDistribution[f[x], {x, 0, 3}];
```


```
In[4]: CDF[loi, x] //Simplify
```

Out[4]:

$$\begin{cases} \frac{x^3}{3} & 0 < x \leq 1 \\ \frac{1}{6}(5 - 6x + 3x^2) & 1 < x \leq 2 \\ 1 & x \geq 3 \\ \frac{1}{6}(69 - 192x + 240x^2 - 160x^3 + 60x^4 - 12x^5 + x^6) & 2 < x < 3 \\ 0 & \text{True} \end{cases}$$

```
In[5]: Plot[CDF[loi, x], {x, -0.5, 3.5}]
```

Out[5]:



```

In[6]: Median[loi] //N
Out[6]: 1.57735
In[7]: Quantile[loi, 0.34]
Out[7]: 1.11547
In[8]: Mean[loi] //N
Out[8]: 1.55952
In[9]: Variance[loi] //N
Out[9]: 0.547647
In[10]: Skewness[loi] //N
Out[10]: 0.420081
In[11]: Kurtosis[loi] //N
Out[11]: 2.33726

```

### Exercice 172

Soit  $X$  une *var* de densité :

$$f(x) = \begin{cases} \frac{2}{\pi(2x^2 - 2x + 1)} & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- vérifier que  $f$  est bien une densité,
- créer la loi de  $X$ ,
- tracer, sur un même graphique, le graphe de  $f$  ainsi que l'aire sous la courbe correspondante à  $\mathbb{P}(0.37 < X < 0.71)$ , le tout en vert,
- calculer  $\mathbb{P}(0.37 < X < 0.71)$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{E}(4X - 2)$ ,
- calculer  $\mathbb{E}(2X^2 - 2X + 1)$ ,
- calculer  $\mathbb{V}(X)$ .

Commandes : Integrate, Pi, ProbabilityDistribution, With, PDF, Plot, If, Filling, PlotStyle, ColorData, Probability, ≈, Mean, Expectation, Variance.

**Solution 172.** On propose :


```
In[1]: Integrate[2 / (Pi (2 x^2 - 2 x + 1)), {x, 0, 1}] == 1
```

```
Out[1]: True
```

```
In[2]: loi = ProbabilityDistribution[2 / (Pi (2 x^2 - 2 x + 1)), {x, 0, 1}];
```

```
In[3]: With[{dens = PDF[loi]}, Plot[{If[0.37 < x < 0.71, dens[x]], dens[x]},  
{x, 0, 1}, Filling -> {1 -> Axis}, PlotStyle -> ColorData[1, 6]]]
```

```
Out[3]:
```



```
In[4]: Probability[0.37 < x < 0.71, x ≈ loi]
```

```
Out[4]: 0.415074
```

```
In[5]: Mean[loi]
```

```
Out[5]:  $\frac{1}{2}$ 
```

```
In[6]: Expectation[4 x - 2, x ≈ loi]
```

```
Out[6]: 0
```

```
In[7]: Expectation[2 x^2 - 2 x + 1, x ≈ loi]
```

```
Out[7]:  $\frac{2}{\pi}$ 
```

```
In[8]: Variance[loi]
```

```
Out[8]:  $\frac{4 - \pi}{4\pi}$ 
```

## Exercice 173

Soit  $X$  une *var* de densité :

$$f(x) = \begin{cases} 4 \frac{\log(x)}{x^3} & \text{si } x \geq 1, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- vérifier que  $f$  est bien une densité,
- calculer  $\mathbb{P}(1.2 \leq X \leq 3.4)$ ,
- déterminer le réel  $x$  tel que  $\mathbb{P}(X \leq x) = 0.16$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{\log(X)}\right)$ .

Commandes : `Integrate`, `Log`, `∞`, `ProbabilityDistribution`, `Probability`, `≈`, `Quantile`, `Mean`, `Expectation`.

**Solution 173.** On propose :

```
In[1]: Integrate[4 Log[x] / x^3, {x, 1, ∞}] == 1
```

```
Out[1]: True
```

```
In[2]: loi = ProbabilityDistribution[4 Log[x] / x^3, {x, 1, ∞}];
```

```
In[3]: Probability[1.2 ≤ x ≤ 3.4, x ≈ loi]
```

```
Out[3]: 0.162359
```

```
In[4]: Quantile[loi, 0.16]
```

```
Out[4]: 2.97109
```

```
In[5]: Mean[loi]
```

```
Out[5]: 1
```

```
In[6]: Expectation[1 / Log[x], x ≈ loi]
```

```
Out[6]: 1/2
```

## Exercice 174

Soit  $X$  une *var* de fonction de répartition :

$$F(x) = \begin{cases} 1 - \left(1 + \frac{x}{2}\right)^2 e^{-x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- déterminer une densité de  $X$  avec la commande D,
- déterminer une densité de  $X$  avec la commande ProbabilityDistribution.

Commandes : x\_, :=, Piecewise, Exp, D, //Simplify, ProbabilityDistribution, CDF, ∞, PDF.

**Solution 174.** On propose :

```
In[1]: F[x_] := Piecewise[{{1 - (1 + x / 2)^2 Exp[-x], x ≥ 0}}];
```

```
In[2]: D[F[x], x] //Simplify
```

```
Out[2]: 
$$\begin{cases} \frac{1}{4}e^{-x}x(2+x) & x > 0 \\ 0 & \text{True} \end{cases}$$

```

```
In[3]: loi = ProbabilityDistribution[{"CDF", F[x]}, {x, -∞, ∞}]
```

```
In[4]: PDF[loi, x] //Simplify
```

```
Out[4]: 
$$\begin{cases} \frac{1}{4}e^{-x}x(2+x) & x > 0 \\ 0 & \text{True} \end{cases}$$

```

### Exercice 175

Soit  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([0, 1])$ , *i.e.* de densité :

$$f(x) = \begin{cases} 1 & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- déterminer une densité de  $Y = -\frac{1}{5} \log(1 - X)$ ,
- déterminer une densité de  $Z = X(1 - X)$ ,
- déterminer une densité de  $W = -\log\left(\frac{X}{1 - X}\right)$ .

Commandes : UniformDistribution, TransformedDistribution, Log, ≈, PDF.

**Solution 175.** On propose :

```
In[1]: loi = UniformDistribution[{0, 1}];
```

```
In[2]: loi2 = TransformedDistribution[-(1 / 5) Log[1 - x], x ≈ loi];
```

In[3]: PDF[loi2, x]

$$\text{Out [3]: } \begin{cases} 5e^{-5x} & x > 0 \ \&\& \ e^{5x} \geq 1 \\ 0 & \text{True} \end{cases}$$

!! On reconnaît une densité associée à la loi exponentielle  $\mathcal{E}(5)$ .

In[4]: loi3 = TransformedDistribution[x (1 - x), x ≈ loi];

In[5]: PDF[loi3, x]

$$\text{Out [5]: } \begin{cases} \text{Indeterminate} & x == 0 \ || \ x == \frac{1}{4} \\ \frac{2}{\sqrt{1-4x}} & 0 < x < \frac{1}{4} \\ 0 & \text{True} \end{cases}$$

In[6]: loi4 = TransformedDistribution[-Log[x / (1 - x)], x ≈ loi];

In[7]: PDF[loi4, x]

$$\text{Out [7]: } \frac{e^x}{(1 + e^x)^2}$$

!! On reconnaît une densité associée à la loi logistique  $\mathcal{L}_{logi}(0, 1)$ .

### Exercice 176

Soit  $X$  une *var* suivant la loi exponentielle  $\mathcal{E}(3)$ , *i.e.* de densité :

$$f(x) = \begin{cases} 3e^{-3x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- déterminer une densité de  $Y = \log(e^X - 1)$ ,
- déterminer la loi de  $Z = [X]$  (partie entière de  $X$ ).

Commandes : ExponentialDistribution, TransformedDistribution, Log, Exp, ≈, PDF, Floor.

**Solution 176.** On propose :

In[1]: loi = ExponentialDistribution[3];

In[2]: loi2 = TransformedDistribution[Log[Exp[x] - 1], x ≈ loi];

In[3]: PDF[loi2, x]

$$\text{Out [3]: } \frac{3e^x}{(1 + e^x)^4}$$

```
In[4]: loi3 = TransformedDistribution[Floor[x], x ≈ loi]
Out[4]: GeometricDistribution[1 - 1/e^3]
In[5]: PDF[loi3, x]
Out[5]: { (1 - 1/e^3) e^{-3x} x ≥ 0
          0 True
```

**Exercice 177**

Soit  $X$  une *var* suivant la loi normale  $\mathcal{N}(0, 1)$ , *i.e.* de densité :

$$f(x) = \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}}, \quad x \in \mathbb{R}.$$

Utiliser *Mathematica* pour

- déterminer une densité de  $T = aX + b$  avec  $(a, b) \in \mathbb{R}^2$ ,
- déterminer une densité de  $U = |X|$ ,
- déterminer une densité de  $V = X(X + 4)$ ,
- déterminer une densité de  $W = e^X$ ,
- déterminer une densité de  $Z = e^{X^2}$ .

Commandes : `NormalDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Abs`, `Exp`.

**Solution 177.** On propose :

```
In[1]: loi = NormalDistribution[0, 1];
In[2]: loi2 = TransformedDistribution[a x + b, x ≈ loi]
Out[2]: NormalDistribution[b, Abs[a]]
In[3]: PDF[loi2, x]
Out[3]: e^{-(-b+x)^2 / (2Abs[a]^2)} / sqrt(2π Abs[a])
In[4]: loi3 = TransformedDistribution[Abs[x], x ≈ loi]
Out[4]: HalfNormalDistribution[sqrt(π/2)]
In[5]: PDF[loi3, x]
Out[5]: { e^{-x^2/2} sqrt(2/π) x > 0
          0 True
```



```
In[6]: loi4 = TransformedDistribution[x (x + 4), x ≈ loi];
```

```
In[7]: PDF[loi4, x]
```

$$\text{Out [7]: } \begin{cases} \text{Indeterminate} & x == -4 \\ \frac{e^{-4-\frac{x}{2}-2\sqrt{4+x}} (1 + e^{4\sqrt{4+x}})}{2\sqrt{2\pi}\sqrt{4+x}} & x > -4 \\ 0 & \text{True} \end{cases}$$

```
In[8]: loi5 = TransformedDistribution[Exp[x], x ≈ loi]
```

```
Out [8]: LogNormalDistribution[0, 1]
```

```
In[9]: PDF[loi5, x]
```

$$\text{Out [9]: } \begin{cases} \frac{e^{-\frac{1}{2}\text{Log}[x]^2}}{\sqrt{2\pi x}} & x > 0 \\ 0 & \text{True} \end{cases}$$

```
In[10]: loi6 = TransformedDistribution[Exp[x^2], x ≈ loi];
```

```
In[11]: PDF[loi6, x]
```

$$\text{Out [11]: } \begin{cases} \text{Indeterminate} & x == 1 \\ \frac{1}{\sqrt{2\pi x^{3/2}} \sqrt{\text{Log}[x]}} & x > 1 \\ 0 & \text{True} \end{cases}$$

### Exercice 178

Soit  $X$  une *var* suivant la loi de Cauchy  $\mathcal{C}(0, 1)$ , *i.e.* de densité :

$$f_X(x) = \frac{1}{\pi(1+x^2)}, \quad x \in \mathbb{R}.$$

Utiliser *Mathematica* pour

- déterminer une densité de  $T = \ln(|X|)$ ,
- déterminer une densité de  $U = \frac{1}{X}$ ,
- déterminer une densité de  $V = \arctan(X)$ ,
- déterminer une densité de  $W = \frac{2X}{1-X^2}$ .

Commandes : `CauchyDistribution`, `TransformedDistribution`, `Log`, `Abs`, `≈`, `PDF`, `ArcTan`.

**Solution 178.** On propose :

In[1]: `loi = CauchyDistribution[0, 1];`

In[2]: `loi2 = TransformedDistribution[Log[Abs[x]], x ≈ loi];`

In[3]: `PDF[loi2, x]`

Out[3]: 
$$\frac{2e^x}{\pi + e^{2x}\pi}$$

In[4]: `loi3 = TransformedDistribution[1 / x, x ≈ loi]`

Out[4]: `CauchyDistribution[0, 1]`

In[5]: `PDF[loi3, x]`

Out[5]: 
$$\frac{1}{\pi(1 + x^2)}$$

In[6]: `loi4 = TransformedDistribution[ArcTan[x], x ≈ loi];`

In[7]: `PDF[loi4, x]`

Out[7]: 
$$\begin{cases} \frac{1}{\pi} & -\frac{\pi}{2} < x < \frac{\pi}{2} \\ 0 & \text{True} \end{cases}$$

!! On reconnaît une densité associée à la loi uniforme  $\mathcal{U}\left(-\frac{\pi}{2}, \frac{\pi}{2}\right]$ .

In[8]: `loi5 = TransformedDistribution[2 x / (1 - x^2), x ≈ loi];`

In[9]: `PDF[loi5, x]`

Out[9]: 
$$\frac{1}{\pi + \pi x^2}$$

!! On reconnaît une densité associée à la loi de Cauchy  $\mathcal{C}(0, 1)$ .

### Exercice 179

Soit  $X$  une *var* de fonction de répartition :

$$F(x) = \frac{1}{1 + e^{-x}}, \quad x \in \mathbb{R}.$$

Utiliser *Mathematica* pour

- créer la loi de  $X$  à partir de  $F$ ,
- vérifier que  $\lim_{x \rightarrow \infty} F(x) = 1$ ,
- vérifier que  $\lim_{x \rightarrow -\infty} F(x) = 0$ ,
- vérifier que  $(x - y)(F(x) - F(y)) \geq 0$  avec la commande `Refine`,
- déterminer une densité de  $Y = \frac{e^X - 1}{e^X + 1}$ .

Commandes : ProbabilityDistribution, CDF, Exp,  $\infty$ , Limit, Refine, Element, Reals, TransformedDistribution.

**Solution 179.** On propose :

```
In[1]: loi = ProbabilityDistribution[{"CDF", 1/(1 + Exp[-x])}, {x, -∞, ∞}];
In[2]: Limit[CDF[loi, x], x → ∞] == 1
Out[2]: True
In[3]: Limit[CDF[loi, x], x → -∞] == 0
Out[3]: True
In[4]: Refine[(x - y) (CDF[loi, x] - CDF[loi, y]) ≥ 0, Element[{x, y}, Reals]]
Out[4]: True
In[5]: loi2 = TransformedDistribution[(Exp[x] + 1) / (Exp[x] - 1), x ≈ loi];
In[6]: PDF[loi2, x]
Out[6]: { Indeterminate x == -1 || x == 1
          { 1/(2x^2) x > 1 || x < -1
          { 0 True
```

### Exercice 180

Soit  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([0, 1])$ , *i.e.* de densité :

$$f(x) = \begin{cases} 1 & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

On pose  $Y = \frac{\sup(X, 1 - X)}{\inf(X, 1 - X)}$ .

Utiliser **Mathematica** pour

- déterminer une densité de  $Y$ ,
- calculer  $\mathbb{P}(2 \leq Y \leq 4)$ ,
- calculer la médiane de  $Y$ , *i.e.* le réel  $x$  tel que  $\mathbb{P}(Y \leq x) = 0.5$ ,
- calculer  $\mathbb{E}(\sqrt{Y})$ .

Commandes : UniformDistribution, TransformedDistribution, Max, Min,  $\approx$ , PDF, Probability, Median, Expectation, Sqrt.

**Solution 180.** On propose :

```
In[1]: loi = UniformDistribution[{0, 1}];
In[2]: loi2 = TransformedDistribution[Max[x, 1 - x] / Min[x, 1 - x], x ≈ loi];
In[3]: PDF[loi2, x]
Out[3]: 
$$\begin{cases} \frac{2}{(1+x)^2} & x > 1 \\ 0 & \text{True} \end{cases}$$

In[4]: Probability[2 ≤ y ≤ 4, y ≈ loi2]
Out[4]:  $\frac{4}{15}$ 
In[5]: Median[loi2]
Out[5]: 3
In[6]: Expectation[Sqrt[y], y ≈ loi2]
Out[6]:  $\frac{2 + \pi}{2}$ 
```

### Exercice 181

Soit  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([0, 2])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{2} & \text{si } x \in [0, 2], \\ 0 & \text{sinon.} \end{cases}$$

On pose  $Y = \inf(X, X^2)$ .

Utiliser **Mathematica** pour

- déterminer une densité de  $Y$ ,
- déterminer la fonction de répartition de  $Y$ ,
- calculer  $\mathbb{E}(Y^2 + 5Y + 2)$ ,
- calculer  $\mathbb{E}_{\{0.8 < Y < 1.5\}}(e^Y)$ .

Commandes : UniformDistribution, TransformedDistribution, Min, ≈, PDF, CDF, Expectation, Exp, ▷.

**Solution 181.** On propose :

```
In[1]: loi = UniformDistribution[{0, 2}];
In[2]: loi2 = TransformedDistribution[Min[x, x^2], x ≈ loi];
```

```

In[3]: PDF[loi2, x]
Out[3]: 
$$\begin{cases} \frac{1}{2} & 1 < x < 2 \\ \frac{1}{4\sqrt{x}} & 0 < x \leq 1 \\ 0 & \text{True} \end{cases}$$

In[4]: CDF[loi2, x]
Out[4]: 
$$\begin{cases} 1 & x \geq 2 \\ \frac{\sqrt{x}}{2} & 0 < x \leq 1 \\ \frac{x}{2} & 1 < x < 2 \\ 0 & \text{True} \end{cases}$$

In[5]: Expectation[y^2 + 5y + 2, y ≈ loi2]
Out[5]:  $\frac{157}{20}$ 
In[6]: Expectation[Exp[y] > 0.8 < y < 1.5, y ≈ loi2]
Out[6]: 3.34068

```

**Exercice 182**

Soient  $\lambda > 0$  et  $X$  une *var* suivant la loi uniforme  $\mathcal{U}([0, 2])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{2}{3}\lambda e^{-\lambda x} & \text{si } x \geq 0, \\ \frac{1}{3}\lambda e^{\lambda x} & \text{sinon.} \end{cases}$$

On pose  $Y = e^{\lambda X}$ .

Utiliser **Mathematica** pour

- vérifier que  $f$  est bien une densité,
- déterminer une densité  $f_Y$  de  $Y$ ,
- tracer le graphe de  $f_Y(x)$  pour  $x \in [0, 6]$ ,
- déterminer la fonction de répartition  $F_Y$  de  $Y$ ,
- tracer le graphe de  $F_Y(x)$  pour  $x \in [0, 6]$ ,
- calculer  $\mathbb{E}_{\{Y < 5.1\}}(Y)$ .

Commandes : `x_`, `:=`, `Piecewise`, `λ`, `Exp`, `Integrate`, `∞`, `Assumptions`, `ProbabilityDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Plot`, `//Evaluate`, `CDF`, `//Simplify`, `Expectation`, `>`.

**Solution 182.** On propose :

```
In[1]: f[x_] := Piecewise[{{2 / 3 λ Exp[-λ x], x ≥ 0},
  {1 / 3 λ Exp[λ x], x < 0}}];
```

```
In[2]: Integrate[f[x], {x, -∞, ∞}, Assumptions → λ > 0] == 1
```

```
Out[2]: True
```

```
In[3]: loi = ProbabilityDistribution[f[x], {x, -∞, ∞}, Assumptions → λ > 0];
```


```
In[4]: loi2 = TransformedDistribution[Exp[λ x], x ≈ loi];
```

```
In[5]: PDF[loi2, x]
```

```
Out[5]: {
  1/3          0 < x < 1
  Indeterminate x == 0 || x == 1
  2/3x^2      x > 1
  0           True
```

```
In[6]: Plot[PDF[loi2, x] //Evaluate, {x, 0, 6}]
```

```
Out[6]:
```




```
In[7]: CDF[loi2, x] //Simplify
```

```
Out[7]: {
  x/3          0 < x ≤ 1
  1 - 2/3x     x > 1
  0           True
```

```
In[8]: Plot[CDF[loi2, x] //Evaluate, {x, 0, 6}]
```

Out [8]:

In[9]: `Expectation[y > y < 5.1, y ≈ loi2]`

Out [9]: 1.44122

**Exercice 183**Soit  $X$  une *var* de densité :

$$f(x) = \frac{3}{5} \times \frac{1}{\sqrt{2\pi \times 0.22^2}} e^{-\frac{(x-2)^2}{2 \times 0.22^2}} + \frac{2}{5} \times \frac{1}{\sqrt{2\pi \times 0.34^2}} e^{-\frac{(x-3)^2}{2 \times 0.34^2}}, \quad x \in \mathbb{R}.$$


Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `MixtureDistribution`,
- tracer le graphe de  $f(x)$  pour  $x \in [1, 4]$ ,
- calculer  $\mathbb{E}(X)$ ,
- calculer  $\mathbb{V}(X)$ ,
- déterminer la fonction génératrice des moments de  $X$ .

Commandes : `NormalDistribution`, `MixtureDistribution`, `Plot`, `PDF`, `Mean`, `Variance`, `MomentGeneratingFunction`.

**Solution 183.** On propose :In[1]: `loi1 = NormalDistribution[2, 0.22];`In[2]: `loi2 = NormalDistribution[3, 0.34];`In[3]: `loi3 = MixtureDistribution[{3, 2}, {loi1, loi2}];`In[4]: `Plot[PDF[loi3, x], {x, 1, 4}]`

Out [4]:



In [5]: Mean[loi3]

Out [5]: 2.4

In [6]: Variance[loi3]

Out [6]: 0.31528

In [7]: MomentGeneratingFunction[loi3, t]

Out [7]:  $\frac{1}{5}(3e^{2t+0.0242t^2} + 2e^{3t+0.0578t^2})$ **Exercice 184**

Exécuter les commandes suivantes une à une et comprendre leurs enjeux :

```
loi[p_] := MixtureDistribution[{p, 1 - p}, {NormalDistribution[-1, 0.5],
NormalDistribution[1, 0.5]}];
Table[Plot[PDF[loi[p], x] //Evaluate, {x, -3, 3}, PlotLabel ->
NumberForm[{p, 1 - p}, 2], Filling -> Axis, PlotRange -> {0, 0.8}],
{p, 0., 1., 0.1}]
```

Commandes : x\_, MixtureDistribution, NormalDistribution, Table, Plot, PDF, //Evaluate, PlotLabel, NumberForm, Filling, PlotRange.


**Solution 184.** On propose :

```
In [1]: loi[p_] := MixtureDistribution[{p, 1 - p}, {NormalDistribution[-1, 0.5],
NormalDistribution[1, 0.5]}];
```

```
In [2]: Table[Plot[PDF[loi[p], x] //Evaluate, {x, -3, 3}, PlotLabel ->
NumberForm[{p, 1 - p}, 2], Filling -> Axis, PlotRange -> {0, 0.8}],
{p, 0., 1., 0.1}]
```



Out [2] :



On obtient alors les graphes de la densité :

$$f(x) = p \frac{1}{\sqrt{2\pi 0.5^2}} e^{-\frac{(x+1)^2}{2 \times 0.5^2}} + (1-p) \frac{1}{\sqrt{2\pi 0.5^2}} e^{-\frac{(x-1)^2}{2 \times 0.5^2}}, \quad x \in \mathbb{R},$$

pour les valeurs de  $p \in \{0, 0.1, 0.2, \dots, 0.9, 1\}$ .**Exercice 185**Soit  $X$  une *var* de densité :

$$f(x) = \frac{1}{4} \times \frac{1}{2} e^{-\frac{|x|}{2}} + \frac{3}{4} \times \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}}, \quad x \in \mathbb{R}.$$

Utiliser Mathematica pour


- tracer le graphe de  $f(x)$  pour  $x \in [-3, 3]$ ,
- calculer la valeur numérique de  $\mathbb{P}(|X| \leq 1)$ ,
- calculer la valeur numérique de  $\mathbb{E}(X)$ ,
- calculer la valeur numérique de  $\mathbb{V}(X)$ .

Commandes : LaplaceDistribution, NormalDistribution, MixtureDistribution, Plot, PDF, NProbability, Abs,  $\approx$ , Mean, //N, Variance.

**Solution 185.** On propose :

```
In[1]: loi1 = LaplaceDistribution[0, 1];
In[2]: loi2 = NormalDistribution[0, 1];
In[3]: loi3 = MixtureDistribution[{1, 3}, {loi1, loi2}];
In[4]: Plot[PDF[loi3, x], {x, -3, 3}]
```

Out[4]:



```
In[5]: NProbability[Abs[x] ≤ 1, x ≈ loi3]
```

Out[5]: 0.670047

```
In[6]: Mean[loi3] //N
```

Out[6]: 0

```
In[7]: Variance[loi3] //N
```

Out[7]: 1.25

### Exercice 186

*Inégalité de Bienaymé-Tchebychev* : Soit  $X$  une var admettant un moment d'ordre 2. Alors, pour tout  $\epsilon > 0$ , on a

$$\mathbb{P}(|X - \mathbb{E}(X)| > \epsilon) \leq \frac{\mathbb{V}(X)}{\epsilon^2}.$$

L'objectif de cet exercice est d'illustrer cette inégalité. Utiliser `Mathematica` pour

- créer une fonction  $f$  a deux arguments : `eps` et `loi`, telle que, pour  $X$  suivant la loi `loi`, on a

$$f(\text{eps}, \text{loi}) = \mathbb{P}(|X - \mathbb{E}(X)| > \text{eps}),$$

- créer une fonction  $g$  a deux arguments : `eps` et `loi`, telle que, pour  $X$  suivant la loi `loi`, on a

$$g(\text{eps}, \text{loi}) = \frac{\mathbb{V}(X)}{\text{eps}^2},$$

- représenter sur un même graphique les graphes de  $f(\text{eps}, \text{loi})$  et  $g(\text{eps}, \text{loi})$  avec  $\text{eps} \in [0.1, 6]$  et successivement :

- ◇ `loi = ExponentialDistribution[2]`,
- ◇ `loi = NormalDistribution[1.3, 0.5]`,
- ◇ `loi = GammaDistribution[2, 0.5]`.

Commandes : `NProbability`, `Abs`, `Expectation`, `≈`, `Plot`, `ExponentialDistribution`, `PlotLabels`, `NormalDistribution`, `GammaDistribution`.


**Solution 186.** On propose :

```
In[1]: f[eps_, loi_] := NProbability[Abs[x - Expectation[x, x ≈ loi]] ≥ eps,
x ≈ loi];
```

```
In[2]: g[eps_, loi_] := Expectation[(x - Expectation[x, x ≈ loi])^2,
x ≈ loi] / eps^2;
```


```
In[3]: Plot[{f[eps, ExponentialDistribution[2]],
g[eps, ExponentialDistribution[2]]}, {eps, 0.1, 6},
PlotLabels → {"f(eps, loi)", "g(eps, loi)"}]
```

Out [3] :




```
In[4]: Plot[{f[eps, NormalDistribution[1.3, 0.5]],
g[eps, NormalDistribution[1.3, 0.5]]}, {eps, 0.1, 6},
PlotLabels -> {"f(eps, loi)", "g(eps, loi)"}]
```

Out[4]:



```
In[5]: Plot[{f[eps, GammaDistribution[2, 0.5]],
g[eps, GammaDistribution[2, 0.5]]}, {eps, 0.1, 6},
PlotLabels -> {"f(eps, loi)", "g(eps, loi)"}]
```

Out[5]:



### Exercice 187

Soit  $X$  une *var* suivant la loi de Cauchy tronquée  $\mathcal{C}_{]-\infty, 0.7]}(0, 1)$  :

$$f(x) = c \frac{1}{1+x^2}, \quad x \in ]-\infty, 0.7],$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\int_{-\infty}^{\infty} f(x) dx = 1$ .

Utiliser *Mathematica* pour

- créer la loi de  $X$  avec la commande `TruncatedDistribution`,


- tracer sur un même graphique le graphe de  $f$  et celui d'une densité associée à la loi de Cauchy  $\mathcal{C}(0,1)$ ,
- déterminer la fonction de répartition de  $X$ ,
- déterminer la médiane de  $X$ , *i.e.* le réel  $x$  tel que  $F(x) = 0.5$ ,
- déterminer le réel  $x$  tel que  $F(x) = 0.67$ .

Commandes : `CauchyDistribution`, `TruncatedDistribution`, `Plot`, `PDF`, `PlotLegends`, `CDF`, `Median`, `Quantile`.

**Solution 187.** On propose :

```
In[1]: loi = CauchyDistribution[0, 1];
In[2]: loi2 = TruncatedDistribution[{-∞, 0.7}, loi];
In[3]: Plot[{PDF[loi, x], PDF[loi2, x]}, {x, -6, 6}, PlotLegends → {"Cauchy",
    "Cauchy tronquée"}]
```

Out[3]:



```
In[4]: CDF[loi2, x]
```

```
Out[4]: { 1.44009 (1/2 + ArcTan[x]/π)  x ≤ 0.7
         0                               True
```

```
In[5]: Median[loi2]
```

```
Out[5]: -0.520656
```

```
In[6]: Quantile[loi2, 0.67]
```

```
Out[6]: -0.109612
```

**Exercice 188**

Comprendre les enjeux des commandes Mathematica suivantes :





```
Manipulate[Plot[PDF[LogNormalDistribution[0,  $\sigma$ ], x], {x, 0, 5}], { $\sigma$ , 0.3, 3}]
```

Commandes : Manipulate, Plot, PDF, LogNormalDistribution,  $\sigma$ .

**Solution 188.** Cela renvoie un graphique qui représente une densité associée à la loi log-normale  $\mathcal{LN}(0, \sigma)$  (notation française usuelle :  $\mathcal{LN}(0, \sigma^2)$ ) :

$$f(x) = \begin{cases} \frac{1}{x\sqrt{2\pi\sigma^2}} e^{-\frac{(\ln(x))^2}{2\sigma^2}} & \text{si } x > 0, \\ 0 & \text{sinon.} \end{cases}$$

Dans ce graphique, on a un curseur défilant qui permet de choisir plusieurs valeurs de  $\sigma$  :





## 10 Couples de var à densité

### Exercice 189

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y) = \begin{cases} b(xe^y + y^2) & \text{si } (x, y) \in [0, 2] \times [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour déterminer  $b$ .

Commandes : `Integrate`, `Exp`, `N[%]`.

**Solution 189.** On propose :

```
In[1]: b = 1 / Integrate[x Exp[y] + y^2, {x, 0, 2}, {y, 0, 1}]
```

```
Out[1]:  $\frac{1}{-\frac{4}{3} + 2e}$ 
```

```
In[2]: N[%]
```

```
Out[2]: 0.24371
```

### Exercice 190

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y) = \begin{cases} \frac{b}{xy} & \text{si } (x, y) \in [1, e]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour déterminer  $b$ .

Commandes : `Integrate`, `E`.

**Solution 190.** On propose :

```
In[1]: b = 1 / Integrate[1 / (x y), {x, 1, E}, {y, 1, E}]
```

```
Out[1]: 1
```



**Exercice 191**

Utiliser *Mathematica* pour vérifier que la fonction  $g$  définie ci-dessous est une densité :

$$g(x, y) = \begin{cases} \frac{1}{3}(2 - xy) & \text{si } (x, y) \in [0, 2] \times [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Commande : `Integrate`.

**Solution 191.** On propose :

```
In[1]: Integrate[1 / 3 (2 - x y), {x, 0, 2}, {y, 0, 1}] == 1
```

```
Out[1]: True
```

**Exercice 192**

Utiliser *Mathematica* pour vérifier que la fonction  $g$  définie ci-dessous est une densité :

$$g(x, y) = \begin{cases} \frac{1}{2} & \text{si } |x| + |y| < 1, \\ 0 & \text{sinon.} \end{cases}$$

Commandes : `Integrate`, `Piecewise`, `Abs`, `Boole`.

**Solution 192.** On propose :

```
In[1]: Integrate[Piecewise[{{1 / 2, Abs[x] + Abs[y] < 1}}, {x, -1, 1},
{y, -1, 1}] == 1
```

```
Out[1]: True
```

!! On aurait aussi pu utiliser la commande `Boole` :

```
In[1]: Integrate[1 / 2 Boole[Abs[x] + Abs[y] < 1], {x, -1, 1}, {y, -1, 1}] == 1
```

```
Out[1]: True
```

**Exercice 193**

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y) = \begin{cases} b \cos(x + y) & \text{si } -\frac{\pi}{4} \leq x \leq y \leq \frac{\pi}{4}, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la fonction  $h(x, y) = \frac{1}{b}g(x, y)$ ,
- déterminer  $b$ .

Commandes : `x_`, `:=`, `Piecewise`, `Cos`, `Pi`, `Integrate`.

**Solution 193.** On propose :

```
In[1]: h[x_, y_] := Piecewise[{{Cos[x + y], -Pi / 4 ≤ x ≤ y ≤ Pi / 4}}];
```

```
In[2]: b = 1 / Integrate[h[x, y], {x, -Pi / 4, Pi / 4}, {y, -Pi / 4, Pi / 4}]
```

```
Out[2]: 1
```

### Exercice 194

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y) = \begin{cases} (bx - y)^2 & \text{si } \sqrt{x} + \sqrt{y} \geq 1 \text{ et } \sqrt{1-x} + \sqrt{1-y} \geq 1, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la fonction  $h(x, y, b) = g(x, y)$  (avec  $b$  considéré comme un argument),
- déterminer la valeur numérique du réel  $b$  recherché en utilisant la fonction  $h(x, y, b)$  et la commande `Solve`,
- donner la valeur numérique de  $b$ .

Commandes : `x_`, `:=`, `Piecewise`, `Sqrt`, `&&`, `Solve`, `Integrate`, `N[%]`.

**Solution 194.** On propose :

```
In[1]: h[x_, y_, b_] := Piecewise[{{(b x - y)^2, Sqrt[x] + Sqrt[y] ≥ 1 &&
Sqrt[1 - x] + Sqrt[1 - y] ≥ 1}}];
```

```
In[2]: Solve[Integrate[h[x, y, b], {x, 0, 1}, {y, 0, 1}] == 1 && b > 0, b]
```

```
Out[2]: {{b →  $\frac{1}{22}(15 + \sqrt{2051})$ }}
```

```
In[3]: N[%]
```

```
Out[3]: {{b → 2.74036}}
```

## Exercice 195

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 2e^{-x}e^{-2y} & \text{si } (x, y) \in [0, \infty[^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$  avec la commande `ProductDistribution`,
- calculer  $\mathbb{P}(\{X < 1.5\} \cap \{Y < 2.1\})$ ,
- calculer  $\mathbb{P}(\{X > 1.4\} \cup \{Y < 0.6\})$ ,
- calculer  $\mathbb{P}(X < Y)$ ,
- calculer  $\mathbb{P}(|X - Y| \geq 0.3)$ .

Commandes : `ExponentialDistribution`, `ProductDistribution`, `CDF`, `Probability`, `||`, `≈`, `Abs`.

**Solution 195.** On propose :

```
In[1]: loi1 = ExponentialDistribution[1];
```

```
In[2]: loi2 = ExponentialDistribution[2];
```

```
In[3]: loi3 = ProductDistribution[loi1, loi2];
```

```
In[4]: CDF[loi3, {1.5, 2.1}]
```

```
Out[4]: 0.76522
```

```
In[5]: Probability[x > 1.4 || y < 0.6, {x, y} ≈ loi3]
```

```
Out[5]: 0.773079
```

```
In[6]: Probability[x < y, {x, y} ≈ loi3]
```

```
Out[6]: 1/3
```

```
In[7]: Probability[Abs[x - y] ≥ 0.3, {x, y} ≈ loi3]
```

```
Out[7]: 0.676816
```

## Exercice 196

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} c(x(2 - y) + y) & \text{si } (x, y) \in [0, 1] \times [1, 2], \\ 0 & \text{sinon,} \end{cases}$$

où  $c$  désigne la constante de normalisation, *i.e.*  $c$  vérifie  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ .

Utiliser Mathematica pour

- calculer  $c$ ,
- créer la loi de  $(X, Y)$  en utilisant la valeur numérique de  $c$  dans la définition de  $f$ ,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ .

Commandes : `Integrate`, `ProbabilityDistribution`, `PDF`, `MarginalDistribution`.

**Solution 196.** On propose :

In[1]: `c = 1 / Integrate[x (2 - y) + y, {x, 0, 1}, {y, 1, 2}]`

Out[1]:  $\frac{4}{7}$

In[2]: `loi = ProbabilityDistribution[c (x (2 - y) + y) , {x, 0, 1}, {y, 1, 2}];`

In[3]: `PDF[MarginalDistribution[loi, 1], x]`

Out[3]: 
$$\begin{cases} \frac{2(3+x)}{7} & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[4]: `PDF[MarginalDistribution[loi, 2], y]`

Out[4]: 
$$\begin{cases} \frac{2(2+y)}{7} & 1 < x < 2 \\ 0 & \text{True} \end{cases}$$

### Exercice 197

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 12xy(1-x) & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes.

Commandes : `Integrate`, `ProbabilityDistribution`, `PDF`, `MarginalDistribution`, `CDF`.

**Solution 197.** On propose :

```
In[1]: Integrate[12 x y (1 - x), {x, 0, 1}, {y, 0, 1}] == 1
Out[1]: True
In[2]: loi = ProbabilityDistribution[12 x y (1 - x), {x, 0, 1}, {y, 0, 1}];
In[3]: PDF[MarginalDistribution[loi, 1], x]
Out[3]: 
$$\begin{cases} -6(-x + x^2) & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[4]: PDF[MarginalDistribution[loi, 2], y]
Out[4]: 
$$\begin{cases} 2y & 0 < y < 1 \\ 0 & \text{True} \end{cases}$$

In[5]: CDF[loi, {0.2, 0.2}] == CDF[MarginalDistribution[loi, 1], 0.2]
CDF[MarginalDistribution[loi, 2], 0.2]
Out[5]: False
```

!! Ainsi, on a montré que  $\mathbb{P}(\{X \leq 0.2\} \cap \{Y \leq 0.2\}) \neq \mathbb{P}(X \leq 0.2)\mathbb{P}(Y \leq 0.2)$ , ce qui entraîne la non indépendance de  $X$  et  $Y$  (les valeurs  $x = 0.2$  et  $y = 0.2$  ont été choisies au hasard dans  $(X, Y)(\Omega) = [0, 1]^2$ ).

### Exercice 198

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{3}{8}(x^2 + y^2) & \text{si } (x, y) \in [-1, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$  avec la commande `ProbabilityDistribution`,
- tracer le graphe de  $f$  avec la commande `Plot3D` et l'option :  
`ColorFunction`  $\rightarrow$  `Function[{x, y, z}, Hue[z]]`,
- déterminer la fonction de répartition de  $(X, Y)$ , *i.e.*  $F(x, y) = \mathbb{P}(\{X \leq x\} \cap \{Y \leq y\})$ ,  $(x, y) \in \mathbb{R}^2$ ,
- tracer le graphe de  $F(x, y)$  pour  $(x, y) \in [-1, 1]^2$ ,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes.


Commandes : `ProbabilityDistribution`, `Plot3D`, `PDF`, `ColorFunction`, `Hue`, `CDF`, `MarginalDistribution`.

**Solution 198.** On propose :

```
In[1]: loi = ProbabilityDistribution[3 / 8 (x^2 + y^2), {x, -1, 1}, {y, -1, 1}];
```

```
In[2]: Plot3D[PDF[loi, {x, y}], {x, -1, 1}, {y, -1, 1}, ColorFunction →  
Function[{x, y, z}, Hue[z]]]
```

Out [2]:




```
In[3]: CDF[loi, {x, y}]
```

```
Out [3]: { 1                x ≥ 1 && y ≥ 1  
          1/4(2 + x + x^3)  -1 < x < 1 && y ≥ 1  
          1/4(2 + y + y^3)  x ≥ 1 && -1 < y < 1  
          1/8(2 + x + x^3 + y + x^3y + y^3 + xy^3) -1 < x < 1 && -1 < y < 1  
          0                True
```

```
In[4]: Plot3D[CDF[loi, {x, y}], {x, -1, 1}, {y, -1, 1}, ColorFunction →  
Function[{x, y, z}, Hue[z]]]
```

Out [4] :



In[5]: PDF[MarginalDistribution[loi, 1], x]

Out [5]: 
$$\begin{cases} \frac{1}{4}(1 + 3x^2) & -1 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[6]: PDF[MarginalDistribution[loi, 2], y]

Out [6]: 
$$\begin{cases} \frac{1}{4}(1 + 3y^2) & -1 < y < 1 \\ 0 & \text{True} \end{cases}$$

In[7]: CDF[loi, {0.2, 0.2}] == CDF[MarginalDistribution[loi, 1], 0.2]

CDF[MarginalDistribution[loi, 2], 0.2]

Out [7]: False

### Exercice 199

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{1}{2}(3x + y) & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$ ,
- déterminer la fonction de répartition de  $(X, Y)$ ,

- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer la matrice de covariance de  $(X, Y)$ ,
- créer la loi de  $X + Y$ ,
- vérifier que  $\mathbb{E}(X + Y) = \mathbb{E}(X) + \mathbb{E}(Y)$ ,
- vérifier que  $\mathbb{V}(X + Y) = \mathbb{V}(X) + \mathbb{V}(Y) + 2C(X, Y)$ .

Commandes : ProbabilityDistribution, CDF, Mean, Covariance, //MatrixForm, TransformedDistribution, ≈, Variance, Covariance, //Simplify.

**Solution 199.** On propose :

```
In[1]: loi = ProbabilityDistribution[1 / 2 (3 x + y), {x, 0, 1}, {y, 0, 1}];
```

```
In[2]: CDF[loi, {x, y}]
```

```
Out[2]: {
  1           x ≥ 1 && y ≥ 1
  1/4 x(1 + 3x) 0 < x < 1 && y ≥ 1
  1/4 y(3 + y)  x ≥ 1 && 0 < y < 1
  1/4 xy(3x + y) 0 < x < 1 && 0 < y < 1
  0           True
```

```
In[3]: Mean[loi]
```

```
Out[3]: {5/8, 13/24}
```

```
In[4]: Covariance[loi] //MatrixForm
```

```
Out[4]: ( 13/192  -1/47
          1/192  1/576 )
```

```
In[5]: loi2 = TransformedDistribution[x + y, {x, y} ≈ loi];
```

```
In[6]: Mean[loi2] == Mean[loi][[1]] + Mean[loi][[2]]
```

```
Out[6]: True
```

```
In[7]: Variance[loi2] == Covariance[loi, 1, 1] + Covariance[loi, 2, 2] +
2 Covariance[loi, 1, 2] //Simplify
```

```
Out[7]: True
```



## Exercice 200

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 3|x - y| & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- tracer le graphe de  $f$ ,
- créer la loi de  $(X, Y)$ ,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer  $\mathbb{C}(X, Y)$ ,
- déterminer la fonction génératrice des moments de  $(X, Y)$  :  $M(s, t) = \mathbb{E}(e^{sX+tY})$ .

Commandes : `Integrate`, `Abs`, `Plot3D`, `ColorFunction`, `Hue`, `ProbabilityDistribution`, `PDF`, `MarginalDistribution`, `Mean`, `Variance`, `Covariance`, `MomentGeneratingFunction`.


**Solution 200.** On propose :

```
In[1]: Integrate[3 Abs[x - y], {x, 0, 1}, {y, 0, 1}] == 1
```

```
Out[1]: True
```

```
In[2]: Plot3D[3 Abs[x - y], {x, 0, 1}, {y, 0, 1}, ColorFunction ->
Function[{x, y, z}, Hue[z]]]
```

```
Out[2]:
```



```

In[3]: loi = ProbabilityDistribution[3 Abs[x - y], {x, 0, 1}, {y, 0, 1}];
In[4]: PDF[MarginalDistribution[loi, 1], x]
Out[4]: 
$$\begin{cases} \frac{3}{2}(1 - 2x + 2x^2) & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[5]: PDF[MarginalDistribution[loi, 2], y]
Out[5]: 
$$\begin{cases} \frac{3}{2}(1 - 2y + 2y^2) & 0 < y < 1 \\ 0 & \text{True} \end{cases}$$

In[6]: Mean[loi]
Out[6]: 
$$\left\{ \frac{1}{2}, \frac{1}{2} \right\}$$

In[7]: Variance[loi]
Out[7]: 
$$\left\{ \frac{1}{10}, \frac{1}{10} \right\}$$

In[8]: Covariance[loi, 1, 2]
Out[8]: 
$$-\frac{1}{20}$$

In[9]: MomentGeneratingFunction[loi, {s, t}]
Out[9]: 
$$-\frac{1}{s^2 t^2 (s + t)} 3(-(-1 + e^s)(1 + e^t)t^2 + (e^s + e^t)st^2 + s^2(1 - e^{s+t} + e^t(-1 + t) + e^s(1 + t)))$$


```

**Exercice 201**

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 2 & \text{si } (x, y) \in \mathcal{D}, \\ 0 & \text{sinon,} \end{cases}$$

avec  $\mathcal{D} = \{(x, y) \in [0, 1]^2; x \geq y + 0.5 \text{ ou } x \leq y \leq x + 0.5\}$ .

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- créer la loi de  $(X, Y)$  avec la commande `ProbabilityDistribution`,
- calculer  $\mathbb{P}_{\{Y < 0.31\}}(X \geq 0.27)$ ,
- calculer  $\mathbb{P}(\inf(X, Y) \geq 0.45)$ ,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,

◦ calculer la matrice de covariance de  $(X, Y)$ .

Commandes : `x_`, `:=`, `Piecewise`, `||`, `ProbabilityDistribution`, `Probability`, `▷`, `≈`, `Min`, `PDF`, `MarginalDistribution`, `//Simplify`, `Mean`, `Covariance`, `//MatrixForm`, `CDF`.

**Solution 201.** On propose :

```
In[1]: f[x_, y_] := Piecewise[{{2, x ≥ y + 0.5 || x ≤ y ≤ x + 0.5}}];
```

```
In[2]: loi = ProbabilityDistribution[f[x, y], {x, 0, 1}, {y, 0, 1}];
```

```
In[3]: Probability[x ≥ 0.27 ▷ y < 0.31, {x, y} ≈ loi]
```

```
Out[3]: 0.695161
```

```
In[4]: Probability[Min[x, y] ≥ 0.45, {x, y} ≈ loi]
```

```
Out[4]: 0.3025
```

```
In[5]: PDF[MarginalDistribution[loi, 1], x] //Simplify
```

```
Out[5]: { 1. 0. < x < 1.
          0. True
```

!! On reconnaît une densité associée à la loi uniforme  $\mathcal{U}([0, 1])$ .

```
In[6]: PDF[MarginalDistribution[loi, 2], y] //Simplify
```

```
Out[6]: { 1. 0. < y < 1.
          0. True
```

!! De nouveau, on reconnaît une densité associée à la loi uniforme  $\mathcal{U}([0, 1])$ .

```
In[7]: Mean[loi]
```

```
Out[7]: {0.5, 0.5}
```

```
In[8]: Covariance[loi] //MatrixForm
```

```
Out[8]: ( 0.0833333  0.
          0.        0.0833333)
```

```
In[9]: CDF[loi, {0.8, 0.2}] == CDF[MarginalDistribution[loi, 1], 0.8]
```

```
CDF[MarginalDistribution[loi, 2], 0.2]
```

```
Out[9]: False
```

**Exercice 202**

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{15}{2}x(2 - x - y) & \text{si } (x, y) \in [-1, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la loi de  $(X, Y)$ ,
- vérifier l'inégalité suivante (inégalité de Cauchy-Schwarz) :  
 $|\mathbb{E}(XY)| \leq (\mathbb{E}(X^2))^{\frac{1}{2}} (\mathbb{E}(Y^2))^{\frac{1}{2}}$ ,
- vérifier l'inégalité suivante (conséquence de l'inégalité de Cauchy-Schwarz) :  
 $|\mathbb{C}(X, Y)| \leq \sigma(X)\sigma(Y)$ .

Commandes : `ProbabilityDistribution`, `Abs`, `Moment`, `Sqrt`, `Covariance`.

**Solution 202.** On propose :

```
In[1]: loi = ProbabilityDistribution[15 / 2 x (2 - x - y), {x, 0, 1}, {y, 0, 1}];
```

```
In[2]: Abs[Moment[loi, {1, 1}]] ≤ Sqrt[Moment[loi, {2, 0}]]
      Sqrt[Moment[loi, {0, 2}]]
```

```
Out[2]: True
```

```
In[3]: Abs[Covariance[loi, 1, 2]] ≤ Sqrt[Covariance[loi, 1, 1]]
      Sqrt[Covariance[loi, 2, 2]]
```

```
Out[3]: True
```

**Exercice 203**

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{3}{4}(1 - \max(|x|, |y|)) & \text{si } (x, y) \in [-1, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- créer la loi de  $(X, Y)$ ,
- tracer le graphe de  $f$  avec les couleurs de **Aquamarine**,
- déterminer une densité de  $X$  (on utilisera la commande `Simplify` pour un résultat concis),

- déterminer une densité de  $Y$  (on utilisera la commande `Simplify` pour un résultat concis),
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer la matrice de covariance  $\Sigma$  de  $(X, Y)$ ,
- extraire de  $\Sigma$  la covariance  $\mathbb{C}(X, Y)$ ,
- vérifier que  $X$  et  $Y$  ne sont pas indépendantes.

Commandes : `Integrate`, `Max`, `Abs`, `ProbabilityDistribution`, `Plot3D`, `PDF`, `ColorFunction`, `MarginalDistribution`, `//Simplify`, `Mean`, `Covariance`, `//MatrixForm`, `CDF`.

**Solution 203.** On propose :


```
In[1]: Integrate[3 / 4 (1 - Max[Abs[x], Abs[y]]), {x, -1, 1}, {y, -1, 1}] == 1
```

```
Out[1]: True
```

```
In[2]: loi = ProbabilityDistribution[3 / 4 (1 - Max[Abs[x], Abs[y]]),
      {x, -1, 1}, {y, -1, 1}];
```

```
In[3]: Plot3D[PDF[loi, {x, y}], {x, -1, 1}, {y, -1, 1}, ColorFunction ->
      "Aquamarine"]
```

```
Out[3]:
```



```
In[4]: PDF[MarginalDistribution[loi, 1], x] //Simplify
```

```
Out[4]: { -3/4(-1 + x^2)  -1 < x < 1
          0                True
```

```

In[5]: PDF[MarginalDistribution[loi, 2], y] //Simplify
Out[5]: 
$$\begin{cases} -\frac{3}{4}(-1+y^2) & -1 < y < 1 \\ 0 & \text{True} \end{cases}$$

In[6]: Mean[loi]
Out[6]: {0, 0}
In[7]: Covariance[loi] //MatrixForm
Out[7]: 
$$\begin{pmatrix} \frac{1}{5} & 0 \\ 0 & \frac{1}{5} \end{pmatrix}$$

In[8]: Covariance[loi, 1, 2]
Out[8]: 0
In[9]: CDF[loi, {0.2, 0.2}] == CDF[MarginalDistribution[loi, 1], 0.2]
CDF[MarginalDistribution[loi, 2], 0.2]
Out[9]: False

```

### Exercice 204

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{896}{127\pi(1+x^2+y^2)^8} & \text{si } x^2 + y^2 \leq 1, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- tracer le graphe de  $f$  avec les couleurs de `IslandColors` et l'option : `PlotRange → Full`,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer la matrice de covariance  $\Sigma$  de  $(X, Y)$ ,
- extraire de  $\Sigma$  la covariance  $\mathbb{C}(X, Y)$ ,
- vérifier que  $\mathbb{P}_{\{Y>0.2\}}(X < 0.5) \neq \mathbb{P}(X < 0.5)$ , ce qui entraîne la non indépendance de  $X$  et  $Y$ .

Commandes : `x_`, `:=`, `Piecewise`, `Pi`, `Integrate`, `Plot3D`, `ColorFunction`, `PlotRange`, `ProbabilityDistribution`, `Mean`, `Covariance`, `//MatrixForm`, `Probability`, `▷`, `≈`.

**Solution 204.** On propose :


```
In[1]: f[x_, y_] := Piecewise[{{896 / (127 Pi (1 + x^2 + y^2)^8), x^2 + y^2 ≤ 1}}];
```

```
In[2]: Integrate[f[x, y], {x, -1, 1}, {y, -1, 1}] == 1
```

```
Out[2]: True
```

```
In[3]: Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, ColorFunction -> "IslandColors",
PlotRange -> Full]
```

```
Out[3]:
```



```
In[4]: loi = ProbabilityDistribution[f[x, y], {x, -1, 1}, {y, -1, 1}];
```

```
In[5]: Mean[loi]
```

```
Out[5]: {0, 0}
```

```
In[6]: Covariance[loi] //MatrixForm
```

```
Out[6]: 
$$\begin{pmatrix} \frac{10}{127} & 0 \\ 0 & \frac{10}{127} \end{pmatrix}$$

```

```
In[7]: Probability[x < 0.5 & y > 0.2, {x, y} ≈ loi] != Probability[x < 0.5,
{x, y} ≈ loi]
```

```
Out[7]: True
```

### Exercice 205

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{1}{4}(1 + xy(x^2 - y^2)) & \text{si } (x, y) \in [-1, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$ ,
- tracer le graphe de  $f$  avec les couleurs de `SolarColors`,
- déterminer la densité de  $X$ ,
- déterminer la densité de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer  $\rho(X, Y)$ .
- déterminer la fonction caractéristique de  $X + Y$  :  $\varphi_{X+Y}(t) = \mathbb{E}(e^{it(X+Y)})$ ,
- vérifier que  $\varphi_{X+Y}(t) = \varphi_X(t)\varphi_Y(t)$  pour tout  $t \in \mathbb{R}$ ,
- vérifier que  $X$  et  $Y$  ne sont (pourtant) pas indépendantes.


Commandes : `ProbabilityDistribution`, `Plot3D`, `PDF`, `ColorFunction`, `MarginalDistribution`, `Mean`, `Variance`, `Correlation`, `TransformedDistribution`, `CharacteristicFunction`, `CDF`.

**Solution 205.** On propose :

```
In[1]: loi = ProbabilityDistribution[1 / 4 (1 + x y (x^2 - y^2)), {x, -1, 1},
      {y, -1, 1}];
```

```
In[2]: Plot3D[PDF[loi, {x, y}], {x, -1, 1}, {y, -1, 1}, ColorFunction ->
      "SolarColors"]
```

Out [2]:



```
In[3]: PDF[MarginalDistribution[loi, 1], x]
```

```
Out [3]: { 1/2  -1 < x < 1
          0    True
```



!! On reconnaît une densité associée à la loi uniforme  $\mathcal{U}([-1, 1])$ .

```
In[4]: PDF[MarginalDistribution[loi, 2], y]
```

```
Out[4]: { 1/2  -1 < y < 1
          0    True
```

!! On reconnaît une densité associée à la loi uniforme  $\mathcal{U}([-1, 1])$ .

```
In[5]: Mean[loi]
```

```
Out[5]: {0, 0}
```

```
In[6]: Variance[loi]
```

```
Out[6]: {1/3, 1/3}
```

```
In[7]: Correlation[loi, 1, 2]
```

```
Out[7]: 0
```

```
In[8]: loi2 = TransformedDistribution[x + y, {x, y} ≈ loi];
```

```
In[9]: CharacteristicFunction[loi2, t]
```

```
Out[9]: Sin[t]^2 / t^2
```

```
In[10]: % == CharacteristicFunction[MarginalDistribution[loi, 1], t]
```

```
CharacteristicFunction[MarginalDistribution[loi, 2], t]
```

```
Out[10]: True
```

```
In[11]: CDF[loi, {0.4, 0.2}] == CDF[MarginalDistribution[loi, 1], 0.4]
```

```
CDF[MarginalDistribution[loi, 2], 0.2]
```

```
Out[11]: False
```

!! On a  $\mathbb{P}(\{X \leq 0.4\} \cap \{Y \leq 0.2\}) \neq \mathbb{P}(X \leq 0.4)\mathbb{P}(Y \leq 0.2)$ , donc  $X$  et  $Y$  ne sont pas indépendantes.

### Exercice 206

Soient  $\alpha \geq 1$  et  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{\alpha}{\log(\alpha + 1)} x^{\alpha y} & \text{si } (x, y) \in ]0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $f$  avec  $\alpha$  en argument,

- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- tracer le graphe de  $f$  pour plusieurs valeurs de  $a$  avec la commande `manipulate`, les couleurs de `BlueGreenYellow` et l'option : `PlotRange → Full`,
- déterminer la densité de  $X$ ,
- déterminer la densité de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer simultanément  $\mathbb{V}(X)$  et  $\mathbb{V}(Y)$ ,
- calculer  $\rho(X, Y)$ .

Commandes : `x_`, `α`, `:=`, `Piecewise`, `Log`, `Integrate`, `Assumptions`, `Manipulate`, `Plot3D`, `ColorFunction`, `PlotRange`, `ProbabilityDistribution`, `PDF`, `MarginalDistribution`, `Mean`, `Variance`, `Correlation`.

**Solution 206.** On propose :



```
In[1]: f[x_, y_, α_] := Piecewise[{{α / Log[α + 1] x^(α y), 0 < x ≤ 1 &&
0 < y ≤ 1}}
```



```
In[2]: Integrate[f[x, y, α], {x, 0, 1}, {y, 0, 1}, Assumptions → α ≥ 1] == 1
```

Out[2]: True

```
In[3]: Manipulate[Plot3D[f[x, y, α], {x, 0, 1}, {y, 0, 1}, ColorFunction →
"BlueGreenYellow", PlotRange → Full], {a, 1, 30}, {a, Range[30]}
```

Out[3]:





```
In[4]: loi = ProbabilityDistribution[f[x, y,  $\alpha$ ], {x, 0, 1}, {y, 0, 1},
Assumptions  $\rightarrow \alpha \geq 1$ ];
```

```
In[5]: PDF[MarginalDistribution[loi, 1], x]
```

```
Out[5]: 
$$\begin{cases} \frac{-1 + x^\alpha}{\text{Log}[x]\text{Log}[1 + \alpha]} & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

```

```
In[6]: PDF[MarginalDistribution[loi, 2], y]
```

```
Out[6]: 
$$\begin{cases} \frac{\alpha}{(1 + y\alpha)\text{Log}[1 + \alpha]} & 0 < y < 1 \\ 0 & \text{True} \end{cases}$$

```

```
In[7]: Mean[loi]
```

```
Out[7]: 
$$\left\{ \frac{\text{Log}\left[1 + \frac{\alpha}{2}\right]}{\text{Log}[1 + \alpha]}, \frac{\alpha - \text{Log}[1 + \alpha]}{\alpha \text{Log}[1 + \alpha]} \right\}$$

```

```
In[8]: Variance[loi]
```

```
Out[8]: 
$$\left\{ \frac{-\text{Log}\left[\frac{2}{2+\alpha}\right]^2 - \text{Log}[1 + \alpha]\text{Log}\left[\frac{3}{3+\alpha}\right]}{\text{Log}[1 + \alpha]^2}, \frac{-2\alpha + 2\text{Log}[1 + \alpha] + \alpha \text{Log}[1 + \alpha]}{2\alpha \text{Log}[1 + \alpha]^2} \right\}$$

```

```
In[9]: Correlation[loi, 1, 2]
```

```
Out[9]: 
$$\frac{\alpha \text{Log}[4] + 2\alpha \text{Log}[1 + \alpha] + \text{Log}[4]\text{Log}[1 + \alpha] - 2\alpha \text{Log}[2 + \alpha] - 2\text{Log}[1 + \alpha]\text{Log}[2 + \alpha]}{\sqrt{2\alpha \text{Log}[1 + \alpha]^2} \sqrt{\frac{-2\alpha + 2\text{Log}[1 + \alpha] + \alpha \text{Log}[1 + \alpha]}{\alpha \text{Log}[1 + \alpha]^2}} \sqrt{\frac{-\text{Log}\left[\frac{2}{2+\alpha}\right]^2 - \text{Log}[1 + \alpha]\text{Log}\left[\frac{3}{3+\alpha}\right]}{\text{Log}[1 + \alpha]^2}}$$

```

## Exercice 207

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{\pi^2}{\pi - 2} \cos\left(\frac{\pi}{2}(x + 2y)\right) & \text{si } x > 0, y > 0 \text{ et } x + 2y \leq 1, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- vérifier que  $\int_{-\infty}^{\infty} \int_{-\infty}^{\infty} f(x, y) dx dy = 1$ ,
- tracer le graphe de  $f(x, y)$  pour  $(x, y) \in [0, 1] \times [0, \frac{1}{2}]$  avec les couleurs de *DeepSeaColors*,
- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ ,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer la matrice de covariance de  $(X, Y)$ .

Commandes : `x_`, `:=`, `Piecewise`, `Pi`, `Cos`, `&&`, `Integrate`, `Plot3D`, `ColorFunction`, `ProbabilityDistribution`, `PDF`, `MarginalDistribution`, `Mean`, `Covariance`, `//MatrixForm`.

**Solution 207.** On propose :


```
In[1]: f[x_, y_] := Piecewise[{{Pi^2 / (Pi - 2) Cos[Pi / 2 (x + 2 y)],
x > 0 && y > 0 && x + 2 y < 1}}];
```

```
In[2]: Integrate[f[x, y], {x, 0, 1}, {y, 0, 1/2}] == 1
```

```
Out[2]: True
```

```
In[3]: Plot3D[f[x, y], {x, 0, 1}, {y, 0, 1/2}, ColorFunction -> "DeepSeaColors"]
```

```
Out[3]:
```



In[4]: `loi = ProbabilityDistribution[f[x, y], {x, 0, 1}, {y, 0, 1}];`

In[5]: `PDF[MarginalDistribution[loi, 1], x]`

$$\text{Out [5]: } \begin{cases} -\frac{\pi(-1 + \text{Sin}[\frac{\pi x}{2}])}{-2 + \pi} & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[6]: `PDF[MarginalDistribution[loi, 2], y]`

$$\text{Out [6]: } \begin{cases} -\frac{2\pi(-1 + \text{Sin}[\pi y])}{-2 + \pi} & 0 < y < \frac{1}{2} \\ 0 & \text{True} \end{cases}$$

In[7]: `Mean[loi]`

$$\text{Out [7]: } \left\{ \frac{-8 + \pi^2}{2(-2 + \pi)\pi}, \frac{-8 + \pi^2}{4(-2 + \pi)\pi} \right\}$$

In[8]: `Covariance[loi] //MatrixForm`

$$\text{Out [8]: } \begin{pmatrix} \frac{(24 - 12\pi + \pi^2)(-24 + 4\pi + \pi^2)}{12(-2 + \pi)^2\pi^2} & -\frac{384 - 192\pi + 4\pi^3 + \pi^4}{24(-2 + \pi)^2\pi^2} \\ -\frac{384 - 192\pi + 4\pi^3 + \pi^4}{24(-2 + \pi)^2\pi^2} & \frac{(24 - 12\pi + \pi^2)(-24 + 4\pi + \pi^2)}{48(-2 + \pi)^2\pi^2} \end{pmatrix}$$

### Exercice 208

Soit  $(X, Y)$  un couple de *var* à densité de fonction de répartition :

$$F(x, y) = \begin{cases} 1 - \frac{1}{1+x} - \frac{1}{1+y} + \frac{1}{1+x+y} & \text{si } (x, y) \in [0, \infty[^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la fonction  $F$ ,
- créer la loi de  $(X, Y)$ ,
- déterminer une densité  $f$  de  $(X, Y)$  avec la commande `PDF`,
- vérifier l'égalité :

$$f(x, y) = \frac{\partial}{\partial x \partial y} F(x, y).$$

- déterminer une densité de  $X$ ,
- déterminer une densité de  $Y$ .

Commandes : `x_`, `:=`, `Piecewise`, `ProbabilityDistribution`, `CDF`, `∞`, `PDF`, `D`, `//Simplify`, `MarginalDistribution`.

**Solution 208.** On propose :

```
In[1]: F[x_, y_] := Piecewise[{{1 - 1 / (1 + x) - 1 / (1 + y)
+ 1 / (1 + x + y), x ≥ 0 && y ≥ 0}}];

In[2]: loi = ProbabilityDistribution[{"CDF", F[x, y]}, {x, 0, ∞}, {y, 0, ∞}];

In[3]: PDF[loi, {x, y}]

Out[3]: 
$$\begin{cases} \frac{2}{(1+x+y)^3} & x > 0 \ \&\& \ y > 0 \\ 0 & \text{True} \end{cases}$$


In[4]: PDF[loi, {x, y}] == D[CDF[loi, {x, y}], x, y] //Simplify

Out[4]: True

In[5]: PDF[MarginalDistribution[loi, 1], x]

Out[5]: 
$$\begin{cases} \frac{1}{(1+x)^2} & x > 0 \\ 0 & \text{True} \end{cases}$$


In[6]: PDF[MarginalDistribution[loi, 2], y]

Out[6]: 
$$\begin{cases} \frac{1}{(1+y)^2} & y > 0 \\ 0 & \text{True} \end{cases}$$

```

### Exercice 209

Soient  $\sigma > 0$ , et  $X$  et  $Y$  deux *var* à densité indépendantes telles que :

◊  $X$  suit la loi normale  $\mathcal{N}(0, \sigma)$ , *i.e.* de densité :

$$f_X(x) = \frac{1}{\sqrt{2\pi\sigma^2}} e^{-\frac{x^2}{2\sigma^2}}, \quad x \in \mathbb{R}.$$

◊  $Y$  suit la loi de Cauchy  $\mathcal{C}(0, 1)$ , *i.e.* de densité :

$$f_Y(y) = \frac{1}{\pi(1+y^2)}, \quad y \in \mathbb{R}.$$

Utiliser **Mathematica** pour

- déterminer une densité  $f$  de  $(X, Y)$  avec la commande **ProductDistribution**,
- tracer le graphe de  $f$  pour plusieurs valeurs de  $\sigma$  avec la commande **Manipulate**,
- déterminer la fonction caractéristique de  $(X, Y)$  :  $\varphi_{(X,Y)}(s, t) = \mathbb{E}(e^{isX+itY})$ .

Commandes : **NormalDistribution**,  $\sigma$ , **CauchyDistribution**, **ProductDistribution**, **PDF**, **Manipulate**, **Plot3D**, **PDF**, **ColorFunction**, **PlotRange**, **CharacteristicFunction**.

**Solution 209.** On propose :

In[1]: `loi1 = NormalDistribution[0,  $\sigma$ ];`

In[2]: `loi2 = CauchyDistribution[0,1];`





In[3]: `loi3 = ProductDistribution[loi1, loi2];`

In[4]: `PDF[loi3, {x, y}]`

Out[4]: 
$$\frac{e^{-\frac{x^2}{2\sigma^2}}}{\sqrt{2\pi^{3/2}}(1+y^2)\sigma}$$

In[5]: `Manipulate[Plot3D[PDF[ProductDistribution[NormalDistribution[0,  $\sigma$ ],  
CauchyDistribution[0, 1]], {x, y}], {x, -5, 5}, {y, -5, 5},  
ColorFunction  $\rightarrow$  "Rainbow", PlotRange  $\rightarrow$  Full], { $\sigma$ , 0.1, 2}]`

Out[5]:



In[6]: `CharacteristicFunction[loi3, {s, t}]`

Out[6]:  $e^{-\frac{1}{2}s^2\sigma^2+t\text{Sign}[t]}$

### Exercice 210

Soient  $\lambda > 0$ , et  $X$  et  $Y$  deux *var* iid suivant chacune la loi exponentielle  $\mathcal{E}(\lambda)$ , i.e. de densité :

$$f(x) = \begin{cases} \lambda e^{-\lambda x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- déterminer une densité  $f$  de  $(X, Y)$  avec la commande `ProductDistribution`,
- calculer  $\mathbb{P}(|Y - X| \geq 1)$ ,
- calculer  $\mathbb{P}(Y \geq |1 - X|)$ ,
- calculer  $\mathbb{P}(2Y \geq (1 - X)^2)$ ,
- calculer  $\mathbb{E}(X^2 - Y^3 + 2XY^2 - X^2Y + 5)$ ,
- calculer  $\mathbb{E}(e^{-(X-Y)^2})$  (on verra apparaître la fonction  $\text{Erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt$ ,  $x \geq 0$ ).

Commandes : `ExponentialDistribution`,  $\lambda$ , `ProductDistribution`, `PDF`, `Probability`, `Abs`, `≈`, `Expectation`, `Exp`.

**Solution 210.** On propose :

In[1]: `loi = ExponentialDistribution[λ];`

In[2]: `loi2 = ProductDistribution[{loi, 2}];`

In[3]: `PDF[loi2, {x, y}]`

Out[3]: 
$$\begin{cases} e^{-x\lambda - y\lambda} \lambda^2 & x \geq 0 \ \&\& \ y \geq 0 \\ 0 & \text{True} \end{cases}$$

In[4]: `Probability[Abs[y - x] ≥ 1, {x, y} ≈ loi2]`

Out[4]:  $e^{-\lambda}$

In[5]: `Probability[y ≥ Abs[1 - x], {x, y} ≈ loi2]`

Out[5]:  $\frac{1}{2}e^{-\lambda}(1 + 2\lambda)$

In[6]: `Probability[2 y ≥ (1 - x)^2, {x, y} ≈ loi2]`

Out[6]:  $e^{-\lambda/2} \sqrt{\frac{\pi}{2}} \sqrt{\lambda}$



In[7]: `Expectation[x^2 - y^3 + 2 x y^2 - x^2 y + 5, {x, y} ≈ loi2]`

Out[7]: 
$$\frac{-4 + 2\lambda + 5\lambda^3}{\lambda^3}$$

In[8]: `Expectation[Exp[-(x - y)^2], {x, y} ≈ loi2]`

Out[8]: 
$$-\frac{1}{2} e^{\frac{\lambda^2}{4}} \sqrt{\pi\lambda} \left( -1 + \operatorname{Erf} \left[ \frac{\lambda}{2} \right] \right)$$

### Exercice 211

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{1}{12} & \text{si } (x, y) \in [-1, 1] \times [1, 7], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$  avec la commande `UniformDistribution`,
- déterminer une densité de  $S = XY$ ,
- déterminer une densité de  $T = (X + Y)^2$ ,
- déterminer une densité de  $U = \frac{1}{X - Y}$ ,
- déterminer la loi de  $V = \begin{cases} 1 & \text{si } Y > 4X, \\ 0 & \text{sinon,} \end{cases}$
- déterminer la loi de  $W = \frac{8X - Y}{|8X - Y|}$ .

Commandes : `UniformDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Boole`, `Abs`, `DiracDelta`.

**Solution 211.** On propose :

In[1]: `loi = UniformDistribution[{{-1, 1}, {1, 7}}];`

In[2]: `loi2 = TransformedDistribution[x y, {x, y} ≈ loi];`

In[3]: `PDF[loi2, x]`

Out[3]: 
$$\begin{cases} \frac{\operatorname{Log}[7]}{12} & -1 < x \leq 1 \\ \frac{1}{12} \operatorname{Log} \left[ -\frac{7}{x} \right] & -7 < x \leq 1 \\ \frac{1}{12} \operatorname{Log} \left[ \frac{7}{x} \right] & 1 < x < 7 \\ 0 & \text{True} \end{cases}$$

```
In[4]: loi3 = TransformedDistribution[(x + y)^2, {x, y} ≈ loi];
```

```
In[5]: PDF[loi3, x]
```

$$\text{Out [5]: } \begin{cases} \frac{1}{24} & 0 < x \leq 4 \\ \text{Indeterminate} & x == 0 \\ -\frac{1}{24} + \frac{1}{3\sqrt{x}} & 36 < x < 64 \\ \frac{1}{12\sqrt{x}} & 4 < x \leq 36 \\ 0 & \text{True} \end{cases}$$

```
In[6]: loi4 = TransformedDistribution[1 / (x - y), {x, y} ≈ loi];
```

```
In[7]: PDF[loi4, x]
```

$$\text{Out [7]: } \begin{cases} -\frac{1}{12x^3} & x \leq -\frac{1}{2} \\ \frac{1}{6x^2} & -\frac{1}{2} < x \leq -\frac{1}{6} \\ \frac{1+8x}{12x^3} & -\frac{1}{6} < x < -\frac{1}{8} \\ 0 & \text{True} \end{cases}$$

```
In[8]: loi5 = TransformedDistribution[Boole[y > 4 x], {x, y} ≈ loi];
```

```
In[9]: PDF[loi5, x]
```

$$\text{Out [9]: } \frac{29}{32}\text{DiracDelta}[-1 + x] + \frac{3\text{DiracDelta}[x]}{32}$$

!! Ainsi, la loi de  $V$  est donnée par :  $\mathbb{P}(V = 0) = \frac{3}{32}$ ,  $\mathbb{P}(V = 1) = \frac{29}{32}$ .

On reconnaît la loi de Bernoulli  $\mathcal{B}(\frac{29}{32})$ .

```
In[10]: loi6 = TransformedDistribution[(8 x - y) / Abs[8 x - y], {x, y} ≈ loi];
```

```
In[11]: PDF[loi6, x]
```

$$\text{Out [11]: } \frac{1}{4}\text{DiracDelta}[-1 + x] + \frac{3}{4}\text{DiracDelta}[1 + x]$$

!! Ainsi, la loi de  $W$  est donnée par :  $\mathbb{P}(W = -1) = \frac{3}{4}$ ,  $\mathbb{P}(W = 1) = \frac{1}{4}$ .

On reconnaît la loi de Bernoulli  $\mathcal{B}(\frac{1}{4})$ .

## Exercice 212

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 2 & \text{si } x \geq 0, y \geq 0 \text{ et } x + y \leq 1, \\ 0 & \text{sinon.} \end{cases}$$

On pose  $Z = X^2 - Y$ .

Utiliser **Mathematica** pour

- créer la fonction  $f$ ,
- créer la loi de  $(X, Y)$ ,
- déterminer une densité  $f_Z$  de  $Z$ ,
- tracer le graphe de  $f_Z$ ,
- déterminer une densité  $f_W$  de  $W = |Z|$ ,
- tracer le graphe de  $f_W$ .

Commandes : `x_`, `:=`, `Piecewise`, `&&`, `ProbabilityDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Plot`, `//Evaluate`, `Abs`.

**Solution 212.** On propose :

```
In[1]: f[x_, y_] := Piecewise[{{2, x ≥ 0 && y ≥ 0 && x + y ≤ 1}}];
```

```
In[2]: loi = ProbabilityDistribution[f[x, y], {x, 0, 1}, {y, 0, 1}];
```


```
In[3]: loi2 = TransformedDistribution[x^2 - y, {x, y} ≈ loi];
```

```
In[4]: PDF[loi2, x]
```

$$\text{Out[4]: } \begin{cases} -1 - 2\sqrt{x} + \frac{5}{\sqrt{5+4x}} + \frac{4x}{\sqrt{5+4x}} & 0 < x < 1 \\ \frac{5+4x - \sqrt{5+4x}}{\sqrt{5+4x}} & -1 < x \leq 0 \\ 0 & \text{True} \end{cases}$$

```
In[5]: Plot[PDF[loi2, x] //Evaluate, {x, -1, 1}]
```


Out [5]:

In[6]: `loi3 = TransformedDistribution[Abs[z], z ≈ loi2];`In[7]: `PDF[loi3, x]`

Out [7]: 
$$\begin{cases} -2 + \frac{5}{\sqrt{5-4x}} - 2\sqrt{x} - \frac{4x}{\sqrt{5-4x}} + \frac{5}{\sqrt{5+4x}} + \frac{4x}{\sqrt{5+4x}} & 0 < x < 1 \\ 0 & \text{True} \end{cases}$$

In[8]: `Plot[PDF[loi3, x] //Evaluate, {x, 0, 1}]`

Out [8]:

**Exercice 213**

Soient  $X$  et  $Y$  deux *var iid* suivant chacune la loi de Laplace  $\mathcal{L}(0, 1)$ , *i.e.* de densité :

$$f(x) = \frac{1}{2}e^{-|x|}, \quad x \in \mathbb{R}.$$

On pose  $Z = X + Y$ .

Utiliser **Mathematica** pour

- déterminer une densité  $f_Z$  de  $Z$ ,
- représenter le graphe de  $f_Z(x)$  pour  $x \in [-7, 7]$ .

Commandes : `LaplaceDistribution`, `TransformedDistribution`, `≈`.

**Solution 213.** On propose :

```
In[1]: loi = LaplaceDistribution[0,1];
```

```
In[2]: loi2 = TransformedDistribution[x + y, {x ≈ loi, y ≈ loi}];
```


```
In[3]: PDF[loi2, x]
```

```
Out[3]: 
$$\begin{cases} -\frac{1}{4}e^x(-1+x) & x \leq 0 \\ \frac{1}{4}e^{-x}(1+x) & \text{True} \end{cases}$$

```

```
In[4]: Plot[PDF[loi2, x], {x, -7, 7}]
```

```
Out[4]:
```



### Exercice 214

Soient  $a > 0$ , et  $X$  et  $Y$  deux *var iid* suivant chacune la loi uniforme  $\mathcal{U}([0, a])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{a} & \text{si } x \in [0, a], \\ 0 & \text{sinon.} \end{cases}$$

On pose  $U = X + Y$  et  $V = \frac{X + Y}{2}$ .

Utiliser **Mathematica** pour

- créer la loi de  $U$  avec la commande **UniformSumDistribution**,
- déterminer une densité  $f_U$  de  $U$ ,
- tracer le graphe de  $f_U(x)$  pour  $a \in \{1, 1.4, 3.2\}$  (soit trois courbes) et  $x \in [0, 7]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée,
- créer la loi de  $V$  avec la commande **BatesDistribution**,
- déterminer une densité  $f_V$  de  $V$ ,
- tracer le graphe de  $f_V(x)$  pour  $a \in \{2, 2.4, 4\}$  (soit trois courbes) et  $x \in [0, 4.5]$  sur un même graphique, avec l'aire sous les courbes et une légende détaillée.

Commandes : UniformSumDistribution, PDF, Plot, Table, //Evaluate, PlotLegends, LineLegend, LegendLabel, Filling, BatesDistribution.

**Solution 214.** On propose :

```
In[1]: loi = UniformSumDistribution[2, {0, a}];
```


```
In[2]: PDF[loi, x]
```

$$\text{Out [2]: } \begin{cases} 0 & x < 0 \\ \frac{x}{a^2} & 0 \leq x < a \\ \frac{2a - x}{a^2} & a \leq x \leq 2a \\ 0 & \text{True} \end{cases}$$

```
In[3]: Plot[Table[PDF[loi, x], {a, {1, 1.4, 3.2}}] //Evaluate, {x, 0, 7},
```

```
PlotLegends → LineLegend[{1, 1.4, 3.2}, LegendLabel → a], Filling → Axis]
```

Out [3]:



```
In[4]: loi2 = BatesDistribution[2, {0, a}];
```


```
In[5]: PDF[loi2, x]
```

$$\text{Out [5]: } \begin{cases} 0 & x < 0 \\ \frac{4x}{a^2} & 0 \leq x < \frac{a}{2} \\ \frac{4(a-x)}{a^2} & \frac{a}{2} \leq x \leq a \\ 0 & \text{True} \end{cases}$$

```
In[6]: Plot[Table[PDF[loi2, x], {a, {2, 2.4, 4}}] //Evaluate, {x, 0, 4.5},
```

```
PlotLegends → LineLegend[{2, 2.4, 4}, LegendLabel → a], Filling → Axis]
```

Out [6] :

**Exercice 215**

Soient  $\lambda > 0$ , et  $X$  et  $Y$  deux *var* iid suivant chacune la loi exponentielle  $\mathcal{E}(\lambda)$ , i.e. de densité :

$$f(x) = \begin{cases} \lambda e^{-\lambda x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- déterminer une densité de  $A = X + Y$ ,
- déterminer une densité de  $B = X - Y$ ,
- déterminer une densité de  $C = |X - Y|$ ,
- déterminer une densité de  $D = \inf(X, Y)$ ,
- déterminer une densité de  $E = \frac{X}{X + Y}$ ,
- déterminer une densité de  $F = \frac{X}{|X - Y|}$ ,
- déterminer une densité de  $G = \frac{X + Y}{|X - Y|}$ ,
- déterminer une densité de  $H = \frac{X}{Y}$ ,
- déterminer une densité de  $I = \frac{XY}{(X - Y)^2}$ .

Commandes : `ExponentialDistribution`,  `$\lambda$` , `TransformedDistribution`,  `$\approx$` , `PDF`, `Abs`, `OrderDistribution`, `.`

**Solution 215.** On propose :

```
In[1]: loi = ExponentialDistribution[ $\lambda$ ];
```

```
In[2]: loi1 = TransformedDistribution[x + y, {x ≈ loi, y ≈ loi}]
```

```
Out[2]: ErlangDistribution[2, λ]
```

```
In[3]: PDF[loi1, x]
```

```
Out[3]: 
$$\begin{cases} e^{-x\lambda}x\lambda & x > 0 \\ 0 & \text{True} \end{cases}$$

```

!! On reconnaît également une densité associée à la loi gamma  $\Gamma(2, \frac{1}{\lambda})$ . En fait, on a : loi de Erlang  $\mathcal{E}_r(\alpha, \lambda) = \text{loi gamma } \Gamma(\alpha, \frac{1}{\lambda})$ .

```
In[4]: loi2 = TransformedDistribution[x - y, {x ≈ loi, y ≈ loi}]
```

```
Out[4]: LaplaceDistribution[0,  $\frac{1}{\lambda}$ ]
```

```
In[5]: PDF[loi2, x]
```

```
Out[5]: 
$$\begin{cases} \frac{1}{2}e^{-x\lambda} & x \geq 0 \\ \frac{1}{2}e^{x\lambda} & \text{True} \end{cases}$$

```

```
In[6]: loi3 = TransformedDistribution[Abs[x - y], {x ≈ loi, y ≈ loi}];
```

```
In[7]: PDF[loi3, x]
```

```
Out[7]: 
$$\begin{cases} e^{-x\lambda} & x > 0 \\ 0 & \text{True} \end{cases}$$

```

!! On reconnaît une densité associée à la loi exponentielle  $\mathcal{E}(\lambda)$ .

```
In[8]: loi4 = OrderDistribution[{loi, 2}, 1]
```

```
Out[8]: ExponentialDistribution[2λ]
```

```
In[9]: PDF[loi4, x]
```

```
Out[9]: 
$$\begin{cases} 2e^{-2x\lambda} & x \geq 0 \\ 0 & \text{True} \end{cases}$$

```

```
In[10]: loi5 = TransformedDistribution[x / (x + y), {x ≈ loi, y ≈ loi}];
```

```
In[11]: PDF[loi5, x]
```

```
Out[11]: 
$$\begin{cases} 1 & 0 \leq x \leq 1 \\ 0 & \text{True} \end{cases}$$

```

!! On reconnaît une densité associée à la loi uniforme  $\mathcal{U}([0, 1])$ .

```
In[12]: loi6 = TransformedDistribution[x / Abs[x - y], {x ≈ loi, y ≈ loi}];
```



In[13]: PDF[loi6, x]

$$\text{Out[13]: } \begin{cases} \frac{1}{(1+2x)^2} & 0 \leq x \leq 1 \\ \frac{2+8x^2}{(1-4x^2)^2} & x > 1 \\ 0 & \text{True} \end{cases}$$

In[14]: loi7 = TransformedDistribution[(x + y) / Abs[x - y], {x ≈ loi, y ≈ loi}];

In[15]: PDF[loi7, x]

$$\text{Out[15]: } \begin{cases} \frac{1}{x^2} & x > 1 \\ 0 & \text{True} \end{cases}$$

!! On reconnaît une densité associée à la loi de Pareto  $\mathcal{P}_{ar}(1,1)$ .

In[16]: loi8 = TransformedDistribution[x / y, {x ≈ loi, y ≈ loi}]

Out[16]: ParetoDistribution[1, 1, 0]

In[17]: PDF[loi8, x]

$$\text{Out[17]: } \begin{cases} \frac{1}{(1+x)^2} & x \geq 0 \\ 0 & \text{True} \end{cases}$$

!! On reconnaît une densité associée à la loi de Pareto généralisée  $\mathcal{P}_{ar}^g(1,1,0)$ .

In[18]: loi9 = TransformedDistribution[x y / (x - y)^2, {x ≈ loi, y ≈ loi}];

In[19]: PDF[loi9, x]

$$\text{Out[19]: } \begin{cases} \text{Indeterminate} & x == 0 \\ \frac{2}{(1+4x)^{3/2}} & x > 0 \\ 0 & \text{True} \end{cases}$$

!! On reconnaît une densité associée à la loi de Pareto généralisée  $\mathcal{P}_{ar}^g(\frac{1}{4}, \frac{1}{2}, 0)$ .

### Exercice 216

Soient  $(\mu_1, \mu_2) \in \mathbb{R}^2$ ,  $(\sigma_1, \sigma_2) \in ]0, \infty[^2$  et  $X$  et  $Y$  deux *var* indépendantes telles que :

◇  $X$  suit la loi normale  $\mathcal{N}(\mu_1, \sigma_1)$ , *i.e.* de densité :

$$f_X(x) = \frac{1}{\sqrt{2\pi\sigma_1^2}} e^{-\frac{(x-\mu_1)^2}{2\sigma_1^2}}, \quad x \in \mathbb{R},$$

◊  $Y$  suit la loi normale  $\mathcal{N}(\mu_2, \sigma_2)$ , i.e. de densité :

$$f_Y(x) = \frac{1}{\sqrt{2\pi\sigma_2^2}} e^{-\frac{(x-\mu_2)^2}{2\sigma_2^2}}, \quad x \in \mathbb{R}.$$

On pose  $S = \frac{X - \mu_1}{\sigma_1}$  et  $T = \frac{Y - \mu_2}{\sigma_2}$ .

Utiliser **Mathematica** pour

- donner la loi de  $S$ ,
- donner la loi de  $T$ ,
- donner la loi de  $U = S + T$ ,
- donner la loi de  $V = \frac{S}{T}$ ,
- donner la loi de  $W = S^2 + T^2$ .

Commandes : `TransformedDistribution`,  $\mu_1$ ,  $\sigma_1$ , `≈`, `NormalDistribution`, `CauchyDistribution`, `ChiSquareDistribution`.

**Solution 216.** On propose :

```
In[1]: loi1 = TransformedDistribution[(x - μ1) / σ1,
x ≈ NormalDistribution[μ1, σ1]]

Out[1]: NormalDistribution[0, 1]

In[2]: loi2 = TransformedDistribution[(y - μ2) / σ2,
y ≈ NormalDistribution[μ2, σ2]]

Out[2]: NormalDistribution[0, 1]

In[3]: loi3 = TransformedDistribution[s + t, {s ≈ loi1, t ≈ loi2}]

Out[3]: NormalDistribution[0, √2]

In[4]: loi4 = TransformedDistribution[s / t, {s ≈ loi1, t ≈ loi2}]

Out[4]: CauchyDistribution[0, 1]

In[5]: loi5 = TransformedDistribution[s^2 + t^2, {s ≈ loi1, t ≈ loi2}]

Out[5]: ChiSquareDistribution[2]
```

**Exercice 217**

Soient  $X$  et  $Y$  deux *var* iid suivant chacune la loi de Cauchy  $\mathcal{C}(0,1)$ , *i.e.* de densité :

$$f(x) = \frac{1}{\pi(1+x^2)}, \quad x \in \mathbb{R}.$$

Utiliser **Mathematica** pour

- déterminer une densité de  $A = X + Y$ ,
- déterminer une densité de  $B = X - Y$ ,
- déterminer une densité de  $C = XY$ ,
- déterminer une densité de  $D = \frac{X}{Y}$ ,
- déterminer une densité de  $E = \inf(X, Y)$ ,
- déterminer une densité de  $F = (X + Y, X - Y)$ ,
- déterminer une densité de  $G = \left(X + Y, \frac{X}{Y}\right)$ ,
- déterminer une densité de  $H = \left(XY, \frac{X}{Y}\right)$ .

Commande : `CauchyDistribution`, `TransformedDistribution`, `≈`, `PDF`, `Min`.

**Solution 217.** On propose :

```
In[1]: loi = CauchyDistribution[0,1];
```

```
In[2]: loi1 = TransformedDistribution[x + y, {x ≈ loi, y ≈ loi}]
```

```
Out[2]: CauchyDistribution[0,2]
```

```
In[3]: PDF[loi1, x]
```

```
Out[3]: 
$$\frac{1}{2\pi \left(1 + \frac{x^2}{4}\right)}$$

```

```
In[4]: loi2 = TransformedDistribution[x - y, {x ≈ loi, y ≈ loi}];
```

```
In[5]: PDF[loi2, x]
```

```
Out[5]: 
$$\frac{2}{\pi(4 + x^2)}$$

```

!! On reconnaît également une densité associée à la loi de Cauchy  $\mathcal{C}(0,2)$ .

```
In[6]: loi3 = TransformedDistribution[x y, {x ≈ loi, y ≈ loi}];
```

```
In[7]: PDF[loi3, x]
```

```
Out[7]: 
$$\frac{\text{Log}[x^2]}{\pi^2(-1 + x^2)}$$

```

```

In[8]: loi4 = TransformedDistribution[x / y, {x ≈ loi, y ≈ loi}];
In[9]: PDF[loi4, x]
Out[9]:  $\frac{\text{Log}[x^2]}{\pi^2(-1+x^2)}$ 
In[10]: loi5 = TransformedDistribution[Min[x, y], {x ≈ loi, y ≈ loi}];
In[11]: PDF[loi5, x]
Out[11]:  $\frac{\pi - 2\text{ArcTan}[x]}{\pi^2(1+x^2)}$ 
In[12]: loi5 = TransformedDistribution[{x + y, x - y}, {x ≈ loi, y ≈ loi}];
In[13]: PDF[loi5, {x, y}]
Out[13]:  $\frac{8}{\pi^2(4+x^2-2xy+y^2)(4+x^2+2xy+y^2)}$ 
In[14]: loi6 = TransformedDistribution[{x + y, x / y}, {x ≈ loi, y ≈ loi}];
In[15]: PDF[loi6, {x, y}]
Out[15]:  $\frac{x^2}{\pi^2 \sqrt{\frac{x^2}{(1+y)^4} (1+2y+(1+x^2)y^2)(x^2+(1+y)^2)}}$ 
In[16]: loi7 = TransformedDistribution[{x y, x / y}, {x ≈ loi, y ≈ loi}];
In[17]: PDF[loi7, {x, y}] //FullSimplify
Out[17]: 
$$\begin{cases} \frac{\text{Sign}[y]}{\pi^2(x+y)(1+xy)} & (x > 0 \ \&\& \ y > 0) \ || \ (x < 0 \ \&\& \ y < 0) \\ 0 & \text{True} \end{cases}$$


```

**Exercice 218**

Soient  $X$  et  $Y$  deux *var* indépendantes. Utiliser Mathematica pour vérifier les résultats en loi du tableau :

loi de $X$	loi de $Y$	loi de $X + Y$
exponentielle $\mathcal{E}(\lambda)$	exponentielle $\mathcal{E}(\lambda)$	Erlang $\mathcal{E}_r(2, \lambda)$
Erlang $\mathcal{E}_r(m, \lambda)$	Erlang $\mathcal{E}_r(n, \lambda)$	Erlang $\mathcal{E}_r(m + n, \lambda)$
normale $\mathcal{N}(a, b)$	normale $\mathcal{N}(c, d)$	normale $\mathcal{N}(a + b, \sqrt{c^2 + d^2})$
Cauchy $\mathcal{C}(a, b)$	Cauchy $\mathcal{C}(c, d)$	Cauchy $\mathcal{C}(a + c, b + d)$
Chi-deux $\chi^2(m)$	Chi-deux $\chi^2(n)$	Chi-deux $\chi^2(m + n)$

Commandes : TransformedDistribution, ≈, ExponentialDistribution, λ, ErlangDistribution, NormalDistribution, CauchyDistribution, ChiSquareDistribution.

**Solution 218.** On propose :

```
In[1]: TransformedDistribution[x + y, {x ≈ ExponentialDistribution[λ],
      y ≈ ExponentialDistribution[λ]}]
Out[1]: ErlangDistribution[2, λ]
In[2]: TransformedDistribution[x + y, {x ≈ ErlangDistribution[m, λ],
      y ≈ ErlangDistribution[n, λ]}]
Out[2]: ErlangDistribution[m + n, λ]
In[3]: TransformedDistribution[x + y, {x ≈ NormalDistribution[a, b],
      y ≈ NormalDistribution[c, d]}]
Out[3]: NormalDistribution[a + c, √(b² + d²)]
In[4]: TransformedDistribution[x + y, {x ≈ CauchyDistribution[a, b],
      y ≈ CauchyDistribution[c, d]}]
Out[4]: CauchyDistribution[a + c, b + d]
In[5]: TransformedDistribution[x + y, {x ≈ ChiSquareDistribution[m],
      y ≈ ChiSquareDistribution[n]}]
Out[5]: ChiSquareDistribution[m + n]
```

### Exercice 219

Soient  $X$  et  $Y$  deux *var* indépendantes. Utiliser Mathematica pour vérifier les résultats en loi du tableau :

loi de $X$	loi de $Y$	transformation	loi
normale $\mathcal{N}(0, 1)$	normale $\mathcal{N}(0, 1)$	$X^2 + Y^2$	Chi-deux $\chi^2(2)$
normale $\mathcal{N}(0, 1)$	Chi-deux $\chi^2(\nu)$	$\frac{X}{\sqrt{\frac{Y}{\nu}}}$	Student $\mathcal{T}(\nu)$
Chi-deux $\chi^2(\nu_1)$	Chi-deux $\chi^2(\nu_2)$	$\frac{\frac{X}{\nu_1}}{\frac{Y}{\nu_2}} = \frac{\nu_2 X}{\nu_1 Y}$	Fisher $\mathcal{F}(\nu_1, \nu_2)$

Commandes : TransformedDistribution, ≈, NormalDistribution, Sqrt, ChiSquareDistribution,  $\nu$ ,  $\nu_1$ .

**Solution 219.** On propose :

```
In[1]: TransformedDistribution[x^2 + y^2, {x ≈ NormalDistribution[0, 1],
      y ≈ NormalDistribution[0, 1]}]
Out[1]: ChiSquareDistribution[2]
```

```

In[2]: TransformedDistribution[x / Sqrt[y / ν], {x ≈ NormalDistribution[0, 1],
      y ≈ ChiSquareDistribution[ν]}]
Out[2]: StudentTDistribution[0, √ν, ν]
In[3]: TransformedDistribution[(x / ν1) / (y / ν2),
      {x ≈ ChiSquareDistribution[ν1], y ≈ ChiSquareDistribution[ν2]}]
Out[3]: FRatioDistribution[ν1, ν2]

```

**Exercice 220**

Soient  $X$  et  $Y$  deux *var iid* suivant chacune la loi uniforme  $\mathcal{U}([-1, 2])$ , *i.e.* de densité :

$$f(x) = \begin{cases} \frac{1}{3} & \text{si } x \in [-1, 2], \\ 0 & \text{sinon.} \end{cases}$$

On pose  $U = \inf(X, Y)$  et  $V = X + Y$ .

Utiliser *Mathematica* pour

- déterminer une densité  $f_{(U,V)}$  de  $(U, V)$ ,
- tracer le graphe de  $f_{(U,V)}$ ,
- calculer la matrice de covariance de  $(U, V)$ ,
- créer la fonction  $g(\alpha) = \mathbb{P}(U \geq \alpha V)$ ,  $\alpha \in \mathbb{R}$ ,
- expliciter  $g(\alpha)$ ,
- tracer le graphe de  $g(\alpha)$  pour  $\alpha \in [-4, 5]$ ,
- déterminer l'unique réel  $\alpha$  qui minimise  $g(\alpha)$
- préciser la valeur minimale de  $g(\alpha)$ .

Commandes : `UniformDistribution`, `TransformedDistribution`, `Min`, `≈`, `PDF`, `Plot3D`, `Covariance`, `//MatrixForm`, `x_`, `:=`, `Probability`, `Plot`, `ArgMin`, `MinValue`.

**Solution 220.** On propose :


```

In[1]: loi = UniformDistribution[{-1,2}];
In[2]: loi2 = TransformedDistribution[{Min[x, y], x + y}, {x ≈ loi, y ≈ loi}];
In[3]: PDF[loi2, {x, y}]
Out[3]: 
$$\begin{cases} \frac{2}{9} & 2x < y \ \&\& \ -1 \leq x \leq 2 \ \&\& \ y \leq 2 + x \ \&\& \ x \leq 1 + y \\ 0 & \text{True} \end{cases}$$

In[4]: Plot3D[PDF[loi2, {x, y}], {x, -1, 2}, {y, -1, 2}]

```

Out [4]:



In[5]: Covariance[loi2] //MatrixForm

Out [5]: 
$$\begin{pmatrix} \frac{1}{3} & \frac{3}{4} \\ \frac{2}{3} & \frac{4}{3} \\ \frac{3}{4} & \frac{4}{3} \\ \frac{4}{3} & \frac{2}{3} \end{pmatrix}$$


In[6]: g[α\_] := Probability[u ≥ α v, {u, v} ≈ loi2];

In[7]: g[α]

Out [7]: 
$$\begin{cases} \frac{2}{9} - \frac{1}{9\alpha} & 2\alpha > 1 \\ \frac{4-8\alpha}{9-9\alpha} & -1 \leq \alpha \leq \frac{1}{2} \\ \frac{1}{9} \left( 7 + \frac{1}{\alpha} \right) & \text{True} \end{cases}$$

In[8]: Plot[g[α], {α, -4, 5}]

Out [8]:



In[9]: ArgMin[g[α], α]

Out[9]:  $\frac{1}{2}$

In[10]: MinValue[g[α], α] (\* ou g[1 / 2] \*)

Out[10]: 0

### Exercice 221

Soient  $\lambda > 0$ , et  $X$  et  $Y$  deux *var iid* suivant chacune la loi exponentielle  $\mathcal{E}(\lambda)$ , *i.e.* de densité :

$$f(x) = \begin{cases} \lambda e^{-\lambda x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

On pose  $U = \sup(X, Y)$  et  $V = X + \frac{Y}{2}$ .

Utiliser *Mathematica* pour

- déterminer la fonction de répartition de  $U$ ,
- déterminer une densité  $f_U$  de  $U$ ,
- tracer le graphe de  $f_U(x)$  avec  $\lambda = 0.7$  pour  $x \in [0, 10]$ ,
- calculer  $\mathbb{E}(U)$ ,
- calculer  $\mathbb{V}(U)$ ,
- vérifier que  $U$  et  $V$  suivent la même loi en étudiant
  - ◇ les densités,
  - ◇ les fonctions caractéristiques.
- déterminer le réel  $x$  tel que  $\mathbb{P}(U \leq x) = \mathbb{P}(V \geq x)$ .

Commandes : `ExponentialDistribution`,  $\lambda$ , `OrderDistribution`, `CDF`, `PDF`, `Plot`, `/.`  $\rightarrow$ , `Mean`, `Variance`, `TransformedDistribution`, `≈`, `//Simplify`, `CharacteristicFunction`, `Median`.

**Solution 221.** On propose :

In[1]: loi = ExponentialDistribution[λ];

In[2]: loi2 = OrderDistribution[{loi, 2}, 2];

In[3]: CDF[loi2, x]

Out[3]: 
$$\begin{cases} (1 - e^{-x\lambda})^2 & x \geq 0 \\ 0 & \text{True} \end{cases}$$




In[4]: PDF[loi2, x]

Out[4]: 
$$\begin{cases} 2e^{-x\lambda}(1 - e^{-x\lambda})\lambda & x \geq 0 \\ 0 & \text{True} \end{cases}$$

In[5]: Plot[PDF[loi2, x] /. λ → 0.7, {x, 0, 10}]

Out[5]:



In[6]: Mean[loi2]

Out[6]:  $\frac{3}{2\lambda}$

In[7]: Variance[loi2]

Out[7]:  $\frac{5}{4\lambda^2}$

In[8]: loi3 = TransformedDistribution[x + y / 2, {x ≈ loi, y ≈ loi}];

In[9]: PDF[loi2, x] == PDF[loi3, x] //Simplify

Out[9]: True

In[10]: CharacteristicFunction[loi2, t] ==  
CharacteristicFunction[loi3, t] //Simplify

Out[10]: True

In[11]: Median[loi2]

Out[11]: 
$$-\frac{\text{Log}\left[1 - \frac{1}{\sqrt{2}}\right]}{\lambda}$$

!! On a considéré la médiane de la loi de  $U$  car, comme  $U$  et  $V$  suivent la même loi, on a :  $\mathbb{P}(U \leq x) = \mathbb{P}(V \geq x) \Leftrightarrow \mathbb{P}(U \leq x) = \mathbb{P}(U \geq x) \Leftrightarrow \mathbb{P}(U \leq x) = 0.5 \Rightarrow x$  est la médiane de la loi de  $U$ .

**Exercice 222**

Soient  $X$  et  $Y$  deux *var iid* suivant chacune la loi normale  $\mathcal{N}(0, 1)$ , *i.e.* de densité :

$$f_X(x) = \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}}, \quad x \in \mathbb{R}.$$

On pose  $Z = \sup(|X|, |Y|)$ .

Utiliser **Mathematica** pour

- déterminer une densité  $f_Z$  de  $Z$  (on verra apparaître la fonction  $\text{Erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt$ ,  $x \geq 0$ ),
- calculer  $\mathbb{E}(Z)$ ,
- calculer  $\mathbb{V}(Z)$ .

Commandes : `NormalDistribution`, `TransformedDistribution`, `Max`, `Abs`, `≈`, `PDF`, `Mean`, `Variance`.

**Solution 222.** On propose :

`In[1]: loi = NormalDistribution[0, 1];`

`In[2]: loi2 = TransformedDistribution[Max[Abs[x], Abs[y]], {x ≈ loi, y ≈ loi}];`

`In[3]: PDF[loi2, x]`

`Out[3]:` 
$$\begin{cases} 2e^{-\frac{x^2}{2}} \sqrt{\frac{2}{\pi}} \text{Erf} \left[ \frac{x}{\sqrt{2}} \right] & x > 0 \\ 0 & \text{True} \end{cases}$$

`In[4]: Mean[loi2]`

`Out[4]:` 
$$\frac{2}{\sqrt{\pi}}$$

`In[5]: Variance[loi2]`

`Out[5]:` 
$$\frac{-2 + \pi}{\pi}$$

**Exercice 223**

Soient  $(m, n) \in (\mathbb{N}^*)^2$ ,  $X$  et  $Y$  deux *var* indépendantes avec

◊  $X$  suivant la loi de Bates  $\mathcal{B}_{ates}(m)$ , *i.e.* de densité :

$$f_X(x) = \begin{cases} \frac{m^m}{(m-1)!} \sum_{k=0}^{\lfloor mx \rfloor} (-1)^k \binom{m}{k} \left(x - \frac{k}{m}\right)^{m-1} & \text{si } x \in ]0, 1[, \\ 0 & \text{sinon,} \end{cases}$$

où  $\lfloor mx \rfloor$  désigne la partie entière de  $mx$ ,

◇  $Y$  suivant la loi de Bates  $\mathcal{B}_{ates}(n)$ .

On pose  $Z = \frac{m}{m+n}X + \frac{n}{m+n}Y$ .

Utiliser **Mathematica** pour

- créer la loi de  $Z$ ,
- déterminer la fonction caractéristique de  $Z$ ,
- déterminer la fonction caractéristique d'une  $var U$  suivant la loi de Bates  $\mathcal{B}_{ates}(m+n)$ ,
- vérifier que  $Z$  suit la loi de Bates  $\mathcal{B}_{ates}(m+n)$ .

Commandes : `BatesDistribution`, `TransformedDistribution`, `≈`, `CharacteristicFunction`, `%`, `%%`.

**Solution 223.** On propose :

```
In[1]: loi1 = BatesDistribution[m];
```

```
In[2]: loi2 = BatesDistribution[n];
```

```
In[3]: loi3 = TransformedDistribution[m / (n + m) x + n / (n + m) y,
      {x ≈ loi1, y ≈ loi2}];
```

```
In[4]: CharacteristicFunction[loi3, t]
```

```
Out[4]: 
$$\left( \frac{i \left( -1 + e^{\frac{it}{m+n}} \right) (m+n)}{t} \right)^{m+n}$$

```

```
In[5]: CharacteristicFunction[BatesDistribution[m + n], t]
```

```
Out[5]: 
$$\left( \frac{i \left( -1 + e^{\frac{it}{m+n}} \right) (m+n)}{t} \right)^{m+n}$$

```

```
In[6]: % - %%
```

```
Out[6]: 0
```

### Exercice 224

Soit  $(X, Y)$  un couple de  $var$  dont la loi est donnée par la commande **Mathematica** :

```
BinormalDistribution[ρ]
```

Utiliser **Mathematica** pour

- vérifier que les commandes suivantes sont équivalentes :

```
PDF[BinormalDistribution[ρ], {x, y}]
```

```
PDF[MultinormalDistribution[{0, 0}, {{1, ρ}, {ρ, 1}}], {x, y}]
```

- expliciter une densité  $f$  de  $(X, Y)$ ,
- représenter graphiquement le comportement de  $f(x, y)$  pour  $(x, y) \in [-3.5, 3.5]^2$  avec plusieurs valeurs de  $\rho$  (on pourra utiliser les commandes `Manipulate`, `Plot3D` ou `DensityPlot` ou les deux).

Commandes : PDF, BinormalDistribution,  $\rho$ , MultinormalDistribution, Manipulate, Plot3D, PlotRange, ColorFunction, PlotRangePadding, PlotPoints, ImageSize, DensityPlot, Deployed, FrameMargins.

**Solution 224.** On propose :

```
In[1]: PDF[BinormalDistribution[ $\rho$ ], {x, y}] == PDF[MultinormalDistribution[{0, 0},
  {{1,  $\rho$ }, { $\rho$ , 1}}], {x, y}] //Simplify
```


Out[1]: True




```
In[2]: PDF[BinormalDistribution[ $\rho$ ], {x, y}]
```



Out[2]: 
$$\frac{e^{-\frac{x^2+y^2-2xy\rho}{2(1-\rho^2)}}}{2\pi\sqrt{1-\rho^2}}$$

```
In[3]: Manipulate[Plot3D[PDF[BinormalDistribution[ $\rho$ ], {x, y}], {x, -3.2, 3.2},
  {y, -3.2, 3.2}, PlotRange → Full, ColorFunction → "Pastel",
  PlotRangePadding → 0, PlotPoints → 40, ImageSize → 300]
  DensityPlot[PDF[BinormalDistribution[ $\rho$ ], {x, y}], {x, -3.2, 3.2},
  {y, -3.2, 3.2}, PlotRange → Full, ColorFunction → "Pastel",
  PlotRangePadding → 0, PlotPoints → 40, ImageSize → 230], { $\rho$ , -0.9, 0.9},
  { $\rho$ , Range[-0.9, 0.9, 0.1]}, Deployed → True, FrameMargins → 0]
```

Out[3]:







!! On constate que :

- plus  $|\rho|$  s'éloigne de 0, plus la silhouette du graphe de  $f$  est étirée dans une direction (laquelle dépend du signe de  $\rho$ ),
- plus  $|\rho|$  est proche de 1, plus la liaison linéaire entre  $Y$  et  $X$  est forte ; plus la probabilité qu'il existe  $a$  et  $b$  tels que  $Y = a + bX$  est forte.

## Exercice 225

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{2}{7} \times 36xy(1-x)(1-y) + \frac{5}{7} & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour


- créer la loi de  $(X, Y)$  avec les commandes `ProductDistribution`, `BetaDistribution`, `UniformDistribution` et `MixtureDistribution`,
- tracer le graphe de  $f$  avec les couleurs de `SunsetColors`,
- calculer simultanément  $\mathbb{E}(X)$  et  $\mathbb{E}(Y)$ ,
- calculer la matrice de covariance de  $(X, Y)$ .

Commandes : `ProductDistribution`, `BetaDistribution`, `UniformDistribution`, `MixtureDistribution`, `Plot3D`, `PDF`, `PlotRange`, `ColorFunction`, `Mean`, `Covariance`, `//MatrixForm`.

**Solution 225.** On propose :

```
In[1]: loi1 = ProductDistribution[{BetaDistribution[2, 2], 2}];
In[2]: loi2 = UniformDistribution[{{0, 1}, {0, 1}}];
In[3]: loi3 = MixtureDistribution[{2, 5}, {loi1, loi2}];
In[4]: Plot3D[PDF[loi3, {x, y}], {x, 0, 1}, {y, 0, 1}, PlotRange -> Full,
ColorFunction -> "SunsetColors"]
```

Out[4]:



```
In[5]: Mean[loi3]
Out[5]:  $\left\{\frac{1}{2}, \frac{1}{2}\right\}$ 
In[6]: Covariance[loi3] //MatrixForm
Out[6]:  $\begin{pmatrix} \frac{31}{420} & 0 \\ 0 & \frac{31}{420} \end{pmatrix}$ 
```

**Exercice 226**

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \frac{3}{5} \times \frac{1}{4} e^{-|x+1|-|y-0.5|} + \frac{2}{5} \times \frac{1}{2\pi\sqrt{0.91}} e^{-\frac{1}{2 \times 0.91}(x^2+y^2-2 \times 0.3xy)}, \quad (x, y) \in \mathbb{R}^2.$$

Utiliser Mathematica pour

- créer la loi de  $(X, Y)$  avec la commande `MixtureDistribution`,
- tracer le graphe de  $f$ ,
- tracer le `DensityPlot` de  $f$ .


Commandes : `ProductDistribution`, `LaplaceDistribution`, `BinormalDistribution`, `MixtureDistribution`, `plot3D`, `PDF`, `PlotRange`, `≈`, `ColorFunction`, `Function`, `Hue`, `DensityPlot`.

**Solution 226.** On propose :

```
In[1]: loi1 = ProductDistribution[LaplaceDistribution[-1, 1],
LaplaceDistribution[0.5, 1]];
In[2]: loi2 = BinormalDistribution[0.3];
In[3]: loi3 = MixtureDistribution[{3, 2}, {loi1, loi2}];
In[4]: Plot3D[PDF[loi3, {x, y}], {x, -4, 4}, {y, -4, 4}, PlotRange -> Full,
ColorFunction -> Function[{x, y, z}, Hue[z]]]
```




Out [4] :



```
In[5]: DensityPlot[PDF[loi3, {x, y}], {x, -4, 4}, {y, -4, 4}, PlotRange → Full,
ColorFunction → "SunsetColors"]
```

Out [5] :



### Exercice 227

Exécuter les commandes suivantes une à une et comprendre leurs enjeux :


```
loi1 = MultinormalDistribution[{2, 2}, {{1.7, 0.3}, {0.3, 1.1}}];
loi2 = MultinormalDistribution[{-2, -2}, {{3.4, 0.6}, {0.6, 2.6}}];
loi3 = MixtureDistribution[{3, 7}, {loi1, loi2}]
Plot3D[PDF[loi3, {x, y}], {x, -6.5, 5.5}, {y, -6.5, 6.5}, ColorFunction →
"SiennaTones", PlotRange → All]
```

Commandes : MultinormalDistribution, MixtureDistribution, Plot3D, PDF, ColorFunction, PlotRange.

**Solution 227.** On propose :

```
In[1]: loi1 = MultinormalDistribution[{2, 2}, {{1.7, 0.3}, {0.3, 1.1}}];
In[2]: loi2 = MultinormalDistribution[{-2, -2}, {{3.4, 0.6}, {0.6, 2.6}}];
In[3]: loi3 = MixtureDistribution[{3, 7}, {loi1, loi2}]
In[4]: Plot3D[PDF[loi3, {x, y}], {x, -6.5, 5.5}, {y, -6.5, 6.5}, ColorFunction ->
"SiennaTones", PlotRange -> All]
```

Out[4]:



On obtient alors le graphe de la densité :

$$f(x, y) = \frac{3}{10}f_1(x, y) + \frac{7}{10}f_2(x, y), \quad (x, y) \in \mathbb{R}^2,$$

où  $f_1$  est une densité associée à la loi normale multivariée  $\mathcal{N}_2(\mu_1, \Sigma_1)$  avec  $\mu_1 = (2, 2)^t$  et  $\Sigma_1 = \begin{pmatrix} 1.7 & 0.3 \\ 0.3 & 1.1 \end{pmatrix}$ , et

$f_2$  est une densité associée à la loi normale multivariée  $\mathcal{N}_2(\mu_2, \Sigma_2)$  avec  $\mu_2 = (-2, -2)^t$  et  $\Sigma_2 = \begin{pmatrix} 3.4 & 0.6 \\ 0.6 & 2.6 \end{pmatrix}$ .



## 11 Vecteurs de *var* à densité et convergence

### Exercice 228

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y, z) = \begin{cases} z(x^2 + y^2) & \text{si } (x, y, z) \in [-1, 1]^2 \times [0, b], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser la commande `Solve` de Mathematica pour déterminer  $b$ .

Commandes : `Solve`, `Integrate`, `&&`.

**Solution 228.** On propose :

```
In[1]: Solve[Integrate[z (x^2 + y^2), {x, -1, 1}, {y, -1, 1}, {z, 0, b}] == 1
      && b > 0, b]
```

```
Out[1]: {{ { b -> sqrt(3)/2 } }}
```

### Exercice 229

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y, z) = \begin{cases} x^y |z| & \text{si } (x, y, z) \in [0, 1]^2 \times [-b, b], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser la commande `Solve` de Mathematica pour déterminer  $b$ .

Commandes : `Solve`, `Integrate`, `Abs`, `&&`.

**Solution 229.** On propose :

```
In[1]: Solve[Integrate[x^y Abs[z], {x, 0, 1}, {y, 0, 1}, {z, -b, b}] == 1
      && b > 0, b]
```

```
Out[1]: {{ { b -> 1/sqrt(Log[2]) } }}
```

**Exercice 230**

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y, z) = \begin{cases} b \sin\left(\frac{y+z}{2}\right) & \text{si } 0 \leq x \leq y \leq z \leq \pi, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- créer la fonction  $h(x, y, z) = \frac{1}{b}g(x, y, z)$ ,
- déterminer  $b$ .

Commandes : `x_`, `:=`, `Piecewise`, `Sin`, `Pi`, `Integrate`.

**Solution 230.** On propose :

```
In[1]: h[x_, y_, z_] := Piecewise[{{Sin[(y + z) / 2], 0 ≤ x ≤ y ≤ z ≤ Pi}}];
```

```
In[2]: b = 1 / Integrate[h[x, y, z], {x, 0, Pi}, {y, 0, Pi}, {z, 0, Pi}]
```

```
Out[2]:  $\frac{1}{4}$ 
```

**Exercice 231**

On sait qu'il existe un unique réel  $b > 0$  tel que la fonction  $g$  définie ci-dessous soit une densité :

$$g(x, y, z) = \begin{cases} b(x + y + z) & \text{si } (x, y, z) \in \mathcal{D}, \\ 0 & \text{sinon,} \end{cases}$$

avec  $\mathcal{D} = \{(x, y, z) \in [0, 1]^2 \times [0, 2]; z \leq x^2 + y^2, y \leq x\}$ .

Utiliser **Mathematica** pour

- créer la fonction  $h(x, y, z) = \frac{1}{b}g(x, y, z)$ ,
- déterminer  $b$ .

Commandes : `x_`, `:=`, `Piecewise`, `&&`, `Integrate`.

**Solution 231.** On propose :

```
In[1]: h[x_, y_, z_] := Piecewise[{{x + y + z, 0 ≤ z ≤ x^2 + y^2 &&
0 ≤ y ≤ x ≤ 1}}];
```

```
In[2]: b = 1 / Integrate[h[x, y, z], {x, 0, 1}, {y, 0, 1}, {z, 0, 2}]
```

```
Out[2]:  $\frac{180}{103}$ 
```

**Exercice 232**

Soient  $X$ ,  $Y$  et  $Z$  trois *var iid* suivant chacune la loi uniforme  $\mathcal{U}([0, 1])$ , i.e. de densité :

$$f(x) = \begin{cases} 1 & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Utiliser **Mathematica** pour

- déterminer une densité de  $(X, Y, Z)$ ,
- déterminer la fonction de répartition de  $(X, Y, Z)$ ,
- calculer  $\mathbb{P}(X < Y < Z)$ ,
- calculer  $\mathbb{P}_{\{X < Y\}}(Y < Z)$ ,
- calculer  $\mathbb{P}(\{X > Y\} \cup \{Y < Z\})$ ,
- calculer  $\mathbb{P}(X > 2 \sup(Y, Z))$ .

Commandes : `UniformDistribution`, `ProductDistribution`, `PDF`, `CDF`, `Probability`, `≈`, `▷`, `||`, `Max`.

**Solution 232.** On propose :

```
In[1]: loi = UniformDistribution[{0, 1}];
```

```
In[2]: loi2 = ProductDistribution[{loi, 3}];
```

```
In[3]: PDF[loi2, {x, y, z}]
```

```
Out[3]: { 1  0 ≤ x ≤ 1 && 0 ≤ y ≤ 1 && 0 ≤ z ≤ 1
          0  True
```

```
In[4]: CDF[loi2, {x, y, z}]
```

```
Out[4]: { 1    x > 1 && y > 1 && z > 1
          x    0 ≤ x ≤ 1 && y > 1 && z > 1
          y    x > 1 && 0 ≤ y ≤ 1 && z > 1
          xy   0 ≤ x ≤ 1 && 0 ≤ y ≤ 1 && z > 1
          z    x > 1 && y > 1 && 0 ≤ z ≤ 1
          xz   0 ≤ x ≤ 1 && y > 1 && 0 ≤ z ≤ 1
          yz   x > 1 && 0 ≤ y ≤ 1 && 0 ≤ z ≤ 1
          xyz  0 ≤ x ≤ 1 && 0 ≤ y ≤ 1 && 0 ≤ z ≤ 1
          0    True
```

```

In[5]: Probability[x < y < z, {x, y, z} ≈ loi2]
Out[5]:  $\frac{1}{6}$ 

In[6]: Probability[y < z & x < y, {x, y, z} ≈ loi2]
Out[6]:  $\frac{1}{3}$ 

In[7]: Probability[x > y || y < z, {x, y, z} ≈ loi2]
Out[7]:  $\frac{2}{3}$ 

In[8]: Probability[x > 2 Max[y, z], {x, y, z} ≈ loi2]
Out[8]:  $\frac{1}{12}$ 

```

**Exercice 233**

Soit  $(X, Y, Z)$  un vecteur de *var* de densité :

$$f(x, y, z) = \begin{cases} \frac{10}{\pi}(x^2 + y^2) & \text{si } (x, y, z) \in \mathcal{D}, \\ 0 & \text{sinon,} \end{cases}$$

avec  $\mathcal{D} = \{(x, y, z) \in [-1, 1]^2 \times [0, 1]; x^2 + y^2 \leq z^2\}$ .

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- créer la loi de  $(X, Y, Z)$ ,
- calculer simultanément  $\mathbb{E}(X)$ ,  $\mathbb{E}(Y)$  et  $\mathbb{E}(Z)$ ,
- vérifier que  $\mathbb{E}(X^2 Z) \neq \mathbb{E}(X^2)\mathbb{E}(Z)$  (cela implique que  $X$  et  $Z$  ne sont pas indépendantes),
- calculer matrice de covariances de  $(X, Y, Z)$ ,
- calculer  $\mathbb{E}\left(\frac{1}{X^2 + Y^2}\right)$ ,
- calculer  $\mathbb{E}_{\{X^2 + Y^2 > 0.5\}}(X^2)$ .

Commandes : `x_`, `:=`, `Piecewise`, `Pi`, `ProbabilityDistribution`, `Mean`, `Moment`, `Covariance`, `//MatrixForm`, `Expectation`, `≈`, `▷`.

**Solution 233.** On propose :

```

In[1]: f[x_, y_, z_] := Piecewise[{{10 / Pi (x^2 + y^2), x^2 + y^2 ≤ z^2}}];
In[2]: loi = ProbabilityDistribution[f[x, y, z], {x, -1, 1}, {y, -1, 1},
    {z, 0, 1}];

```

```

In[3]: Mean[loi]
Out[3]: {0, 0, 5/6}
In[4]: Moment[loi, {2, 0, 1}] != Moment[loi, {2, 0, 0}] Moment[loi, {0, 0, 1}]
Out[4]: True
In[5]: Covariance[loi] //MatrixForm
Out[5]: 
$$\begin{pmatrix} \frac{5}{21} & 0 & 0 \\ 0 & \frac{5}{21} & 0 \\ 0 & 0 & \frac{5}{252} \end{pmatrix}$$

In[6]: Expectation[1 / (x^2 + y^2), {x, y, z} ≈ loi]
Out[6]: 10/3
In[7]: Expectation[x^2 > x^2 + y^2 ≥ 0.5, {x, y, z} ≈ loi]
Out[7]: 0.341345

```

**Exercice 234**

Soit  $(X, Y, Z)$  un vecteur de *var* de densité :

$$f(x, y, z) = \begin{cases} c \frac{1}{(x + y + z + 1)^3} & \text{si } (x, y, z) \in \mathcal{D}, \\ 0 & \text{sinon,} \end{cases}$$

avec  $c = \frac{16}{8 \ln(2) - 5}$  et  $\mathcal{D} = \{(x, y, z) \in [0, 1]^3; x + y + z \leq 1\}$ .

Utiliser *Mathematica* pour

- créer la fonction  $f$ ,
- vérifier que  $f$  est une densité,
- créer la loi de  $(X, Y, Z)$ ,
- calculer simultanément  $\mathbb{E}(X)$ ,  $\mathbb{E}(Y)$  et  $\mathbb{E}(Z)$ ,
- calculer simultanément  $\mathbb{V}(X)$ ,  $\mathbb{V}(Y)$  et  $\mathbb{V}(Z)$ ,
- calculer  $\rho(X, Y)$ .

Commandes : `x_`, `:=`, `Piecewise`, `Log`, `Integrate`, `ProbabilityDistribution`, `Mean`, `Variance`, `Correlation`.



**Solution 234.** On propose :

```
In[1]: f[x_, y_, z_] := Piecewise[{{(16 / (8 Log[2] - 5))
(1 / (x + y + z + 1)^3), x + y + z ≤ 1}}];
In[2]: Integrate[f[x, y, z], {x, 0, 1}, {y, 0, 1}, {z, 0, 1}] == 1
Out[2]: True
In[3]: loi = ProbabilityDistribution[f[x, y, z], {x, 0, 1}, {y, 0, 1}, {z, 0, 1}]
Out[3]: True
In[4]: Mean[loi]
Out[4]: { (17 - 24Log[2]) / (3(-5 + Log[256])), (17 - 24Log[2]) / (3(-5 + Log[256])), (17 - 24Log[2]) / (3(-5 + Log[256])) }
In[5]: Variance[loi]
Out[5]: { (-83 + 120Log[2]) / (18(-5 + Log[256])^2), (-83 + 120Log[2]) / (18(-5 + Log[256])^2), (-83 + 120Log[2]) / (18(-5 + Log[256])^2) }
In[6]: Correlation[loi, 1, 2]
Out[6]: (-661 + 1752Log[2] - 1152Log[2]^2) / (2(-83 + 120Log[2]))
```

### Exercice 235

Soit  $(X_1, X_2, X_3, X_4)$  un vecteur de *var* de densité :

$$f(x_1, x_2, x_3, x_4) = \begin{cases} 24e^{-(x_1+2x_2+3x_3+4x_4)} & \text{si } (x_1, x_2, x_3, x_4) \in [0, \infty[^4, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- créer la loi de  $(X_1, X_2, X_3, X_4)$ ,
- déterminer une densité de  $(X_1, X_3)$ ,
- calculer  $\mathbb{P}(\inf(X_1, X_2, X_3, X_4) > 0.3)$ ,
- calculer  $\mathbb{P}(X_1 + X_2 + X_3 + X_4 < 3.6)$ ,
- calculer  $\mathbb{E}(X_1 X_2^2 X_3^3 X_4^4)$ .

Commandes : ProbabilityDistribution, Exp, x<sub>1</sub>, ∞, PDF, MarginalDistribution, Probability, Min, ≈, Moment, Expectation.

**Solution 235.** On propose :

```
In[1]: loi = ProbabilityDistribution[24 Exp[-(x1 + 2 x2 + 3 x3 + 4 x4)],
{x1, 0, ∞}, {x2, 0, ∞}, {x3, 0, ∞}, {x4, 0, ∞}];
```

```

In[2]: PDF[MarginalDistribution[loi, {1, 3}], {x1, x3}]
Out[2]: 
$$\begin{cases} 3e^{-x_1-3x_3} & x_1 > 0 \ \&\& \ x_3 > 0 \\ 0 & \text{True} \end{cases}$$

In[3]: Probability[Min[x1, x2, x3, x4] > 0.3, {x1, x2, x3, x4} ≈ loi]
Out[3]: 0.0497871
In[4]: Probability[x1 + x2 + x3 + x4 < 3.6, {x1, x2, x3, x4} ≈ loi]
Out[4]: 0.895104
In[5]: Moment[loi, {1, 2, 3, 4}]
Out[5]:  $\frac{1}{96}$ 

```

### Exercice 236

Soient  $n \in \mathbb{N}^*$  et  $X_1, \dots, X_n$   $n$  var indépendantes. On pose :  $S_n = \sum_{i=1}^n X_i$ .

Utiliser Mathematica pour

- créer une fonction `sumloi` à deux arguments : `loi` et `n`, qui caractérise la loi de  $S_n$  avec  $X_1, \dots, X_n$  suivant chacune la loi `loi`,
- utiliser la fonction `sumloi` pour vérifier les résultats en loi du tableau avec  $n = 5$  :

loi de $X_i$	loi de $S_n$
exponentielle $\mathcal{E}(\lambda)$	Erlang $\mathcal{E}_r(n, \lambda)$
Erlang $\mathcal{E}_r(m, \lambda)$	Erlang $\mathcal{E}_r(nm, \lambda)$
normale $\mathcal{N}(\mu, \sigma)$	normale $\mathcal{N}(n\mu, \sqrt{n} \sigma )$
Cauchy $\mathcal{C}(a, b)$	Cauchy $\mathcal{C}(na, nb)$
Chi-deux $\chi^2(\nu)$	Chi-deux $\chi^2(n\nu)$

- proposer des commandes pour vérifier les résultats en loi du tableau avec  $n = 5$  :

loi de $X_i$	loi de $S_n$
normale $\mathcal{N}(\mu_i, \sigma_i)$	normale $\mathcal{N}\left(\sum_{i=1}^n \mu_i, \sqrt{\sum_{i=1}^n \sigma_i^2}\right)$
Chi-deux $\chi^2(\nu_i)$	Chi-deux $\chi^2\left(\sum_{i=1}^n \nu_i\right)$

Commandes : `TransformedDistribution`, `Sum`, `Table`, `≈`, `ExponentialDistribution`, `λ`, `ErlangDistribution`, `NormalDistribution`, `μ`, `σ`, `CauchyDistribution`, `ChiSquareDistribution`, `ν`, `μi`, `σi`, `νi`.

**Solution 236.** On propose :

```
In[1]: sumloi[loi_, n_] := TransformedDistribution[Sum[x[i], {i, 1, n}],
      Table[x[i] ≈ loi, {i, 1, n}]];

In[2]: sumloi[ExponentialDistribution[λ], 5]
Out[2]: ErlangDistribution[5, λ]

In[3]: sumloi[ErlangDistribution[m, λ], 5]
Out[3]: ErlangDistribution[5m, λ]

In[4]: sumloi[NormalDistribution[μ, σ], 5]
Out[4]: NormalDistribution[5μ, √5 Abs[σ]]

In[5]: sumloi[CauchyDistribution[a, b], 5]
Out[5]: CauchyDistribution[5a, 5b]

In[6]: sumloi[ChiSquareDistribution[ν], 5]
Out[6]: ChiSquareDistribution[5ν]

In[7]: TransformedDistribution[Sum[x[i], {i, 5}], Table[x[i] ≈
      NormalDistribution[μi, σi], {i, 5}]]
Out[7]: NormalDistribution[μ1 + μ2 + μ3 + μ4 + μ5, √σ12 + σ22 + σ32 + σ42 + σ52]

In[8]: TransformedDistribution[Sum[x[i], {i, 10}], Table[x[i] ≈
      ChiSquareDistribution[νi], {i, 10}]]
Out[8]: ChiSquareDistribution[ν1 + ν2 + ν3 + ν4 + ν5]
```

### Exercice 237

Soient  $(\mu_1, \dots, \mu_5) \in \mathbb{R}^5$ ,  $(\sigma_1, \dots, \sigma_5) \in ]0, \infty[^5$ , et  $X_1, \dots, X_5$  5 var indépendantes telles que, pour tout  $i \in \{1, \dots, 5\}$ ,  $X_i$  suit la loi log-normale  $\mathcal{LN}(\mu_i, \sigma_i)$ , i.e. de densité :

$$f_{X_i}(x) = \begin{cases} \frac{1}{x\sqrt{2\pi\sigma_i^2}} e^{-\frac{(\ln(x)-\mu_i)^2}{2\sigma_i^2}} & \text{si } x > 0, \\ 0 & \text{sinon.} \end{cases}$$

On pose  $P = \prod_{i=1}^5 X_i$ .

Utiliser Mathematica pour vérifier que  $P$  suit une loi log-normale.

Commandes : TransformedDistribution, Product, Table, ≈, LogNormalDistribution,  $\mu_i$ ,  $\sigma_i$ .

**Solution 237.** On propose :

```
In[1]: TransformedDistribution[Product[x[i], {i, 1, 5}], Table[x[i] ≈
      LogNormalDistribution[μi, σi], {i, 1, 5}]]
```

```
Out[1]: LogNormalDistribution[μ1 + μ2 + μ3 + μ4 + μ5, √(σ12 + σ22 + σ32 + σ42 + σ52) ]
```

### Exercice 238

Soit  $(X, Y, Z)$  un vecteur de *var* dont la loi est donnée par l'objet `loi` défini par les commandes Mathematica :

```
μ = {0, 0, 0}
Σ = {{3, 1 / 3, 0}, {1 / 3, 2 / 3, -1}, {0, -1, 4}}
loi = MultinormalDistribution[μ, Σ]
```

Utiliser Mathematica pour

- expliciter une densité  $f_{(X,Y,Z)}$  de  $(X, Y, Z)$ ,
- déterminer une densité  $f_{(X,Z)}$  de  $(X, Z)$ ,
- tracer le graphe de  $f_{(X,Z)}(x, z)$  pour  $(x, z) \in [-5, 5]^2$  avec les couleurs de `AuroraColors`,
- déterminer la loi de  $(U, V, W)$  avec  $U = X + 2Y + 3Z + 1$ ,  $V = 4X + 5Y - 6Z + 2$  et  $W = 7X + 8Y + 9Z + 3$ .

Commandes : `μ`, `Σ`, `MultinormalDistribution`, `PDF`, `MarginalDistribution`, `Plot3D`, `ColorFunction`, `TransformedDistribution`, `≈`.

**Solution 238.** On propose :

```
In[1]: μ = {0, 0, 0};
```

```
In[2]: Σ = {{3, 1 / 3, 0}, {1 / 3, 2 / 3, -1}, {0, -1, 4}};
```

```
In[3]: loi = MultinormalDistribution[μ, Σ];
```

```
In[4]: PDF[loi, {x, y, z}]
```

```
Out[4]: 
$$\frac{e^{\frac{1}{2} \left( -x \left( \frac{15x}{41} - \frac{12y}{41} - \frac{3z}{41} \right) - y \left( -\frac{12x}{41} + \frac{108y}{41} + \frac{27z}{41} \right) - \left( -\frac{3x}{41} + \frac{27y}{41} + \frac{17z}{41} \right) z \right)}}{2\sqrt{82}\pi^{3/2}}$$

```

```
In[5]: loi2 = MarginalDistribution[loi, {1, 3}]
```

```
Out[5]: MultinormalDistribution[{0, 0}, {{3, 0}, {0, 4}}]
```


```
In[6]: PDF[loi2, {x, z}]
```

```
Out[6]: 
$$\frac{e^{\frac{1}{2} \left( -\frac{x^2}{3} - \frac{z^2}{4} \right)}}{4\sqrt{3}\pi}$$

```

```
In[7]: Plot3D[PDF[loi2, {x, z}], {x, -5, 5}, {z, -5, 5}, ColorFunction →
      "AuroraColors"]
```

Out [7]:



```
In[8]: A = {{1, 2, 3}, {4, 5, -6}, {7, 8, 9}};
```

```
In[9]: c = {1, 2, 3};
```

```
In[10]: loi3 = TransformedDistribution[A.{x, y, z} + c, {x, y, z} ≈ loi];
```

```
Out[10]: MultinormalDistribution[{{1, 2, 3}}, {{31, -52, 105}, {-52, 282, -80},
{105, -80, 407}}]
```

### Exercice 239

Soient  $X_1, X_2, X_3, X_4$  4 var iid suivant chacune la loi normale  $\mathcal{N}(0, 1)$ , i.e. de densité :

$$f(x) = \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}}, \quad x \in \mathbb{R}.$$

On pose  $Y = X_1 X_2 + X_3 X_4$ .

Utiliser *Mathematica* pour

- créer la loi de  $Y$ ,
- déterminer la fonction caractéristique de  $Y$ ,
- vérifier que  $Y$  suit la loi de Laplace  $\mathcal{L}(0, 1)$ .

Commandes : `NormalDistribution`, `ProductDistribution`, `TransformedDistribution`, `x1`, `≈`, `CharacteristicFunction`, `%`, `LaplaceDistribution`.

**Solution 239.** On propose :

```
In[1]: loi = NormalDistribution[0, 1];
In[2]: loi2 = ProductDistribution[{loi, 4}];
In[3]: loi3 = TransformedDistribution[x1 x2 + x3 x4, {x1, x2, x3, x4} ≈ loi2];
In[4]: CharacteristicFunction[loi3, t]
Out[4]:  $\frac{1}{1+t^2}$ 
In[5]: % == CharacteristicFunction[LaplaceDistribution[0,1], t]
Out[5]: True
```

### Exercice 240

Soient  $n \in \mathbb{N}^*$  et  $X_1, \dots, X_n$   $n$  var iid suivant chacune la loi uniforme  $\mathcal{U}([0, 3])$ , i.e. de densité :

$$f(x) = \begin{cases} \frac{1}{3} & \text{si } x \in [0, 3], \\ 0 & \text{sinon.} \end{cases}$$

On pose  $U_n = \frac{2}{n} \sum_{i=1}^n X_i$  et  $V_n = \frac{n+1}{n} \sup(X_1, \dots, X_n)$ .

Utiliser Mathematica pour

- créer la loi de  $U_n$  avec la commande `BatesDistribution`,
- créer la loi de  $V_n$  avec la commande `OrderDistribution`,
- déterminer une densité de  $V_n$ ,
- vérifier que  $\mathbb{E}(U_n) = \mathbb{E}(V_n) = 3$ ,
- déterminer les entiers  $n$  tels que  $\mathbb{V}(V_n) < \mathbb{V}(U_n)$ .

Commandes : `BatesDistribution`, `TransformedDistribution`, `≈`, `OrderDistribution`, `UniformDistribution`, `PDF`, `Mean`, `Reduce`, `Variance`, `&&`.

**Solution 240.** On propose :

```
In[1]: loi1 = BatesDistribution[n, {0, 3}];
In[2]: loi2 = TransformedDistribution[2 x, x ≈ loi1];
In[3]: loi3 = OrderDistribution[{UniformDistribution[{0, 3}], n}, n];
In[4]: loi4 = TransformedDistribution[(n + 1) / n y, y ≈ loi3];
```

```

In[5]: PDF[loi4, x]
Out[5]: 
$$\begin{cases} \frac{3^{-n} n^2 \left(\frac{n}{1+n}\right)^{-1+n} x^{-1+n}}{1+n} & 0 \leq \frac{nx}{1+n} \leq 3 \\ 0 & \text{True} \end{cases}$$

In[6]: Mean[loi2] == Mean[loi4] == 3
Out[6]: True
In[7]: Reduce[Variance[loi4] < Variance[loi2] && n ≥ 1, n]
Out[7]: n > 1
!! Ainsi,  $V_n$  a une variance plus petite que celle de  $U_n$  dès lors que  $n \geq 2$ .

```

**Exercice 241**

Soient  $\lambda > 0$  et  $X_1, \dots, X_6$  6 var iid suivant la loi exponentielle  $\mathcal{E}(\lambda)$ , i.e. de densité :

$$f(x) = \begin{cases} \lambda e^{-\lambda x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

On pose  $U = \sum_{i=1}^6 \frac{X_i}{i}$  et  $V = \sup(X_1, \dots, X_6)$ .

Utiliser **Mathematica** pour

- créer la loi de  $U$ ,
- créer la loi de  $V$ ,
- déterminer une densité de  $U$ ,
- déterminer la fonction caractéristique de  $U$ ,
- vérifier que  $U$  et  $V$  suivent la même loi en étudiant
  - ◇ les densités,
  - ◇ les fonctions caractéristiques.
- déterminer le réel  $x$  tel que  $\mathbb{P}(U \leq x) = \mathbb{P}(V \geq x)$ .

Commandes : ExponentialDistribution,  $\lambda$ , TransformedDistribution, Sum, Table,  $\approx$ , OrderDistribution, PDF, CharacteristicFunction, //Simplify, CharacteristicFunction, Median.

**Solution 241.** On propose :

```

In[1]: loi = ExponentialDistribution[λ];
In[2]: loi2 = TransformedDistribution[Sum[x[i] / i, {i, 1, 6}],
Table[x[i] ≈ loi, {i, 1, 6}]];

```

```

In[3]: loi3 = OrderDistribution[{loi, 6}, 6];
In[4]: PDF[loi2, x]
Out[4]: 
$$\begin{cases} 6e^{-6x\lambda}(-1 + e^{x\lambda})^5\lambda & x \geq 0 \\ 0 & \text{True} \end{cases}$$

In[5]: CharacteristicFunction[loi2, t]
Out[5]: 
$$\frac{720\lambda^6}{(t + i\lambda)(t + 2i\lambda)(t + 3i\lambda)(t + 4i\lambda)(t + 5i\lambda)(t + 6i\lambda)}$$

In[6]: PDF[loi2, x] == PDF[loi3, x] //Simplify
Out[6]: True
In[7]: CharacteristicFunction[loi2, t] ==
CharacteristicFunction[loi3, t] //Simplify
Out[7]: True
In[8]: Median[loi2]
Out[8]: 
$$-\frac{\text{Log}\left[1 - \frac{1}{2^{1/6}}\right]}{\lambda}$$


```

!! On a considéré la médiane de la loi de  $U$  car, comme  $U$  et  $V$  suivent la même loi, on a :  $\mathbb{P}(U \leq x) = \mathbb{P}(V \geq x) \Leftrightarrow \mathbb{P}(U \leq x) = \mathbb{P}(U \geq x) \Leftrightarrow \mathbb{P}(U \leq x) = 0.5 \Rightarrow x$  est la médiane de la loi de  $U$ .

#### Exercice 242

Pour tout  $n \in \mathbb{N}^*$ , soit  $X_n$  une *var* de densité :

$$f_{X_n}(x) = \frac{4ne^{-4nx}}{(1 + e^{-4nx})^2}, \quad x \in \mathbb{R}.$$

Utiliser **Mathematica** pour

- vérifier que  $f_{X_n}$  est une densité,
- créer la loi de  $X_n$ ,
- créer la fonction :  $g(n, \epsilon) = \mathbb{P}(|X_n - 0| \geq \epsilon)$ ,  $\epsilon > 0$ ,
- utiliser  $g$  pour vérifier que  $(X_n)_{n \in \mathbb{N}^*}$  converge en probabilité vers 0.

Commandes : `Integrate`, `Exp`, `∞`, `Assumptions`, `ProbabilityDistribution`, `x_`, `ϵ`, `:=`, `Probability`, `Abs`, `&&`, `≈`, `Limit`.

**Solution 242.** On propose :

```

In[1]: Integrate[4 n Exp[-4 n x] / (1 + Exp[-4 n x])^2, {x, -∞, ∞},
Assumptions → n ≥ 1] == 1

```



Out[1]: True

In[2]: loi = ProbabilityDistribution[4 n Exp[-4 n x] / (1 + Exp[-4 n x])^2,  
{x, -∞, ∞}, Assumptions → n ≥ 1];

In[3]: g[n\_, ε\_] := Probability[Abs[x - 0] > ε && ε > 0, x ≈ loi];

In[4]: Limit[g[n, ε], n → ∞]

Out[4]: 0

!! La densité  $f_{X_n}$  est associée à la loi Logistique  $\mathcal{L}_{og}(0, \frac{1}{4n})$ .

### Exercice 243

Pour tout  $n \in \mathbb{N}^*$ , soit  $X_n$  une var suivant la loi exponentielle  $\mathcal{E}(\frac{1}{n})$ , i.e. de densité :

$$f(x) = \begin{cases} \frac{1}{n} e^{-\frac{1}{n}x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

◦ créer la fonction :  $g(n, \epsilon) = \mathbb{P}\left(\left|\frac{X_n}{n \log(n)} - 0\right| \geq \epsilon\right)$ ,  $\epsilon > 0$ ,

◦ utiliser  $g$  pour vérifier que  $\left(\frac{X_n}{n \log(n)}\right)_{n \in \mathbb{N}^*}$  converge en probabilité vers 0.

Commandes : ExponentialDistribution, x\_, ε, :=, Probability, Abs, Log, &&, ≈, Limit, ∞.

**Solution 243.** On propose :

In[1]: loi = ExponentialDistribution[1 / n];

In[2]: g[n\_, ε\_] := Probability[Abs[x / (n Log[n]) - 0] > ε && ε > 0, x ≈ loi];

In[3]: Limit[g[n, ε], n → ∞]

Out[3]: 0

!! On constate que, pour tout  $\epsilon > 0$ , on a  $\lim_{n \rightarrow \infty} \mathbb{P}\left(\left|\frac{X_n}{n \log(n)} - 0\right| \geq \epsilon\right) = 0$ . Cela entraîne la convergence en probabilité de  $(X_n)_{n \in \mathbb{N}^*}$  vers 0 (la var constante).

**Exercice 244**

Soient  $\lambda > 0$  et  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* suivant chacune la loi uniforme  $\mathcal{U}([0, 1])$ , *i.e.* de densité :

$$f(x) = \begin{cases} 1 & \text{si } x \in [0, 1], \\ 0 & \text{sinon.} \end{cases}$$

Pour tout  $n \in \mathbb{N}^*$ , on pose  $W_n = \frac{n}{\lambda} \inf(X_1, \dots, X_n)$ .

Utiliser *Mathematica* pour

- déterminer la fonction de répartition  $F_{W_n}$  de  $W_n$ ,
- calculer  $\lim_{n \rightarrow \infty} F_{W_n}(x)$ ,
- vérifier que  $(W_n)_{n \in \mathbb{N}^*}$  converge en loi vers une *var*  $W$  suivant la loi exponentielle  $\mathcal{E}(\lambda)$ .

Commandes : `OrderDistribution`, `UniformDistribution`, `TransformedDistribution`, `λ`, `≈`, `Assumptions`, `CDF`, `Limit`, `∞`, `ExponentialDistribution`, `%`, `%%`, `//Simplify`.

**Solution 244.** On propose :

```
In[1]: loi1 = OrderDistribution[{UniformDistribution[{0, 1}], n}, 1];
```

```
In[2]: loi2 = TransformedDistribution[n / λ x, x ≈ loi1, Assumptions →
      {λ > 0, n ≥ 1}];
```

```
In[3]: CDF[loi2, x]
```

```
Out[3]: 
$$\begin{cases} 1 & \frac{x\lambda}{n} \geq 1 \\ 1 - \left(1 - \frac{x\lambda}{n}\right)^n & 0 < \frac{x\lambda}{n} < 1 \\ 0 & \text{True} \end{cases}$$

```

```
In[4]: Limit[CDF[loi2, x], n → ∞, Assumptions → λ > 0]
```

```
Out[4]: 
$$\begin{cases} 1 - e^{-x\lambda} & x > 0 \\ 0 & \text{True} \end{cases}$$

```

```
In[5]: CDF[ExponentialDistribution[λ], x]
```

```
Out[5]: 
$$\begin{cases} 1 - e^{-x\lambda} & x \geq 0 \\ 0 & \text{True} \end{cases}$$

```

```
In[6]: % - %% //Simplify
```

```
Out[6]: 0
```

**Exercice 245**

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* suivant chacune la loi exponentielle  $\mathcal{E}(2)$ , *i.e.* de densité :

$$f(x) = \begin{cases} 2e^{-2x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Pour tout  $n \in \mathbb{N}^*$ , on pose  $W_n = \frac{\log(n)}{2} - \sup(X_1, \dots, X_n)$ .

Utiliser **Mathematica** pour

- déterminer la fonction de répartition  $F_{W_n}$  de  $W_n$ ,
- calculer  $\lim_{n \rightarrow \infty} F_{W_n}(x)$ ,
- vérifier que  $(W_n)_{n \in \mathbb{N}^*}$  converge en loi vers une *var*  $W$  suivant la loi de Gumbell  $\mathcal{G}_{umb}(0, \frac{1}{2})$ .

Commandes : `OrderDistribution`, `ExponentialDistribution`, `TransformedDistribution`, `Log`, `≈`, `CDF`, `Limit`, `∞`, `GumbelDistribution`, `%`, `%%`, `//Simplify`.

**Solution 245.** On propose :

```
In[1]: loi1 = OrderDistribution[{ExponentialDistribution[2], n}, n];
```

```
In[2]: loi2 = TransformedDistribution[Log[n] / 2 - x, x ≈ loi1];
```

```
In[3]: CDF[loi2, x]
```

```
Out[3]: 
$$\begin{cases} 1 - \left(1 - e^{-2\left(-x + \frac{\text{Log}[n]}{2}\right)}\right)^n & 2x - \text{Log}[n] \leq 0 \\ 1 & \text{True} \end{cases}$$

```

```
In[4]: Limit[CDF[loi2, x], n → ∞]
```

```
Out[4]:  $1 - e^{-e^{2x}}$ 
```

```
In[5]: CDF[GumbelDistribution[0, 1 / 2], x]
```

```
Out[5]:  $1 - e^{-e^{2x}}$ 
```

```
In[6]: % - %% //Simplify
```

```
Out[6]: 0
```

!! On a montré que, pour tout point de continuité  $x$  de  $F_W(x)$ ,  $\lim_{n \rightarrow \infty} F_{W_n}(x) = F_W(x)$ . Cela entraîne la convergence en loi de  $(W_n)_{n \in \mathbb{N}^*}$  vers une *var*  $W$  suivant la loi de Gumbell  $\mathcal{G}_{umb}(0, \frac{1}{2})$ .

**Exercice 246**

*Théorème central limite : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right).$$

*Alors la suite de var  $(Z_n)_{n \in \mathbb{N}^*}$  converge en loi vers une var  $Z$  suivant la loi normale  $\mathcal{N}(0, 1)$ .*

En termes mathématiques, le théorème central limite est équivalent à la convergence simple :

$$\lim_{n \rightarrow \infty} \varphi_{Z_n}(t) = e^{-\frac{t^2}{2}}, \quad t \in \mathbb{R},$$

où  $\varphi_{Z_n}(t) = \mathbb{E}(e^{itZ_n})$  est la fonction caractéristique de  $Z_n$  (notons que  $e^{-\frac{t^2}{2}} = \varphi_Z(t) = \mathbb{E}(e^{itZ})$  est la fonction caractéristique de  $Z$ ). Après développement et utilisant les hypothèses, on peut montrer que

$$\varphi_{Z_n}(t) = \left( \varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}} \left( \frac{t}{\sqrt{n}} \right) \right)^n,$$

où  $\varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}}(t)$  est la fonction caractéristique de  $\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}$ .

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- créer une fonction `tcl` ayant deux arguments : `loi` et `t`, telle que, pour  $X_1$  suivant la loi `loi` :

$$\text{tcl}(\text{loi}, t) = \lim_{n \rightarrow \infty} \left( \varphi_{\frac{X_1 - \mathbb{E}(X_1)}{\sigma(X_1)}} \left( \frac{t}{\sqrt{n}} \right) \right)^n.$$

- vérifier que `tcl(loi, t) = e-t2/2` avec l'argument `loi` correspondant à :
  - ◊ la loi normale tordue donnée par la commande `SkewNormalDistribution[0, 2, 3]`,
  - ◊ la loi de Nakagami donnée par la commande `NakagamiDistribution[1, 3]`,

Commandes : `x_`, `:=`, `Limit`, `CharacteristicFunction`, `TransformedDistribution`, `Mean`, `StandardDeviation`, `≈`, `Sqrt`, `∞`, `SkewNormalDistribution`, `NakagamiDistribution`.

**Solution 246.** On propose :

```
In[1]: tcl[loi_, t_] := Limit[(CharacteristicFunction[
      TransformedDistribution[(x - Mean[loi]) / StandardDeviation[loi],
      x ≈ loi], t / Sqrt[n]])^n, n → ∞];
```

In[2]: `tcl[SkewNormalDistribution[0, 2, 3], t]`

Out[2]:  $e^{-\frac{t^2}{2}}$

In[3]: `tcl[NakagamiDistribution[1, 3], t]`

Out[3]:  $e^{-\frac{t^2}{2}}$

### Exercice 247

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var* telle que, pour tout  $n \in \mathbb{N}^*$ ,  $X_n$  suit la loi de Student  $\mathcal{T}(n)$ , *i.e.* de densité :

$$f_{X_n}(x) = \frac{1}{\sqrt{n\pi}} \frac{\Gamma\left(\frac{n+1}{2}\right)}{\Gamma\left(\frac{n}{2}\right)} \left(1 + \frac{x^2}{n}\right)^{-\frac{n+1}{2}}, \quad x \in \mathbb{R},$$

où  $\Gamma$  désigne la fonction gamma d'Euler définie par :  $\Gamma(x) = \int_0^\infty t^{x-1} e^{-t} dt$ ,  $x > 0$ .

Utiliser *Mathematica* pour

- déterminer  $\lim_{n \rightarrow \infty} f_{X_n}(x)$ ,
- vérifier que  $(X_n)_{n \in \mathbb{N}^*}$  converge en loi vers une *var*  $X$  suivant la loi normale  $\mathcal{N}(0, 1)$ .

Commandes : `Limit`, `PDF`, `StudentTDistribution`, `∞`, `NormalDistribution`, `%`, `%%`, `//Simplify`.

**Solution 247.** On propose :

In[1]: `Limit[PDF[StudentTDistribution[n], x], n → ∞]`

Out[1]:  $\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\pi}}$

In[2]: `PDF[NormalDistribution[0, 1], x]`

Out[2]:  $\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\pi}}$

In[3]: `% - %% //Simplify`

Out[3]: `0`

### Exercice 248

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* suivant chacune la loi exponentielle  $\mathcal{E}(1)$ , *i.e.* de densité :

$$f(x) = \begin{cases} e^{-x} & \text{si } x \geq 0, \\ 0 & \text{sinon.} \end{cases}$$

Pour tout  $n \in \mathbb{N}^*$ , on pose

$$S_n = \sum_{k=1}^n X_k, \quad Y_n = 3^n e^{-2S_n}.$$

- créer la loi de  $S_n$  en utilisant un résultat en loi connu,
- créer la loi de  $Y_n$ ,
- créer la loi de  $\sqrt{Y_n}$ ,
- vérifier que  $(Y_n)_{n \in \mathbb{N}^*}$  ne converge pas en  $\mathbb{L}^1$  vers 0,
- vérifier que  $(\sqrt{Y_n})_{n \in \mathbb{N}^*}$  converge en  $\mathbb{L}^1$  vers 0.

Commandes : `ErlangDistribution`, `TransformedDistribution`, `Exp`, `≈`, `Limit`, `Expectation`, `Abs`, `∞`.

**Solution 248.** On propose :

```
In[1]: loi = ErlangDistribution[n, 1];
In[2]: loi2 = TransformedDistribution[3^n Exp[- 2 x], x ≈ loi];
In[3]: loi3 = TransformedDistribution[3^(n / 2) Exp[- x], x ≈ loi];
In[4]: Limit[Expectation[Abs[x - 0], x ≈ loi2], n → ∞] != 0
Out[4]: True
In[5]: Limit[Expectation[Abs[x - 0], x ≈ loi3], n → ∞] == 0
Out[5]: True
```

### Exercice 249

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* suivant chacune la loi de Rademacher  $\mathcal{R}(1)$  :

$$\mathbb{P}(X_1 = -1) = \mathbb{P}(X_1 = 1) = \frac{1}{2}.$$

Pour tout  $n \in \mathbb{N}^*$ , on pose  $Y_n = \sum_{k=1}^n \frac{X_k}{2^k}$  et on admet que la fonction caractéristique de  $Y_n$  vérifie :

$$\varphi_{Y_n}(t) = \prod_{k=1}^n \cos\left(\frac{t}{2^k}\right) = \sin(t) \left(2^n \sin\left(\frac{t}{2^n}\right)\right)^{-1}.$$

Utiliser *Mathematica* pour

- déterminer  $\lim_{n \rightarrow \infty} \varphi_{Y_n}(t)$ ,
- vérifier que  $(Y_n)_{n \in \mathbb{N}^*}$  converge en loi vers une *var*  $Y$  suivant la loi uniforme  $\mathcal{U}([-1, 1])$ .

Commandes : `UniformDistribution`, `Limit`, `Sin`, `∞`, `CharacteristicFunction`, `%`, `%%`, `//Simplify`.

**Solution 249.** On propose :

```
In[1]: loi = UniformDistribution[{-1, 1}];
In[2]: Limit[Sin[t] / (2^n Sin[t / 2^n]), n -> ∞]
Out[2]:  $\frac{\text{Sin}[t]}{t}$ 
In[3]: CharacteristicFunction[loi, t]
Out[3]:  $-\frac{i(-e^{-it} + e^{it})}{2t}$ 
In[4]: % - %% //Simplify
Out[4]: 0
```

### Exercice 250

Soient  $\alpha \in ]0, 1[$  et  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var* indépendantes avec, pour tout  $n \in \mathbb{N}^*$ ,  $X_n$  suivant la loi la loi géométrique tronquée en 0  $\mathcal{G}_{\{1, \dots, \infty\}}(\frac{\alpha}{n})$  :

$$\mathbb{P}(X_n = k) = \left(1 - \frac{\alpha}{n}\right)^{k-1} \frac{\alpha}{n}, \quad k \in \mathbb{N}^*.$$

Utiliser *Mathematica* pour

- créer la loi de  $X_n$ ,
- créer la loi de  $\frac{X_n}{n}$ ,
- déterminer  $\lim_{n \rightarrow \infty} \varphi_{\frac{X_n}{n}}(t)$ , où  $\varphi_{\frac{X_n}{n}}$  désigne la fonction caractéristique de  $\frac{X_n}{n}$ ,
- vérifier que  $\left(\frac{X_n}{n}\right)_{n \in \mathbb{N}^*}$  converge en loi vers une *var*  $X$  suivant la loi exponentielle  $\mathcal{E}(\alpha)$  en utilisant les fonctions caractéristiques.

Commandes : `TruncatedDistribution`, `∞`, `GeometricDistribution`, `α`, `TransformedDistribution`, `≈`, `Limit`, `CharacteristicFunction`, `ExponentialDistribution`, `α`, `%`, `%%`, `//Simplify`.

**Solution 250.** On propose :

```
In[1]: loi1 = TruncatedDistribution[{0, ∞}, GeometricDistribution[α / n]];
In[2]: loi2 = TransformedDistribution[x / n, x ≈ loi1];
In[3]: Limit[CharacteristicFunction[loi2, t], n -> ∞]
Out[3]:  $\frac{i\alpha}{t + i\alpha}$ 
```

```

In[4]: CharacteristicFunction[ExponentialDistribution[α, t]
Out[4]:  $\frac{\alpha}{-it + \alpha}$ 
In[5]: % - %% //Simplify
Out[5]: 0

```

**Exercice 251**

Soit  $(X_n)_{n \in \mathbb{N} \setminus \{0,1\}}$  une suite de *var*. Pour tout  $n \in \mathbb{N} \setminus \{0,1\}$ , la loi de  $X_n$  est donnée par

$$\mathbb{P}(X_n = k) = \frac{2k}{n(n+1)}, \quad k \in \{1, \dots, n\}.$$

Utiliser Mathematica pour vérifier que  $\left(\frac{X_n}{n}\right)_{n \in \mathbb{N} \setminus \{0,1\}}$  converge en loi vers une *var*  $X$  de densité :

$$f(x) = \begin{cases} 2x & \text{si } x \in [0, 1], \\ 0 & \text{sinon,} \end{cases}$$

en utilisant les fonctions génératrices des moments.

Commandes : ProbabilityDistribution, Assumptions, TransformedDistribution,  $\approx$ , Limit, MomentGeneratingFunction,  $\infty$ , %, %%, //Simplify.

**Solution 251.** On propose :

```

In[1]: loi1 = ProbabilityDistribution[{2 k / (n (n + 1))}, {k, 1, n, 1},
Assumptions -> n >= 2];
In[2]: loi2 = TransformedDistribution[x / n, x ≈ loi1];
In[3]: loi3 = ProbabilityDistribution[2 x, {x, 0, 1}];
In[4]: Limit[MomentGeneratingFunction[loi2, t], n -> ∞]
Out[4]:  $\left\{ \frac{2 + 2e^t(-1 + t)}{t^2} \right\}$ 
In[5]: MomentGeneratingFunction[loi3, t]
Out[5]:  $\frac{2(1 - e^t + e^t t)}{t^2}$ 
In[6]: % - %% //Simplify
Out[6]: {0}

```





## 12 Simulation de var à densité

### Exercice 252

Utiliser Mathematica pour

- simuler 6 réalisations d'une var  $X$  suivant la loi uniforme  $\mathcal{U}([-2, 1])$ ,
- simuler 4 réalisations d'une var  $X$  suivant la exponentielle  $\mathcal{E}(2.2)$ ,
- simuler 5 réalisations d'une var  $X$  suivant la loi de Erlang  $\mathcal{E}_r(3, 1.7)$ ,
- simuler 6 réalisations d'une var  $X$  suivant la loi normale  $\mathcal{N}(0, 1)$ ,
- simuler 4 réalisations d'une var  $X$  suivant la loi de Weibull  $\mathcal{W}(1.3, 0.4)$ ,
- simuler 5 réalisations d'une var  $X$  suivant la loi log-normale  $\mathcal{LN}(1.1, 0.3)$ .

Commandes : `RandomVariate`, `UniformDistribution`, `ExponentialDistribution`, `ErlangDistribution`, `NormalDistribution`, `WeibullDistribution`, `LogNormalDistribution`.

**Solution 252.** On propose :

```
In[1]: RandomVariate[UniformDistribution[{-2,1}], 6]
Out[1]: {-1.38837, -1.96905, 0.478181, -0.0302768, -0.163489, -0.00371731}
In[2]: RandomVariate[ExponentialDistribution[2.2], 4]
Out[2]: {0.753109, 0.322629, 0.499243, 0.0666377}
In[3]: RandomVariate[ErlangDistribution[3, 1.7], 5]
Out[3]: {0.847942, 1.25061, 1.37847, 3.6658, 0.767797}
In[4]: RandomVariate[NormalDistribution[0,1], 6]
Out[4]: {-0.365716, 0.445301, -0.344021, -0.621036, -0.709934, 0.469766}
In[5]: RandomVariate[WeibullDistribution[1.3, 0.4], 4]
Out[5]: {0.191419, 0.131188, 0.537281, 0.383478}
In[6]: RandomVariate[LogNormalDistribution[1.1, 0.3], 5]
Out[6]: {2.25156, 2.27117, 2.78261, 3.77056, 2.381}
```

### Exercice 253

Utiliser Mathematica pour

- simuler 9000 réalisations d'une var  $X$  suivant la loi uniforme  $\mathcal{U}([1, 3])$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de Aquamarine.


Commandes : `RandomVariate`, `UniformDistribution`, `Histogram`, `ColorFunction`.

**Solution 253.** On propose :

```
In[1]: simu = RandomVariate[UniformDistribution[{1,3}], 9000];
```

```
In[2]: Histogram[simu, ColorFunction -> "Aquamarine"]
```

Out[2]:



### Exercice 254

Utiliser Mathematica pour

- simuler 3000 réalisations d'une var  $X$  suivant la loi exponentielle  $\mathcal{E}(1.4)$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de FuchsiaTones.


Commandes : RandomVariate, ExponentialDistribution, Histogram, ColorFunction.

**Solution 254.** On propose :

```
In[1]: simu = RandomVariate[ExponentialDistribution[1.4], 3000];
```

```
In[2]: Histogram[simu, ColorFunction -> "FuchsiaTones"]
```

Out[2]:



### Exercice 255

Utiliser *Mathematica* pour

- simuler 50000 réalisations d'une *var*  $X$  dont la loi est donnée par les commandes :  
`TruncatedDistribution[{1, 4}, TriangularDistribution[{0, 7}]]`
- représenter les valeurs obtenues par un histogramme avec les couleurs de `BrownCyanTones`.

Commandes : `TruncatedDistribution`, `TriangularDistribution`, `RandomVariate`, `Histogram`, `ColorFunction`.


**Solution 255.** On propose :

```
In[1]: loi = TruncatedDistribution[{1, 4}, TriangularDistribution[{0, 7}]];
```

```
In[2]: simu = RandomVariate[loi, 50000];
```

```
In[3]: Histogram[simu, ColorFunction -> "BrownCyanTones"]
```

Out [3]:



### Exercice 256

Utiliser *Mathematica* pour

- simuler 10000 réalisations d'une *var*  $X$  suivant la loi normale  $\mathcal{N}(1, 2)$ ,
- représenter ces réalisations par un histogramme des fréquences à 22 classes avec les couleurs de `DeepSeaColors`,
- tracer le graphe de la densité associée à la loi normale  $\mathcal{N}(1, 2)$ , *i.e.*

$$f(x) = \frac{1}{\sqrt{2\pi \cdot 2}} e^{-\frac{(x-1)^2}{2 \times 2^2}}, \quad x \in \mathbb{R},$$

avec un trait jaune épais et  $x \in [-8, 8]$ ,

- superposer à l'historgramme le graphe obtenus.


Commandes : `RandomVariate`, `NormalDistribution`, `Histogram`, `ColorFunction`, `Plot`, `PDF`, `PlotStyle`, `Thickness`, `Show`.

**Solution 256.** On propose :

```
In[1]: simu = RandomVariate[NormalDistribution[1, 2], 10000];
```


```
In[2]: h = Histogram[simu, 22, "ProbabilityDensity", ColorFunction →  
"DeepSeaColors"]
```

Out[2]:




```
In[3]: g = Plot[PDF[NormalDistribution[1, 2], x], {x, -8, 8},  
PlotStyle → {Yellow, Thickness[0.007]}]
```

Out[3]:



```
In[4]: Show[h, g]
```

Out [4] :

**Exercice 257**Soit  $X$  une var de densité

$$f(x) = \begin{cases} \frac{2}{5}e^{2+x} & \text{si } x \leq -2, \\ \frac{3}{5}e^{-1-x} & \text{si } x \geq -1, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser Mathematica pour

- créer la fonction  $f$ ,
- simuler 50000 réalisations de  $X$ ,
- représenter ces réalisations par un histogramme des fréquences à 30 classes avec les couleurs de `CherryTones`,
- superposer à l'historgramme le graphe de  $f(x)$  avec  $x \in [-6, 4]$  en jaune.

Commandes : `x_`, `:=`, `Piecewise`, `Exp`, `ProbabilityDistribution`, `∞`, `RandomVariate`, `Histogram`, `ColorFunction`, `Show`, `Plot`, `PDF`, `PlotStyle`, `Thickness`.

**Solution 257.** On propose :


```
In[1]: f[x_] := Piecewise[{{2 / 5 Exp[2 + x], x ≤ -2}, {3 / 5 Exp[-1 - x],
x ≥ -1}}];
```

```
In[2]: loi = ProbabilityDistribution[f[x], {x, -∞, ∞}];
```

```
In[3]: simu = RandomVariate[loi, 50000];
```


```
In[4]: h = Histogram[simu, 30, "ProbabilityDensity", ColorFunction → "CherryTones"]
```

Out [4] :



```
In[5]: Show[h, Plot[PDF[loi, x], {x, -6, 4}, PlotStyle -> {Yellow,
Thickness[0.006]}]]
```

Out [5] :



### Exercice 258

Soit  $X$  une var de densité :

$$f(x) = \frac{1}{3} \times \frac{1}{\sqrt{2\pi \times 0.5^2}} e^{-\frac{(x-1)^2}{2 \times 0.5^2}} + \frac{2}{3} \times \frac{1}{\sqrt{2\pi \times 1.7^2}} e^{-\frac{(x-4)^2}{2 \times 1.7^2}}, \quad x \in \mathbb{R}.$$

Utiliser Mathematica pour


- créer la loi de  $X$  avec la commande `MixtureDistribution`,
- simuler 100000 réalisations de  $X$ ,
- représenter ces réalisations par un histogramme des fréquences à 65 classes avec les couleurs de `CoffeeTones`,
- superposer à l'historgramme le graphe de  $f(x)$  avec  $x \in [-2, 10]$  en jaune.

Commandes : `NormalDistribution`, `MixtureDistribution`, `RandomVariate`, `Histogram`, `ColorFunction`, `Plot`, `PDF`, `PlotStyle`, `Thickness`, `Show`.

**Solution 258.** On propose :


```
In[1]: loi1 = NormalDistribution[1, 0.5];
In[2]: loi2 = NormalDistribution[4, 1.7];
In[3]: loi3 = MixtureDistribution[{1, 2}, {loi1, loi2}];
In[4]: simu = RandomVariate[loi3, 100000];
In[5]: h = Histogram[simu, 65, "ProbabilityDensity", ColorFunction →
"CoffeeTones"]
```

Out [5]:



```
In[6]: g = Plot[PDF[loi3, x], {x, -2, 10}, PlotStyle → {Yellow,
Thickness[0.007]}];
In[7]: Show[h, g]
```

Out [7]:





**Exercice 259**

Soient  $X$  et  $Y$  deux *var* indépendantes telles que

- ◊  $X$  suit la loi normale  $\mathcal{N}(0,1)$ ,
- ◊  $Y$  suit la loi exponentielle  $\mathcal{E}(0.3)$ .

On pose  $Z = X - Y$ .

Utiliser Mathematica pour


- simuler 100000 réalisations de  $Z$ ,
- représenter les valeurs obtenues par un histogramme avec les couleurs de `PearlColors`.

Commandes : `NormalDistribution`, `ExponentialDistribution`, `TransformedDistribution`, `≈`, `RandomVariate`, `Histogram`, `ColorFunction`.

**Solution 259.** On propose :


```
In[1]: loi1 = NormalDistribution[0,1];
In[2]: loi2 = ExponentialDistribution[0.3];
In[3]: loi3 = TransformedDistribution[x - y, {x ≈ loi1, y ≈ loi2}];
In[4]: simu = RandomVariate[loi3, 100000];
In[5]: Histogram[simu, ColorFunction → "PearlColors"]
```

**Out [5]:**



**Exercice 260**

On simule 10000 réalisations d'une var  $X$ . À partir des QuantilePlot suivant, déduire la loi de  $X$  :





Commande : `QuantilePlot`.

**Solution 260.** La courbe qui se confond le plus avec la droite diagonale  $y = x$  est celle du QuantilePlot considérant la loi de Cauchy  $\mathcal{C}(0,1)$ . Par conséquent,  $X$  suit plutôt la loi de Cauchy  $\mathcal{C}(0,1)$ .

**Exercice 261**

On simule 10000 réalisations d'une var  $X$ . À partir des ProbabilityPlot suivant, déduire la loi de  $X$  :





**Solution 261.** La courbe qui se confond le plus avec la droite diagonale  $y = x$  est celle du `ProbabilityPlot` considérant la loi logistique  $\mathcal{L}_{og}(0, 1)$ . Par conséquent,  $X$  suit plutôt la loi logistique  $\mathcal{L}_{og}(0, 1)$ .

### Exercice 262

Soient  $X$  et  $Y$  deux *var* indépendantes telles que  $X$  suit la loi exponentielle  $\mathcal{E}(0.6)$  et  $Y$  suit la loi exponentielle  $\mathcal{E}(0.6)$ . Alors on sait que  $Z = X + Y$  suit la loi de Erlang  $\mathcal{E}_r(1 + 1, 0.6) = \mathcal{E}_r(2, 0.6)$ .

Utiliser `Mathematica` pour illustrer ce résultat avec

- les commandes `RandomVariate` et `QuantilePlot`,
- les commandes `RandomVariate` et `ProbabilityPlot`.

Commandes : `RandomVariate`, `ExponentialDistribution`, `QuantilePlot`, `ErlangDistribution`, `ProbabilityPlot`.

**Solution 262.** On propose :


```
In[1]: x = RandomVariate[ExponentialDistribution[0.6], 10000];
```

```
In[2]: y = RandomVariate[ExponentialDistribution[0.6], 10000];
```

```
In[3]: z = x + y;
```


```
In[4]: QuantilePlot[z, ErlangDistribution[2, 0.6]]
```

Out[4]:



```
In[5]: ProbabilityPlot[z, ErlangDistribution[2, 0.6]]
```

Out[5]:



!! Dans les deux graphiques, le nuage de points est bien ajusté par la droite diagonale  $y = x$ . Cela illustre le fait que  $X + Y$  suit la loi de Erlang  $\mathcal{E}_r(2, 0.6)$ .

### Exercice 263

Soient  $X$  et  $Y$  deux *var* indépendantes. On a les résultats en loi du tableau :

loi de $X$	loi de $Y$	transformation	loi
normale $\mathcal{N}(0, 1)$	normale $\mathcal{N}(0, 1)$	$X^2 + Y^2$	Chi-deux $\chi^2(2)$
normale $\mathcal{N}(0, 1)$	Chi-deux $\chi^2(6.5)$	$\frac{X}{\sqrt{\frac{Y}{6.5}}}$	Student $\mathcal{T}(6.5)$

Utiliser **Mathematica** pour illustrer ces deux résultats avec

- les commandes `RandomVariate` et `QuantilePlot`,
- les commandes `RandomVariate` et `ProbabilityPlot`.

Commandes : `RandomVariate`, `NormalDistribution`, `QuantilePlot`, `ChiSquareDistribution`, `ProbabilityPlot`, `StudentTDistribution`.

**Solution 263.** On propose :


```
In[1]: x = RandomVariate[NormalDistribution[0,1], 10000];
```

```
In[2]: y = RandomVariate[NormalDistribution[0,1], 10000];
```

```
In[3]: z = x^2 + y^2;
```


```
In[4]: QuantilePlot[z, ChiSquareDistribution[2]]
```

Out [4] :



In[5]: `ProbabilityPlot[z, ChiSquareDistribution[2]]`

Out [5] :




In[6]: `x = RandomVariate[NormalDistribution[0,1], 10000];`

In[7]: `y = RandomVariate[ChiSquareDistribution[6.5], 10000];`

In[8]: `z = x / Sqrt[y / 6.5];`


In[9]: `QuantilePlot[z, StudentTDistribution[6.5]]`

Out [9] :



```
In[10]: ProbabilityPlot[z, StudentTDistribution[6.5]]
```

Out[10]:



!! Dans chacun des graphiques, le nuage de points est bien ajusté par la droite diagonale  $y = x$ . Cela illustre les résultats en loi demandés.

#### Exercice 264

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} 4xy & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

On pose  $Z = \log(1 + X + Y)$ .

Utiliser *Mathematica* pour

- simuler 60000 réalisations de  $Z$ ,
- représenter les valeurs obtenues par un histogramme à 40 classes.

Commandes : `ProbabilityDistribution`, `TransformedDistribution`, `Log`, `≈`, `RandomVariate`, `Histogram`.

**Solution 264.** On propose :


```
In[1]: loi = ProbabilityDistribution[4 x y, {x, 0, 1}, {y, 0, 1}];
```

```
In[2]: loi2 = TransformedDistribution[Log[1 + x + y], {x, y} ≈ loi];
```

```
In[3]: simu = RandomVariate[loi2, 60000];
```

```
In[4]: Histogram[simu, 40]
```

Out [4] :

**Exercice 265**

Soit  $(X, Y)$  un couple de *var* de densité :

$$f(x, y) = \begin{cases} \frac{3}{4} + xy & \text{si } (x, y) \in [0, 1]^2, \\ 0 & \text{sinon.} \end{cases}$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,
- représenter le graphe de  $f$  avec les couleurs de *ValentineTones*,
- simuler 60000 réalisations de  $(X, Y)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de *ValentineTones*,
- simuler 60000 réalisations de  $(X - Y, X + Y)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de *ValentineTones*.


Commandes : *ProbabilityDistribution*, *Plot3D*, *PDF*, *ColorFunction*, *RandomVariate*, *Histogram3D*, *TransformedDistribution*.

**Solution 265.** On propose :

```
In[1]: loi = ProbabilityDistribution[3 / 4 + x y, {x, 0, 1}, {y, 0, 1}];
```

```
In[2]: Plot3D[PDF[loi, {x, y}], {x, 0, 1}, {y, 0, 1}, ColorFunction ->
      "ValentineTones"]
```


Out [2] :



```
In[3]: simu = RandomVariate[loi, 60000];
```

```
In[4]: Histogram3D[simu, ColorFunction → "ValentineTones"]
```

Out [4] :




```
In[5]: loi2 = TransformedDistribution[{x - y, x + y}, {x, y} ≈ loi];
```

```
In[6]: simu2 = RandomVariate[loi, 60000];
```

```
In[7]: Histogram3D[simu2, ColorFunction → "ValentineTones"]
```



Out [7] :

**Exercice 266**

Soit  $(X, Y)$  un couple de *var* suivant la loi normale multivariée  $\mathcal{N}_2(\mu, \Sigma)$  avec  $\mu = (0, 0)^t$  et

$$\Sigma = \begin{pmatrix} 0.7 & 0.3 \\ 0.3 & 1.1 \end{pmatrix}.$$

Utiliser *Mathematica* pour

- créer la loi de  $(X, Y)$ ,
- tracer le graphe d'une densité associée avec les couleurs de *GreenPinkTones*,
- simuler 100000 réalisations de  $(X, Y)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec un grand nombre de classes et les couleurs de *GreenPinkTones*.


Commandes : *MultinormalDistribution*, *Plot3D PDF*, *ColorFunction*, *PlotRange*, *RandomVariate*, *histogram3D*.

**Solution 266.** On propose :

```
In[1]: loi = MultinormalDistribution[{0, 0}, {{0.7, 0.3},{0.3, 1.1}}];
```

```
In[2]: Plot3D[PDF[loi, {x, y}], {x, -2.5, 2.5}, {y, -2.5, 2.5},
ColorFunction -> "GreenPinkTones", PlotRange -> All]
```


Out [2] :



```
In[3]: simu = RandomVariate[loi, 100000];
```

```
In[4]: Histogram3D[simu, 40, ColorFunction -> "GreenPinkTones"]
```

Out [4] :



**Exercice 267**

*Loi forte des grands nombres : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i.$$

*Alors  $(\bar{X}_n)_{n \in \mathbb{N}^*}$  converge presque sûrement vers  $\mathbb{E}(X_1)$ . L'objectif de cet exercice est d'illustrer ce résultat.*

Utiliser **Mathematica** pour

- créer une fonction `fsimu` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $|\bar{X}_n - \mathbb{E}(X_1)|$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- tester graphiquement cette fonction en faisant :
 


```
Histogram[fsimu[ExponentialDistribution[2], 100], ColorFunction -> "NeonColors"]
Histogram[fsimu[ExponentialDistribution[2], 1000], ColorFunction -> "NeonColors"]
Histogram[fsimu[ExponentialDistribution[2], 10000], ColorFunction -> "NeonColors"]
```
- refaire l'analyse graphique précédente avec la loi caractérisée par les commandes `HalfNormalDistribution[0.3]`.

Commandes : `x_`, `:=`, `Table`, `Abs`, `Mean`, `RandomVariate`, `Histogram`, `ExponentialDistribution`, `ColorFunction`.

**Solution 267.** On propose :


```
In[1]: fsimu[loi_, n_] := Table[Abs[Mean[RandomVariate[loi, n]] - Mean[loi]], 1000];
In[2]: Histogram[fsimu[ExponentialDistribution[2], 100], ColorFunction ->
"NeonColors"]
```

Out [2]:




```
In[3]: Histogram[fsimu[ExponentialDistribution[2], 1000], ColorFunction ->
"NeonColors"]
```

Out [3] :



```
In[4]: Histogram[fsimu[ExponentialDistribution[2], 10000], ColorFunction ->
"NeonColors"]
```


Out [4] :



!! En regardant l'axe des abscisses pour chaque graphique, on remarque que plus  $n$  est grand, plus les différences entre les réalisations de  $\bar{X}_n$  et  $\mathbb{E}(X_1) = 0.5$  s'amenuisent (en moyenne).


```
In[5]: Histogram[fsimu[HalfNormalDistribution[0.3], 100], ColorFunction ->
"NeonColors"]
```

Out [5] :




```
In[6]: Histogram[fsimu[HalfNormalDistribution[0.3], 1000], ColorFunction ->
"NeonColors"]
```

Out [6] :



```
In[7]: Histogram[fsimu[HalfNormalDistribution[0.3], 10000], ColorFunction ->
"NeonColors"]
```

Out [7] :



!! La conclusion est la même que pour la loi `ExponentialDistribution[2]` (et sera la même pour n'importe qu'elle autre loi, du moment que le moment d'ordre 1 existe).

### Exercice 268

Résultat : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad S_n^2 = \frac{1}{n-1} \sum_{i=1}^n (X_i - \bar{X}_n)^2.$$

Alors  $(S_n^2)_{n \in \mathbb{N}^*}$  converge presque sûrement vers  $\mathbb{V}(X_1)$  (et  $\mathbb{E}(S_n^2) = \mathbb{V}(X_1)$ ). L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser `Mathematica` pour

- créer une fonction `fsimu` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $|S_n^2 - \mathbb{V}(X_1)|$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- tester graphiquement cette fonction en faisant :

```
Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 100], ColorFunction -> "Pastel"]
Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 1000], ColorFunction ->
"Pastel"]
Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 10000], ColorFunction ->
"Pastel"]
```


Commandes : `x_`, `:=`, `Table`, `Abs`, `Variance`, `RandomVariate`, `Histogram`, `TriangularDistribution`, `ColorFunction`.

**Solution 268.** On propose :

```
In[1]: fsimu[loi_, n_] := Table[Abs[Variance[RandomVariate[loi, n]] -
Variance[loi]], 1000];
```


```
In[2]: Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 100], ColorFunction →  
"Pastel"]
```

Out[2]:




```
In[3]: Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 1000], ColorFunction →  
"Pastel"]
```

Out[3]:



```
In[4]: Histogram[fsimu[TriangularDistribution[{0, 4}, 1], 10000], ColorFunction →  
"Pastel"]
```

Out [4] :



!! En regardant l'axe des abscisses pour chaque graphique, on remarque que plus  $n$  est grand, plus les différences entre les réalisations de  $S_n^2$  et  $\mathbb{V}(X_1) = 0.722222$  s'amenuisent (en moyenne).

### Exercice 269

*Théorème central limite : Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de var iid admettant un moment d'ordre 2. Pour tout  $n \in \mathbb{N}^*$ , on pose*

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i, \quad Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right) = \frac{1}{\sqrt{n}} \sum_{i=1}^n \left( \frac{X_i - \mathbb{E}(X_1)}{\sigma(X_1)} \right).$$

*Alors la suite de var  $(Z_n)_{n \in \mathbb{N}^*}$  converge en loi vers une var  $Z$  suivant la loi normale  $\mathcal{N}(0, 1)$ .*

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- créer une fonction `fsimutcl` ayant deux arguments : `loi` et `n`, qui renvoie un vecteur de taille 1000 dont chaque élément est une réalisation de  $Z_n$ , avec  $X_1, \dots, X_n$  iid suivant chacune la loi `loi`,
- appliquer la fonction `fsimutcl` avec  $n = 100000$  et, pour `loi`, la loi exponentielle  $\mathcal{E}(1.7)$ ,
- représenter ces 100000 réalisations par un histogramme des fréquences à 30 classes avec les couleurs de `LakeColors`,
- superposer à l'histogramme le graphe de la densité  $f(x)$  associée à la loi normale  $\mathcal{N}(0, 1)$  pour  $x \in [-4, 4]$  en rouge,
- reprendre les points précédents avec la loi de Weibull  $\mathcal{W}(2.1, 3.8)$ .

Commandes : `x_`, `:=`, `Table`, `Sqrt`, `Total`, `RandomVariate`, `TransformedDistribution`, `Mean`, `StandardDeviation`, `≈`, `//N`, `ExponentialDistribution`, `Histogram`, `ColorFunction`, `Show`, `Plot`, `NormalDistribution`, `PlotStyle`, `WeibullDistribution`,



**Solution 269.** On propose :


```
In[1]: fsimutcl[loi_, n_] := Table[1 / Sqrt[n] Total[RandomVariate[
  TransformedDistribution[(x - Mean[loi]) / StandardDeviation[loi],
  x ≈ loi], n] //N], 1000];
```

```
In[2]: a = fsimutcl[ExponentialDistribution[1.7], 100000];
```

```
In[3]: h = Histogram[a, 30, "ProbabilityDensity", ColorFunction → "LakeColors"];
```

```
In[4]: Show[h, Plot[PDF[NormalDistribution[0, 1], x], {x, -4, 4}, PlotStyle → Red]]
```

Out[4]:




```
In[5]: b = fsimutcl[WeibullDistribution[2.1, 3.8], 100000];
```

```
In[6]: h2 = Histogram[b, 30, "ProbabilityDensity", ColorFunction → "LakeColors"];
```

```
In[7]: Show[h2, Plot[PDF[NormalDistribution[0, 1], x], {x, -4, 4}, PlotStyle → Red]]
```

Out[7]:



!! Les graphiques obtenus précédemment montrent que le graphe de  $f$  ajuste bien les sommets des bâtons de l'histogramme. Cela illustre le théorème central limite et plus particulièrement l'approximation :  $Z_n = \sqrt{n} \left( \frac{\bar{X}_n - \mathbb{E}(X_1)}{\sigma(X_1)} \right) \approx \mathcal{N}(0, 1)$  pour  $n$  suffisamment grand.

### Exercice 270

Soit  $(X_n)_{n \in \mathbb{N}^*}$  une suite de *var iid* admettant un moment d'ordre 2. Le théorème central limite nous assure que, pour  $n$  suffisamment grand, on a l'approximation :

$$\bar{X}_n = \frac{1}{n} \sum_{i=1}^n X_i \approx \mathcal{N} \left( \mathbb{E}(X_1), \frac{\sigma(X_1)}{\sqrt{n}} \right).$$

L'objectif de cet exercice est d'illustrer ce résultat.

Utiliser **Mathematica** pour

- reproduire et comprendre l'enjeu de la fonction `tclhist` (sophistiquée) à un argument `loi` :

```
tclhist[loi_] := Table[Show[Histogram[(1 / n) Plus @@@
RandomVariate[ProductDistribution[{loi, n}], 10^5], Automatic, "PDF", ColorFunction
→ "MintColors", PlotLabel → n],
Plot[Boole[n ≥ 2] PDF[NormalDistribution[Mean[loi],
StandardDeviation[loi] / Sqrt[n]], x], {x, -20, 20}, PlotStyle → Red,
PlotRange → Full]], {n, Range[1, 78, 7]}]
```

- appliquer la fonction `fsimutcl` avec, pour `loi` :

- ◊ la loi exponentielle  $\mathcal{E}(1.8)$ ,
- ◊ la loi de Weibull  $\mathcal{W}(5, 3)$ ,
- ◊ la loi de Pareto  $\mathcal{P}(8, 6)$ .

Commandes : `Table`, `Show`, `Histogram`, `Plus`, `RandomVariate`, `ProductDistribution`, `ColorFunction`, `PlotLabel`, `Plot`, `Boole`, `PDF`, `NormalDistribution`, `Mean`, `StandardDeviation`, `Sqrt`, `PlotStyle`, `PlotRange`, `Range`, `ExponentialDistribution`, `WeibullDistribution`, `ParetoDistribution`.


**Solution 270.** On propose :

```
In[1]: tclhist[loi_] := Table[Show[Histogram[(1 / n) Plus @@@
RandomVariate[ProductDistribution[{loi, n}], 10^5], Automatic, "PDF",
ColorFunction → "MintColors", PlotLabel → n],
Plot[Boole[n ≥ 2] PDF[NormalDistribution[Mean[loi],
StandardDeviation[loi] / Sqrt[n]], x], {x, -20, 20}, PlotStyle → Red,
PlotRange → Full]], {n, Range[1, 78, 7]}];
```

L'enjeu de la fonction `tclhist` est de simuler  $10^5$  réalisations de  $\bar{X}_n$  pour plusieurs valeurs de  $n$  ( $n \in \{1, 6, 11, 16, \dots, 50, 56\}$ ), tracer l'histogramme associé pour chacune de ces valeurs et superposer la loi normale  $\mathcal{N}\left(\mathbb{E}(X_1), \frac{\sigma(X_1)}{\sqrt{n}}\right)$ . Le théorème central limite nous assure alors que, plus  $n$  est grand, plus les sommets de l'histogramme seront bien ajustés par le graphe d'une densité de la loi normale considérée. À cela s'ajoute plusieurs options graphiques secondaires (tailles, couleurs...).


```
In[2]: tclhist[ExponentialDistribution[1.8]]
```

Out [2]:




In[3]: `tclhist[WeibullDistribution[5,3]]`

Out[3]:



In[4]: `tclhist[ParetoDistribution[8,6]]`

Out[4]:



**Exercice 271**

Soient  $X_1, \dots, X_8$  8 var iid suivant chacune la loi uniforme  $\mathcal{U}([-5, 6])$ . On pose  $U = \inf(X_1, \dots, X_8)$  et  $V = \sup(X_1, \dots, X_8)$ .

Utiliser Mathematica pour


- créer la loi de  $(U, V)$ ,
- simuler 10000 réalisations de  $(U, V)$ ,
- représenter les valeurs obtenues par un histogramme adapté avec les couleurs de `BeachColors`,
- vérifier que les moyennes des valeurs obtenues pour  $U$  et  $V$  sont proches de  $(\mathbb{E}(U), \mathbb{E}(V))$ .

Commandes : `UniformDistribution`, `OrderDistribution`, `RandomVariate`, `Histogram3D`, `ColorFunction`, `Mean`.

**Solution 271.** On propose :

```
In[1]: loi = UniformDistribution[{-5, 6}];
In[2]: loi2 = OrderDistribution[{loi, 8}, {1, 8}]
In[3]: simu = RandomVariate[loi2, 10000];
In[4]: Histogram3D[simu, ColorFunction -> "BeachColors"]
```

Out[4]:



```
In[5]: Mean[simu] - Mean[loi2] //N
```

Out[5]: {0.0103856, 0.0223894}



## 13 Ressources

### Sur les probabilités avec Mathematica :

- Le guide Probability And Statistics du site de Wolfram :  
<http://reference.wolfram.com/language/guide/ProbabilityAndStatistics.html>
- Introduction to Probability with Mathematica de Kevin J. Hastings :  
<https://www.crcpress.com/Introduction-to-Probability-with-Mathematica-Second-Edition>

### Sur les notions de Probabilités :

- Formulaire de Probabilités de l'auteur :  
<http://www.math.unicaen.fr/~chesneau/form-prob-MATHS.pdf>
- Probabilités de Stéphane Ducay :  
<http://www.lamfa.u-picardie.fr/ducaey/Accueil.html>
- Introduction aux Probabilités de Arnaud Guyader :  
<http://www.lsta.lab.upmc.fr/modules/resources/download/labsta/Pages/Guyader/IntroProba.pdf>
- Éléments de cours de Probabilités de Jean-François Marckert :  
[http://eric.univ-lyon2.fr/~ricco/cours/cours/La\\_regression\\_dans\\_la\\_pratique.pdf](http://eric.univ-lyon2.fr/~ricco/cours/cours/La_regression_dans_la_pratique.pdf)
- Probabilités et Statistiques de Alain Yger :  
<https://www.math.u-bordeaux.fr/~ayger/Proba6031.pdf>
- Cours de Théorie des probabilités de Bruno Saussereau :  
<http://bsauss.perso.math.cnrs.fr/CTU-1314-SAUSSEREAU-THEORIE-DES-PROBABILITES.pdf>

### Autres ouvrages de l'auteur (aux éditions Ellipses) :

- Probabilités et variables aléatoires discrètes - exercices corrigés :  
[http://www.editions-ellipses.fr/product\\_info.php?products\\_id=10826](http://www.editions-ellipses.fr/product_info.php?products_id=10826)
- Intervalles de confiance et tests statistiques - exercices corrigés (avec Maher Kachour) :  
[http://www.editions-ellipses.fr/product\\_info.php?products\\_id=10674](http://www.editions-ellipses.fr/product_info.php?products_id=10674)
- Probabilités et variables aléatoires à densité - exercices corrigés (à venir).